

# **gnuplot**

An Interactive Plotting Program

Thomas Williams & Colin Kelley

Version 4.2 organized by: Hans-Bernhard Bröker and others

Major contributors (alphabetic order):

Hans-Bernhard Bröker  
John Campbell  
Robert Cunningham  
David Denholm  
Gershon Elber  
Roger Fearick  
Carsten Grammes  
Lucas Hart  
Lars Hecking  
Thomas Koenig  
David Kotz  
Ed Kubaitis  
Russell Lang  
Alexander Lehmann  
Alexander Mai  
Ethan A Merritt  
Petr Mikulík  
Carsten Steger  
Tom Tkacik  
Jos Van der Woude  
Alex Woo  
James R. Van Zandt  
Johannes Zellner

Copyright (C) 1986 - 1993, 1998, 2004 Thomas Williams, Colin Kelley

Mailing list for comments: [gnuplot-info@lists.sourceforge.net](mailto:gnuplot-info@lists.sourceforge.net)

Mailing list for bug reports: [gnuplot-bugs@lists.sourceforge.net](mailto:gnuplot-bugs@lists.sourceforge.net)

This manual was prepared by Dick Crawford.

Last edited: 2006/03/06 17:42:29

## Contents

<b>I Gnuplot</b>	<b>14</b>
<b>1 Copyright</b>	<b>14</b>
<b>2 Introduction</b>	<b>15</b>
<b>3 Seeking-assistance</b>	<b>16</b>
<b>4 New features introduced in version 4.2</b>	<b>16</b>
4.1 新しい描画スタイル . . . . .	16
4.1.1 ヒストグラム . . . . .	16
4.1.2 ラベル描画 . . . . .	16
4.1.3 画像データ . . . . .	17
4.1.4 閉曲線の塗りつぶし . . . . .	17
4.1.5 ベクトル . . . . .	17
4.2 バイナリデータファイルからの入力 . . . . .	17
4.3 新しい描画要素 . . . . .	17
4.3.1 RGB 色指定 . . . . .	17
4.3.2 任意の長方形描画 . . . . .	17
4.4 文字列処理 . . . . .	17
4.4.1 データファイルからの文字列や文字データの入力 . . . . .	17
4.4.2 ユーザ定義文字列変数、文字列用の演算子や関数 . . . . .	17
4.5 マクロ . . . . .	17
4.6 複数グラフの 1 ページ自動配置描画 . . . . .	18
4.7 内部変数 . . . . .	18
4.8 新しい/修正された出力形式 . . . . .	18
4.8.1 wxt . . . . .	18
4.8.2 emf . . . . .	18
4.8.3 gif, jpeg, png . . . . .	18
4.8.4 postscript . . . . .	18
4.8.5 ai . . . . .	18
4.8.6 epslatex, pslatex, pstex . . . . .	18
4.8.7 windows . . . . .	18
4.9 Canvas size . . . . .	19
<b>5 Backwards compatibility</b>	<b>19</b>
<b>6 Features introduced in version 4.0</b>	<b>20</b>
6.1 対話型出力形式上のマウスとホットキーのサポート . . . . .	20
6.2 新しい出力形式 . . . . .	20
6.3 新しい描画スタイル pm3d . . . . .	20
6.4 塗りつぶされた箱 (filled boxes) . . . . .	20

6.5 新しい描画オプション smooth frequency . . . . .	20
6.6 文字列オプションの改良 . . . . .	20
6.7 文字コード化 (encodings) のサポートの追加 . . . . .	21
6.8 矢 (arrows) . . . . .	21
6.9 データファイル形式 . . . . .	21
6.10 新しいコマンド . . . . .	21
6.11 他の変更と追加 . . . . .	21
6.12 付属ドキュメント . . . . .	21
<b>7 Batch/Interactive Operation</b>	<b>22</b>
<b>8 Command-line-editing</b>	<b>22</b>
<b>9 Comments</b>	<b>23</b>
<b>10 Coordinates</b>	<b>23</b>
<b>11 Datastrings</b>	<b>24</b>
<b>12 Environment</b>	<b>24</b>
<b>13 Expressions</b>	<b>25</b>
13.1 Functions . . . . .	26
13.1.1 Random number generator . . . . .	27
13.2 Operators . . . . .	27
13.2.1 Unary . . . . .	27
13.2.2 Binary . . . . .	28
13.2.3 Ternary . . . . .	28
13.3 Gnuplot-defined variables . . . . .	29
13.4 User-defined variables and functions . . . . .	29
<b>14 Glossary</b>	<b>30</b>
<b>15 Linetype, colors, and styles</b>	<b>31</b>
15.1 Colorspec . . . . .	32
<b>16 Mouse input</b>	<b>32</b>
16.1 Bind . . . . .	32
16.2 Mouse variables . . . . .	34
<b>17 Plotting</b>	<b>34</b>
<b>18 Start-up</b>	<b>34</b>
<b>19 String constants and string variables</b>	<b>35</b>
<b>20 Substitution and Command line macros</b>	<b>35</b>

20.1	Substitution of system commands in backquotes . . . . .	36
20.2	Substitution of string variables as macros . . . . .	36
20.3	String variables, macros, and command line substitution . . . . .	36
<b>21 Syntax</b>		<b>37</b>
21.1	Quote Marks . . . . .	37
<b>22 Time/Date data</b>		<b>38</b>
<b>II Commands</b>		<b>39</b>
<b>23 Cd</b>		<b>39</b>
<b>24 Call</b>		<b>39</b>
<b>25 Clear</b>		<b>40</b>
<b>26 Exit</b>		<b>40</b>
<b>27 Fit</b>		<b>41</b>
27.1	Adjustable parameters . . . . .	42
27.2	Short introduction . . . . .	42
27.3	Error estimates . . . . .	43
27.3.1	Statistical overview . . . . .	43
27.3.2	Practical guidelines . . . . .	44
27.4	Control . . . . .	45
27.4.1	Control variables . . . . .	45
27.4.2	Environment variables . . . . .	45
27.5	Multi-branch . . . . .	46
27.6	Starting values . . . . .	46
27.7	Tips . . . . .	46
<b>28 Help</b>		<b>47</b>
<b>29 History</b>		<b>47</b>
<b>30 If</b>		<b>48</b>
<b>31 Load</b>		<b>48</b>
<b>32 Lower</b>		<b>49</b>
<b>33 Pause</b>		<b>49</b>
<b>34 Plot</b>		<b>50</b>
34.1	Data . . . . .	50
34.1.1	Binary . . . . .	52

34.1.2	Binary general . . . . .	52
34.1.2.1	Array . . . . .	53
34.1.2.2	Record . . . . .	53
34.1.2.3	Format . . . . .	53
34.1.2.4	Endian . . . . .	53
34.1.2.5	Filetype . . . . .	54
34.1.2.5.1	Avs . . . . .	54
34.1.2.5.2	Edf . . . . .	54
34.1.2.6	Keywords . . . . .	54
34.1.2.6.1	Scan . . . . .	54
34.1.2.6.2	Transpose . . . . .	54
34.1.2.6.3	Dx, dy, dz . . . . .	54
34.1.2.6.4	Flipx flipy, flipz . . . . .	54
34.1.2.6.5	Origin . . . . .	55
34.1.2.6.6	Center . . . . .	55
34.1.2.6.7	Rotate . . . . .	55
34.1.2.6.8	Perpendicular . . . . .	55
34.1.2.7	Binary examples . . . . .	55
34.1.3	Every . . . . .	56
34.1.4	Example datafile . . . . .	56
34.1.5	Index . . . . .	57
34.1.6	Smooth . . . . .	57
34.1.6.1	Acsplines . . . . .	57
34.1.6.2	Bezier . . . . .	58
34.1.6.3	Csplines . . . . .	58
34.1.6.4	Sbezier . . . . .	58
34.1.6.5	Unique . . . . .	58
34.1.6.6	Frequency . . . . .	58
34.1.7	Special-filenames . . . . .	58
34.1.8	Thru . . . . .	59
34.1.9	Using . . . . .	60
34.1.9.1	Using title . . . . .	61
34.1.9.2	Xticlabels . . . . .	61
34.1.9.3	X2ticlabels . . . . .	61
34.1.9.4	Yticlabels . . . . .	62
34.1.9.5	Y2ticlabels . . . . .	62
34.1.9.6	Zticlabels . . . . .	62
34.2	Errorbars . . . . .	62
34.3	Errorlines . . . . .	62
34.4	Parametric . . . . .	63
34.5	Ranges . . . . .	63
34.6	Title . . . . .	64

34.7 With . . . . .	65
<b>35 Print</b>	<b>67</b>
<b>36 Pwd</b>	<b>67</b>
<b>37 Quit</b>	<b>67</b>
<b>38 Raise</b>	<b>67</b>
<b>39 Replot</b>	<b>68</b>
<b>40 Reread</b>	<b>68</b>
<b>41 Reset</b>	<b>69</b>
<b>42 Save</b>	<b>69</b>
<b>43 Set-show</b>	<b>70</b>
43.1 Angles . . . . .	70
43.2 Arrow . . . . .	70
43.3 Autoscale . . . . .	72
43.3.1 Parametric mode . . . . .	73
43.3.2 Polar mode . . . . .	74
43.4 Bars . . . . .	74
43.5 Bmargin . . . . .	74
43.6 Border . . . . .	74
43.7 Boxwidth . . . . .	75
43.8 Clabel . . . . .	76
43.9 Clip . . . . .	76
43.10 Cntrparam . . . . .	77
43.11 Color box . . . . .	78
43.12 Contour . . . . .	79
43.13 Data style . . . . .	80
43.14 Datafile . . . . .	80
43.14.1 Set datafile fortran . . . . .	80
43.14.2 Set datafile missing . . . . .	80
43.14.3 Set datafile separator . . . . .	81
43.14.4 Set datafile commentschars . . . . .	81
43.14.5 Set datafile binary . . . . .	82
43.15 Decimalsign . . . . .	82
43.16 Dgrid3d . . . . .	83
43.17 Dummy . . . . .	83
43.18 Encoding . . . . .	84
43.19 Fit . . . . .	84

43.20 Fontpath . . . . .	85
43.21 Format . . . . .	85
43.21.1 Gprintf . . . . .	86
43.21.2 Format specifiers . . . . .	86
43.21.3 Time/date specifiers . . . . .	87
43.22 Function style . . . . .	88
43.23 Functions . . . . .	88
43.24 Grid . . . . .	88
43.25 Hidden3d . . . . .	89
43.26 Historysize . . . . .	90
43.27 Isosamples . . . . .	90
43.28 Key . . . . .	91
43.29 Label . . . . .	94
43.30 Lmargin . . . . .	95
43.31 Loadpath . . . . .	96
43.32 Locale . . . . .	96
43.33 Logscale . . . . .	96
43.34 Macros . . . . .	97
43.35 Mapping . . . . .	97
43.36 Margin . . . . .	97
43.37 Mouse . . . . .	98
43.37.1 X11 mouse . . . . .	99
43.38 Multiplot . . . . .	99
43.39 Mx2tics . . . . .	100
43.40 Mxtics . . . . .	101
43.41 My2tics . . . . .	101
43.42 Mytics . . . . .	101
43.43 Mztics . . . . .	101
43.44 Offsets . . . . .	102
43.45 Origin . . . . .	102
43.46 Output . . . . .	102
43.47 Parametric . . . . .	103
43.48 Plot . . . . .	103
43.49 Pm3d . . . . .	103
43.50 Palette . . . . .	106
43.50.1 Rgbformulae . . . . .	108
43.50.2 Defined . . . . .	109
43.50.3 Functions . . . . .	109
43.50.4 File . . . . .	110
43.50.5 Gamma correction . . . . .	110
43.50.6 Postscript . . . . .	111
43.50.7 Colornames . . . . .	111

43.51 Pointsize . . . . .	111
43.52 Polar . . . . .	111
43.53 Print . . . . .	112
43.54 Object . . . . .	112
43.55 Rmargin . . . . .	113
43.56 Rrange . . . . .	113
43.57 Samples . . . . .	113
43.58 Size . . . . .	114
43.59 Style . . . . .	115
43.59.1 Set style arrow . . . . .	115
43.59.2 Set style data . . . . .	116
43.59.3 Set style fill . . . . .	116
43.59.4 Set style function . . . . .	116
43.59.5 Set style increment . . . . .	117
43.59.6 Set style line . . . . .	117
43.59.7 Plotting styles . . . . .	118
43.59.8 Set style rectangle . . . . .	119
43.59.8.1 Boxerrorbars . . . . .	119
43.59.8.2 Boxes . . . . .	119
43.59.8.3 Boxyerrorbars . . . . .	120
43.59.8.4 Candlesticks . . . . .	120
43.59.8.5 Dots . . . . .	121
43.59.8.6 Filledcurves . . . . .	121
43.59.8.7 Financebars . . . . .	122
43.59.8.8 Fsteps . . . . .	122
43.59.8.9 Histeps . . . . .	122
43.59.8.10 Histograms . . . . .	122
43.59.8.10.1 Newhistogram . . . . .	124
43.59.8.11 Image . . . . .	124
43.59.8.12 Impulses . . . . .	124
43.59.8.13 Labels . . . . .	125
43.59.8.14 Lines . . . . .	125
43.59.8.15 Linespoints . . . . .	125
43.59.8.16 Points . . . . .	125
43.59.8.17 Steps . . . . .	125
43.59.8.18 Rgbimage . . . . .	125
43.59.8.19 Vectors . . . . .	125
43.59.8.20 Xerrorbars . . . . .	125
43.59.8.21 Xyerrorbars . . . . .	126
43.59.8.22 Yerrorbars . . . . .	126
43.59.8.23 Xerrorlines . . . . .	126
43.59.8.24 Xyerrorlines . . . . .	126

43.59.8.25 Yerrorlines . . . . .	126
43.60 Surface . . . . .	126
43.61 Table . . . . .	127
43.62 Terminal . . . . .	127
43.63 Termoption . . . . .	127
43.64 Tics . . . . .	128
43.65 Ticslevel . . . . .	128
43.66 Ticscale . . . . .	128
43.67 Timestamp . . . . .	128
43.68 Timefmt . . . . .	129
43.69 Title . . . . .	130
43.70 Tmargin . . . . .	130
43.71 Trange . . . . .	130
43.72 Urange . . . . .	130
43.73 Variables . . . . .	130
43.74 Version . . . . .	131
43.75 View . . . . .	131
43.76 Vrange . . . . .	131
43.77 X2data . . . . .	131
43.78 X2dtics . . . . .	132
43.79 X2label . . . . .	132
43.80 X2mtics . . . . .	132
43.81 X2range . . . . .	132
43.82 X2tics . . . . .	132
43.83 X2zeroaxis . . . . .	132
43.84 Xdata . . . . .	132
43.85 Xdtics . . . . .	133
43.86 Xlabel . . . . .	133
43.87 Xmtics . . . . .	134
43.88 Xrange . . . . .	134
43.89 Xtics . . . . .	135
43.90 Xyplane . . . . .	138
43.91 Xzeroaxis . . . . .	138
43.92 Y2data . . . . .	138
43.93 Y2dtics . . . . .	138
43.94 Y2label . . . . .	138
43.95 Y2mtics . . . . .	138
43.96 Y2range . . . . .	138
43.97 Y2tics . . . . .	138
43.98 Y2zeroaxis . . . . .	139
43.99 Ydata . . . . .	139
43.100 Ydtics . . . . .	139

43.10 <code>Ylabel</code>	139
43.10 <code>Ymtics</code>	139
43.10 <code>Yrange</code>	139
43.10 <code>Ytics</code>	139
43.10 <code>Yzeroaxis</code>	139
43.10 <code>Zdata</code>	139
43.10 <code>Zdtics</code>	139
43.10 <code>Zzeroaxis</code>	139
43.10 <code>Cbdata</code>	139
43.11 <code>Cbdtics</code>	140
43.11 <code>Zero</code>	140
43.11 <code>Zzeroaxis</code>	140
43.11 <code>Zlabel</code>	140
43.11 <code>Zmtics</code>	140
43.11 <code>Zrange</code>	141
43.11 <code>Ztics</code>	141
43.11 <code>Cblabel</code>	141
43.11 <code>Cbmtics</code>	141
43.11 <code>Cbrange</code>	141
43.12 <code>Cbtics</code>	141
<b>44 Shell</b>	<b>141</b>
<b>45 Splot</b>	<b>142</b>
45.1 Data-file	142
45.1.1 Binary matrix	143
45.1.2 Example datafile	144
45.1.3 Matrix_ascii	144
45.1.4 Matrix	145
45.2 Grid data	145
45.3 Splot overview	146
<b>46 System</b>	<b>146</b>
<b>47 Test</b>	<b>146</b>
<b>48 Unset</b>	<b>146</b>
<b>49 Update</b>	<b>146</b>
<b>III Terminal types</b>	<b>147</b>
<b>50 Terminal</b>	<b>147</b>
50.1 Aed767	147

50.2	Aifm . . . . .	147
50.3	Amiga . . . . .	148
50.4	Apollo . . . . .	148
50.5	Aqua . . . . .	148
50.6	Atari ST (via AES) . . . . .	148
50.7	Be . . . . .	149
50.7.1	Command-line_options . . . . .	149
50.7.2	Monochrome_options . . . . .	150
50.7.3	Color_resources . . . . .	150
50.7.4	Grayscale_resources . . . . .	150
50.7.5	Line_resources . . . . .	150
50.8	Cgi . . . . .	151
50.9	Cgm . . . . .	151
50.9.1	Font . . . . .	152
50.9.2	Fontsize . . . . .	154
50.9.3	Linewidth . . . . .	154
50.9.4	Rotate . . . . .	154
50.9.5	Solid . . . . .	154
50.9.6	Size . . . . .	154
50.9.7	Width . . . . .	154
50.9.8	Nofontlist . . . . .	155
50.10	Corel . . . . .	155
50.11	Debug . . . . .	155
50.12	Dospc . . . . .	155
50.13	Dumb . . . . .	155
50.14	Dxf . . . . .	156
50.15	Dxy800a . . . . .	156
50.16	Eepic . . . . .	156
50.17	Emf . . . . .	157
50.18	Emxvga . . . . .	157
50.19	Epslatex . . . . .	158
50.20	Epson-180dpi . . . . .	161
50.21	Excl . . . . .	161
50.22	Fig . . . . .	161
50.23	Ggi . . . . .	162
50.24	Gif . . . . .	163
50.25	Gnugraph(GNU plotutils) . . . . .	164
50.26	Gpic . . . . .	165
50.27	Gpr . . . . .	165
50.28	Grass . . . . .	166
50.29	Hercules . . . . .	166
50.30	Hp2623a . . . . .	166

50.31 Hp2648 . . . . .	166
50.32 Hp500c . . . . .	166
50.33 Hpgl . . . . .	166
50.34 Hpljii . . . . .	167
50.35 Hppj . . . . .	167
50.36 Imagen . . . . .	167
50.37 Iris4d . . . . .	168
50.38 Jpeg . . . . .	168
50.39 Kyo . . . . .	169
50.40 Latex . . . . .	169
50.41 Linux . . . . .	170
50.42 Macintosh . . . . .	170
50.43 Mf . . . . .	171
50.43.1 METAFONT Instructions . . . . .	171
50.44 Mgr . . . . .	172
50.45 Mif . . . . .	172
50.46 Mp . . . . .	172
50.46.1 Metapost Instructions . . . . .	174
50.47 Mtos . . . . .	175
50.48 Next . . . . .	175
50.49 Openstep (next) . . . . .	175
50.50 Pbm . . . . .	176
50.51 Pdf . . . . .	176
50.52 Pm . . . . .	177
50.53 Png . . . . .	177
50.54 Postscript . . . . .	178
50.54.1 Enhanced postscript . . . . .	180
50.54.2 Editing postscript . . . . .	181
50.54.3 Postscript fontfile . . . . .	182
50.54.4 Postscript prologue . . . . .	183
50.55 Pslatex and pstex . . . . .	183
50.56 Pstricks . . . . .	185
50.57 Qms . . . . .	185
50.58 Regis . . . . .	185
50.59 Rgip . . . . .	185
50.60 Sun . . . . .	186
50.61 Svg . . . . .	186
50.62 Svga . . . . .	186
50.63 Tek40 . . . . .	186
50.64 Tek410x . . . . .	187
50.65 Texdraw . . . . .	187
50.66 Tgif . . . . .	187

50.67 Tkcanvas . . . . .	188
50.68 Tpic . . . . .	189
50.69 Unixpc . . . . .	189
50.70 Unixplot . . . . .	189
50.71 Atari ST (via VDI) . . . . .	189
50.72 Vgagl . . . . .	190
50.73 VWS . . . . .	190
50.74 Vx384 . . . . .	190
50.75 Windows . . . . .	190
50.75.1 Graph-menu . . . . .	191
50.75.2 Printing . . . . .	191
50.75.3 Text-menu . . . . .	191
50.75.4 Wgnuplot.ini . . . . .	192
50.75.5 Windows3.0 . . . . .	193
50.76 Wxt . . . . .	193
50.77 X11 . . . . .	194
50.77.1 X11_fonts . . . . .	195
50.77.2 Command-line_options . . . . .	196
50.77.3 Monochrome_options . . . . .	197
50.77.4 Color_resources . . . . .	197
50.77.5 Grayscale_resources . . . . .	198
50.77.6 Line_resources . . . . .	198
50.77.7 X11 pm3d_resources . . . . .	199
50.77.8 X11 other_resources . . . . .	199
50.78 Xlib . . . . .	200
<b>IV Graphical User Interfaces</b>	<b>200</b>
<b>V Bugs</b>	<b>200</b>
<b>51 Gnuplot limitations</b>	<b>200</b>
<b>52 Specific terminals</b>	<b>201</b>
<b>53 External libraries</b>	<b>201</b>
<b>VI Index</b>	<b>201</b>

## Part I

# Gnuplot

## 1 Copyright

Copyright (C) 1986 - 1993, 1998, 2004, 2007 Thomas Williams, Colin Kelley

Permission to use, copy, and distribute this software and its documentation for any purpose with or without fee is hereby granted, provided that the above copyright notice appear in all copies and that both that copyright notice and this permission notice appear in supporting documentation.

Permission to modify the software is granted, but not the right to distribute the complete modified source code. Modifications are to be distributed as patches to the released version. Permission to distribute binaries produced by compiling modified sources is granted, provided you

1. distribute the corresponding source modifications from the released version in the form of a patch file along with the binaries,
2. add special version identification to distinguish your version in addition to the base release version number,
3. provide your name and address as the primary contact for the support of your modified version, and
4. retain our contact information in regard to use of the base software.

Permission to distribute the released version of the source code along with corresponding source modifications in the form of a patch file is granted with same provisions 2 through 4 for binary distributions.

This software is provided "as is" without express or implied warranty to the extent permitted by applicable law.

### AUTHORS

#### Original Software:

Thomas Williams, Colin Kelley.

#### Gnuplot 2.0 additions:

Russell Lang, Dave Kotz, John Campbell.

#### Gnuplot 3.0 additions:

Gershon Elber and many others.

#### Gnuplot 4.0 additions:

See list of contributors at head of this document.

(以下おおまかな訳; 訳は正しくないかも知れませんので詳しくは上記の原文を当たってください。訳者は責任を持ちません。)

Copyright (C) 1986 - 1993, 1998, 2004 Thomas Williams, Colin Kelley

このソフトウェアとその付属文書の使用、複製、配布の許可は、上記の著作権 (copyright) 表示が、全ての複製物に書かれていること、および著作権表示とこの許諾文の両方がその支援文書に書かれていることを条件とした上で、この文書により保証されます。

このソフトウェアの修正も認められています。しかし、修正を含む全ソースコードの配布の権利は認められません。修正はリリース版に対するパッチの形で配布しなければなりません。修正されたソースをコンパイルして作られたバイナリの配布は、以下の条件の元で認められます：

1. リリース版からのソースの修正部分を、パッチの形でバイナリと共に配布すること
2. ベースとなるリリース版と区別するために、そのバージョン番号に特別なバージョン指定子を付加すること
3. その修正版のサポート用に、あなたの名前とアクセス可能なアドレスとを提供すること
4. ベースとなるソフトウェアの使用に関しては、我々の連絡情報を保持し続けること

リリース版のソースコードを、パッチの形でのソースの修正と一緒に配布することは、バイナリ配布に関する条項 2 から 4 までの条件の元で許されます。

このソフトウェアは "あるがまま" 提供され、適用可能な法律で許められる範囲の保証を表明あるいは暗示していません。

#### 著者

##### オリジナルソフトウェア:

Thomas Williams, Colin Kelley.

##### Gnuplot 2.0 追加:

Russell Lang, Dave Kotz, John Campbell.

##### Gnuplot 3.0 追加:

Gershon Elber とその他の人々。

##### Gnuplot 4.0 追加:

この文書の最初にある寄与者 (contributors) のリスト参照。

## 2 Introduction

gnuplot は、コマンド入力方式の対話的な関数描画プログラムです。コマンドや関数名は大文字小文字を区別します。いずれのコマンドも、あいまいさの無い限りにおいて省略することができます。1行中にはセミコロン (;) で区切って複数のコマンドを書くことができます (ただし、load と call は最後のコマンドでなければなりません)。文字列は引用符を使って表します。引用符は、一重でも、二重でも構いません。例えば

```
load "filename"  
cd 'dir'
```

しかし、両者には微妙な違いがあります (詳細は syntax を参照してください)。

コマンドラインでの引数は gnuplot 用のコマンドの書かれたファイルの名前であるものとします。但し標準の X11 の引数は例外で、まず最初に処理されます。各ファイルはコマンドライン上で指定された順に load コマンドでロードされます。gnuplot は、最後に指定されたファイルを処理し終ると終了します。ファイルが 1 つも指定されていない場合は、gnuplot は対話モードになります。特別なファイル名 "-" は標準入力を表します。詳細は batch/interactive に関するヘルプを参照してください。

gnuplot のコマンドの多くは複数のオプションを持っています。Version 4 は以前の版に比べて、それらのオプションの順序に関しては許容的になっていますが、しかし順序に依存するものも残っています。オプションが認識されないといったエラーメッセージが現われたら、正確にドキュメントに書かれている通りの順番で指定して試してください。

コマンドは、複数行にまたがることができます。その場合は、最終行以外の全ての行の行末にバックスラッシュ (\) を書く必要があります。バックスラッシュは必ず各行 \*最後\* の文字でなくてはなりません。その結果としてバックスラッシュと、それに続く改行文字が存在しなかったかのように扱われます。つまり、改行文字がスペースの役をすることもありませんし、改行によってコメントが終了することもありません。ですから複数行にまたがる行の先頭をコメントアウトすると、そのコマンド全体がコメントアウトされることになります (comment 参照)。なお注意しますが、もし、複数行のコマンドのどこかでエラーが起きたとき、パーサはその場所を正確には指示することができませんし、また、正しい行に指示する必要もないでしょう。

このドキュメントにおいて、中括弧 ({} ) は省略可能な引数を表すものとし、縦棒 (|) は、互いに排他的な引数を区切るものとします。gnuplot のキーワードや help における項目名は、バッククオート (`) または可能な場合には boldface (太字) で表します。角括弧 (<>) は、それに対応するものに置き換えられるべきものを表します。多くの場合、オプションの引数にはそれが省略されるとデフォルトの値が使用されます。しかし、これらの場合必ずしも角括弧が中括弧で囲まれて書かれているわけではありません。

ある項目についてのヘルプが必要なときには、help に続けてその項目名を入力して下さい。または単に help や ? でもヘルプの項目のメニューが現われます。

初めて gnuplot を使う方は、plotting に関する説明から読みはじめると良いでしょう（現在使用中であれば help plotting と入力して下さい）。

simple.dem デモを参照してください。あるいは以下の Web ページにも他のデモと一緒にあります。

<http://www.gnuplot.info/demo/simple.html>

### 3 Seeking-assistance

gnuplot ユーザのためのメーリングリストがあります。しかし、ニュースグループ  
`comp.graphics.apps.gnuplot`

は、そのメーリングリストと同等であることに注意してください（どちらにも同じメッセージが流れます）。私達はメーリングリストに参加するより、むしろニュースグループのメッセージを読むことを勧めます。メーリングリストに参加する方法に関しては、SourceForge にある gnuplot の開発 Web サイト

<http://sourceforge.net/projects/gnuplot>

を参照してください。

メーリングリストメンバーへのメールアドレス:

`gnuplot-info@lists.sourceforge.net`

バグリポート、ソースの改良等は次のところへ:

`gnuplot-bugs@lists.sourceforge.net`

テスト版に関するメーリングリスト:

`gnuplot-beta@lists.sourceforge.net`

公式の（そしてたまに更新が遅れることもある）WWW ページもあります。

<http://www.gnuplot.info>

助けを求める前に、次をチェックしてください:

FAQ（度々聞かれる質問； Frequently Asked Questions）の一覧

何か質問を投稿するときは、あなたが使用している gnuplot のバージョン、実行マシン、オペレーティングシステム、といった全ての情報を含むようにしてください。その問題を引き起こす「小さい」スクリプトがあればなお良いです。その場合、データファイルのプロットよりも関数のプロットの方がより良いです。もし、gnuplot-info メールをするなら、そのメーリングリストの購読をしているかどうかを述べてください。そうすれば、ニュースを見たユーザはあなたへの返事をメールで出せば良いことが分かるでしょうから。そのような記事のポストの form が WWW サイトにあります。

### 4 New features introduced in version 4.2

gnuplot version 4.2 は、直前の公式バージョン 4.0 以降に追加されたたくさんの機能を提供します。この節では、主な追加機能の一覧と、変更や細かい新機能の一部を紹介します。より網羅的な一覧は、ファイル NEWS を参照してください。

#### 4.1 新しい描画スタイル

##### 4.1.1 ヒストグラム

ヒストグラムや棒グラフが生成できます。histograms 参照。

##### 4.1.2 ラベル描画

以下に述べる新しい datastrings の機能との組み合わせによって、gnuplot に曲線の各頂点にラベルを付けさせることができます。labels 参照。

### 4.1.3 画像データ

スタイル `image` や `rgbimage` により、2 次元や 3 次元グラフ内に 2 次元画像 (ASCII 形式、または binary 形式のファイル) を描画、あるいは写像することができます。`image`, `rgbimage` 参照。

### 4.1.4 閉曲線の塗りつぶし

描画スタイル `fillstyle` は、入力された 2 つの曲線で囲まれた領域を、色やパターンで塗りつぶすことを可能にします。`filledcurves` 参照。

### 4.1.5 ベクトル

`gnuplot` は、2 次元グラフでは 4 列、3 次元グラフでは 6 列のデータを使って、小さい矢先を持つベクトルの描画が行えます。`vectors` 参照。

## 4.2 バイナリデータファイルからの入力

現在の `gnuplot` は、行列 (matrix) 形式のバイナリや general binary (これまで `gnuplot` は binary matrix 形式のみをサポートしていました) を含む一般的な binary データの入力が可能になっています。色々な matrix ファイルフォーマット (`gpbin`, `edf`, `avs`) が自動的に認識されるようになっています。バイナリデータファイルは、`image` や `rgbimage` の描画で特に有用です。`binary`, `binary general filetype` 参照。

## 4.3 新しい描画要素

### 4.3.1 RGB 色指定

すべての描画要素に対して、定義済み線種 (`linetype`) の指定の代わりに明示的に RGB の色を指定することができます。`colorspec` 参照。

### 4.3.2 任意の長方形描画

2 次元グラフ内の任意の位置に、お好みの塗り潰しスタイルや境界を持つ長方形を描画することができます。`set object rectangle` 参照。

## 4.4 文字列処理

### 4.4.1 データファイルからの文字列や文字データの入力

現在、`gnuplot` はデータファイルからテキスト部分を読み、処理できます。`datastrings` 参照。

### 4.4.2 ユーザ定義文字列変数、文字列用の演算子や関数

文字列変数と文字列関数が導入されました。これまで文字列定数を要求していた `gnuplot` コマンドは、現在はそのほとんどが、文字列変数、文字列式、文字列を返す関数を受けつけます。`string variables` 参照。

## 4.5 マクロ

`gnuplot` は '@stringvariablename' によるコマンドラインでのマクロの展開をサポートします。`macros` 参照。

## 4.6 複数グラフの 1 ページ自動配置描画

multiplot モードは現在は、各グラフに対してもいちいちサイズや位置を設定することなしに、単純な複数グラフの描画を生成することができます。multiplot 参照。

## 4.7 内部変数

gnuplot は現在、GPVAL\_TERM, GPVAL\_X\_MIN 等のいくつかの "参照専用" の変数の値を見せます。gnuplot-defined variables 参照。

## 4.8 新しい/修正された出力形式

### 4.8.1 wxt

wxt 出力形式は、複数のプラットホームで動作する、スクリーン描画用の対話型出力形式です。これはユーザインターフェースには wxWidgets ライブリを使用し、実際の描画には Pango と関連する Cairo を使用します。これは線や文字にアンチエイリアスを利用した素晴らしい描画を生成します。この出力形式はマウス、pm3d 描画、画像 (image) の描画、拡張文字処理 (enhanced text) を含む、gnuplot の能力のすべてをサポートしています。

### 4.8.2 emf

emf 出力形式は Enhanced Metafile Format (EMF) ファイルを生成します。このファイル形式は MS Win32 システムでは標準的なメタファイル形式です。

### 4.8.3 gif, jpeg, png

gd ライブリを利用するこれらの出力形式のソースコードは統合されました。gif 出力形式には、複数のグラフからアニメーション GIF を作成する機能もあります。

### 4.8.4 postscript

postscript 出力形式は、prologue ファイルを読みこむことができます。prologue ファイルには、ユーザ定義セクション、例えば文字エンコーディングに関するものを追加することができます。postscript prologue 参照。

### 4.8.5 ai

Adobe Illustrator 用のドライバ ai は時代遅れです。Adobe Illustrator は PostScript ファイルを認識するので、set terminal post level1 ... を代わりに使うべきです。

### 4.8.6 epslatex, pslatex, pstex

PostScript コマンドを伴う latex ファイルへの出力をサポートするこれらの出力形式のコードは統合され、多くのオプションが postscript 出力形式のものと同一になっています。

### 4.8.7 windows

windows 出力形式は、現在は拡張文字処理 (enhanced text) モードをサポートしています。

## 4.9 Canvas size

gnuplot の以前の版では、`set size` の値を、出力する描画領域（キャンバス）のサイズを制御するのにも使っていた出力形式もありました（すべての出力形式がそうだというわけではありません）。この仕様は現在は非推奨です。gnuplot の将来の版では、オプション `set size` と `set term <foo> size` は、お互いを補う効果を持つことになるでしょう：

`set term <terminal_type> size <XX>, <YY>` は、出力ファイルのサイズ、または "キャンバス" のサイズを制御します。サイズパラメータの有効な値については、個々の出力形式のヘルプを参照してください。デフォルトでは、グラフはそのキャンバス全体に描画されます。

`set size <XX>, <YY>` は、描画自体をキャンバスのサイズに対して相対的に伸縮させます。1 より小さい伸縮値を指定すると、グラフはキャンバス全体を埋めず、1 より大きい伸縮値を指定すると、グラフの一部分のみがキャンバス全体に合うように描画されます。1 より大きい伸縮値を指定すると、ある出力形式では問題が起こるかもしれませんことに注意してください。

例：

```
set term pbm size 600, 400
set size 0.5, 0.5
set output "figure.pbm"
plot "data" with lines
```

このコマンドは、幅 600 ピクセル、高さ 400 ピクセルの出力ファイル"figure.pbm" を生成します。グラフはキャンバスの中の左下に置かれます。これは、`multiplot` モードが常に行ってきた方法とは矛盾していませんが、しかし単一のグラフに対する version 4.0 のあるドライバの挙動とは異っています。

4.2 のほとんどの出力形式は、この新しい規則に従っていますが、png/jpeg/gif ドライバに関してはコンパイル時のオプションによって変わります。`backwards compatibility` 参照。

## 5 Backwards compatibility

以前の版で使われていたいくつかの書式は gnuplot 4.0 で非推奨となりましたが、それは今でも認識はされ続けています。バージョン 4.2 もデフォルトでは非推奨の書式を認識していますが、これは現在コンパイル時に制御され、以下のようにして無効にできます：

```
./configure --disable-backwards-compatibility
```

注意：非推奨の書式は、将来の版では、完全に無効になってしまうかもしれません。

一つの主な違いは、複雑なコマンドのあいまいさを避けるような必須キーワードの導入です。特に注意すべき点は、オフセット、線種、点種に対する生の数の使用でした。具体例：

非推奨：

```
set data linespoints
plot <foo> 2 4
```

新：

```
set style data linespoints
plot <foo> linetype 2 pointtype 4
```

その他の主な互換性の部分は、`set size` と `set term <foo> size` の相互作用に関する部分です。gnuplot の以前の版では、`set size` に関するさまざまな出力形式の挙動には、不合理な点が含まれていました。今後、すべての出力形式で統一性が図られる予定です。

バージョン 4.2 では、ほとんどの出力形式が新しい規則（`set size` は出力される描画領域の大きさには影響を与えない）に従っていますが、png/gif/jpeg 出力形式は、`--disable-backwards-compatibility` なしで `configure` したのであれば、古い非推奨の仕様になります。gnuplot の次の版では、すべての出力形式が新しい規則に従います。よって、`set size` を png/gif/jpeg 出力ファイルのピクセルサイズを制御するのには使用しないことを強く勧めます。代わりに `set term png size <XX>, <YY>` を使ってください。

`set size`, `set term size`、および個々の出力形式のヘルプも参照してください。

## 6 Features introduced in version 4.0

gnuplot version 4.0 は、直前の公式バージョン 3.7 以降に追加されたたくさんの特徴を含んでいます。以下に要約します。

### 6.1 対話型出力形式上のマウスとホットキーのサポート

マウス、またはホットキーによる現在の描画への作用が、X11, OS/2 プレゼンテーションマネージャ、ggi, Windows, wxWidgets 出力ドライバでサポートされています。マウス動作に関する詳細は `mouse input` を、ホットキーに関する詳細は `bind` を参照してください。また、ggi, pm, windows, wxt, x11 それぞれのマウス動作に関する記述も参照してください。

サンプルスクリプト: `mousevariables.dem`

### 6.2 新しい出力形式

`aqua`: Mac OS X 用の新しい出力形式。AquaTerm 1.0 以降が必要。

`epslatex`: 新しい出力形式。`latex` 文書中に取り込むための `eps` 画像の生成を行ないます。

`gif`: `png/jpeg` 出力形式に統一。`libgd` が必要。

`ggi`: Linux のための新しいフルスクリーン対話型出力形式。General Graphics Interface ライブラリへのインターフェース。

`pdf`: Adobe Portable Document Format の出力を行なう新しい出力形式。`ibpdf` が必要です。

`png` および `jpeg`: `libgd` を使う新しいドライバによる GIF, PNG, JPEG 画像出力のサポート。新しいドライバは、TrueType フォントなど、以前の PNG ドライバよりもより多くの機能をサポートします。`libgd` が必要です。

`svg`: 縮尺可能なベクトルグラフィックス (Scalable Vector Graphics) を出力する新しい出力形式。

### 6.3 新しい描画スタイル `pm3d`

コマンド `splot` は現在では、2 次元の色地図や 3 次元曲面を、灰色階調やカラーパレットで連続階調で塗ることができます。`set pm3d`, `set palette`, `set cbrange`, `set view map`, `set colorbox`, `test palette` 参照。

サンプルスクリプト: `pm3d.dem` `pm3dcolors.dem` `pm3dgamma.dem`

### 6.4 塗りつぶされた箱 (filled boxes)

`boxes` 描画スタイルを含む幾つかの描画スタイルに対して、カラーによるベタ塗り、またはパターンによる塗りつぶしを設定できます。`boxes`, `boxerrorbars`, `boxxyerrorbars`, `candlesticks`, `set style fill` 参照。

サンプルスクリプト: `fillstyle.dem` `candlesticks.dem`

### 6.5 新しい描画オプション `smooth frequency`

データの近似や補間にに関する幾つかのルーチンが用意されていて、それで入力データをフィルタできます。`smooth`, `frequency`, `unique` 参照。

サンプルスクリプト: `step.dem` `mgr.dem`

### 6.6 文字列オプションの改良

文字列のラベルを生成するほとんどの gnuplot の描画コマンドは、現在は文字色、フォント、大きさ、回転角度の指定を受け付けます。`set label` 参照。しかし、全ての出力形式がこれらをサポートしているわけ

ではありません。拡張テキストモードは postscript, pm 出力形式では既に有効だったものですがそれが、他の出力形式にも拡張されました。この機能をサポートしているのは、aqua, dump, jpeg, pdf, pm, png, postscript, x11, windows, wxt などです。enhanced text を参照。

サンプルスクリプト: textcolor.dem textrotate.dem

## 6.7 文字コード化 (encodings) のサポートの追加

postscript, x11, pm などのいくつかの出力形式で、文字コード化 (encodings) のサポートがいくつか追加されています: ISO 8859-1 (Latin 1), ISO 8859-2 (Latin 2), ISO 8859-15 (ユーロ記号を含む 8859-1 の亜種), KOI8-R と KOI8-U (キリル文字), および幾つかのコードページ。詳細は encoding 参照。

## 6.8 矢 (arrows)

片方、あるいは両端に矢先のついた矢を、plot with vectors スタイルを用いて、データファイルから、あるいはコマンドラインから別々に描画グラフ上に配置することができます。set style arrow, plotting styles vectors 参照。

サンプルスクリプト: arrowstyle.dem vector.dem

## 6.9 データファイル形式

新しい set datafile コマンドは、入力データの形式に関する情報を指定するのに使えますが、例えばフィールドを分離する文字やコメント行を意味する文字、欠けたデータを示す文字の指定などに使えます。gnuplot は、現在は datafile 形式の設定に基づいて、空白を持つような文字列フィールドも単一項目とみなすことができます。これにより表計算ソフトが出力するような csv (コンマ分離形式) ファイルの入力も可能になります。set datafile 参照。オプション binary も参照のこと (version 4.2 で導入されました)。

## 6.10 新しいコマンド

set view map は、3D 曲面描画の真上からの 2D 射影を選択します。

set term push と set term pop は現在の出力形式の保存/復帰です。

load と save コマンドはそれぞれパイプ入力/出力を受け付けます。

## 6.11 他の変更と追加

現在は set no<something> よりむしろ unset <something> を使ってください。以前の形は推奨されていません。Version 4.2 は古い書式も受け付けますが、そのような後方互換性は将来の版では失われるでしょう。

set <something> <style> の形式のコマンドも、より一般的な形式である set style <something> <options> に置き換えられています。現在は多くの描画要素自身がスタイルオプションを持つようになっています。例えば矢 (arrows)、塗りつぶされた領域 (filled area)、線 (lines)、点 (points) などです。入力データの設定や、書式に関するスタイルもあります。set style, set decimalsign, set datastyle 参照。

MS Windows パッケージには実行ファイル pgnuplot.exe が追加されていますが、これは標準入力からのパイプをサポートするためのものです。そのようなパイプ機能は、そのシステムの標準ではグラフィックアプリケーションには提供されていません。

## 6.12 付属ドキュメント

docs/psdocs/ ディレクトリには、gnuplot の PostScript 出力ガイドといいくつかのコード化 (encoding) におけるポストスクリプトの記号の一覧がありますが、そこからも新しい情報を得ることができます。

FAQ も改良されています。公の場で質問する前に、是非それを一読してください。

demo/ ディレクトリにはとてもたくさんの新しいデモファイル \*.dem があります。助けを請う前に、例えば

```
load "all.dem"
```

のようにして是非それらを実行してみてください。デモスクリプトで生成される描画は以下でも見ることができます。

```
http://www.gnuplot.info/demo/
```

## 7 Batch/Interactive Operation

gnuplot は多くのシステム上で、バッチ処理形式、あるいは対話型のどちらの形式でも実行でき、それらを組み合わせることも可能です。

コマンドライン引数は gnuplot コマンドを含むファイルのファイル名であると解釈されます（先に指定される標準的な X11 用コマンドの引数を除いて）。各ファイルは、指定された順に load コマンドで読み込まれます。最後のファイルを実行した後は gnuplot は終了します。ロードファイルを指定しない場合は、gnuplot は対話モードに入ります。特別なファイル名 “-” は標準入力を指定するのに使われます。

exit と quit はどちらも現在のコマンドファイルを終了し、まだ全てのファイルが終っていなければ、次のファイルを load するのに使われます。

例:

対話を開始する:

```
gnuplot
```

2 つのコマンドファイル "input1", "input2" を使ってバッチ処理を行なう:

```
gnuplot input1 input2
```

初期化ファイル "header" の後、対話型モードを起動し、その後別のコマンドファイル "tailer" を実行する:

```
gnuplot header - trailer
```

## 8 Command-line-editing

コマンドライン編集は Unix, Atari, VMS, MS-DOS and OS/2 上の gnuplot でサポートされています。履歴（ヒストリ）機能で、以前のコマンドを編集し再実行することも出来ます。コマンドラインの編集後は、カーソルがどこにいても改行や復帰キーによって行全体が入力されます。

(gnuplot における readline 関数は、GNU Bash や GNU Emacs で使われる readline 関数と全く同じではありません。もし、GNU 版を望むなら、コンパイル時に gnuplot 版の代わりに選択できます)

編集コマンドは以下の通りです:

コマンド行編集コマンド	
文字	機能
行編集	
^B	1 文字前へ戻す
^F	1 文字先へ進める
^A	行の先頭へ移動
^E	行の最後へ移動
^H, DEL	直前の文字を削除
^D	現在位置の文字を削除
^K	現在位置から行末まで削除
^L, ^R	壊れた表示の行を再表示
^U	行全体の削除
^W	カーソル手前の単語から行末まで削除
履歴	
^P	前の履歴へ移動
^N	次の履歴へ移動

IBM PC では、行編集用に DOSEDIT とか CED などの TSR (常駐) プログラムを使いたいと思うかも知れません。デフォルトの makefile はこれを仮定していて、gnuplot はデフォルトでは行編集機能無しでコンパイルされます。もし gnuplot の行編集機能を使用したければ、makefile の READLINE をセットしてリンクファイルとして readline.obj を追加してください。IBM PC と Atari 版で readline を使う場合は以下のキーも使えます。

矢印キー	機能
左 ( )	$\text{^B}$ と同じ
右 ( )	$\text{^F}$ と同じ
Ctrl + 左	$\text{^A}$ と同じ
Ctrl + 右	$\text{^E}$ と同じ
上 ( )	$\text{^P}$ と同じ
下 ( )	$\text{^N}$ と同じ

Atari 版の readline は更にいくつかのエイリアスが定義されています:

キー	機能
Undo	$\text{^L}$ と同じ
Home	$\text{^A}$ と同じ
Ctrl Home	$\text{^E}$ と同じ
Esc	$\text{^U}$ と同じ
Help	'help' + return
Ctrl Help	'help'

## 9 Comments

コメントは次のように実装されています: 文字 '#' は、行中のたいていの場所に書くことができます。このとき gnuplot はその行の残りの部分を無視します。ただし、引用符の中、数(複素数を含む)の中、コマンド置換(command substitution)の中などではこの効果がありません。簡単に言うと、意味のあるような使い方をしさえすれば、正しく動作すると言うことです。

データファイル中のコメント文字の指定は、set datafile commentschars を参照してください。

## 10 Coordinates

コマンド set arrow, set key, set label, set object はグラフ上の任意の位置が指定できます。その位置は以下の書式で指定します:

```
{<system>} <x>, {<system>} <y> ,{<system>} <z>}
```

各座標系指定 <system> には、first, second, graph, screen, character のいずれかが入ります。

first は左と下の軸で定義される x,y (3D の場合は z も) の座標系を使用します。second は第 2 軸(上と右の軸)を使用します。graph はグラフ描画領域内の相対的位置を指定し、左下が 0,0 で右上が 1,1 (splot の場合はグラフ描画領域内の左下が 0,0,0 で、土台の位置は負の z の値を使用します。set ticslevel 参照)となります。screen は表示範囲内(範囲全体であり、set size で選択される一部分ではありません)を指定し、左下が 0,0 で右上が 1,1 となります。character は、画面領域の左下(screen 0,0)からの、文字の幅、文字の高さでの位置を与えます。よって、character 座標は、選択されたフォントのサイズに依存します。

x の座標系が指定されていない場合は first が使われます。y の座標系が指定されていない場合は x に対する座標系が使用されます。

与える座標が絶対的な位置ではなくて相対的な値である場合もあります(例えば set arrow ... rto の 2 番目の数値)。そのほとんどが、与えられた数値を最初の位置に対する差として使います。与えられた座標が対数軸内にある場合は、その相対的な値は積因子として解釈されます。例えば

```
set logscale x
set arrow 100,5 rto 10,2
```

は、*x* 軸が対数軸で *y* 軸が線形の軸なので、100,5 の位置から 1000,7 の位置への矢印を書くことになります。

一つ（あるいはそれ以上）の軸が時間軸である場合、timefmt の書式文字列に従って、引用符で囲まれた時間文字列で適切な座標を指定する必要があります。set xdata, set timefmt を参照してください。また、gnuplot は整数表記も認めていて、その場合その整数は 2000 年 1 月 1 日からの秒数と解釈されます。

## 11 Datastrings

configure 時のオプション `-enable-datastrings` は、gnuplot にデータファイル中のテキスト部分を (text field) を読み込ませ、処理させることを許可します。テキスト部分は、ホワイトスペース (空白やタブ) を含まない任意の印字可能な文字列、あるいは 2 重引用符で囲まれた任意の文字列 (ホワイトスペースが含まれても良い)、のいずれかの形です。データファイルに次の例のような行が含まれている場合、それは 4 つの列を含み、3 列目がテキスト部分であると見なされます:

```
1.000 2.000 "Third column is all of this text" 4.00
```

テキスト部分は 2 次元や 3 次元描画内で例えば以下のように使用されます:

```
plot 'datafile' using 1:2:4 with labels
splot 'datafile' using 1:2:3:4 with labels
```

テキスト部分の列データは 1 つ、または複数の描画軸の目盛りのラベルとして使用できます。次の例は、入力データの 3 列目と 4 列目を (X,Y) 座標として取り出し、それらの点の列を結ぶ線分を描画します。しかしこの場合 gnuplot は、*x* 軸に沿って標準的に間の空いた数字ラベルのついた目盛り刻みをつけるのではなく、入力データファイルの 1 行目の X 座標の位置に、目盛り刻みと文字列を *x* 軸に沿ってつけて行きます。

```
set xtics
plot 'datafile' using 3:4:xticlabels(1) with linespoints
```

入力データの列の最初のエントリをテキスト部分と解釈するもう一つのオプションがあり、それはテキスト部分を、その描画した列のデータの説明 (key) のタイトル部分として使用します。次の例は、先頭の行の 2 列目の部分を説明ボックス内のタイトルを生成するのに使用し、その後の列の 2,4 列目は要求された曲線を描画するのに処理されます:

```
plot 'datafile' using 1:(f($2)/$4) title 2 with lines
```

```
set style labels, using xticlabels, plot title, using 参照。
```

## 12 Environment

gnuplot は多くのシェル環境変数を認識します。必須のものはありませんが、使えば便利になるかも知れません。

GNUTERM が定義されている場合、それは使用される出力形式 (terminal) の名前として使われます。これは gnuplot が起動時に見つけた出力形式に優先して使用されますが、.gnuplot (またはそれに相当する) スタートアップファイル (start-up 参照) による指定や、当り前のことですが、その後に明示的に指定した物の方が優先されます。

Unix, AmigaOS, AtariTOS, MS-DOS, OS/2 では、GNUHELP にヘルプファイル (gnuplot.gih) のパス名を定義しておくことができます。

VMS では、論理名 GNUPLOT\$HELP を gnuplot のヘルプライブラリの名前として定義します。gnuplot のヘルプは任意のシステムのヘルプライブラリに入れることができ、gnuplot の内部からでも外部からでも参照して構いません。

Unixにおいては、カレントディレクトリに .gnuplot というファイルがない場合には、HOME に定義されたディレクトリを探します。AmigaOS, AtariTOS, MS-DOS, Windows, OS/2 では GNUPLOT がその役割に使われます。Windows では、NT 専用の変数 USERPROFILE も参照されます。VMS では SYS\$LOGIN です。help start-up と打ってみてください。

Unixにおいては、PAGERがヘルプメッセージの出力用のフィルタとして使われます。

Unix, AtariTOS, AmigaOSでは、SHELLがshellコマンドの際に使われます。MS-DOS, OS/2ではCOMSPECがshellコマンドの際に使われます。

MS-DOSでBGIまたはWatcomインターフェースが使われている場合、PCTRMが、使用するモニタがサポートする最大解像度を指示するのに使われます。PCTMはS<最大水平解像度>のように指定します。例えば、モニタの最大解像度が800x600ならば、以下のように指定します:

```
set PCTRM=S800
```

PCTRMが設定されていなければ、標準的なVGA(640x480)が使われます。

FIT\_SCRIPTは、当てはめ(fit)が中断されたときに実行するgnuplotコマンドの指定に使われます。fitを参照してください。FIT\_LOGは当てはめによるログファイルのデフォルトのファイル名の指定に使われます。

GNUPLOT\_LIBは、データやコマンドファイルの検索ディレクトリを追加定義するのに使われます。その変数は、一つのディレクトリ名かまたは複数のディレクトリ名を書くことができますが、ディレクトリの区切りはプラットホーム毎に違います。例えばUnixでは':'で、DOS,Windows,OS/2,Amigaでは';'です。GNUPLOT\_LIBの値は変数loadpathに追加されますが、それはsaveやsave setコマンドでは保存されません。

出力ドライバの中にはgdライブラリ経由でTrueTypeフォントを扱えるものもいくつかあります。これらのドライバのフォント検索パスは、環境変数GDFONTPATHで制御できます。さらに、それらのドライバでのデフォルトのフォントは環境変数GNUPLOT\_DEFAULT\_GDFONTで制御できます。

postscript出力ドライバは自分で持っているフォント検索パスを使いますが、それは環境変数GNUPLOT\_FONTPATHで制御できます。書式はGNUPLOT\_LIBと同じです。GNUPLOT\_FONTPATHの値は変数fontpathに追加されますが、それはsaveやsave setコマンドでは保存されません。

PostScriptドライバは、外部(組み込まれていない)定義ファイルを使うために環境変数GNUPLOT\_PS\_DIRを利用します。インストール時の作業により、gnuplotにはそれらのファイルのコピーが組み込まれているか、または単にそれらのファイルのあるデフォルトのパスが埋め込まれるかが変わります。カスタマイズした定義ファイルでPostScript出力をテストする場合はこの環境変数を使ってください。postscript prologueも参照してください。

## 13 Expressions

基本的にはC,FORTRAN,Pascal,BASICにおいて利用可能な数学表現を使用できます。演算子の優先順位はC言語の仕様に従います。数式中の空白文字とタブ文字は無視されます。

複素数の定数は{<real>,<imag>}と表現します。ここで<real>と<imag>(実部、虚部)は数値定数である必要があります。例えば{3,2}は $3 + 2i$ をあらわし、{0,1}は'i'自身を表します。これらには明示的に中カッコを使う必要があります。

gnuplotは"実数"と"整数"演算をFORTRANやCのように扱うということに注意してください。"1","-10"などは整数と見なされ、"1.0","-10.0","1e1",3.5e-1などは実数と見なされます。これら2つのもっとも重要な違いは割算です。整数の割算は切り捨てられます: $5/2 = 2$ 。実数はそうではありません: $5.0/2.0 = 2.5$ 。それらが混在した式の場合、計算の前に整数は実数に"拡張"されます: $5/2e0 = 2.5$ 。負の整数を正の整数で割る場合、その値はコンパイラによって変わります。"print -5/2"として、あなたのシステムが-2と-3のどちらを答えとするかを確認してください。

数式"1/0"は"未定義値(undefined)"フラグを生成し、それによりその点は無視されます。ternary演算子(三項演算子)の項にその例があります。

複素数表現の実数部分、虚数部分は、どんな形で入力されても常に実数です:{3,2}の"3"と"2"は実数であり、整数ではありません。

gnuplotは文字列に対する単純な演算、および文字列変数も利用できます。例えば式("A" . "B" eq "AB")は真と評価されますが、これは文字列の結合演算子と文字列の等号演算子を意味しています。

数としての値を含む文字列は、それが数式で利用された場合は、対応する整数や実数に変換されます。よって、("3" + "4" == 7)や(6.78 == "6.78")はどちらも真になります。整数は、それが文字列結合演算子で使われた場合は文字列に変換されますが、実数や複素数はダメです。典型的な例は、ファイル名や他の文字列内に整数を使う場合でしょう: 例えば("file" . 4 eq "file4")は真です。

後置指定する範囲記述子 `[beg:end]` によって、部分文字列を指定することができます。例えば、`"ABCDEF"[3:4] == "CD"` で、`"ABCDEF"[4:]* == "DEF"` です。書式 `"string"[beg:end]` は、文字列値の組み込み関数 `substr("strings",beg,end)` を呼ぶこととほぼ同じですが、関数呼び出しでは `beg`, `end` は省略することはできません。

### 13.1 Functions

gnuplot の関数は、Unix 数学ライブラリの関数とほぼ同じですが、特に注意がなければ全ての関数が整数、実数、複素数の引数を取ることができます。

度、あるいはラジアンのどちらかで角度を引数としたり戻り値としたりする関数 (`sin(x)`, `cos(x)`, `tan(x)`, `asin(x)`, `acos(x)`, `atan(x)`, `atan2(x)`, `arg(z)`) に対しては、その単位は `set angles` で選択でき、デフォルトはラジアンです。

数学ライブラリ関数		
関数	引数	戻り値
<code>abs(x)</code>	任意	$x$ の絶対値, $ x $ ; 同じ型
<code>abs(x)</code>	複素数	$x$ の長さ, $\sqrt{\text{real}(x)^2 + \text{imag}(x)^2}$
<code>acos(x)</code>	任意	$\cos^{-1} x$ (アークコサイン)
<code>acosh(x)</code>	任意	ラジアンでの $\cosh^{-1} x$ (逆双曲余弦)
<code>arg(x)</code>	複素数	$x$ の偏角
<code>asin(x)</code>	任意	$\sin^{-1} x$ (アークサイン)
<code>asinh(x)</code>	任意	ラジアンでの $\sinh^{-1} x$ (逆双曲正弦)
<code>atan(x)</code>	任意	$\tan^{-1} x$ (アークタンジェント)
<code>atan2(y,x)</code>	整数または実数	$\tan^{-1}(y/x)$ (アークタンジェント)
<code>atanh(x)</code>	任意	ラジアンでの $\tanh^{-1} x$ (逆双曲正接)
<code>besj0(x)</code>	整数または実数	$x$ の $j_0$ 次ベッセル関数
<code>besj1(x)</code>	整数または実数	$x$ の $j_1$ 次ベッセル関数
<code>besy0(x)</code>	整数または実数	$x$ の $y_0$ 次ベッセル関数
<code>besy1(x)</code>	整数または実数	$x$ の $y_1$ 次ベッセル関数
<code>ceil(x)</code>	任意	$\lceil x \rceil$ , $x$ 以上の最小の整数 (real part)
<code>cos(x)</code>	任意	$x$ のコサイン $\cos x$
<code>cosh(x)</code>	任意	$\cosh x$ , $x$ のハイパボリックコサイン
<code>erf(x)</code>	任意	$\text{erf}(\text{real}(x))$ , $x$ の実部の誤差関数
<code>erfc(x)</code>	任意	$\text{erfc}(\text{real}(x))$ , $1.0 - (\text{erf}(\text{real}(x)))$
<code>exp(x)</code>	任意	$e^x$ , $x$ の指数関数
<code>floor(x)</code>	任意	$\lfloor x \rfloor$ , $x$ (の実部) 以下の最大の整数
<code>gamma(x)</code>	任意	$\text{gamma}(\text{real}(x))$ , $x$ の実部のガンマ関数
<code>ibeta(p,q,x)</code>	任意	$\text{ibeta}(\text{real}(p, q, x))$ , $p, q, x$ の実部の不完全ベータ関数
<code>inverf(x)</code>	任意	$x$ の実部の逆誤差関数
<code>igamma(a,x)</code>	任意	$\text{igamma}(\text{real}(a, x))$ , $a, x$ の実部の不完全ガンマ関数
<code>imag(x)</code>	複素数	$x$ の虚数部分 (実数)
<code>invnorm(x)</code>	任意	$x$ の実部の逆正規分布関数
<code>int(x)</code>	実数	$x$ の実数部分 (0 に向かって丸め)
<code>lambertw(x)</code>	実数	Lambert W 関数
<code>lgamma(x)</code>	任意	$\text{lgamma}(\text{real}(x))$ , $x$ の実部のガンマ対数関数
<code>log(x)</code>	任意	$\log_e x$ , $x$ の自然対数 (底 $e$ )
<code>log10(x)</code>	任意	$\log_{10} x$ , $x$ の対数 (底 10)
<code>norm(x)</code>	任意	$x$ の実部の正規分布 (ガウス分布) 関数
<code>rand(x)</code>	任意	$\text{rand}(\text{real}(x))$ , 疑似乱数生成器
<code>real(x)</code>	任意	$x$ の実部
<code>sgn(x)</code>	任意	$x > 0$ なら 1, $x < 0$ なら -1, $x = 0$ なら 0. $x$ の虚部は無視
<code>sin(x)</code>	任意	$\sin x$ , $x$ のサイン
<code>sinh(x)</code>	任意	$\sinh x$ , $x$ のハイパボリックサイン
<code>sqrt(x)</code>	任意	$\sqrt{x}$ , $x$ の平方根
<code>tan(x)</code>	任意	$\tan x$ , $x$ のタンジェント
<code>tanh(x)</code>	任意	$\tanh x$ , $x$ のハイパボリックタンジェント

文字列関数		
関数	引数	返り値
gprintf("format",x)	任意	gnuplot の書式解析器を適用した結果の文字列
sprintf("format",x,...)	複数個	C 言語の sprintf の返す文字列
strlen("string")	文字列	文字列の長さ (整数)
strstrt("string","key")	文字列	部分文字列 "key" が現れる先頭位置
substr("string",beg,end)	複数個	文字列 "string"[beg:end]
system("command")	文字列	出力を持つシェルコマンド文字列
word("string",n)	文字列, 整数	文字列 "string" の n 番目の文字列
words("string")	文字列	文字列 "string" 中の単語数

他の gnuplot の関数		
関数	引数	返り値
column(x)	整数	データファイル操作での x 列目
defined(X)	変数名	[非推奨] X が定義されていれば 1 そうでなければ 0 を返す
exists("X")	"変数名"	変数名 X が定義されていれば 1 そうでなければ 0 を返す
stringcolumn(x)	整数	文字列としての x 列目の内容
timecolumn(x)	整数	データファイル処理中の日時データ列 x
tm_hour(x)	整数	時
tm_mday(x)	整数	日
tm_min(x)	整数	分
tm_mon(x)	整数	月
tm_sec(x)	整数	秒
tm_wday(x)	整数	その週の何日目
tm_yday(x)	整数	その年の何日目
tm_year(x)	整数	西暦
valid(x)	整数	データ中の column(x) の正当性

以下も参照してください。

airfoil.dem: 関数と複素変数を翼に使ったデモ

### 13.1.1 Random number generator

組み込み関数 rand(x) の挙動は version 3.81 で変更されました。古い gnuplot スクリプトで、rand(x>0) により同じ種からなる疑似乱数列が生成されることを期待しているような場合は、代わりに rand(0) を使うように修正する必要があります。現在の挙動は以下の通りです: rand(0) 内部に持つ 2 つの 32bit の種 (seed) の現在の値から生成され

る [0:1] 区間内の疑似乱数値を返す

rand(-1) 2 つの種の値を標準値に戻す rand(x) x>0 ならば両方の種に x の値に基づく値を設定する  
rand({x,y}) x>0 ならば seed1 を x に、seed2 を y に設定する

## 13.2 Operators

gnuplot の演算子は、C 言語の演算子とほぼ同じですが、特に注意がなければ全ての演算子が整数、実数、複素数の引数を取ることができます。また、FORTRAN で使える \*\* (累乗) 演算子もサポートされています。

演算の評価の順序を変更するにはかっこを使います。

### 13.2.1 Unary

以下は、単項演算子とその使用法の一覧です:

単項演算子		
記号	例	説明
-	-a	マイナス符号
+	+a	プラス符号 (何もしない)
~	~a	* 1 の補数 (ビット反転)
!	!a	* 論理的否定
!	a!	* 階乗
\$	\$3	* ‘using‘ 内での引数/列指定

説明に星印 (\*) のついた演算子の引数は整数でなければなりません。

演算子の優先順位は Fortran や C と同じです。それらの言語同様、演算の評価される順序を変えるためにかっこが使われます。よって  $-2^{**}2 = -4$  で、 $(-2)^{**}2 = 4$  です。

階乗演算子は、大きな値を返せるように実数を返します。

### 13.2.2 Binary

以下は、二項演算子とその使用法の一覧です:

二項演算子		
記号	例	説明
**	a**b	累乗
*	a*b	積
/	a/b	商
%	a%b	* 余り
+	a+b	和
-	a-b	差
==	a==b	等しい
!=	a!=b	等しくない
<	a<b	より小さい
<=	a<=b	以下
>	a>b	より大きい
>=	a>=b	以上
&	a&b	* ビット積 (AND)
^	a^b	* ビット排他的論理和 (XOR)
	a b	* ビット和 (OR)
&&	a&&b	* 論理的 AND
	a  b	* 論理的 OR
.	A.B	文字列の連結
eq	A eq B	文字列が等しいかどうか
ne	A ne B	文字列が等しくないかどうか

説明に星印 (\*) のついた演算子の引数は整数でなければなりません。大文字の A,B は演算子が文字列引数を要求することを意味します。

論理演算子の AND (&&) と OR (||) は C 言語同様に必要最小限の評価しかしません。すなわち、&& の第 2 引数は、第 1 引数が偽ならば評価されませんし、|| の第 2 引数は、第 1 引数が真ならば評価されません。

### 13.2.3 Ternary

一つだけ三項演算子があります:

三項演算子		
記号	例	説明
?:	a?b:c	三項演算子

三項演算子は C のものと同じ働きをします。最初の引数 (a) は整数でなければいけません。この値が評価され、それが真 (ゼロでない) ならば 2 番目の引数 (b) が評価されその値が返され、そうでなければ 3 番目の引数 (c) が評価され、その値が返されます。

三項演算子は、区別的に定義された関数や、ある条件が満たされた場合にのみ点を描画する、といったことを行なう場合に有用です。

例:

$0 \leq x < 1$  では  $\sin(x)$  に、 $1 \leq x < 2$  では  $1/x$  に等しくて、それ以外の  $x$  では定義されない関数を描画:

```
f(x) = 0<=x && x<1 ? sin(x) : 1<=x && x<2 ? 1/x : 1/0
plot f(x)
```

gnuplot は未定義値に対しては何も表示せずにただ無視するので、最後の場合の関数 (1/0) は点を何も出力しないことに注意してください。また、この関数描画の描画スタイルが lines (線描画) の場合、不連続点 ( $x=1$ ) の所も連続関数として線が結ばれてしまうことにも注意してください。その点を不連続になるようにするには、関数を 2 つの部分それぞれに分けてください (このような場合、媒介変数関数を使うのが便利です)。

ファイル 'file' のデータで、4 列目のデータが負でないときだけ、1 列目のデータに関する 2 列目と 3 列目のデータの平均値を描画:

```
plot 'file' using 1:( $4<0 ? 1/0 : ($2+$3)/2 )
```

using の書式の説明に関しては plot datafile using を参照してください。

### 13.3 Gnuplot-defined variables

変数 pi は と定義されています。以下のようにしてみてください。

```
print pi
```

さらに、gnuplot は色々な操作に対してあらかじめいくつかの変数を定義しています。

mouse 機能が使える対話型入出力形式の作業上では、"MOUSE\_" で始まる変数をいくつか定義しています。詳細は mouse variables を参照してください。

また、GPVAL\_TERM, GPVAL\_X\_MIN, GPVAL\_X\_MAX, GPVAL\_Y\_MIN,... のような、"GPVAL\_" で始まる「読み込み専用」の変数があります。これらの一覧とその値を見るには、show variables all と入力してください。ただし、軸のパラメータに関連する値 (範囲、対数軸であるか等) は、現在 set したものではなく、最後に描画されたものが使用されます。

fit 機能は、"FIT\_" で始まるいくつかの変数を使用しますので、そのような名前を使うのは避けるべきでしょう。しかし、fit の使用に際しては、例えば "FIT\_LIMIT" のように再定義をする必要があるような変数はあります。set fit errorvariables とすると各当てはめ変数のエラーは、そのパラメータ名に "\_err" を追加した変数に保存されます。詳細は fit に関する説明書きを参照してください。

user-defined variables, mouse variables, fit も参照してください。

### 13.4 User-defined variables and functions

新たなユーザ定義変数と 1 個から 5 個までの引数を持つユーザ定義関数を、任意の場所で定義したり使ったりすることができます。それは plot コマンド上でも可能です。

ユーザ定義関数書式:

```
<func-name>( <dummy1> {,<dummy2>} ... {,<dummy5>} ) = <expression>
```

ここで <expression> は仮変数 <dummy1> から <dummy5> で表される数式です。

ユーザ定義変数書式:

```
<variable-name> = <constant-expression>
```

例:

```
w = 2
q = floor(tan(pi/2 - 0.1))
f(x) = sin(w*x)
sinc(x) = sin(pi*x)/(pi*x)
delta(t) = (t == 0)
ramp(t) = (t > 0) ? t : 0
min(a,b) = (a < b) ? a : b
comb(n,k) = n!/(k!*(n-k)!)
len3d(x,y,z) = sqrt(x*x+y*y+z*z)
plot f(x) = sin(x*a), a = 0.2, f(x), a = 0.4, f(x)

file = "mydata.inp"
file(n) = sprintf("run_%d.dat",n)
```

最後の 2 行の例は、ユーザ定義文字列変数と、ユーザ定義文字列関数を意味しています。

円周率 pi は既に定義されています。しかしこれは決して手品のようなものではなく、好きなように再定義することができます。他にもいくつかの変数が、例えば対話型入出力形式でのマウス操作や当てはめ (fit) などの gnuplot の動作状態に応じて定義されます。詳細は [gnuplot-defined variables](#) を参照してください。

ある変数 V が既に定義されているかどうかは、式 `exists("V")` でチェックできます。例:

```
a = 10
if (exists("a")) print "a is defined"
if (!exists("b")) print "b is not defined"
```

変数名や関数名の命名規則は、大抵のプログラミング言語と同じで、先頭はアルファベットで、その後の文字はアルファベット、数字、"\$", "\_" が使えます。

`show functions`, `functions`, [gnuplot-defined variables](#), `macros` も参照してください。

## 14 Glossary

このドキュメント全体に渡って、用語に関する一貫性の維持が考えられています。しかしこの試みは完全には成功していません。それは gnuplot が時間をかけて進化してきたように、コマンドやキーワードの名前もそのような完全性を排除するかのように採用されて来ているからです。この節では、これらのキーワードのいくつかがどのように使われているかを説明します。

"ページ (page)" または "表示画面 (screen)" は gnuplot がアクセス可能な領域全体を指します。ディスプレイモニタでは、これは画面全体を指し、プロッタでは、一枚の紙全体になります。

表示画面は、一つ、またはそれ以上の "グラフ描画 (plot)" を含みます。グラフ描画は一つの横座標と一つの縦座標で定義されますが、余白 (margin) やその中に書かれる文字列 (text) 同様、それらは実際にその上に表示されている必要はありません。

グラフ描画は一つの "グラフ" を含みます。グラフは一つの横座標と一つの縦座標で定義されますが、これらは実際にその上に表示されている必要はありません。

グラフは一つまたはそれ以上の "曲線 (line)" を含みます。曲線は一つの関数、またはデータセットです。用語 "line" は描画スタイルとしても使われます。さらにこの用語は "文字列の一行 (a line of text)" のように使われることもあります。多分文脈からそれらは区別できるでしょう。

一つのグラフ上の複数の曲線はそれぞれ名前を持ちます。その名前は、その曲線の表現に使われる描画スタイルのサンプルとともに "(説明) key" 内に一覧表示されます。説明は、時には "(表題) legend" とも呼ばれます。

用語 "タイトル (title)" は gnuplot では複数の意味で使われます。このドキュメントではそれらを区別するために、形容詞として "描画の (plot)"、"曲線の (line)"、"説明の (key)" を頭につけたりもします。

2 次元のグラフは 4 つまでの見出し付けされる軸を持つことができます。これらの使われ方の中の 4 つの軸の名前はそれぞれ、グラフ描画の下の境界に沿う軸である "x"、左の境界に沿う軸 "y"、上の境界に沿う軸 "x2"、右の境界に沿う軸 "y2" となっています。

3 次元のグラフは 4 つまでの見出し付けされる軸 "x", "y", "z" を持つことができます。どの特定の軸に関してもそれがグラフ上でどこに書かれるかを述べることはできません。それは、set view でグラフを見る方向を変更できるからです。

データファイルに関する議論では、用語 "行 (record)" を復活し、ファイルの一行の文字列、すなわち、改行文字や行末文字同士の間の文字列、を指示するのに使います。"点 (point)" は行から取り出した一つのデータです。"データブロック (datablock)" は、空行で区切られた連続した複数の行からなる点の集合です。データファイルの議論の中で "line" が参照される場合は、これはデータブロックの部分集合を指します。

(訳注: この日本語訳の中ではここに書かれているような用語の統一は考慮されてはおらず、よって混乱を引き起こす可能性があります。厳密には原文を参照すべきでしょう。)

## 15 Linetype, colors, and styles

gnuplot の各出力形式は "線種 (linetype)" をある程度用意しています。それらは色、太さ、点線/破線のパターン、または色と点線/破線の組合せで違いを表現しています。特定の出力形式のデフォルトの線種は、その出力形式を設定した (set terminal) 後で test コマンドを発行することで確認できます。定義されている色、点線/破線のパターンはすべての出力形式で同じものであるという保証は何もありませんが、線種 -1 は特別に全ての出力形式で、最も普通の表示色 (通常は黒) の実線を意味しています。デフォルトでは、一つの描画コマンド内での関数やデータファイルの並びには、順番に線種が割り当てられます。そのデフォルトは、関数、データファイル、またはその他の描画要素に個別に線種を指定することで上書きできます。

例:

```
plot "foo", "bar"          # 線種 1, 2 で 2 ファイルを描画
plot sin(x) linetype 4    # 出力形式に依存する線種色 4
plot sin(x) lt -1         # 黒
```

多くの出力形式で、ユーザ定義色を定義できるようになっています。色の定義は明示的な rgb (赤、緑、青) を指定するか、色名か、現在の PM3D パレットを参照する色の値のいずれかで行ないます。

例:

```
plot sin(x) lt rgb "violet"      # gnuplot の持つ色名の一つ
plot sin(x) lt rgb "#FF00FF"     # 明示的な 16 進 RGB 3 つ組
plot sin(x) lt palette cb -45   # 現在のパレットの cbrange の
                                # -45 に対応する色
plot sin(x) lt palette frac 0.3 # パレットに対応する小数値
```

show palette colornames, set palette, cbrange 参照。

点線/破線のパターンをサポートする出力形式では、デフォルトの線種は点線 / 破線のパターンと色の両方が定義されています。しかし、デフォルトの色は、キーワード linecolor (省略形は lc) を使うことで上書きできます。例えば、postscript 出力形式のデフォルトの線種 3 は青の破線です。以下の描画コマンドは同じ破線パターンで 3 つの描画を行ないますが、一つは青 (デフォルト)、一つは赤 (線種 1 のデフォルト色)、もう一つは金色になります。

例:

```
set term postscript dashed color
plot 'foo' lt 3, 'baz' lt 3 linecolor 1, 'bar' lt 3 lc rgb 'gold'
```

線は、線幅のような他の属性も持つことができます。点 (point) の記号に対する同様の属性とともに、これらの属性をコマンド set style line を使ってユーザ定義 "ラインスタイル (linestyle)" として結びつけることができます。一度定義された linestyle は、plot コマンド中で、一つまたは多くの描画要素に対してその表示を制御するのに使うことができます。

例:

```
# 新しいラインスタイルを、出力形式に依存しない色 cyan、線幅が 3,
# 点種 6 (丸の中に点) と定義
set style line 5 lt rgb "cyan" lw 3 pt 6
plot sin(x) with linespoints ls 5           # 定義スタイル 5 で
```

`linestyle`, `set style line` 参照。

## 15.1 Colorspec

多くのコマンドで、明示的な色の指定をともなった線種を指定することができます。このオプションは、使用している出力形式が RGB カラーか pm3d パレットをサポートしている場合にのみ利用可能です。

書式:

```
... {linetype | lt} <colorspec>
```

<colorspec> は以下の形式のいずれかです:

```
rgbcolor "colorname"
rgbcolor "#RRGGBB"
rgbcolor variable
palette frac <val>      # <val> は 0 から 1 の値
palette cb <value>       # <val> は cbrange の範囲の値
palette z
```

"colorname" は gnuplot が内部に持っている色の名前のうちの一つを指定します。有効な名前の一覧に関しては `show palette colormames` を参照してください。

"#RRGGBB" は "#" 記号の後に 16 進定数を並べます。RRGGBB は、赤、緑、青の色の成分を意味し、それぞれ 0 - 255 の範囲の数です。例えば、マゼンタ (紫) = 最も明るい赤 + 最も明るい青、なので、#FF00FF と表され、これは 16 進数で、 $(255 \ll 16) + (0 \ll 8) + (255)$  を意味しています。

"rgb variable" は、3D 描画モード (splot) のみで有効で、`using` 指定子に追加の列を要求します。その列は 24-bit 形式の RGB の 3 つ組であるとみなされます。これはデータファイル中で 16 進値として (上を参照) 簡単に指定できます。

例:

```
rgb(r,g,b) = 65536 * int(r) + 256 * int(g) + int(b)
splot "data" using 1:2:3:(rgb($1,$2,$3)) with points lc rgb variable
```

カラーパレットとは、色の線型な勾配で、単一の数値を特定の色に滑らかに対応づけます。常にそのような 2 つの対応付けが効力を持ちます。`palette frac` は 0 から 1 までの小数値を、カラーパレットの全範囲に対応付けるもので、`palette cb` は、色軸の範囲を同じカラーパレットへ割り当てるものです。`set cbrange` を参照してください。`set colorbox` も参照してください。これらの対応付けのどちらかを使って、現在のパレットから定数色を選び出すことができます。

"`palette z`" は、各描画線分や描画要素の `z` の値を、パレットへ対応づけられている `cbrange` の範囲に対応づけます。これにより、3 次元の曲線や曲面に沿って色を滑らかに変化させることができます。このオプションは 3D 描画 (splot) のみに適用できます。

## 16 Mouse input

出力形式 x11, pm, windows, ggi, wxt では、現在の描画にマウスを使って作用をすることが可能になっています。これらはホットキーの定義もサポートしていて、マウスカーソルが有効な描画ウィンドウにあるときに、あるキーを押すことであらかじめ定義した関数を実行させることができます。マウス入力を batch コマンドスクリプトと組み合わせることも可能で、例えば `pause mouse` として、その後にマウスクリックによってパラメータとして返って来るマウス変数をその後のスクリプト動作に反映させることができます。`bind`, `mouse variables`, も参照してください。また、コマンド `set mouse` も参照してください。

## 16.1 Bind

`bind` は、ホットキーの定義、再定義に使用します。ホットキーとは、入力カーソルがドライバのウィンドウ内にあるときに、あるキー、または複数のキーを押すことで、gnuplot のコマンド列を実行させる機能のことを言います。`bind` は、gnuplot が `mouse` をサポートするようにコンパイルされていてかつマウス

が有効な出力形式上で使われての場合にのみ有効であることに注意してください。キー割当 (binding) は、組み込み (builtin) キー割当 (通常のエディタに似た形に設定されている) を上書きしますが、<space> と 'q' だけは再定義はできません (下で述べる 1 例を除いて)。マウスボタンも再定義はできません。

ホットキーの一覧を得るには bind とタイプするか、グラフィック上で'h' と入力してください。

修飾キーを含む複数のキーの定義は引用符で囲む必要があることに注意してください。

標準ではホットキーは現在の描画ウィンドウ上に入力カーソルがある場合のみ認識されます。bind all-windows <key> ... (bind all <key> ... と省略可) は、<key> の割当を、それが現在の有効なものか否かに関わらず、すべての gnuplot の描画ウィンドウ上で可能にします。この場合、gnuplot 変数 MOUSE\_KEY\_WINDOW にそれが行なわれたウィンドウの ID が保存されるのでそれをキーに割り当てたコマンドで使用することができます。

デフォルトでは、<space> ホットキーは gnuplot のコマンドウィンドウを前面に出します。いくつかの出力形式 (例えば x11, wxt, pm) では 'q' はグラフィックを閉じます。これらのデフォルトは、'gnuplot -ctrlq' のようにして gnuplot を起動する (x11 command-line-options も参照)、または X リソースの 'gnuplot\*ctrlq' で、ctrl-space や ctrl-q に変更できます。注意: <space> (または ctrl-space) で X11 の gnuplot ウィンドウが前面に出ない場合は raise の項をよく読んでください。

書式:

```
bind {allwindows} [<key-sequence>] ["<gnuplot commands>"]
bind!
```

例:

- キー割当の設定:

```
bind a "replot"
bind "ctrl-a" "plot x*x"
bind "ctrl-alt-a" 'print "great"'
bind Home "set view 60,30; replot"
bind all Home 'print "This is window ",MOUSE_KEY_WINDOW'
```

- キー割当を表示:

```
bind "ctrl-a"          # ctrl-a に対するキー割当を表示
bind                  # 全てのキー定義を表示
```

- キー割当を削除:

```
bind "ctrl-alt-a" ""   # ctrl-alt-a のキー割当を削除
                        (組み込みキー定義は削除されません)
bind!                  # デフォルト (組み込み) のキー定義を導入
```

- トグルスイッチ形式にキー割当:

```
v=0
bind "ctrl-r" "v=v+1;if(v%2)set term x11 noraise; else set term x11 raise"
```

修飾キー (ctrl / alt) は大文字小文字の区別はありませんが、キーはそうではありません:

```
ctrl-alt-a == CtRl-altA
ctrl-alt-a != ctrl-altA
```

修飾キー (alt == meta) の一覧:

```
ctrl, alt
```

サポートされている特殊キーの一覧:

```
"BackSpace", "Tab", "Linefeed", "Clear", "Return", "Pause", "Scroll_Lock",
"Sys_Req", "Escape", "Delete", "Home", "Left", "Up", "Right", "Down",
"PageUp", "PageDown", "End", "Begin",
"KP_Space", "KP_Tab", "KP_Enter", "KP_F1", "KP_F2", "KP_F3", "KP_F4",
"KP_Home", "KP_Left", "KP_Up", "KP_Right", "KP_Down", "KP_PageUp",
"KP_PageDown", "KP_End", "KP_Begin", "KP_Insert", "KP_Delete", "KP_Equal",
"KP_Multiply", "KP_Add", "KP_Separator", "KP_Subtract", "KP.Decimal",
"KP_Divide",
"KP_1" - "KP_9", "F1" - "F12"
```

mouse, if の項目も参照してください。

## 16.2 Mouse variables

マウス機能が有効な場合、現在のウィンドウ上でのマウスクリックによって gnuplot のコマンドライン上で使うことができる色々なユーザ変数が設定されます。クリック時のマウスの座標は変数 MOUSE\_X, MOUSE\_Y, MOUSE\_X2, MOUSE\_Y2 に代入されます。クリックされたボタンや、そのときのメタキーの状態は MOUSE\_BUTTON, MOUSE\_SHIFT, MOUSE\_ALT, MOUSE\_CTRL に代入されます。これらの変数は任意の描画の開始時には未定義で、有効な描画ウィンドウ中でのマウスクリックイベントによって初めて定義されます。有効な描画ウィンドウ中でマウスが既にクリックされたかどうかをスクリプトから調べるには、これらの変数のうちのどれか一つが定義されているかどうかをチェックすれば十分です。

```
plot 'something'
pause mouse
if (defined(MOUSE_BUTTON)) call 'something_else'; \
else print "No mouse click."
```

描画ウィンドウ上での一連のキー入力を追跡することも、マウスコードを使うことで可能となります。

```
plot 'something'
pause mouse keypress
print "Keystroke ", MOUSE_KEY, " at ", MOUSE_X, " ", MOUSE_Y
```

pause mouse keypress が、キー入力で終了した場合は MOUSE\_KEY には押されたキーの ascii コードが保存されます。MOUSE\_CHAR にはその文字自身が文字列値として保存されます。pause コマンドが(例えば ctrl-C や描画ウィンドウが外部から閉じられるなどして)異常終了した場合は MOUSE\_KEY は -1 になります。

マウスによる拡大の後の新しい描画範囲は、GPVAL\_X\_MIN, GPVAL\_X\_MAX, GPVAL\_Y\_MIN, GPVAL\_Y\_MAX で参照できることに注意してください。gnuplot-defined variables も参照してください。

## 17 Plotting

gnuplot には描画を生成する 3 つのコマンド、plot, splot, replot があります。plot は 2 次元描画を生成し、splot は 3 次元描画(もちろん実際には 2 次元面への射影)を生成します。replot は与えられた引数を、直前の plot または splot コマンドに追加し、それを実行します。

描画に関する一般的な情報の大半は、plot に関する項で見つかります。3 次元描画に固有の情報は splot の項にあります。

plot は xy 直交座標系と極座標系が使えます。極座標系の詳細に関しては set polar を参照してください。splot は xyz 直交座標系のみしか扱えませんが、コマンド set mapping で他の 2, 3 の座標系を使用することが出来ます。さらに、オプション using を使えば、plot でも splot でもほとんどどんな座標系でもそれを定義して使うことが出来ます。

plot では、4 つの境界 x(下), x2(上), y(左), y2(右) をそれぞれ独立な軸として扱うこともできます。オプション axes で、与えられた関数やデータ集合をどの軸のペアで表示させるかを選べます。また、各軸の縮尺や見出しづけを完全に制御するために十分な補佐となる set コマンド群が存在します。いくつかのコマンドは、set xlabel のように軸の名前をその中に持っていますし、それ以外のものは set logscale xy のように、1 つ、または複数の軸の名前をオプションとしてとります。z 軸を制御するオプションやコマンドは 2 次元グラフには効力を持ちません。

splot は、曲面の描画と、点や線を加えて等高線を書くことも出来ます。3 次元の関数の格子定義に関する情報については、splot と set isosamples の項目を、3 次元データのファイルに必要な形態に関しては splot datafile の項目を、等高線に関する情報については set contour と set cntrparam の項目を参照してください。

splot での縮尺や見出し付けの制御は、x2 軸と y2 軸を制御するコマンドやオプションは効果がなく、z 軸を制御するものにはもちろん効果がある、ということを除けば plot と全く同じです。

## 18 Start-up

gnuplot が起動されるとき、初期設定ファイルを読み込もうとします。このファイルは Unix と AmigaOS

では, .gnuplot であり, その他の処理系では GNUPLOT.INI となっています。このファイルがカレントディレクトリに無い場合, gnuplot はホームディレクトリを検索します (AmigaOS, Atari (single)TOS, MS-DOS, Windows, OS/2 では, 環境変数 GNUPLOT にホームディレクトリに対応するディレクトリを指定します; Windows NT では、GNUPLOT が定義されていなければ USERPROFILE が使われます)。注意: インストールの時に NOCWDRC を定義した場合には, gnuplot はカレントディレクトリからは読みません。

初期設定ファイルが見つかると, gnuplot はこのファイルに書かれているコマンドを実行します。ここには任意の正しい gnuplot コマンドを書くことが可能ですが、一般的には、出力装置の指定や、よく使う関数や変数の定義を設定する程度に抑えておきます。

## 19 String constants and string variables

文字列定数に加えて、ほとんどの gnuplot コマンドは文字列変数、文字列式または文字列を返す関数も受け付けます。例えば、以下の 4 つの plot のやり方は結果として全て同じ描画タイトルを生成します:

```
four = "4"
graph4 = "Title for plot #4"
graph(n) = sprintf("Title for plot #%d",n)

plot 'data.4' title "Title for plot #4"
plot 'data.4' title graph4
plot 'data.4' title "Title for plot #".four
plot 'data.4' title graph(4)
```

整数は、それが文字列結合演算子によって作用された場合は、文字列に変換されますので、以下の例も上と同様に動作します:

```
N = 4
plot 'data.'.N title "Title for plot #".N
```

一般に、コマンドラインの各要素は、それらが標準的な gnuplot への命令文法の一部分と認識されるもの以外は、有効な文字列変数としての評価のみが行なわれます。よって、以下のコマンド列は、恐らくは混乱を引き起こさないように避けられるべきですが、文法的には間違ってはいません:

```
plot = "my_datafile.dat"
title = "My Title"
plot plot title title
```

文字列に対する 3 つの二項演算子が用意されています: 文字列の結合演算子 ".", 文字列の等号演算子 "eq", および文字列の不等号演算子 "ne" です。以下の例では TRUE が表示されます。

```
if ("A"."B" eq "AB") print "TRUE"
```

2 つの文字列式関数 gprintf と sprintf も参照してください。

任意の文字列、文字列変数、文字列値関数に、範囲指定子をつけることにより部分文字列を指定できます。範囲指定子は [begin:end] の形で、begin は部分文字列の先頭位置、end は最後の位置です。位置指定は、最初の文字を 1 番目と見ます。先頭の位置、最後の位置は空、あるいは '\*' でも構いません。その場合、それは元の文字列自体の先頭、あるいは最後を意味します。例えば、str[:] や str[\*:\*] はどちらも str の文字列全体を意味します。

## 20 Substitution and Command line macros

gnuplot への命令文字列が最初に読み込まれた時点、すなわちまだそれが解釈され、もしくは実行される前の段階で、2 つの形式の単語の置換が実行されます。それらはバッククオート (ASCII 番号 96) で囲まれているか、または @ (ASCII 番号 64) が頭についていた文字列に対して行なわれます。

## 20.1 Substitution of system commands in backquotes

シェルコマンドをバッククオートで囲むことによってコマンド置換を行うことができます。このコマンドは子プロセスで実行され、その出力結果でコマンドラインのバッククオートで囲まれたコマンドを置き換えます。処理系によってはパイプがサポートされている場合もあります。plot datafile special-filenames を参照してください。

コマンド置換は、単一引用符内の文字列以外は、gnuplot のコマンドライン中、どこででも使用可能です。

例:

以下の例は、leastsq というプログラムを実行し、その出力結果で、leastsq を（まわりの引用符こみで）置き換えます:

```
f(x) = `leastsq`
```

ただし VMS では、

```
f(x) = `run leastsq`
```

以下は現在の日付とユーザー名のラベルを生成します:

```
set label "generated on `date +%Y-%m-%d` by `whoami`" at 1,1
set timestamp "generated on %Y-%m-%d by `whoami`"
```

## 20.2 Substitution of string variables as macros

コマンドラインのマクロ置換はデフォルトでは無効になっていますが、set macros で有効にできます。マクロ置換が有効である場合、文字 @ は、コマンドライン上でその文字列変数の値への置換を行なうのに使われます。文字列変数の値としての文は、複数の単語からなることも可能です。これにより文字列変数をコマンドラインマクロとして使うことが可能になります。この機能により展開できるのは文字列定数のみで、文字列を値に取る式を使うことはできません。例:

```
set macros
style1 = "lines lt 4 lw 2"
style2 = "points lt 3 pt 5 ps 2"
range1 = "using 1:3"
range2 = "using 1:5"
plot "foo" @range1 with @style1, "bar" @range2 with @style2
```

この @ 記号を含む行は、その入力時に展開され、それが実際に実行されるときには次のように全部打ち込んだ場合と同じことになります。

```
plot "foo" using 1:3 with lines lt 4 lw 2,
      "bar" using 1:5 with points lt 3 pt 5 ps 2
```

関数 exists() はマクロの評価に関して有用でしょう。以下の例は、C が安全にユーザ定義変数の名前に展開できるかどうかをチェックします。

```
C = "pi"
if (exists(C)) print C, " = ", @C
```

マクロの展開は、単一引用符内、または二重引用符内では行なわれませんが、バッククオート内ではマクロ展開されます。

## 20.3 String variables, macros, and command line substitution

文字列変数やバッククオートによる置換、マクロによる置換の相互関係は少しややこしいです。バッククオートはマクロ置換を妨げないので、

```
filename = "mydata.inp"
lines = ` wc --lines @filename | sed "s/ .*/" '
```

は、mydata.ipn の行数を整数変数 lines に保存することになります。また、二重引用符はバッククオートの置換を妨げないので、

```
mycomputer = "'uname -n'"
```

は、システムコマンド uname -n の返す文字列を文字列変数 mycomputer に保存することになります。

しかし、マクロ置換は二重引用符内では機能しないので、システムコマンドをマクロとして定義してそれをマクロとして利用しかつバッククオート置換を行なうことはできません。

```
machine_id = "uname -n"
mycomputer = "@machine_id" # うまくいかない！
```

この失敗は、二重引用符が @machine\_id をマクロとして解釈することを妨げているからです。システムコマンドをマクロとして保存し、その後それを実行するには、バッククオート自体もマクロ内に含める必要があります。これは以下のようにマクロを定義することで実現できます。sprintf の書式には 3 種類の引用符全てが入れ子になっていることに注意してください。

```
machine_id = sprintf("'`uname -n`'")
mycomputer = @machine_id
```

## 21 Syntax

gnuplot Version 4 では、以前の範囲比べてキーワードやオプションに関する順序はだいぶ許容的になりました。しかし、もしあなたが正しく動くべきだと思うオプション指定がエラーメッセージを出したなら、正確にドキュメントに書かれている通りの順序で指定して試してみてください。

リストや座標がコンマ (,) 区切りであるのに対し、オプションやそれに伴うパラメータはスペース ( ) 区切りです。範囲はコロン (: ) で区切ってかぎかっこ ([] ) でくくりますし、文字列やファイル名は引用符でくくり、他にいくつかカッコ (( )) でくくるものがあります。中カッコ ({} ) は特別な目的で使われます。

コンマは以下の区切りで使用されます。set コマンドの arrow, key, label の座標; 当てはめ (fit) られる変数のリスト (コマンド fit のキーワード via に続くリスト); コマンド set cntrparam で指定されるるびとびの等高線の値やそのループパラメータのリスト; set コマンドの dgrid3d dummy, isosamples, offsets, origin, samples, size, time, view の引数; 目盛りの位置やそのループパラメータのリスト; タイトルや軸の見出しの位置; plot, replot, splot コマンドの x,y,z 座標の計算に使われる媒介変数関数のリスト; plot, replot, splot コマンドの複数の描画 (データ、または関数) のそれぞれの一連のキーワードのリスト。

(丸) カッコは、目盛りの見出しを (ループパラメータではなく) 明示的に集合与える場合の区切りとして、または fit, plot, replot, splot コマンドの using フィルタでの計算を指示するために使われます。

(カッコやコンマは通常の関数の表記でも使われます。)

かぎかっこは、set, plot, splot コマンドでは範囲を区切るのに使われます。

コロンは range (範囲) 指定 (set, plot, splot コマンドで使われる) の両端の値を区切るのに、または plot, replot, splot, fit コマンドの using フィルタの各エントリを区切るのに使われます。

セミコロン (; ) は、一行のコマンド行内で与えられる複数のコマンドを区切るのに使われます。

中カッコは、postscript のようないくつかの出力形式で特別に処理される文字列内で使用されます。または複素数を記述するのにも使われます: {3,2} = 3 + 2i となります。

PostScript 出力形式で enhanced text モード (拡張文字列処理モード) を使う場合、現在は、{} の内部に \n を入れてはいけません。

EEPIC, Imagen, Uniplex, LaTeX, TPIC の各出力形式では、单一引用符内の \\ または二重引用符内の \\ \\ で改行を示すことが可能ですが。

### 21.1 Quote Marks

gnuplot は、文字列を区切るのに、二重引用符 (ASCII コード 34 番)、単一引用符 (ASCII コード 39 番)、およびバッククオート (ASCII コード 96 番) の 3 種類の引用符を使います。

ファイル名は单一引用符、あるいは二重引用符内で囲みます。このマニュアルでは一般にコマンドの例示では、わかりやすくするためにファイル名は单一引用符でくくり、他の文字列は二重引用符でくくります。見出し (label)、タイトル (title)、またはその他の描画要素で使用される文字列定数や複数行文字列は单一引用符、あるいは二重引用符内で囲みます。引用符で囲まれた文字列のさらなる処理の結果は、どの引用符記号を選びかによって変わります。

\n (改行) や \345 (8 進表記の文字コード) のようなバックスラッシュ (\) による特殊文字表現は、2 重引用符内の文字列では効力を持ります。单一引用符内では、バックスラッシュ 자체が通常の文字と見なされます。单一引用符内の文字列で单一引用符自体 (ASCII コード 39 番) を使うには、それを重ねて書く必要があります。つまり、文字列 "d\" s' b\\\" と、'd" s'" b\' は完全に同じものとなります。

1 つの複数行文字列に関する位置合わせは各行に同等に働きます。よって、中央に位置合わせされた文字列  
 "This is the first line of text.\nThis is the second line."

は次のように表示されます:

```
This is the first line of text.  
This is the second line.
```

しかし

```
'This is the first line of text.\nThis is the second line.'
```

だと次のようになります。

```
This is the first line of text.\nThis is the second line.
```

拡張文字列処理 (enhanced text processing) は二重引用符に対しても単一引用符に対しても機能します。しかし、そのモードをサポートしている出力形式でのみ働きます。enhanced text 参照。

バッククオート (``) はコマンドライン中の置換のためにシステムコマンドを囲むのに使います。substitution を参照。

## 22 Time/Date data

gnuplot は入力データとして時間/日付情報の使用をサポートしています。この機能は set xdata time, set ydata time などのコマンドによって有効になります。

内部では全ての時間/日付は 2000 年からの秒数に変換されます。コマンド set timefmt は全ての入力書式を定義します。データファイル、範囲、軸の目盛りの見出し、ラベルの位置 - 手短に言えば、データの値を受けとる全てのものがこの書式にしたがって受けとらなければいけません。一時には一つの入力書式のみが有効なので、同じときに入力される全ての時間/日付のデータは同じ書式である必要があります。よって、ファイル内の x と y の両方が時間/日付データである場合は、それらは同じ書式でなければいけません。

秒数へ (秒数から) の変換は国際標準時 (UT; グリニッジ標準時 (GMT) と同じ) が使われます。各国標準時や夏時間への変換の機能は何も持ち合っていません。もしデータがすべて同じ標準時間帯に従っているなら (そして全てが夏時間か、そうでないかのどちらか一方にのみ従うなら) これに関して何も心配することはありません。しかし、あなたが使用的するアプリケーションで絶対的な時刻を厳密に考察しなければいけない場合は、あなた自身が UT に変換すべきでしょう。

show xrange のようなコマンドは、その整数値を timefmt に従って解釈し直します。timefmt を変更しても一度 show でその値を表示させるとそれは新しい timefmt に従って表示されます。このため、もし機能を停止させるコマンド (set xdata のような) を与えると、その値は整数値として表示されることになります。

コマンド set format は、指定された軸が時間/日付であるなしに関わらず目盛りの見出しに使われる書式を定義します。

時間/日付情報がファイルから描画される場合、plot, splot コマンドでは using オプションを「必ず」使う必要があります。plot, splot では各行のデータ列の分離にスペースを使いますが、時間/日付データはその中にスペースを含み得るからです。もしタブ区切りを使用しているのなら、あなたのシステムがそれをどう扱うか確かめるために何度もテストする必要があるでしょう。

次の例は時間/日付データの描画の例です。

ファイル "data" は以下のような行からなるとします:

---

03/21/95 10:00 6.02e23

このファイルは以下のようにして表示されます:

```
set xdata time
set timefmt "%m/%d/%y"
set xrange ["03/21/95":"03/22/95"]
set format x "%m/%d"
set timefmt "%m/%d/%y %H:%M"
plot "data" using 1:3
```

ここで、x 軸の目盛りの見出しが "03/21" のように表示されます。

各コマンドの詳細はそれぞれの項の記述を参照してください。

## Part II

# Commands

このセクションでは gnuplot が受け付けるコマンドをアルファベット順に並べています。このドキュメントを紙に印刷したものは全てのコマンドを含んでいますが、オンラインドキュメントの方は完全ではない可能性があります。実際、この見出しの下に何のコマンドも表示されないシステムがあります。

ほとんどの場合、コマンド名とそのオプションは、紛らわしくない範囲で省略することが可能ですが。すなわち、"plot f(x) with lines" の代わりに "p f(x) w li" とすることができます。

書式の記述において、中カッコ ({} ) は追加指定できる引数を意味し、縦棒 (|) は互いに排他的な引数を区切るものとします。

## 23 Cd

cd コマンドはカレントディレクトリを変更します。

書式:

```
cd '<ディレクトリ名>'
```

ディレクトリ名は引用符に囲まれていなければなりません。

例:

```
cd 'subdir'
cd '...'
```

バックスラッシュ (\) は二重引用符内 ("") では特別な意味を持つためにエスケープする必要がありますので、DOS と Windows ユーザには単一引用符を使うことを勧めます。例えば、

```
cd "c:\newdata"
```

では失敗しますが、

```
cd 'c:\newdata'
cd "c:\\\\newdata"
```

なら期待通りに動くでしょう。

## 24 Call

call コマンドは、1 つの機能以外は load コマンドと等価です。その機能は、10 個までのパラメータをコマンドに追加できることです (パラメータは標準的な構文規則によって区切られます)。これらのパラメー

タは、ファイルから読まれる行に代入することができます。call した入力ファイルから各行が読まれる時に、\$ (ドル記号) に続く数字 (0-9) の並びを走査します。もし見つかれば、その並びは call のコマンド行の対応するパラメータで置き換えられます。call の行でそのパラメータが文字列として指定されているならば、取り囲んでいる引用符が省かれて代入されます。文字列 \$# はパラメータの個数に置き換えられます。文字が後に続く \$ はその文字になります。例えば、一つの \$ を得るには \$\$ を使います。call のコマンド行に 10 個より多いパラメータを与えるとエラーが起こります。与えられなかったパラメータは、何も無しとして扱われます。call 中のファイルの中にさらに load または call コマンドがあっても構いません。

call コマンドは、複数のコマンドからなる行の中では最後のコマンドでなければなりません。

書式:

```
call "<入力ファイル>" <パラメータ 0> <パ 1> ... <パ 9>
```

入力ファイル名は引用符で囲まなければなりません。そして、パラメータも引用符で囲むことを推奨します (gnuplot の将来のバージョンでは引用符で囲んである部分と囲んでない部分に対しては違う取り扱いをする予定です)。

例:

ファイル 'calltest.gp' は以下の行を含んでいるとすると:

```
print "argc=$# p0=$0 p1=$1 p2=$2 p3=$3 p4=$4 p5=$5 p6=$6 p7=x$7x"
```

次の行を入力すると:

```
call 'calltest.gp' "abcd" 1.2 + "'quoted'" -- "$2"
```

以下のように表示されるでしょう:

```
argc=7 p0=abcd p1=1.2 p2=+ p3='quoted' p4=- p5=- p6=$2 p7=xx
```

注意: using を使用しているデータファイルでは文法的に重なってしまいます。その場合、call されたデータファイルからプロットするときは、データの n カラム目の指示には \$\$n または column(n) を使用してください。

## 25 Clear

clear コマンドは、set output で選択された画面または出力装置をクリアします。通常、ハードコピー装置に対しては改ページを行います。出力装置を選択するには set terminal を使用して下さい。

いくつかの出力装置は clear コマンドでは set size で定義された描画領域のみを消去します。そのため、set multiplot とともに使用することで挿入図を一つ作ることができます。

例:

```
set multiplot
plot sin(x)
set origin 0.5,0.5
set size 0.4,0.4
clear
plot cos(x)
unset multiplot
```

これらのコマンドの詳細については set multiplot, set size, set origin を参照してください。

## 26 Exit

exit と quit の両コマンドは END-OF-FILE 文字 (通常 Ctrl-D) 同様、現在の入力ストリーム、すなわち端末の対話やパイプ入力、ファイル入力 (パイプ) からの入力を終了させます。

入力ストリームが入れ子 (階層的な load のスクリプトで) になっている場合、読み込みは親のストリームで継続されます。トップレベルのストリームが閉じられると、プログラムはそれ自身終了します。

コマンド `exit gnuplot` は、直ちに、無条件に、そして例え入力ストリームが多段階にネストされていても、`gnuplot` を終了させます。その場合、開かれていた全ての出力ファイルはきれいに完全な形では閉じられない可能性があります。使用例:

```
bind "ctrl-x" "unset output; exit gnuplot"
```

詳細は `batch/interactive` に関するヘルプをご覧ください。

## 27 Fit

`fit` コマンドはユーザ定義関数を  $(x,y)$  または  $(x,y,z)$  の形式のデータ点の集合への当てはめを可能にします。それには Marquardt-Levenberg 法による非線形最小自乗法 (NLLS) の実装が用いられます。関数内部に現われるユーザ定義変数はいずれも当てはめのパラメータとして使うことができます。ただ、その関数の返り値は実数である必要があります。

書式:

```
fit {[xrange] {[yrange]}} <function> '<datafile>'  
<datafile-modifiers>  
via '<parameter file>' | <var1>{,<var2>,...}
```

範囲 (`xrange,yrange`) は、当てはめられるデータ点を一時的に制限するのに使うことができ、その範囲を超えたデータは全て無視されます。その書式は `plot` コマンド同様

```
[{dummy_variable}=>{<min>}{:<max>}],
```

です (`plot ranges` 参照)。

`<function>` は通常はあらかじめユーザ定義された  $f(x)$  または  $f(x,y)$  の形の関数ですが、`gnuplot` で有効な任意の式を指定できます。

`<datafile>` は `plot` コマンドと同じ扱われます。`plot datafile` の修飾子 (`using, every, ...`) は、`smooth` とあまり勧められない `thru` を除いて、全て `fit` に使うことができます。`plot datafile` を参照してください。

当てはめる 1 变数関数  $y=f(x)$  へのデフォルトのデータの書式は  $\{x\}y$  か  $x:y:s$  で、これらはデータファイルへの `using` 指定子で変更できます。この 3 番目の項目 (列番号、または数式) が与えられた場合は、それは対応する  $y$  の値の標準偏差として解釈され、それはそのデータへの重み ( $=1/s^{**2}$ ) を計算するのに使われます。そうでなければ、全てのデータは同じ重み (1) で計算されます。`using` オプションを全く指定しなかった場合、3 列目のデータがあった場合でもデータから  $y$  の偏差は読まれませんので、その場合は重み 1 になります。

2 变数関数  $z=f(x,y)$  を当てはめる場合、データの書式は `using` による 4 つの項目  $x:y:z:s$  が要求されます。これは完全に全てが与えられなければなりません—不足する項目に対してはどの列もデフォルトは仮定されていません。各データ点の重みは上と同様に ' $s$ ' から計算されます。もし誤差評価を持っていなければ、定数値を定数式として指定すればいいでしょう (`plot datafile using` 参照)。例えば `using 1:2:3:(1)` のように。

複数のデータ集合も複数の 1 变数関数に同時に当てはめることも、 $y$  を '仮变数' とすれば可能です。例えればデータ行番号を使い、2 变数関数への当てはめ、とすればいいでしょう。`fit multi-branch` を参照してください。

`via` 指定子はパラメータの調節を直接か、またはパラメータファイルを参照することによって行うかを指定します。

例:

```
f(x) = a*x**2 + b*x + c  
g(x,y) = a*x**2 + b*y**2 + c*x*y  
FIT_LIMIT = 1e-6  
fit f(x) 'measured.dat' via 'start.par'  
fit f(x) 'measured.dat' using 3:($7-5) via 'start.par'  
fit f(x) './data/trash.dat' using 1:2:3 via a, b, c  
fit g(x,y) 'surface.dat' using 1:2:3:(1) via a, b, c
```

反復の個々のステップの後で、当てはめの現在の状態についての詳細な情報が画面に表示されます。そして最初と最後の状態に関する同じ情報が "fit.log" というログファイルにも書き出されます。このファイルは前の当てはめの履歴を消さないように常に追加されていきます。これは望むなら削除、あるいは別な名前でできます。コマンド `set fit logfile` を使ってログファイルの名前を変更することもできます。

`gnuplot` が、`set fit errorvariables` を使えるようにインストールされていて、そのコマンドを使用した場合、各当てはめパラメータの誤差は、そのパラメータと似た名前 ("err" が追加された名前) の変数に保存されます。よってその誤差を更なる計算の入力として使用することができます。

当てはめの反復は Ctrl-C を押すことで中断できます (MSDOS と Atari マルチタスクシステムでは Ctrl-C 以外の任意のキー)。現在の反復が正常に終了した後、(1) 当てはめを止めて現在のパラメータの値を採用する、(2) 当てはめを続行する、(3) 環境変数 `FIT_SCRIPT` で指定した `gnuplot` コマンドを実行する、のいずれかを選ぶことができます。`FIT_SCRIPT` のデフォルトは `replot` であり、よってもしデータと当てはめ関数を一つのグラフにあらかじめ描画してあれば、現在の当てはめの状態を表示することができます。

`fit` が終了した後は、最後のパラメータの値を保存するのに `update` コマンドを使います。その値は再びパラメータの値として使うことができます。詳細は `update` を参照。

## 27.1 Adjustable parameters

`via` はパラメータを調節するための 2 つの方法を指定できます。一つはコマンドラインから直接指示するもので、もう一つはパラメータファイルを参照して間接的に行うものです。この 2 つは初期値の設定で違った方法を取ります。

調整するパラメータは、`via` キーワードの後ろにコンマで区切られた変数名のリストを書くことで指定できます。定義されていない変数は初期値 1.0 として作られます。しかし当てはめは、変数の初期値があらかじめ適切な値に設定されている方が多分速く収束するでしょう。

パラメータファイルは個々のパラメータを、個別に 1 行に一つずつ、初期値を次のような形で指定して書きます。

変数名 = 初期値

'#' で始まるコメント行や空行も許されます。特別な形式として

変数名 = 初期値 # FIXED

は、この変数が固定されたパラメータであることを意味し、それはこのファイルで初期化されますが、調節はされません。これは、`fit` でレポートされる変数の中で、どれが固定された変数であるかを明示するのに有用でしょう。なお、# FIXED と言うキーワードは厳密にこの形でなくてはなりません。

## 27.2 Short introduction

`fit` は、与えられたデータ点を与えられたユーザ定義関数にもっとも良く当てはめるようなパラメータを見つけるのに使われます。その当てはめは、同じ場所での入力データ点と関数値との自乗誤差、あるいは残差 (SSR:Sum of the Squared Residuals) の和を基に判定されます。この量は通常 (カイ) 自乗と呼ばれます。このアルゴリズムは SSR を最小化することをしようとします。もう少し詳しく言うと、データ誤差 (または 1.0) の重みつき残差の自乗和 (WSSR) の最小化を行っています。詳細は `fit error_estimate` 参照。

これが、(非線形) 最小自乗当てはめ法と呼ばれるゆえんです。非線形 が何を意味しているのかを見るための例を紹介しますが、その前にいくつかの仮定について述べておきます。ここでは簡単のため、1 変数のユーザー定義関数は  $z=f(x)$ 、2 変数の関数は  $z=f(x,y)$  のようにし、いずれも従属変数として  $z$  を用いることにします。パラメータとは `fit` が調整して適切な値を決定するユーザ定義変数で、関数の定義式中の未知数です。ここで言う、線形性/非線形性とは、従属変数  $z$  と `fit` が調整するパラメータとの関係に対するものであり、 $z$  と独立変数  $x$  (または  $x$  と  $y$ ) との関係のことではありません (数学的に述べると、線形最小自乗問題では、当てはめ関数のパラメータによる 2 階 (そして更に高階の) 導関数は 0、ということになります)。

線形最小自乗法 (LLS) では、ユーザ定義関数は単純な関数の和であり、それぞれは一つのパラメータの定数倍で他のパラメータを含まない項になります。非線形最小自乗法 (NLLS) ではより複雑な関数を扱い、パラメータは色々な使われ方をされます。フーリエ級数は線形と非線形の最小自乗法の違いを表す一つの例です。フーリエ級数では一つの項は

```
z=a*sin(c*x) + b*cos(c*x).
```

のように表されます。もし、*a* と *b* が未知なパラメータで *c* は定数だとすればパラメータの評価は線形最小自乗問題になります。しかし、*c* が未知なパラメータならばそれは非線形問題になります。

線形の場合、パラメータの値は比較的簡単な線形代数の直接法によって決定できます。しかしそのような LLS は特殊な場合であり、'gnuplot' が使用する反復法は、もちろんそれも含めて、より一般的な NLLS 問題を解くことができます。fit は検索を行うことで最小値を探そうとします。反復の各ステップは、パラメータの新しい値の組に対して WSSR を計算します。Marquardt- Levenberg のアルゴリズムは次のステップのパラメータの値を選択します。そしてそれはあらかじめ与えた基準、すなわち、(1) 当てはめが "収束した" (WSSR の相対誤差が FIT\_LIMIT より小さくなった場合)、または (2) あらかじめ設定された反復数の限界 FIT\_MAXITER (fit control variables 参照) に達した場合、のいずれかを満たすまで続けられます。キーボードからその当てはめの反復は中断できますし、それに続けて中止することもできます (fit 参照)。ユーティリティ FIT\_CONVERGED は、直前の fit コマンドが収束により終了した場合は 1 を含み、それ以外の理由で中断した場合は 0 を含みます。

当てはめに使われる関数はしばしばあるモデル (またはある理論) を元にしていて、それはデータの振舞を記述したり、あるいは予測しようとします。よって fit は、データがそのモデルにどれくらいうまく当てはまっているのかを決定するため、そして個々のパラメータの誤差の範囲を評価するために、モデルの自由なパラメータの値を求めるのに使われます。fit error\_estimates も参照してください。

そうでなければ、曲線による当てはめにおける関数は、モデルとは無関係に選ばれています (それは十分な表現力と最も少ない数のパラメータを持ち、データの傾向を記述しそうな関数として経験に基づいて選ばれるでしょう)。

しかし、もしあなたが全てのデータ点を通るような滑らかな曲線を欲しいなら fit ではなく、むしろ plot の smooth オプションでそれを行なうべきでしょう。

### 27.3 Error estimates

fitにおいて "誤差" という用語は 2 つの異なった文脈で用いられます。一つはデータ誤差、もう一つはパラメータ誤差です。

データ誤差は、平方残差の重み付きの和 WSSR、すなわち 自乗を決定する際個々のデータ点の相対的な重みを計算するのに用いられます。それらはパラメータの評価に影響を与えます。それは、それらが、当てはめられた関数からの個々のデータ点の偏差が最終的な値に与える影響の大きさを決定することによります。正確なデータ誤差評価が与えられている場合には、パラメータの誤差評価等の fit が出力する情報はより役に立つでしょう。

'statistical overview' では fit の出力のいくつかを説明し、'practical guidelines' に対する背景を述べています。

#### 27.3.1 Statistical overview

非線形最小自乗法 (Non-Linear Least-Squares; NLLS) の理論は、誤差の正規分布の点から一般的に記述されています。すなわち、入力データは与えられた平均とその平均に対する与えられた標準偏差を持つガウス (正規) 分布に従う母集団からの標本と仮定されます。十分大きい標本、そして母集団の標準偏差を知ることに対しては、自乗分布統計を用いて、通常「自乗」と呼ばれる値を調べることにより「当てはめの良さ」を述べることができます。減らされた自由度の 自乗 ( 自乗の自由度は、データ点の数から当てはめられるパラメータの個数だけ引いた数) が 1.0 である場合は、データ点と当てはめられた関数との偏差の重みつき自乗和が、現在のパラメータ値に対する関数と与えられた標準偏差によって特徴付けられた母集団の、ランダムなサンプルに対する自乗和とが全く同じであることを意味します。

分散 = 総計である数え上げ統計学同様、母集団の標準偏差が定数でない場合、各点は観測される偏差の和と期待される偏差の和を比較するときに個別に重みづけされるべきです。

最終段階で fit は 'stdfit'、すなわち残差の RMS (自乗平均平方根) で求められる当てはめの標準偏差と、データ点が重みづけられている場合に '減らされた 自乗' とも呼ばれる残差の分散をレポートします。自由度 (データ点の数から当てはめパラメータの数を引いたもの) はこれらの評価で使用されます。なぜなら、データ点の残差の計算で使われるパラメータは同じデータから得られるものだからです。これらの値は以下の変数に代入されます:

```

FIT_NDF = 自由度の数
FIT_WSSR = 重みつき残差の自乗和
FIT_STDFIT = sqrt(WSSR/NDF)

```

パラメータに関する信頼レベルを評価することで、当てはめから得られる最小の 自乗と、要求する信頼レベルの 自乗の値を決定するための 自乗の統計を用いることが出来ます。しかし、そのような値を生成するパラメータの組を決定するには、相当のさらなる計算が必要となるでしょう。

`fit` は信頼区間の決定よりむしろ、最後の反復後の分散-共分散行列から直ちに得られるパラメータの誤差評価を報告します。これらの評価は、標準偏差として計算される量の指定に関する統計上の条件が、一般には非線形最小自乗問題では保証されないのですが、線形最小自乗問題での標準誤差（各パラメータの標準偏差）と同じ方法で計算されます。そしてそのため慣例により、これらは "標準誤差" とか "漸近標準誤差" と呼ばれています。漸近標準誤差は一般に楽観過ぎ、信頼レベルの決定には使うべきではありませんが、定性的な指標としては役に立つでしょう。

最終的な解は相関行列も生成します。それは解の範囲におけるパラメータの相関の表示を与えてくれます：もし一つのパラメータが変更されると、 自乗の増加が、他の補正の変更を行なう？主対角成分、すなわち自己相関はすべて 1 で、もし全てのパラメータが独立ならば他の成分はすべて 0 に近い値になります。完全に他を補いあう 2 つ変数は、大きさが 1 で、関係が正の相関か負の相関かによって正か負になる符号を持つ非対角成分を持ちます。非対角要素の大きさが小さいほど、各パラメータの標準偏差の評価は、漸近標準誤差に近くなります。

### 27.3.2 Practical guidelines

個々のデータ点への重みづけの割り当ての基礎を知っているなら、それが測定結果に対するより詳しい情報を使用させようとするでしょう。例えば、幾つかの点は他の点より当てになるということを考慮に入れることが可能です。そして、それらは最終的なパラメータの値に影響します。

データの重み付けは、最後の反復後の `fit` の追加出力に対する解釈の基礎を与えます。各点に同等に重み付けを行なうにしても、重み 1 を使うことよりもむしろ平均標準偏差を評価することが、 自乗が定義によりそうであるように、WSSR を無次元変数とすることになります。

当てはめ反復の各段階で、当てはめの進行の評価に使うことが出来る情報が表示されます（'\*' はより小さい WSSR を見つけられなかったこと、そして再試行していることを意味します）。'sum of squares of residuals'（残差の自乗和）は、'chisquare'（ 自乗）とも呼ばれます、これはデータと当てはめ関数との間の WSSR を意味していて、`fit` はこれを最小化しようとします。この段階で、重み付けされたデータによって、 自乗の値は自由度 (= データ点の数 - パラメータの数) に近付くことが期待されます。WSSR は補正された 自乗値 (WSSR/ndf; ndf = 自由度)、または当てはめ標準偏差 (stdfit = sqrt(WSSR/ndf)) を計算するのに使われます。それらは最終的な WSSR に対してレポートされます。

データが重み付けされていなければ、`stdfit` は、ユーザの単位での、データと当てはめ関数の偏差の RMS (自乗平均平方根) になります。

もし妥当なデータ誤差を与え、データ点が十分多く、モデルが正しければ、補正 自乗値はほぼ 1 になります（詳細は、適当な統計学の本の ' 自乗分布' の項を参照してください）。この場合、この概要に書かれていること以外に、モデルがデータにどれくらい良く当てはっているかを決定するための追加の試験方法がいくつかあります。

補正 自乗が 1 よりはるかに大きくなったら、それは不正なデータ誤差評価、正規分布しないデータ誤差、システム上の測定誤差、孤立した標本値 (outliers)、または良くないモデル関数などのためでしょう。例えば `plot 'datafile' using 1:(\$2-f(\$1))` などとして残差を描画することは、それらのシステム的な傾向を知るための手がかりとなります。データ点と関数の両者を描画することは、他のモデルを考えための手がかりとなるでしょう。

同様に、1.0 より小さい補正 自乗は、WSSR が、正規分布する誤差を持つランダムなサンプルと関数に対して期待されるものよりも小さいことを意味します。データ誤差評価が大きすぎるのか、統計的な仮定が正しくないのか、またはモデル関数が一般的すぎて、内在的傾向に加えて特殊なサンプルによる変動の当てはめになっているのでしょうか。最後の場合は、よりシンプルな関数にすればうまく行くでしょう。

標準的なエラーを、パラメータの不確定性に関する、あるより現実的な評価に関係付けること、および相関行列の重要性を評価することができるようになる前に、あなたは `fit` と、それを適用しようとするある種の問題に慣れておく必要があるでしょう。

`fit` は、大抵の非線形最小自乗法の実装では共通して、距離の自乗  $(y-f(x))^{**2}$  の重み付きの和を最小化しようとすることに注意してください。それは、`x` の値の "誤差" を計算に関してはどんな方法も与えてはおらず、単に `y` に関する評価のみです。また、"孤立点" (正規分布のモデルのから外れているデータ点) は常に解を悪化させる可能性があります。

## 27.4 Control

`fit` に影響を与えるために定義できるたくさんの `gnuplot` の変数があります。それらは `gnuplot` の動作中に一度定義できますが、それは `control_variable` で紹介し、`gnuplot` が立ち上がる前に設定する変数は `environment_variables` で紹介します。

### 27.4.1 Control variables

デフォルトのもっとも小さい数字の限界 (`1e-5`) は、変数

`FIT_LIMIT`

で変更できます。残差の平方自乗和が 2 つの反復ステップ間で、この数値より小さい数しか変化しなかった場合、当てはめルーチンは、これを '収束した' と見なします。

反復数の最大値は変数

`FIT_MAXITER`

で制限されます。0 (または定義しない場合) は制限無しを意味します。

更にそのアルゴリズムを制御したい場合で、かつ Marquardt-Levenberg アルゴリズムを良く知っている場合は、さらにそれに影響を与える変数があります。`lambda` ( ) の最初の値は、通常 ML 行列から自動的に計算されますが、もしそれをあらかじめ用意した値にセットしたければ

`FIT_START_LAMBDA`

にセットしてください。`FIT_START_LAMBDA` を 0 以下にセットすると、自動的に計算されるようになります。変数

`FIT_LAMBDA_FACTOR`

は、自乗化された関数が増加、あるいは減少するにつれて `lambda` が増加あるいは減少する因数を与えます。`FIT_LAMBDA_FACTOR` を 0 とすると、それはデフォルトの因数 10.0 が使用されます。

`fit` には `FIT_` から始まる変数が他にもありますから、ユーザ定義変数としてはそのような名前で始まる変数は使わないようにするのが安全でしょう。

変数 `FIT_SKIP` と `FIT_INDEX` は、以前の版の `gnuplot` の、`gnufit` と呼ばれていた `fit` パッチで使われていたもので、現在は使用されていません。`FIT_SKIP` の機能はデータファイルに対する `every` 指定子で用意されています。`FIT_INDEX` は複数当てはめ法 (multi-branch fitting) で使われていたのですが、1 变数の複数当てはめ法は、今では疑似 3 次元当てはめとして行なわれていて、そこでは枝の指定には 2 变数と `using` が使われています。`fit multi-branch` を参照してください。

### 27.4.2 Environment variables

環境変数は `gnuplot` が立ち上がる前に定義しなければなりません。その設定方法はオペレーティングシステムに依存します。

`FIT_LOG`

は、当てはめのログが書かれるファイル名 (およびパス) を変更します。デフォルトでは、作業ディレクトリ上の "fit.log" となっています。そのデフォルトの値はコマンド `set fit logfile` を使って上書きできます。

`FIT_SCRIPT`

は、ユーザが中断した後に実行するコマンドを指定します。デフォルトでは `replot` ですが、`plot` や `load` コマンドとすれば、当てはめの進行状況の表示をカスタマイズするのに便利でしょう。

## 27.5 Multi-branch

複数当てはめ法 (multi-branch fitting) では、複数のデータ集合を、共通のパラメータを持つ複数の 1 变数の関数に、WSSR の総和を最小化することによって同時に当てはめることができます。各データセットに対する関数とパラメータ (枝) は '疑似变数' を使うことで選択できます。例えば、データ行番号 (-1; 'データ列' の番号) またはデータファイル番号 (-2) を 2 つ目の独立变数とします。

例: 2 つの指数減衰形  $z=f(x)$  が与えられていて、それぞれ異なるデータ集合を記述しているが、共通した減衰時間を持ち、そのパラメータの値を評価する。データファイルが  $x:z:s$  の形式であったとすると、この場合以下のようにすればよい。

```
f(x,y) = (y==0) ? a*exp(-x/tau) : b*exp(-x/tau)
fit f(x,y) 'datafile' using 1:-1:2:3 via a, b, tau
```

より複雑な例については、デモファイル "fit.dem" で使われる "hexa.fnc" を参照してください。

もし從属変数のスケールに差がある場合、単位の重み付けでは 1 つの枝が支配してしまう可能性があるので、適当な重み付けが必要になります。各枝をバラバラに当てはめるのに複数当てはめ法の解を初期値として用いるのは、全体を合わせた解の各枝に対する相対的な影響に関する表示を与えることになるでしょう。

## 27.6 Starting values

非線形当てはめは、大域的な最適値 (残差の自乗和 (SSR) の最小値を持つ解) への収束は保証はしませんが、局所的な極小値を与えることはできます。そのサブルーチンはそれを決定する方法を何も持ち合っていないので、これが起こったかどうかを判断するのはあなたの責任となります。

fit は、解から遠くから始めると失敗するかも知れませんし、しばしばそれは起こり得ます。遠くというのは、SSR が大きく、パラメータの変化に対してその変化が小さい、あるいは数値的に不安定な領域 (例えば数値が大きすぎて浮動小数の桁あふれを起こす) に到達してしまって、その結果 "未定義値 (undefined value)" のメッセージが gnuplot の停止を引き起こしてしまう場合を意味します。

大域的な最適値を見つける可能性を改善するには、最初の値をその解に少なくともほぼ近くに取るべきでしょう。例えば、もし可能ならば一桁分の大きさの範囲内で。最初の値が解に近いほど他の解で終了してしまう可能性は低くなります。最初の値を見つける一つの方法は、データと当てはめ関数を同じグラフの上に描画して適当な近さに達するまで、パラメータの値を変更して `replot` することを繰り返すことです。その描画は、よくない当てはめの極小値で当てはめが終了したかどうかをチェックするのも有用です。

もちろん、適度に良い当てはめが、"それよりよい" 当てはめ (ある改良された当てはめの良さの基準によって特徴付けられた統計学的な意味で、あるいはそのモデルのより適切な解である、という物理的な意味で) が存在しないことの証明にはなりません。問題によっては、各パラメータの意味のある範囲をカバーするような様々な初期値の集合に対して fit することが望ましいかも知れません。

## 27.7 Tips

ここでは、fit を最大限に利用するためにいくつか覚えておくべきヒントを紹介します。それらは組織的ではないので、その本質がしみ込むまで何回もよく読んでください。

fit の引数の `via` には、2 つの大きく異なる目的のための 2 つの形式があります。`via "file"` の形式は、バッチ処理 (非対話型での実行が可能) で最も良く使われ、そのファイルで初期値を与え、またその後で結果を他の (または同じ) パラメータファイルにコピーするために `update` を使うことも出来ます。

`via var1, var2, ...` の形式は対話型の実行で良く使われ、コマンドヒストリの機構が使ってパラメータリストの編集を行い、当てはめを実行したり、あるいは新しい初期値を与えて次の実行を行なったりします。これは難しい問題に対しては特に有用で、全てのパラメータに対して 1 度だけ当てはめを直接実行しても、良い初期値でなければうまくいかないことが起こり得るからです。それを見つけるには、いくつかのパラメータのみに対して何回か反復を行ない、最終的には全てのパラメータに対する 1 度の当てはめがうまくいくところに十分近くなるまでそれを繰り返すことです。

当てはめを行なう関数のパラメータ間に共通の依存関係がないことは確認しておいてください。例えば、 $a \cdot \exp(x+b)$  を当てはめに使ってはいけません。それは  $a \cdot \exp(x+b) = a \cdot \exp(b) \cdot \exp(x)$  だからです。よってこの場合は  $a \cdot \exp(x)$  または  $\exp(x+b)$  を使ってください。

技術的なお話し: パラメータの大きさはあまり違いすぎてはいけません。絶対値が最も大きいパラメータと最も小さいパラメータの比が大きい程当てはめの収束は遅くなります。その比が、マシンの浮動小数の精度の逆数に近いか、またはそれ以上ならば、ほとんど永久に収束しないか、拒否されるでしょう。よってその関数をこれを避けるように改良しなければいけません。例えば、関数の定義で 'parameter' を '1e9\*parameter' にするとか、または最初の値を 1e9 で割るとか。

もし、関数を、当てはめるパラメータを係数とする、単純な関数の線形結合で書けるなら、それはとてもいいのでは非そうしてください。何故なら、問題がもはや非線形ではないので、反復は少ない回数で収束するでしょう。もしかしたらたった一回ですむかもしれません。

実際の実験の講義ではデータ解析に対するいくつかの指示が与えられ、それでデータへの最初の関数の当てはめが行なわれます。もしかすると、基礎理論の複数の側面にひとつずつ対応する複数回のプロセスが必要かも知れませんが、そしてそれらの関数の当てはめのパラメータから本当に欲しかった情報を取り出すでしょう。しかし、fit を使えば、求めるパラメータの視点から直接モデル関数を書くことにより、それはしばしば 1 回で済むのです。時々はより難しい当てはめ問題の計算コストがかかりますが、データ変換もかなりの割合で避けることが出来ます。もしこれが、当てはめ関数の単純化に関して、前の段落と矛盾していると思うなら、それは正解です。

"singular matrix" のメッセージは、この Marquardt-Levenberg アルゴリズムのルーチンが、次の反復に対するパラメータの値の計算が出来ないことを意味します。この場合、別な初期値から始めるか、関数を別な形で書き直すか、より簡単な関数にしてみてください。

最後に、他の当てはめパッケージ (fudgit) のマニュアルから、これらの文書を要約するようないい引用を上げます: "Nonlinear fitting is an art! (非線形当てはめ法は芸術だ !)"

## 28 Help

help コマンドは、オンラインヘルプを表示します。ある項についての説明を指定したいときには、次の書式を使って下さい:

```
help {<項目名>}
```

もし <項目名> が指定されなかった場合は、gnuplot についての簡単な説明が表示されます。指定した項目についての説明が表示された後、それに対する細目のメニューが表示され、その細目名を入力することで細目に対するヘルプを続けることができます。そして、その細目の説明が表示された後に、さらなる細目名の入力を要求されるか、または 1 つ前の項目のレベルへ戻ります。これを繰り返すとやがて、gnuplot のコマンドラインへと戻ります。

また、疑問符 (?) を項目として指定すると、現在のレベルの項目のリストが表示されます。

## 29 History

コマンド history は、過去に編集したコマンドラインの履歴 (history) の一覧を表示したり、その一部を保存したり、その一つを実行したりします。

具体例で使用法を紹介します:

```
history          # 履歴全体を表示
history 5        # 履歴内の直前の 5 つを表示
history quiet 5  # エントリ番号なしで直前の 5 つを表示
history "hist.gp" # 履歴全体をファイル hist.gp に書き出す
history "hist.gp" append # 履歴全体をファイル hist.gp に追加する
history 10 "hist.gp" # 直前の 10 個をファイル hist.gp に出力
history 10 "|head -5 >>diary.gp" # パイプで履歴を 5 つ書き出す
history ?load    # 履歴内の "load" で始まるものすべてを表示
history ?"set c" # 上と同様 (複数の語は引用符で囲む)
hi !reread       # "reread" で始まる最も新しい行を実行
hist !"set xr"   # 上と同様 (複数の語は引用符で囲む)
hi !hi           # 当ててみて :-)
```

`popen` 関数をサポートするようなシステム (Unix など) では、履歴の出力をパイプ経由で他の外部プログラムに渡すことができます。その場合、上の例の一つにあるように、ファイル名としてコマンド名の先頭に '`'`' をつけたものを使います。

## 30 If

`if` コマンドは、条件付でコマンドを実行させることができます。

書式:

```
if (<条件>) <コマンド行> [; else if (<条件>) ...; else ...]
```

<条件> が評価され、もしそれが真 (ゼロでない) ならば、<コマンド行> のコマンドが実行されます。もし、<条件> が偽 (ゼロ) ならば、次の `else` までの <コマンド行> 全体が無視されます。; を使うと同じ行に複数のコマンド置くことが可能ですが、これで条件付きのコマンド (`if` の構文自体) は終っていないことに注意してください。

例:

```
pi=3
if (pi!=acos(-1)) print "?Fixing pi!"; pi=acos(-1); print pi
```

を実行すると、

```
?Fixing pi!
3.14159265358979
```

と表示されますが、

```
if (1==2) print "Never see this"; print "Or this either"
```

とすると、何も表示されません。

その他:

```
v=0
v=v+1; if (v%2) print "2" ; else if (v%3) print "3" ; else print "fred"
```

(何度も最後の行を繰り返してみてください !)

`if` と `reread` を使ってループを構成する例については `reread` を参照してください。

## 31 Load

`load` コマンドは、指定された入力ファイルの各行を、それが対話的に入力されたかのように実行します。`save` コマンドでつくられたファイルは、`load` することができます。有効なコマンドの書かれたテキストファイルをつくれば、それは、`load` コマンドによって、実行することができます。`load` 中のファイルの中にさらに `load` または `call` コマンドがあっても構いません。コマンド中のコメントについては、`comments` を参照して下さい。`load` するときに引数を与える方法については `call` を参照してください。

`load` コマンドは、複数のコマンドからなる行の中では最後のコマンドでなければなりません。

書式:

```
load "<入力ファイル名>"
```

入力ファイル名は引用符で囲まなければなりません。

`load` コマンドは、標準入力からのコマンドの入力のために、特別なファイル名 "`-`" を用意しています。これは、`gnuplot` のコマンドファイルが、いくつかのコマンドを標準入力から受け付けることを意味します。詳細については `batch/interactive` に関するヘルプを参照してください。

Unix のように `popen` 関数をサポートするようなシステムでは、'`<`' で始まるファイル名にすることで、入力ファイルをパイプから読み込むことができます。

例:

```
load 'work.gnu'
load "func.dat"
load "< loadfile_generator.sh"
```

gnuplot への引数として与えられたファイル名は、暗黙のうちに load コマンドによって実行されます。これらは、指定された順にロードされ、その後 gnuplot は終了します。

## 32 Lower

書式:

```
lower {plot_window_nb}
```

コマンド lower (raise の反対) は、pm, win, wxt, x11 等の gnuplot の対話型出力形式の実行中に、描画ウィンドウを下(背後)に下げます。描画ウィンドウを、デスクトップ上のウィンドウマネージャの z 方向のウィンドウの重なりの下に置きます。

x11 や wxt のように複数の描画ウィンドウをサポートしている場合、デフォルトではこのコマンドはそれらの複数のウィンドウを降順に下げ、最初に作られたウィンドウを一番下に、最後に作られたウィンドウを一番上に並べます。オプション引数の描画番号が与えられた場合、それに対応する描画ウィンドウが存在すればそれのみが下げられます。

オプション引数は、単一の描画ウィンドウの出力形式、すなわち pm と win では無視されます。

## 33 Pause

pause コマンドは、コマンドに続く任意の文字列を表示した後、指定された時間または、改行キーが押されるまで待ちます。 pause コマンドは、load 用のファイルと共に使用すると、便利になるでしょう。

書式:

```
pause <time> {"<string>"}
pause mouse {<endcondition>}{, <endcondition>} {"<string>"}
```

<time> は、任意の定数または式です。 -1 を指定すると改行キーが押されるまで待ちます。 0 を指定すると一切待たず、正の数を指定するとその秒数だけ待ちます。 実行環境が秒以下の時間指定をサポートしていない場合、その時間は整数の秒数に丸められます。 pause 0 は print と同じです。

使用している出力形式がマウス機能をサポートしている場合、 pause mouse は、マウスクリックがあるか ctrl-C が押されるまで待つようになります。 そうでない出力形式、またはマウス機能が有効になってない場合 pause mouse は pause -1 と同じです。

一つ、あるいは複数の終了条件が pause mouse の後に与えられた場合、そのうちのどの一つでも pause は終了します。 指定できる終了条件は、keypress, button1, button2, button3, any のいずれかです。 pause がキー入力によって終了した場合、押されたキーの ascii コードは MOUSE\_KEY に保存され、文字それ自身は、1 文字の文字列値として MOUSE\_CHAR に返されます。

どの場合でもマウスの座標は変数 MOUSE\_X, MOUSE\_Y, MOUSE\_X2, MOUSE\_Y2 に保存されます。 mouse variable 参照。

注意: pause コマンドは OS へのコマンドであり描画の一部ではないので、異なる出力装置では異なる動作をする可能性があります。(これは、テキストとグラフィックスが、どのように混在するかによります。)

例:

```
pause -1      # 改行キーが押されるまで待つ
pause 3       # 3 秒待つ
pause -1 "続けるには return を打ってください"
pause 10 "これは美しいですか？ 3 次の spline です"
pause mouse "選択したデータ点上で任意のボタンをクリックしてください"
pause mouse keypress "有効なウィンドウ内で A-F の文字を入力してください"
pause mouse button1, keypress
pause mouse any "任意のキー、ボタンで終了します"
```

並種である "pause mouse key" は、有効な描画ウィンドウ内の任意のキー入力によって再開されます。特別なキー入力まで待つようにしたい場合は、以下のような reread によるループを使うことができます:

```
printf "描画ウィンドウ内で Tab キーを打つと復帰します。"
load "wait_for_tab"
```

ファイル "wait\_for\_tab" は以下のよう�습니다:

```
pause mouse key
if (MOUSE_KEY != 9) reread
```

## 34 Plot

plot は gnuplot で図を描くための基本的なコマンドです。それは関数やデータを実際に多くの方法で表示します。plot は 2 次元の関数やデータを描くのに使われ、splot は 3 次元の曲面やデータの 2 次元投影を描きます。plot と splot は多くの共通の特徴点を持ちますが、その違いについては splot の項を参照してください。特に注意しておきますが、binary <binary list> オプションは plot, splot の両方で動作しますが、両者には小さな違いがありますし、plot の axes オプションは splot には存在しません。

書式:

```
plot {<範囲 (ranges)>}
  { <関数 (function)>
    | {"<データファイル (datafile)>" {データファイル修飾子}}}
  {axes <軸 (axes)>} {<表題 (title)>} {with <スタイル (style)>}
  {, {定義,} <関数> ...}
```

<関数> または引用符で囲まれたデータファイル名のどちらか一方を与えます。関数は一本の数式、または parametric mode においては 2 つの数式の組です。数式は完全に定義してもいいですし、前の方の gnuplot のコマンド列で部分的に定義してもいいです (user-defined の項目参照)。

関数とパラメータは plot コマンド自身の上で定義することも可能です。これは単に他の項目とコンマで分離して記述することでなされます。

軸は、4 種類の組が利用できます; キーワード <軸> は、特定の直線をどの軸に尺度を合わせるか、ということを選択するのに使われます。x1y1 は下の軸と左の軸を指定; x2y2 は上と右の軸の指定; x1y2 は下と右の軸の指定; x2y1 は上と左の軸の指定です。plot コマンドで指定された範囲は、この最初の軸の組 (下と左) にのみ適用されます。

例:

```
plot sin(x)
plot f(x) = sin(x*a), a = .2, f(x), a = .4, f(x)
plot [t=1:10] [-pi:pi*2] tan(t), \
      "data.1" using (tan($2)):(${3/$4}) smooth csplines \
      axes x1y2 notitle with lines 5
```

show plot も参照してください。

### 34.1 Data

ファイルに納められた離散的なデータは、plot コマンドライン上で、そのデータファイル名 (单一引用符または二重引用符で囲まれた) を指定することによって表示できます。

書式 :

```
plot '<file_name>' {binary <binary list>}
  {matrix}
  {index <index list>}
  {every <every list>}
  {thru <thru expression>}
  {using <using list>}
  {smooth <option>}
```

修正子の `binary`, `index`, `every`, `thru`, `using`, `smooth` は、それぞれに分けて説明します。簡単に言うと、`binary` はデータ列をバイナリファイルから取得できるようにし（デフォルトは ASCII ファイル）、`index` はマルチデータセットファイルからどのデータセットを表示するのかを選び、`every` が、一つのデータセットからどの点を表示するのかを選び、`using` は一行からどの列を解釈するのかを決定し（`thru` は、`using` の特別な場合である）、そして `smooth` が、単純な補間と近似を行います。（`splot` は、よく似た書式を持っていますが、`smooth` オプションと `thru` オプションはサポートしていません）

#### ASCII データファイル:

データファイルは、一行につき少なくとも一つのデータ点を含む必要があります（`using` は一行から一つのデータポイントを選ぶことができます）。#（VMS では !）で始まる行は、コメントとして扱われ、無視されます。各データ点は、(x,y) の組を表します。エラーバー、または折れ線表示付エラーバーの `plot` では（`set style errorbars` または `set style errorlines` 参照）、各データ点は、(x,y,ydelta), (x,y,ylow,yhigh), (x,y,xdelta), (x,y,xlow,xhigh), (x,y,xlow,xhigh,ylow,yhigh) のいずれかを意味します。

どんな場合でも、書式の指定子が `using` オプションによって与えられていなければ、データファイルの各行の数字は、ホワイトスペース（一つまたは複数の空白かタブ）によって区切られている必要があります。このホワイトスペースは、各行を列の項目に区切れます。ただし、二重引用符で囲まれたホワイトスペースは列の勘定では無視され、よって次のようなデータ行は 3 列、と見なされます：

```
1.0 "second column" 3.0
```

データは、指数部に e, E, d, D, q, Q の文字をつけた指数表記で書かれても構いません。

必要であるのはただ一つの列（y の値）のみです。もし x の値が省略されたら、`gnuplot` はそれを 0 で始まる整数値として用意します。

データファイルにおいて、ブランク行（空白と改行、復帰以外に文字を含まない行）は重要です — - ブランク行の対は、`index`（`plot datafile index` 参照）を区切れます。2 つのブランク行で分離されたデータは、別々のデータファイルのデータであるかのように扱われます。

一つのブランク行は、`plot` に不連続を指示します；ブランク行によって区切られた点は線で結ばれることはありません（line style で書かれている場合には）。

もし `autoscale` の状態であれば（`set autoscale` 参照）、軸は全てのデータポイントを含むように自動的に引き伸ばされて、目盛りが書かれる状態ならば全ての目盛りがマークされます。これは、2 つの結果を引き起こします： i) `splot` では、曲面の角は底面の角に一致していないことがあります。この場合、縦の線は書かれることはできません。ii) 2 種類の軸での、同じ x の範囲のデータの表示の際、もし x2 の軸に対する目盛りが書かれていらない場合は、x 座標があつてないことがあります。これは x 軸 (x1) は全ての目盛りにまで自動的に引き延ばされるのに対し、x2 軸はそうではないからです。次の例でその問題を見るすることができます：

```
reset; plot '-' '-' axes x2y1
1 1
19 19
e
1 1
19 19
e
```

これを避けるには、`set autoscale` コマンドの `fixmin/fixmax` オプションを使うことができます。これは、軸の範囲を自動的に拡張して、次の目盛りの刻みに合うようにする機能を無効にします。

#### バイナリデータファイル:

`gnuplot` はバイナリデータファイルを読むこともできます。しかし、そのファイルフォーマットに関する十分詳細な情報は、ユーザがコマンドラインから与えるか、またはサポートされている `filetype` のバイナリ形式のファイルそれ自身から抜き出される必要があります。バイナリファイルには 2 つの形式、`matrix` バイナリ形式と `general` バイナリ形式があります。

`matrix` バイナリ形式は、32 ビット IEEE 規格の浮動小数値（float）が 2 次元配列の形で並び、それらの座標値を表す行と列が追加されています。ASCII `matrix` 同様、`using` リストにおいては、座標用の列の並び（1 列目）が `using` の 1 番目の要素（x 座標）に対応し、座標用の行の並び（1 行目）が `using` の 2 番目の要素（y 座標）に対応し、配列の各値が `using` の 3 番目の要素（z 座標）に対応します。

`general` バイナリ形式は、任意個の列のデータを含み、それらの情報はコマンドラインで指定する必要があります。例えば `array`, `record`, `format`, `using` などでサイズや形式、データの次元を指定できます。他

にも、ファイルヘッダ読み飛ばしたり、エンディアン (endian) を変更するための有用なコマンドがありまし、配置、データの変換を行なうコマンドの組があります。それは、一様に標本化されたデータの場合、その座標がファイルには含まれないことが良くあるからです。matrix バイナリや ASCII データと違っているところですが、general バイナリは 1,2,3 といった using リストで生成される列番号を使わず、むしろ 1 列目はファイルの 1 列目、あるいは format リストで指定されたもの、になります。

さまざまな binary オプションに対する大域的なデフォルトの設定も可能で、それは (s)plot <filename> binary ... コマンドに与えるオプションと全く同じ書式で指定できます。その書式は set datafile binary ... です。一般的な規則として、デフォルトのパラメータはファイルから抜き出されたパラメータで上書きされ、それはコマンドラインで指定された共通なパラメータで上書きされます。

例えば array, record, format, filetype の general バイナリ形式を特定するようなキーワードが何もついていなければ、デフォルトのバイナリ形式は matrix バイナリです。

general バイナリデータは、特別なファイル名 `'` を使ってコマンドラインから入力することもできます。しかし、これはキーボードからの入力を意図したものではなく、パイプを使ってプログラムにバイナリ形式を変換させるためのものです。バイナリデータには最後を表す記号がありませんので、gnuplot はパイプからデータを読み込む場合、array 指定子で指定した数の点数になるまでデータを読み込み続けます。

詳細に関しては datafile binary を参照してください。

### 34.1.1 Binary

キーワード binary はデータファイルを ASCII ではなくてバイナリファイルと認識させます。バイナリファイルフォーマットには、matrix バイナリと general binary の 2 つがあります。matrix バイナリは固定フォーマットで、データは、2 次元配列の形で並び、その座標を表す余分な列と行がついています。general バイナリは柔軟なフォーマットで、そのファイルに関する詳細はコマンドラインで指示する必要があります。

詳細に関しては binary matrix と binary general を参照してください。

### 34.1.2 Binary general

general バイナリデータは、その形式に関する情報を必ずしもファイル内にはなくとも良く、ファイルの形式に関する詳細な情報をコマンドラインから与えることで、そのデータを読み込むことが可能になります。そのオプションの書式は一時的なユーザにはやや難解ですが、general バイナリは特に gnuplot を使い、多くのデータを送るようなアプリケーションに取っては有用です。

書式:

```
plot '<file_name>' {binary <binary list>} ...
splot '<file_name>' {binary <binary list>} ...
```

general バイナリ形式は、ファイル構造に関する情報に関するキーワード、すなわち array, record, format, filetype などを <binary list> 内に与えることで有効になります。それ以外の場合は matrix バイナリ形式と見なされます。(詳細に関しては binary matrix 参照。)

バイナリ形式に関する情報が自動的にファイルから読み取ることができるような標準的なファイル形式もいくつかあります(その一覧を見るには、gnuplot の対話画面から show datafile binary してください)。それ以外の場合はコマンドライン、あるいはデフォルトの値を設定する必要があります。キーワードに関しては以下で説明します。

<binary list> に与えるキーワード filetype は、ファイルを読み込むのに使用されるルーチン、すなわちデータの形式の扱いを制御します。サポートしている filetype の一覧を見るには、show datafile binary filetypes としてください。filetype を指定しない場合の規則は、splot 用に binary キーワードが単独で与えられた場合は古典的な gnuplot バイナリ形式だと見なされ、その他全ての状況、すなわち plot 用かまたは少なくとも一つの <binary list> キーワードが与えられたような場合は、raw バイナリファイルとみなされ、そのキーワードによってバイナリ形式が指定されます。

general バイナリデータファイルは 2 つの基本的なクラスに分かれますが、それがどのように扱われるのかによって両方のクラスに入る場合もあります。一つのクラスは、データは一様にサンプリングされていると見なされ、点の座標は必然的にそこから生成されます。このクラスでは <binary list> キーワードを用いて全ての制御が行なわれます。そして、このクラスでの設定の優先順位は、コマンドラインパラメー

タがファイル内のパラメータを上書きし、それはデフォルトの設定を上書きします。もう一つのクラスは、座標情報がファイル内に含まれた複数のファイルの組のためのもの、あるいは gnuplot バイナリのような非一様な標本化を含み得る一つのファイルのためのものです。

gnuplot バイナリのような特殊なデータファイル以外では、一般にバイナリデータは概念上は ASCII データと同様に考えられます。各点は情報の列を持っていて、それは using によって結びつけられる <using list> で選択されます。format 文字列が何も指定されなかった場合、gnuplot はバイナリ変数の数を <using list> で与えられる最大の列番号に等しく取ります。例えば using 1:3 の場合は 3 列ずつデータが読み取られ、2 番目のものは無視されます。各描画スタイルにはその典型的なパラメータの個数があり、それに基づく <using list> がデフォルトとして選択されます。例えば with image では デフォルトで using 1 が、with rgbimage ではデフォルトで using 1:2:3 が使われます。using で 点/線/index を表す特別な文字は、一般にはバイナリデータでは使うべきではないことに注意してください。これを制御する <binary list> のキーワードがあります。

**34.1.2.1 Array** バイナリファイルの標本の配列の大きさを設定します。座標は gnuplot が生成してくれます。各方向の次元を表す数、すなわち配列のサイズを少なくとも一つは指定しなければいけません。例えば array=10x20 は、2 次元で最初の次元方向 (x) には 10 点、2 番目の次元方向 (y) には 20 点の標本化データがあることを意味します。特別な "数字" として Inf を、ファイルの終了までデータが続くことを示すのに使えます。複数のデータのサイズ指定を分離するのに、コロンを使うことができます。例えば array=25:35 は 2 つの 1 次元データがファイルの中にあることを意味します。そのコロンの意味は他の <binary list> 内のキーワードにも影響を与え、それが個々のデータに対応するように意味を持つようになります。

今のところ、3 次元までの配列が書式として許されていますが、3 次元配列を扱うための規則はまだ何も用意されていません。

**34.1.2.2 Record** このキーワードは array と同じ書式で、同じ機能を提供します。しかし record は gnuplot に座標情報を自動生成させません。これは、そのような座標情報が、バイナリデータファイルのある列に含まれている場合のためのものです。

**34.1.2.3 Format** デフォルトのバイナリ形式は、単精度浮動小数 (float) が一つ、です。それをより柔軟に設定するために、この format で変数のサイズに関する詳細な情報を指定できます。例えば format="uchar%int%float" は、最初の using 列として符号なし文字型変数 (unsigned char) を、2 番目の列は符号つき整数 (int) を、3 番目の列は単精度浮動小数 (float) を指定しています。もしサイズ指定子の数が最大列数より小さい場合は、残りの列の変数サイズは暗黙のうちに最後に与えた変数サイズに等しく取られます。

さらに、\* 文字がついた、"読み捨てられる" 列を書式に含めることも可能です。例えば前の例で真中の列のデータを飛ばすのに format="%uchar%\*int%float" と書いて gnuplot は真中の整数データを読み捨てさせることができます。使用できる変数サイズの一覧は、show datafile binary datasizes で見ることができます。それらは、それぞれのコンパイルによってそのバイトサイズとともにマシンに依存する変数名のグループと、マシンに依存しない変数名のグループに分かれています。

**34.1.2.4 Endian** ファイルのバイナリデータのエンディアンは、gnuplot が動作するプラットホームのエンディアンとは異なる場合も良くあります。いくつかの指定で gnuplot がバイトをどのように扱うかを制御できます。例えば endian=little は、バイナリファイルを、そのバイトの並びが小さい桁から大きい桁へ並んでいると見なされます。オプションは以下のものが使えます。

```
little: 小さい桁から大きな桁へ並ぶ
big: 大きな桁から小さな桁へ並ぶ
default: compiler と同じエンディアンと見なす
swap (swab): エンディアンを変更する (おかしいようならこれを
               使ってみてください)
```

gnuplot は、コンパイル時にオプションが指定されていれば、"middle" (や"pdp") エンディアンもサポートできます。

**34.1.2.5 Filetype** gnuplot は、いくつか標準的なバイナリファイル形式については必要な情報をそのファイルから抜き出すことができます。例えば "format=edf" は ESRF ヘッダーファイル形式のファイルとして読み込みます。現在サポートしているファイル形式については、show datafile binary filetypes で見てください。

特別なファイル形式として auto があり、この場合 gnuplot はバイナリファイルの拡張子が、サポートされている形式の標準的な拡張子であるかをチェックします。

コマンドラインキーワードはファイルから読み取る設定を上書きするのに使われ、ファイルから読み取る設定はデフォルトの設定を上書きします（詳細は set datafile binary 参照）。

**34.1.2.5.1 Avs** avs は、自動的に認識される画像イメージに対するバイナリファイルの型の一つです。AVS は非常単純なフォーマットで、アプリケーション間でやりとりするのに最も適しています。これは、2つの long (xwidth と ywidth) と、その後続くピクセルの列から成り、その各ピクセルは alpha/red/green/blue の 4 バイトから成ります。

**34.1.2.5.2 Edf** edf は、自動的に認識される画像イメージに対するバイナリファイルの型の一つです。EDF は ESRF データフォーマット (ESRF Data Format) を意味していて、それは edf と ehf の両方の形式をサポートしています（後者は ESRF Header Format）。画像の使用に関する詳しい情報は以下で見つかるでしょう：

<http://www.esrf.fr/computing/expg/subgroups/general/format/Format.html>

binary も参照してください。

**34.1.2.6 Keywords** 以下のキーワード (keyword) は、座標の生成のときにのみ適用されます。よって、これらは array キーワードが使われるときのためのものです。

**34.1.2.6.1 Scan** gnuplot がバイナリファイルをどのように走査するか、ということと実際の描画で見られる軸の方向との間の関係については多くの混乱が起こり得ます。その混乱を減らすには、gnuplot はバイナリファイルを "常に" 点/線/面、または速い/普通/遅い、と走査すると考えるといいでしょう。このキーワードは gnuplot に、その走査の方向を描画内のどの座標方向 (x/y/z) に割り当てるかを指定します。指定は 2 つ、または 3 つの文字の並びで表現し、最初の文字が点に、次の文字が線に、3 つ目の文字が面に対応します。例えば、scan=yx は、最も速い走査（点の選択）は y 方向に対応し、普通の速さの走査（線の選択）が x 方向に対応することを意味します。

描画モードが plot の場合、指定には x と y の 2 つの文字を使うことができ、splot に対しては x, y, z の 3 つの文字使うことができます。

割り当てに関しては、点/線/面から直交座標方向へのみに制限する内部事情は別にありません。この理由で、円柱座標への割り当てのための指定子も用意されていて、それらは直交座標の x, y, z に類似した形で t (角度), r, z となっています。

**34.1.2.6.2 Transpose** scan=yx、または scan=yxz と同じです。

**34.1.2.6.3 Dx, dy, dz** gnuplot が座標を生成する場合、その間隔はこれらのキーワードで指定されたものが使用されます。例えば dx=10 dy=20 は x 方向に 10、y 方向に 20 の間隔で標本化されたことを意味します。dy は dx がなければ使えません。同様に dz は dy がなければ使えません。もしデータの次元が指定したキーワードの次元よりも大きい場合、残りの次元方向の間隔は、指定された最も高い次元のものと同じ値が使用されます。例えば画像がファイルから読み込まれ、dx=3.5 のみ指定された場合、gnuplot は x 方向の間隔も y 方向の間隔も 3.5 を使用します。

以下のキーワードも座標の生成時にのみ適用されます。しかし、以下のものは matrix バイナリファイルにも使われます。

**34.1.2.6.4 Flipx flipy, flipz** バイナリデータファイルの走査方向が gnuplot の走査方向と一致しないことがたまにあります。これらのキーワードは、それぞれ x, y, z 方向のデータの走査方向を逆向きにします。

**34.1.2.6.5 Origin** gnuplot は転置 (transpose) や反転 (flip) において座標を生成する場合、常に配列の左下の点が原点になるようにします。すなわち、データが、転置や反転の行なわれた後の直交座標系の第 1 象限に来るようになります。

配列をグラフのその他の場所に配置したい場合、origin キーワードで指定した場所に gnuplot は配列の左下の点を合わせます。その指定は、plot では 2 つの座標の組、splot では 3 つの座標の組を指定してください。例えば origin=(100,100):(100,200) は、一つのファイルに含まれる 2 つのデータに対する指定で、2 次元の描画に対する指定です。2 つ目の例として origin=(0,0,3.5) をあげると、これは 3 次元描画用の指定です。

**34.1.2.6.6 Center** origin と似ていますが、このキーワードは、配列の中心がこのキーワードで指定した点になるように配置します。例えば center=(0,0) のようにします。配列のサイズが Inf のときは center は適用されません。

**34.1.2.6.7 Rotate** 転置 (transpose) と反転 (flip) コマンドは座標の生成と座標軸の方向にある種の柔軟性を与えてくれます。しかし、角度に関する完全な制御は、2 次元の回転角を記述した回転角ベクトルを与えることにより行なうことが可能になります。

キーワード rotate は、plot, splot の両方で、2 次元面に対して適用されます。回転は座標平面の正の角度に関して行なわれます。

角度は、ラジアン単位ですが、pi や degrees の倍数としてのラジアンでも表現できます。例えば、rotate=1.5708, rotate=0.5pi, rotate=90deg はすべて同じ意味です。

origin が指定された場合、回転は平行移動の前に左下の点を中心にして行なわれます。それ以外では回転は配列の中心 (center) に関して行なわれます。

**34.1.2.6.8 Perpendicular** splot に関して回転ベクトルの設定が、ベクトルを表現する 3 つの数字の組を指定することで実装されていて、このベクトルは 2 次元の xy 平面に対して向き付けられた法線ベクトル (perpendicular) を表しています。もちろんそのデフォルトは (0,0,1) です。rotate と perpendicular の両方を指定することにより、3 次元空間内で無数の方向へデータを向き付けられることになります。

まず最初に 2 次元の回転が行なわれ、その次に 3 次元の回転が行なわれます。つまり、R' をある角による  $2 \times 2$  の回転行列とし、P を  $(0,0,1)$  を  $(xp,yp,zp)$  へ子午線方向に回転させる  $3 \times 3$  の行列とし、R' を左上の部分行列として持ち 3,3 成分が 1 でその他の成分が 0 であるような行列（つまり z 軸周りの回転行列）とすれば、この変換を表す行列による関係式は  $v' = P R v$  となります。ここで、v はデータファイルから読み込まれた  $3 \times 1$  の位置ベクトルです。ファイルのデータが 3 次元的なものでない場合は、論理的なルールが適用されて 3 次元空間内のデータと見なされます（例えば、通常は z 座標は 0 とされ、xy 平面内の 2 次元データと見なされます）。

#### 34.1.2.7 Binary examples 例:

```
# 2 つの float の値を選択し (2 つ目の値は無意味)、一方を読み捨て、
# 一つおきの float 値を無限に長く続く 1 次元データとして使用
plot '<file_name>' binary format="%float%*float" using 1:2 with lines

# データファイルから座標を生成するのに必要な情報をすべてそのヘッ
# ダに含んでいる EDF ファイルの場合
plot '<file_name>' binary filetype=edf with image
plot '<file_name>.edf' binary filetype=auto with image

# 3 つの符号なし文字型整数値 (unsigned char) を生の RGB 画像の色
# 成分として選択し、y 方向は反転させ画像の方向を座標平面上で変更
# する（左上が原点になるように）。ピクセルの間隔も指定し、ファイ
# ルには 2 つの画像が含まれていて、そのうち一つは origin で平行
# 移動する。
plot '<file_name>' binary array=512x1024:1024x512 format='%uchar' \
      dx=2:1 dy=1:2 origin=(0,0):(1024,1024) flipy u 1:2:3 w rgbimage
```

```
# 4 つの別のデータからなり、座標情報もデータファイルに含まれてい  
# る。ファイルは gnuplot が実行されているシステムとは異なるエン  
# ディアンで生成されている。  
splot '<file_name>' binary record=30:30:29:26 endian=swap u 1:2:3
```

binary matrix も参照してください。

### 34.1.3 Every

キーワード every は、描画するデータをデータセットから周期的にサンプリングすることを可能にします。ここでは「ポイント」はファイル中の 1 つの行によって定義されるデータとし、ここでの「ブロック」は「データ・ブロック」(glossary 参照)と同じものを意味することとします。

書式：

```
plot 'file' every {<ポイント増分>}  
    {:{<ブロック増分>}  
     {:{<開始ポイント>}  
      {:{<開始ブロック>}  
       {:{<終了ポイント>}  
        {:{<終了ブロック>}}}}}}
```

プロットされるデータポイントは、<開始ポイント> から <終了ポイント> まで <ポイント増分> の増加で選ばれ、ブロックは <開始ブロック> から <終了ブロック> まで <ブロック増分> の増加で選ばれます。

各ブロックの最初のデータは、ファイル中の最初のブロックと同じように、「0 番」とされます。

プロットできない情報を含んでいる行もカウントされることに注意して下さい。

いくつかの数字は省略することができます；増分のデフォルトは 1、開始の値は最初のポイントか最初のブロック、そして終了の値は最後のポイントか最後のブロックに設定されます。every が指定されないなら、全ての行の全てのポイントがプロットされます。

例：

```
every ::::3::3 # 4 番目のブロックだけ選ばれます (0 番が最初)  
every ::::::9 # 最初の 10 ブロックが選ばれます  
every 2:2 # 1 つおきのブロックで 1 つおきのポイントが選ば  
# れます  
every ::5:::15 # それぞれのブロックでポイント 5 から 15 までが  
# 選ばれます
```

参照：

単純な plot デモ (simple.dem)

,

非パラメータモードでの splot デモ

,

パラメータモードでの splot デモ

### 34.1.4 Example datafile

次の例は、ファイル "population.dat" 中のデータと理論曲線を図にするものです。

```
pop(x) = 103*exp((1965-x)/10)  
plot [1960:1990] 'population.dat', pop(x)
```

ファイル "population.dat" は次のようなファイルです。

```
# Gnu population in Antarctica since 1965  
1965 103
```

```

1970    55
1975    34
1980    24
1985    10

```

### 34.1.5 Index

キーワード `index` はマルチデータセットファイルの中の、いくつかのデータセットのみを選び出すのに使われます。

書式:

```
plot 'file' index <m>{<n>}:<p>
```

データセットは 2 行の空白で分離されています。`index <m>` は `<m>` 番目のセットだけを選択します;`index <m>:<n>` は `<m>` から `<n>` までのデータセットの選択; `index <m>:<n>:<p>` は、`<m>`, `<m>+<p>`, `<m>+2<p>`, など、`<p>` おきのセットを選択し、セット `<n>` で終了します。C 言語の添字 (`index`) の付け方に従い、`index 0` はそのファイルの最初のデータセットを意味します。大きすぎる `index` の指定にはエラーメッセージが返されます。`index` が指定されない場合は、全てのデータセットが単一のデータセットとして描かれます。

例:

```
plot 'file' index 4:5
```

### 34.1.6 Smooth

`gnuplot` は、データの補間と近似を行う汎用的なルーチンをいくつか持っています。これ `smooth` オプションの中にグループ化されています。より洗練されたデータ処理をしたければ、外部においてデータの前処理をするか、または適切なモデルで `fit` を使うのがいいでしょう。

書式:

```
smooth {unique | frequency | csplines | acsplines | bezier | sbezier}
```

`unique` と `frequency` は、データを単調に揃えた後でそれらを `plot` します。他のルーチンはいずれも、データの両端の点の間を結ぶ、ある連続曲線の係数を決定するためにデータを使います。この曲線は、関数として同じ方法で描画されます。すなわち、それらの値は `x` 座標に沿う同じ幅の区間ごとに選ばれ (`set samples` 参照)、それらの点を線分でつなぐことにより (もし `line style` が選ばれているのならば) 描画されます。

もし `autoscale` の状態であれば、描画範囲はグラフの境界線の中に曲線が収まるように計算されます。

もし `autoscale` の状態でなく、`smooth` のオプションが `acspline` か `cspline` であれば、生成する曲線の標本化は、入力データを含むような `x` の範囲と、`set xrange` などで定義される固定された横座標の範囲の共通部分の上で行なわれます。

選択されたオプションを適用するのにデータの点数が少なすぎる場合は、エラーメッセージが表示されます。その最小のデータ数は `unique` と `frequency` では 1 つ、`acsplines` では 4 つ、他のオプションでは 3 つです。

`smooth` オプションは、関数の描画のときには無視されます。

**34.1.6.1 Acsplines** `acsplines` オプションは「自然な滑らかなスプライン」でデータを近似します。データが `x` に関して単調にされた後 (`smooth unique` 参照)、1 つの曲線が、いくつかの 3 次多項式の一部分により区分的に構成されます。それらの 3 次式の係数は、いくつかのデータポイントの重み付けによって求められます。重みは、データファイルの 3 列目に与えます。そのデフォルトの値は、`using` の 3 番目の項目によって変更することができます。例えば次のようにします。

```
plot 'data-file' using 1:2:(1.0) smooth acsplines
```

性質上、重みの絶対的な大きさは、曲線を構成するのに使われる区分の数を決定します。もし重みが大きければ、個々のデータの影響は大きくなり、そしてその曲線は、隣り合う点同志を自然 3 次スプラインでつないで得られるものに近づきます。もし重みが小さければ、その曲線はより少ない区分で構成され、それによってより平滑的になります。その最も極端な場合はただ 1 つの区分からなる場合であり、それは全てのデータに重みの付き線形最小 2 乗近似によって作られます。誤差の立場から言えば、平滑さの重みは、その曲線に対する「平滑化因子」によって分割された各点への、統計的な重みと見ることができます。それにより、そのファイル中の（標準的な）誤差は平滑さの重みとして使うことができます。

例:

```
sw(x,S)=1/(x*x*S)
plot 'data_file' using 1:2:(sw($3,100)) smooth acsplines
```

**34.1.6.2 Bezier bezier** オプションは、n 次（データ点の個数）のベジェ曲線でデータを近似します。この曲線は両端の点をつなぎます。

**34.1.6.3 Csplines csplines** オプションはデータを単調に揃えた後で（smooth unique 参照）自然 3 次スプライン曲線で引き続く点をつなぎます。

**34.1.6.4 Sbezier sbezier** オプションは、最初にデータを単調に揃え（unique 参照）そして bezier アルゴリズムを適用します。

**34.1.6.5 Unique unique** オプションは、データを x 方向に単調にします。同じ x を持つデータ点は y の値を平均して一つの点で置き換えます。そしてその結果として得られる点を線分で結びます。

デモ

**34.1.6.6 Frequency** オプション frequency は、データを x に関して単調にします。x 座標が同じ点は、それらの y の値の合計を y の値として持つ一つの点に置き換えられます。結果としてこれらの点は真っすぐな線分で結ばれることになります。

### 34.1.7 Special-filenames

'-' という特別なファイル名は、データがオンラインであることを指示します。すなわち、データをコマンドの後に続けて指定します。このときはデータのみがコマンドに続き得ます。よって、plot コマンドに対するフィルター、タイトル、ラインスタイルといったオプションは、plot のコマンドラインの方に書かないといけません。これは、unix シェルスクリプトにおける << (ヒアドキュメント)、あるいは VMS DCL における \$DECK と同様です。そのデータは、それらがファイルから読み込まれたかのように、1 行につき 1 つずつのデータ点が入力されます。そしてデータの終りは、1 列目の始めに文字 "e" を置くことで指示します。using オプションをこれらのデータに適用することは可能ですが、ある関数を通してデータをフィルターすることに使うのは意味があるでしょうが、列を選ぶのに使うことは多分意味がないでしょう。

'-' は、データとコマンドと一緒に持つことが有用である場合のためにあります。例えば、gnuplot があるフロントアプリケーションのサブプロセスとして起動される場合などがこれにあたります。例として、デモンストレーションでこの機能を使うものがあるでしょう。index や every のような plot のオプションが与えられていると、それらはあなたに使われることのないデータを入力する事を強要します。次の例を見てください。

```
plot '-' index 0, '-' index 1
2
4
6
10
12
14
e
```

```
2
4
6

10
12
14
e
```

これは、実際に動作しますが、

```
plot '-' , '-'
2
4
6
e
10
12
14
e
```

とタイプする方が楽でしょう。

もし、replot コマンドで '-' を使うなら、あなたは 1 度以上データを入力する必要があれでしょう。

空のファイル名 ('') は、直前のファイル名が再び使われることを指示します。これは、

```
plot 'ある/とても/長い/ファイル名' using 1:2, '' using 1:3, '' using 1:4
```

のようなときに便利です。(もし同じ plot コマンド上で、'-' と '' の両方を使用すると、上の例にあるように、インラインデータの 2 つのセットを与える必要があります。)

popen 関数を持っているコンピュータシステム (Unix) の上では、データファイルは、'<' で始まるファイル名によって、シェルコマンドからパイプ入力することができます。例えば

```
pop(x) = 103*exp(-x/10)
plot "< awk '{print $1-1965, $2}' population.dat", pop(x)
```

は、最初の人口の例と同じ情報を描画します。ただし、x 座標は 1965 年からの経過年を表すようになります。この例を実行するときは、上のデータファイルのコメント行をすべて削除しなければなりませんが、または上のコマンドの最初の部分を次のように変えることもできます (コンマに続く部分):

```
plot "< awk '$0 !~ /#/ {print $1-1965, $2}' population.dat"
```

このアプローチは最も柔軟性がありますが、using あるいは thru キーワードを用いた単純なフィルタリングで行うことも可能です。

### 34.1.8 Thru

thru 関数は前のバージョンとの互換性のために用意されています。

書式:

```
plot 'file' thru f(x)
```

これは次と同様です:

```
plot 'file' using 1:(f($2))
```

後者の方がより複雑に見えますが、この方が柔軟性を持っています。さらに自然な

```
plot 'file' thru f(y)
```

も動作します (すなわち、y をダミー変数として使うことができます)。

thru は splot と fit でも通りますが、何の効果も持ちません。

### 34.1.9 Using

最もよく使われるデータファイルの修飾子は `using` です。

書式:

```
plot 'file' using {<entry> {:<entry> {:<entry> ...}}}' {'format'}
```

もし、フォーマット (`format`) が指定されれば、C のライブラリ関数 `'scanf'` を使ってデータファイルの各行をそのフォーマット文字列に従って読み込みます。そうでなければ、行はスペースまたはタブの所で列に分割されて読み込まれます。時系列フォーマットデータ (time-format data) の書式はこの方法では指定できません (代わりに `set xdata time` を使ってください)。

データは `entry` の指定に従った列に並び直されます。各 `<entry>` には、データを選び出すための単なる列の番号、カッコで囲まれた数式を指定するか、あるいは何も指定しません。数式中では、最初の列の値を読み込むために `$1`、`2` 番目の列の項目を使うために `$2`、といった書き方を使用できます。また、`column(x)` や `valid(x)` といったものも使うことができます。ここで、`x` は結果として整数になる任意の数式です。`column(x)` は `x` 番目のデータを返します。`valid(x)` は `x` 番目のデータが有効な値かをテストします。列番号の `0` は、各点毎に `0` から始まる番号を表し、それは `2` 行の空行が来たところでリセットされます。列番号の `-1` は `0` から始まるデータ行の番号を意味します。これは `1` 行の空行毎に `1` ずつ増加し、`2` 行の空行が来たところでリセットされます。列番号の `-2` は `index` を意味します。これは `2` 行の空行が来たところで `1` ずつ増加します。`<entry>` に何も書かなければその `entry` のリストの順にデフォルトの値が使われます。例えば `using ::4` は `using 1:2:4` と解釈されます。

注: `call` コマンドも `$` を特別な文字として使います。`call` の引数リストの中に列番号を含ませる方法の詳細については `call` の項目を参照してください。

`using` にただ一つの `entry` を指定した場合はその `<entry>` は `y` の値として使われ、データ点の番号が `x` として使われます。例えば "`plot 'file' using 1`" は "`plot 'file' using 0:1`" と同じ意味です。`using` に `2` つの `entry` を与えた場合、それらは `x, y` として使われます。さらに `entry` を追加すると、それらは `x` および/または `y` の誤差に使われます。誤差情報を使った `plot` スタイルの詳細については `set style` を、そして、回帰曲線法での誤差情報の使用については `fit` を参照してください。

`'scanf'` 関数では色々なデータ形式の数値入力が使えますが、`gnuplot` は全ての入力データを倍精度浮動小数とみなしますから、`gnuplot` では `%lf` が本質的に唯一の数値入力指定、ということになります。書式文字列には、少なくとも一つ、そして `7` つ以下の、そのような入力指定子を入れる必要があります。`'scanf'` は数と数の間にホワイトスペース – 空白、タブ ("`\t`")、改行 ("`\n`")、または改ページ ("`\f`") – があると期待します。それ以外の入力は明示的にスキップされるべきです。

`"\t", "\n", "\f"` を使うときはシングルクオートよりむしろダブルクオートを使うべきであることに注意してください。

例:

次の例は、`1` 番目のデータに対する `2` 番目と `3` 番目の和の値を `plot` します。書式文字列は、各列データがスペース区切りでなく、カンマ区切りであることを指示していますが、同じことが `set datafile separator ","` を指定することでも可能です。

```
plot 'file' using 1:(\$2+\$3) '%lf,%lf,%lf'
```

次の例は、より複雑な書式指定でデータをファイル "MyData" から読み込みます。

```
plot 'MyData' using "%*1f%1f%*20[^\\n]%1f"
```

この書式指定の意味は以下の通りです:

<code>%*1f</code>	数値を無視
<code>%1f</code>	倍精度浮動小数を読み込む (デフォルトでは <code>x</code> の値)
<code>%*20[^\\n]</code>	20 個の改行以外の文字を無視
<code>%1f</code>	倍精度浮動小数を読み込む (デフォルトでは <code>y</code> の値)

3 項演算子 `?:` を使ってデータをフィルタする一つの芸当を紹介します。

```
plot 'file' using 1:(\$3>10 ? \$2 : 1/0)
```

これは、`1` 列目のデータに対して、`3` 列目のデータが `10` 以上であるような `2` 列目のデータを `plot` します。`1/0` は未定義値であり、`gnuplot` は未定義の点を無視するので、よって適切でない点は隠されることになります

カッコで始まっている限りは定数式を列番号として使うことができます。例えば `using 0+(複雑な式)` の様なことができます。そして、その数式は、カッコでスタートしていなければ数式の値が一度評価され、カッコでスタートしていれば個々のデータ点を読み込むためにその値が一度評価される、という点が重要です。

時系列フォーマットデータを使っている場合、その時間のデータは複数の列に渡らせることができます。その場合、他のデータの開始位置を計算するとき、時間のデータに空白が含まれていることに注意してください。例えば、データ行の最初の要素がスペースが埋め込まれた時間データであるならば、`y` の値は 3 列目の値として指定されるべきです。

`plot 'file'` と `plot 'file' using 1:2`、そして `plot 'file' using ($1):($2)` には微妙な違いがあることに注意してください。1) `file` が 1 列と 2 列のデータを持つ行をそれぞれ含んでいるとすると、データが 1 列のみの行に対しては、最初のものは `x` の値を作り出し、2 番目のものはその行は無視し、3 番目のものはそれを未定義の値として保存します(折れ線で `plot` している場合(`plot with lines`)、その未定義の点を通過する線を結ばないように)。2) 1 列目に文字列を含んでいるような行がある場合、最初のものはエラーとして `plot` を中止しますが、2 番目と 3 番目のものはその不要な行を読みとばします。

実際、最初に単に

```
plot 'file' using 1:2
```

と指定することで、大抵の場合どんなにゴミのデータを含む行を持つファイルをも `plot` することが可能になります。しかし、どうしてもデータファイルに文字列を残しておきたいならば、そのテキスト行の第一列にコメント文字 (#) を置く方がより安全でしょう。

#### 弱々しい using のデモ

gnuplot が configure 時に `-enable-datastrings` オプションをつけてインストールされているなら、さらにデータファイル中のテキストファイルを処理する `using` の追加修飾子を使うことができます。`datastrings`, `using xticlabels`, `using title` を参照してください。

**34.1.9.1 Using title** gnuplot が configure 時に `-enable-datastrings` オプションをつけてインストールされているなら、入力データの最初の行の列を、説明(key) ボックスの描画タイトルを与える文字列として使用できます。その指定を含む列と、描画自信に使用される列(複数の列もありうる)は独立しています。

```
plot 'data' using 1:(\$2/\$3) title 2
```

この場合、最初の行の第 2 列が、2 列目を 3 列目で割ったデータから作られる描画グラフの説明に使用されます。先頭行の 3 列目は無視されます。

**34.1.9.2 Xticlabels** gnuplot が configure 時に `-enable-datastrings` オプションをつけてインストールされているなら、入力データのある列を軸の刻みのラベルとして使えます。そのような描画コマンドの書式は以下の通りです。

```
plot 'datafile' using <xcol>:<ycol>:xticlabels(<labelcol>) with <plotstyle>
```

目盛りの見出し、任意の描画軸 `x,x2,y,y2,z` 用に読みだすことができます。`ticlabels(<labelcol>)` 指定は、`using` 指定の中で、そのデータの座標指定が全て済んだ後に行う必要があります。有効な `X,Y[,Z]` 座標の組を持つ各データ点に対して、`<labelcol>` 列に見つかるテキスト部分は、その点の X 座標と同じ位置の x 軸の刻みのラベルに追加されます。`xticlabels(<labelcol>)` は `xtic(<labelcol>)` と省略することもできます。

例:

```
splot "data" using 2:4:6:xtic(1):ytic(3):ztic(6)
```

この例では、x 軸、y 軸の見出しあは x,y 座標値とは別の列から取り出されますが、z 軸の見出しあは、対応する点の z 座標値から生成されます。

**34.1.9.3 X2ticlabels** `plot using xticlabels` 参照。

34.1.9.4 Yticlabels plot using xticlabels 参照。

34.1.9.5 Y2ticlabels plot using xticlabels 参照。

34.1.9.6 Zticlabels plot using xticlabels 参照。

## 34.2 Errorbars

エラーバーは、1から4個の追加されたデータを読む（またはエントリを using で追加選択する）ことにより、2次元データの描画において実現されています。これら追加される値は、それぞれのエラーバースタイルで異なった形で使われます。

デフォルトでは、gnuplot はデータファイルの各行に以下のようないつれかの列があることを期待しています：

```
(x, y, ydelta),
(x, y, ylow, yhigh),
(x, y, xdelta),
(x, y, xlow, xhigh),
(x, y, xdelta, ydelta),
(x, y, xlow, xhigh, ylow, yhigh)
```

x 座標は必ず指定しなければいけません。各数値を書く順序も上で挙げた通りでなくてはなりません。ただ、using 修飾子を使えばその順序を操作できますし、欠けている列の値も補うことは可能ですが。例えば、

```
plot 'file' with errorbars
plot 'file' using 1:2:(sqrt($1)) with xerrorbars
plot 'file' using 1:2:($1-$3):($1+$3):4:5 with xyerrorbars
```

最後の例は、相対的な x の誤差と絶対的な y の誤差、という、サポートされていない組のファイルに対するものです。using エントリが相対的な x の誤差から絶対的な x の最小値と最大値を生成しています。

y のエラーバーは、(x, ylow) から (x, yhigh) への鉛直な線として描かれます。ylow と yhigh の代わりに ydelta が指定されたときは、ylow = y - ydelta, yhigh = y + ydelta となります。ある行にデータが2つしかなければ、ylow と yhigh はともに y となります。x エラーバーは同様に計算された水平線です。データの各点を結ぶ折れ線を引きたい場合は、with errorbars と with lines を指定して、同じデータファイルを2回 plot して下さい（ただし、キーの中に2つのエントリを作らないように、その一方には notitle オプションを使うことを忘れないで下さい）。他の選択肢として、errorlines コマンドもあります（errorlines 参照）。

エラーバーには、もし set bars を使っていなければ、そのそれぞれの端に垂直な線分がつきます（詳細は set bars をご覧下さい）。

自動範囲指定が有効であれば、その描画範囲はエラーバーも含むように調整されます。

以下も参照

### エラーバーのデモ

さらなる情報に関して、plot using, plot with, set style も参照して下さい。

## 34.3 Errorlines

誤差線（errorbar）を伴う線描画は、2次元データファイルの描画でサポートされていて、それは1個から4個の追加の（または using で指定する）列データを与えることで行なわれます。これらの追加される値は、様々な errorline スタイルのそれ各自で異なった形で使われます。

デフォルトの状態では、gnuplot は、データファイルの各行に3個、4個、6個のいずれかの個数のデータがあることを期待し、それぞれ以下のいずれかに対応します。

```
(x, y, ydelta),
(x, y, ylow, yhigh),
```

```
(x, y, xdelta),
(x, y, xlow, xhigh),
(x, y, xdelta, ydelta),
(x, y, xlow, xhigh, ylow, yhigh)
```

x 座標は指定する必要がありますし、データの順番も上の形式である必要がありますが、using 修飾子でその順番を操作したり、欠けている列に対する値を与えたりすることができます。例えば

```
plot 'file' with errorlines
plot 'file' using 1:2:(sqrt($1)) with xerrorlines
plot 'file' using 1:2:($1-$3):($1+$3):4:5 with xyerrorlines
```

最後の例は、相対的な x の誤差と絶対的な y の誤差、というサポートされていない組合せのデータのファイルに対するもので、using で相対的な誤差から絶対的な x の最小値と最大値を生成しています。

y 誤差線は (x, ylow) から (x, yhigh) へ描画される縦線です。ylow, yhigh 代わりに ydelta が指定された場合は、ylow = y - ydelta, yhigh = y + ydelta と扱われます。ある行に 2 つのデータしかない場合、yhigh, ylow は両方とも y になります。x 誤差線は同様の方法で計算される水平線です。

誤差線には、set bars が指定されていない場合、その両端で垂直に交わる線分が付きます（詳細は set bars 参照）。

自動縮尺 (autoscaling) が ON の場合、描画範囲は誤差線が入るように調整されます。

更なる情報については、plot using, plot with, set style を参照してください。

## 34.4 Parametric

媒介変数モード (set parametric) では、plot では 2 つの数式の組を、splot では 3 つの数式の組を与える必要があります。

例:

```
plot sin(t),t**2
splot cos(u)*cos(v),cos(u)*sin(v),sin(u)
```

データファイルは前と同じように描画されます。ただし、データファイルが描画のために与えられる前に、任意の媒介変数関数が先に完全に指定された場合を除いてです。言い換えると、x の媒介変数関数（上の例では sin(t)）と y の媒介変数関数（上の例では t\*\*2）との間に、他の修飾子やデータ関数をはさみこんではいけません。そのようなことをすると、構文エラーになり、媒介変数関数が完全には指定されていない、と表示されます。

with や title のような他の修飾子は、媒介変数関数の指定が完了した後に指定しなければいけません。

```
plot sin(t),t**2 title 'Parametric example' with linespoints
```

以下も参照

媒介変数モードのデモ。

## 34.5 Ranges

オプションの範囲は、表示されるグラフの領域範囲を指定します。

書式:

```
[{<dummy-var>}={}<最小値>:{<最大値>}]
```

```
[<最小値>:{<最大値>}]
```

最初の範囲指定は独立変数の範囲 (xrange またはパラメトリックモードでは trange) で、2 番目のものは従属変数の範囲 yrange (パラメトリックモードでは xrange) となります。<dummy-var> には独立変数の新しい別名を指定します（デフォルトの変数名は set dummy で変更できます）。<最小値>, <最大値> には定数式、あるいは \* を書くことができます。

パラメトリックモードでなければ、与えられるべき範囲指定は xrange, yrange の順になります。

パラメトリックモードでは、plot コマンドに対してはその順序は trange, xrange, yrange になります。以下の plot コマンドは、trange を [-pi:pi], xrange を [-1.3:1.3], yrange を [-1:1] に設定する例です。

```
plot [-pi:pi] [-1.3:1.3] [-1:1] sin(t),t**2
```

$x_2$  の範囲と  $y_2$  の範囲はここでは指定できることに注意してください。それには `set x2range` や `set y2range` が使われます。

範囲は適切なモードに対して、上に示した順序で解釈されます。必要な範囲指定が一度全て指定されると、再び指定し直すことはありませんが、必要な部分を全く指定しないようにはできません – その代わりそこに空の範囲指定 `[]` を置きます。

\* は、最小値や最大値に自動範囲指定 (`autoscale`) の機能を使うことを可能にします。`set autoscale` も参照してください。

`plot` や `splot` のコマンド行で指定された範囲はそのグラフにのみ影響を及ぼします。よって、その後のグラフのデフォルトの範囲を変更するには、`set xrange` や `set yrange` を使用してください。

時間データに対しては、範囲はクオートで囲んで指定する必要があります (データファイルに現われる時間データと同じ形式の)。gnuplot はその範囲を読みこむのに時間書式文字列 (`timefmt`) を使用します。詳しくは `set timefmt` を参照してください。

例:

以下は現在の範囲を使用します:

```
plot cos(x)
```

以下は  $x$  の範囲のみの指定です:

```
plot [-10:30] sin(pi*x)/(pi*x)
```

以下は上と同じですが、仮変数として  $t$  を使います:

```
plot [t = -10 :30] sin(pi*t)/(pi*t)
```

以下は  $x$  と  $y$  の両方の範囲の指定です:

```
plot [-pi:pi] [-3:3] tan(x), 1/x
```

以下は、 $y$  の範囲のみの指定で、両方の軸の自動範囲指定機能を無効にします:

```
plot [ ] [-2:sin(5)*-8] sin(x)**besj0(x)
```

以下は  $x$  の最大値と  $y$  の最小値のみの指定です。

```
plot [:200] [-pi:] exp(sin(x))
```

以下は  $x$  の範囲を時系列データとして指定しています:

```
set timefmt "%d/%m/%y %H:%M"
plot ["1/6/93 12:00":"5/6/93 12:00"] 'timedata.dat'
```

## 34.6 Title

各関数やデータに対する曲線のタイトルは、その曲線のサンプル、および(または)それを表示されるのに使われる記号とともにキーの中に表示されます。それは `title` オプションで変更できます。

書式:

```
title "<title>" | notitle ["<ignored title>"]
```

ここで `<title>` はその曲線の新しいタイトルで、クオートで囲む必要があります。クオートはキーには表示されません。特殊文字も、バックスラッシュに続く 8 進値 (例えば "`\345`" のように) を使うことで用いることができます。タブ文字 "`\t`" は認識されます。バックスラッシュのそのような作用はダブルクオートで囲まれた文字列でしか効きません。逆にその作用を働かさないようにするにはシングルクオートを使ってください。改行文字 "`\n`" はどちらの型のクオートでもキーでは働きません。

曲線タイトルとサンプルは予約語 `notitle` を使うことでキーから削除できます。何もないタイトル (`title ''`) は `notitle` と同じ意味を持ちます。サンプルだけが欲しいときは、一つ以上の空白をタイトルの後ろに入れてください (`title ''`)。`notitle` の後ろに文字列をつけた場合、その文字列は無視されます。

`key autotitles` が設定されて（デフォルト）、かつ `title` も `notitle` も指定されなかつた場合、曲線のタイトルは `plot` コマンド上にある関数名かデータファイル名になります。ファイル名の場合は、指定される任意のデータファイル修飾子もそのデフォルトタイトルに含まれます。

位置やタイトルの位置揃えなどのキーのレイアウトは、`set key` で制御できます。詳細は `set key` の項目を参照してください。

例:

以下は `y=x` をタイトル '`x`' で表示します:

```
plot x
```

以下は、`x` の 2 乗をタイトル "`x^2`" で、ファイル "`data.1`" をタイトル "measured data" で表示します:

```
plot x**2 title "x^2", 'data.1' t "measured data"
```

以下は、極座標グラフの周りに円形の境界を書き、タイトルなしで表示します:

```
set polar; plot my_function(t), 1 notitle
```

## 34.7 With

関数やデータの表示にはたくさんのスタイルのうちの一つを使うことができます。キーワード `with` がその選択のために用意されています。

書式:

```
with <style> { {linestyle | ls <line_style>}
    | {{linetype | lt <line_type>}
        {linewidth | lw <line_width>}
        {linecolor | lc <colorspec>}
        {pointtype | pt <point_type>}
        {pointsize | ps <point_size>}
        {fill | fs <fillstyle>}
        {nohidden3d}
        {palette}}}
```

}

ここで、`<style>` は `lines`, `points`, `linespoints`, `impulses`, `dots`, `steps`, `fsteps`, `histeps`, `errorbars`, `labels`, `xerrorbars`, `yerrorbars`, `xyerrorbars`, `errorlines`, `xerrorlines`, `yerrorlines`, `xyerrorlines`, `boxes`, `histograms`, `filledcurves`, `boxerrorbars`, `boxxyerrorbars`, `financebars`, `candlesticks`, `vectors`, `image`, `rgbimage`, `pm3d` の中のいずれかです。これらのいくつかに対してはデータを付け足す必要があります。それぞれのスタイルの詳細については plotting styles をご覧ください。`fill` は 2 次元描画のみに関係があります（今のところは `boxes`, `boxxyerrorbars` と `candlesticks` のみ）。`filledcurves` と `pm3d` (`pm3d` は `splot` でのみ使用される) には、上には書かれていない追加のオプションを与えることもできます。詳細は、それらのヘルプや下にある例を参照してください。

デフォルトのスタイルは `set style function` や `set style data` コマンドで決定されます。

デフォルトでは、それぞれの関数やデータファイルは、使うことができる型の最大数に達するまで異なる線種、点種を使います。すべての端末用ドライバは最低 6 つの異なる点種をサポートしていて、もしくは要求された場合、それらを順に再利用していきます。LaTeX ドライバは、それより 6 つ多く点種（いずれも円の変種）を持っていて、よって点での曲線の描画は 12 種類の曲線が繰り返されるのです。PostScript ドライバは (postscript) 全部で 64 種類の点種を持っています。

一つの描画で線種や点種を選びたいならば、`<line_type>` や `<point_type>` を指定してください。これらの値は、その描画で使われる線種や点種を指定する正の整定数（または数式）です。使用する端末で使える線種、点種を表示するには `test` コマンドを使ってください。

描画の線の幅や点の大きさは `<line_width>` や `<point_size>` で変更できます。これらはその各々の端末のデフォルトの値に対する相対的な値として指定します。点の大きさは全体に適用するように変更できます。詳細は `set pointsize` を参照してください。しかし、ここでセットされる `<point_size>` と、`set pointsize` でセットされる大きさは、いずれもデフォルトのポイントサイズに掛けられることに注意してください – すなわち、それらの効果は累積しません。例えば、`set pointsize 2; plot x w p ps 3` は、デフォルトのサイズの 3 倍であって、6 倍ではありません。

ラインスタイルの一部分、あるいは各 plotにおいて `pointsize variable` という指定も可能です。この場合、入力には追加の 1 列が要求されます。例えば 2D 描画では 3 列、3D 描画では 4 列のデータが必要になります。個々の点のサイズは、全体を通しての `pointsize` に、データファイルからの入力による値をかけたものとして決定されます。

`set style line` を使って線種/線幅、点種/点幅の組を定義すれば、そのスタイルの番号を `<line_style>` にセットすることでそれらを使うことができます。

gnuplot が pm3d をサポートするようにインストールされているならば、`splots` において `lines`, `points`, `dots` の色を滑らかに変化させるための特別なキーワード `palette` が使えます。その色は、コマンド `set palette` であらかじめ設定された滑らかに変化するカラーパレットから選択します。色の値は、点の z 座標の値か、または `using` で 4 番目のパラメータとして指定される色座標に対応します。2 次元、3 次元の描画 (`plot` と `splot` コマンド) の両方で、パレット色を小数値かまたはカラー・ボックスの範囲へ対応づけられた値のいずれかで指定することができます。2 次元描画では Z の値に対応づけられたパレット色を使うことはできません。`colors`, `set palette`, `linetype` 参照。

キーワード `nohidden3d` は、`splot` コマンドで生成される描画にのみ適用されます。通常、グローバルなオプション `set hidden3d` はグラフ上の全ての描画に適用されますが、各々の描画に `nohidden3d` オプションをつけることで、それを `hidden3d` の処理から除外することができます。`nohidden3d` がマークされた個々の描画要素（線分、点、ラベル等）は、通常は他の何らかの描画要素で隠されてしまう場合も全て描画されます。

キーワードは暗示するような形で省略可能です。

`linewidth`, `pointsize`, `palette` オプションは全ての端末装置でサポートされているわけではないことに注意してください。

例:

以下は、 $\sin(x)$  を鉛直線で描画します:

```
plot sin(x) with impulses
```

以下は、x を点で描画し、 $x^{**2}$  をデフォルトの方式で描画します:

```
plot x w points, x**2
```

以下は、 $\tan(x)$  を関数のデフォルトの方式で、"data.1" を折れ線で描画します:

```
plot [ ] [-2:5] tan(x), 'data.1' with l
```

以下は、"leastsq.dat" を鉛直線で描画します:

```
plot 'leastsq.dat' w i
```

以下は、データファイル "population" を矩形で描画します:

```
plot 'population' with boxes
```

以下は、"exper.dat" をエラーバー付きの折れ線で描画します（エラーバーは 3 列、あるいは 4 列のデータを必要とします）:

```
plot 'exper.dat' w lines, 'exper.dat' notitle w errorbars
```

もう一つの "exper.dat" のエラーバー付きの折れ線 (errorlines) での描画方法（エラーバーは 3 列、あるいは 4 列のデータが必要）:

```
plot 'exper.dat' w errorlines
```

以下は、 $\sin(x)$  と  $\cos(x)$  をマーカー付きの折れ線で描画します。折れ線は同じ線種ですが、マーカーは異なったものを使います:

```
plot sin(x) with linesp lt 1 pt 3, cos(x) with linesp lt 1 pt 4
```

以下は、"data" を点種 3 で、点の大きさを通常の 2 倍で描画します:

```
plot 'data' with points pointtype 3 pointsize 2
```

以下は、"data" を描画しますが、4 列目から読んだデータを `pointsize` の値として使用します:

```
plot 'data' using 1:2:4 with points pt 5 pointsize variable
```

以下は、2つのデータ集合に対して、幅のみ異なる線を用いて描画します:

```
plot 'd1' t "good" w 1 lt 2 lw 3, 'd2' t "bad" w 1 lt 2 lw 1
```

以下は、 $x^x$  の曲線の内部の塗りつぶしと色の帯を描画します:

```
plot x*x with filledcurve closed, 40 with filledcurve y1=10
```

以下は、 $x^x$  の曲線と色の箱を描画します:

```
plot x*x, (x>=-5 && x<=5 ? 40 : 1/0) with filledcurve y1=10 lt 8
```

以下は、滑らかに変化する色の線で曲面を描画します:

```
splot x*x-y*y with line palette
```

以下は、2つの色のついた曲面を、異なる高さで表示します:

```
splot x*x-y*y with pm3d, x*x+y*y with pm3d at t
```

## 35 Print

`print` コマンドは <式> の値を画面に表示します。これは `pause 0` と同じです。<式> は、数を生成する gnuplot の数式か、または文字列です。

書式:

```
print <式>
```

`expressions` を参照して下さい。出力ファイルは `set print` で設定できます。

## 36 Pwd

`pwd` コマンドはカレントディレクトリの名前を画面に表示します。

## 37 Quit

`exit` と `quit` の両コマンドと END-OF-FILE 文字は、gnuplot を終了させます。これらのコマンドは、出力装置を (`clear` コマンドと同様に) クリアしてから終了させます。

## 38 Raise

書式:

```
raise {plot_window_nb}
```

コマンド `raise` (`lower` の反対) は、`pm`, `win`, `wxt`, `x11` 等の gnuplot の対話型出力形式の実行中に、描画ウィンドウを上(前面)に上げます。描画ウィンドウを、デスクトップ上のウィンドウマネージャの `z` 方向のウィンドウの重なりの前(上)に置きます。

`x11` や `wxt` のように複数の描画ウィンドウをサポートしている場合、デフォルトではこのコマンドはそれらの複数のウィンドウを降順に上げ、最初に作られたウィンドウを一番下に、最後に作られたウィンドウを一番上に並べます。オプション引数の描画番号が与えられた場合、それに対応する描画ウィンドウが存在すればそれのみが上げられます。

オプション引数は、単一の描画ウィンドウの出力形式、すなわち `pm` と `win` では無視されます。

ウィンドウが X11 で前面に出ない場合、(1) それらが同じ X11 セッションで動作していない(例えば `telnet` や `ssh` セッションなどによって)、または(2) 前面に出すことがウィンドウマネージャによって防害されているかのどちらかでしょう。KDE では、KDE Control Center => Desktop => Window Behaviour => Advanced と進んで、"Focus stealing prevention level" を `None` (デフォルトは `Low`) に設定するといいでしょう。

## 39 Replot

`replot` コマンドを引数なしで実行すると、最後に実行した `plot` または `splot` コマンドを再実行します。これは、あるプロットを異なる `set` オプションでみたり、同じプロットを異なる装置に出力したりするときに便利でしょう。

`replot` コマンドに対する引数は最後に実行した `plot` または `splot` コマンドの引数に（暗黙の `,` と共に）追加され、それから再実行されます。`replot` は、範囲 (`range`) を除いては、`plot` や `splot` と同じ引数をとることができます。よって、直前のコマンドが `splot` ではなく `plot` の場合は、関数をもう一つの軸刻度でプロットするのに `replot` を使うことができます。

注意:

```
plot '-' ; ... ; replot
```

という使い方は推奨されません。`gnuplot` はインラインデータを保存しないので、`replot` によって新たな情報が直前の `plot` に追加されて修正されたコマンドを実行することになったとしても、最初の `plot` の `'-'` は再びインラインデータを読もうとするからです。

`replot` コマンドは `multiplot` モードでは働きません。それは、それが画面全体にではなく直前のプロットのみを再実行するものだからです。

最後に実行した `plot` (`splot`) コマンドの内容を修正する方法については `command line-editing` を参照して下さい。

直前の描画コマンドの全体を表示させることや、それを `history` の中にコピーする方法については `show plot` を参照してください。

## 40 Reread

`reread` コマンドは、`load` コマンドまたはコマンドラインで指定した `gnuplot` のコマンドファイルを、その次のコマンドが読まれる前に、開始点に再設定します。これは、コマンドファイルの最初から `reread` コマンドまでのコマンドの無限ループを本質的に実装することになります。（しかし、これは何も悪いことではありません。`reread` は `if` と組み合わせることでとても有用なコマンドとなります。詳細は `if` を参照してください。）標準入力からの入力の場合は、`reread` コマンドは何も影響を与えません。

例:

ファイル "looper" が次のようなファイルで

```
a=a+1
plot sin(x*a)
pause -1
if(a<5) reread
```

そして、`gnuplot` から次のように実行するとします。

```
a=0
load 'looper'
```

すると、`pause` のメッセージで分割された 4 回のプロットが行われることになります。

ファイル "data" が、各行に、0 から 10 までの範囲 (`yrange`) の 6 つのデータを持ち、最初が `x` 座標で、その他は 5 つの異なる関数の、その `x` での値であるとします。そして、ファイル "plotter" が

```
c_p = c_p+1
plot "$0" using 1:c_p with lines linetype c_p
if(c_p < n_p) reread
```

で、`gnuplot` から次のように実行するとします。

```
n_p=6
c_p=1
unset key
set yrange [0:10]
```

```
set multiplot
call 'plotter' 'data'
unset multiplot
```

すると、5つのプロットを合わせた1つのグラフができます。yrangeは、multiplotモードで最初のものに続けて書かれる5つのグラフが、同じ軸を持つように、明示的に指定する必要があります。線種も指定しなければなりません。さもなくとも、全てのグラフが同じ線種で書かれることになります。アニメーションのサンプルとして、demoディレクトリの animate.dem を参照してください。

## 41 Reset

コマンド reset は set コマンドで定義できる、グラフに関する全てのオプションをデフォルトの値に設定します。このコマンドは、例えばコマンドファイルの最後にグラフのデフォルトの設定に復帰する、あるいはコマンドファイル内でたくさんの設定を行なった場合に元の状態に戻すなどに便利です。様々なオプションの取るデフォルトの値を知るには、set コマンドの項を参照してください。

以下の set コマンドはグラフの状態を変更しないので、reset もそれらを変化させません: set term による出力形式の設定、および set output による出力ファイルの指定、および set loadpath や set fontpath によるディレクトリパスの設定。

## 42 Save

save コマンドは、ユーザ定義関数、変数、set term の状態、set で設定する全てのオプションのいずれかか、あるいはこれらすべてと、それに加えて最後に実行した plot (または splot) コマンドを、指定したファイルに保存します。

書式:

```
save {<オプション>} '<ファイル名>'
```

ここで、<オプション>は、functions、variables、terminal、set のいずれかです。どれも指定されなかった場合には、gnuplotは、ユーザ定義関数、変数、set で設定するオプション、最後に実行した plot (または splot) コマンドの全てを保存します。

save は、テキスト形式で出力します。また、このファイルは load コマンドで読み込むことができます。set オプション付き、または何もオプションをつけずに save を実行した場合、terminal の選択と output のファイル名はコメント記号つきで書き出されます。これはその出力ファイルを他の環境にインストールされた gnuplot 上で動かす場合に、修正なしに使えるようにする、あるいはうっかりファイルを上書きしてしまったりする危険性を避ける、といった意味があります。

save terminal は、terminal の状態を、コメント記号をつけずに書き出します。これは主に、ちょっとの間だけ terminal の設定を入れ替え、その後保存しておいた terminal の状態を読み込むことで以前の terminal の設定に戻す場合などに役立ちます。ただ、単一の gnuplot セッションでは、現在の terminal を保存/復元する他の方法であるコマンド set term push と set term pop を使う方がむしろいいかもしれません。set term を参照してください。

ファイル名は引用符に囲われていなければなりません。

特別なファイル名 "-" により save コマンドに標準出力に出力させることができます。popen 関数をサポートするようなシステム (Unix など) では、save の出力をパイプ経由で他の外部プログラムに渡すことができます。その場合、ファイル名としてコマンド名の先頭に '!'をつけたものを使います。これは、gnuplot とパイプを通して通信するプログラムに、gnuplot の内部設定に関する首尾一貫したインターフェースを提供します。詳細は batch/interactive に関するヘルプを参照してください。

例:

```
save 'work.gnu'
save functions 'func.dat'
save var 'var.dat'
save set 'options.dat'
save term 'myterm.gnu'
```

```
save '-'
save '|grep title >t.gp'
```

## 43 Set-show

set コマンドは実際に多くのオプションを設定するのに使われます。しかし、plot, splot, replot コマンドが与えられるまで何も表示しません。

show コマンドはそれらの設定値を表示します。show all でそれら全てを表示します。

set コマンドで変更されたオプションは、それに対応する unset コマンドを実行することでデフォルトの状態に戻すことができます。reset コマンドも参照してください。これは全てのパラメータの設定をデフォルトの値に戻します。

もし変数が日時のデータを含むならば、show は、set timefmt によって現在設定されている書式に従って表示します。それは変数が最初に設定されていてその書式が効果を持たなかったとしてもです。

### 43.1 Angles

デフォルトでは gnuplot は極座標グラフの独立変数の単位はラジアンを仮定します。set polar の前に set angles degrees を指定すると、その単位は度になり、デフォルトの範囲は [0:360] となります。これはデータファイルの描画で特に便利でしょう。角度の設定は、set mapping コマンドを設定することにより 3 次元でも有効です。

書式:

```
set angles {degrees | radians}
show angles
```

set grid polar で指定される角度も、set angles で指定した単位で読み込まれます。

set angles は組み込み関数 sin(x), cos(x), tan(x) の引数や asin(x), acos(x), atan8x, atan2(x), arg(x) の出力にも影響を与えます。双曲線関数や、ベッセル関数の引数には影響を与えません。しかし、複素数を引数とする逆双曲線関数の出力には影響が出ます。それらの関数が使われるときは、set angles radians は入出力の引数の間に一貫性を持った管理を実現していかなければなりません。

```
x={1.0,0.1}
set angles radians
y=sinh(x)
print y          #{1.16933, 0.154051} と表示
print asinh(y)  #{1.0, 0.1} と表示
```

しかし、

```
set angles degrees
y=sinh(x)
print y          #{1.16933, 0.154051} と表示
print asinh(y)  #{57.29578, 5.729578} と表示
```

以下も参照

poldat.dem: set angles を用いた極座標描画のデモ

### 43.2 Arrow

set arrow コマンドを使うことにより、グラフ上の任意の位置に矢印を表示することができます。

書式:

```
set arrow {<tag>} {from <position>} {to|rto <position>}
{ {arrowstyle | as <arrow_style>}
| {nohead | head | backhead | heads}
```

```

{size <length>,<angle>{,<backangle>}}
{filled | empty |nofilled}
{front | back}
{ {linestyle | ls <line_style>}
| {linetype | lt <line_type>}
{linewidth | lw <line_width>} } }
}

unset arrow {<tag>}
show arrow {<tag>}

```

タグ `<tag>` は各矢印を識別する整数です。タグを指定しない場合は、その時点で未使用の最も小さい数が自動的に割り当てられます。タグを使うことで、特定の矢印を変更したり、削除したりできます。既に存在する矢印の属性を変更する場合は、タグを明示した `set arrow` コマンドで変更箇所を指定してください。

`<position>` は `x,y` あるいは `x,y,z` で指定します。そしてその前に座標系を選択するために `first`, `second`, `graph`, `screen`, `character` を置くことができます。座標を指定しなければデフォルトでは `0` と見なされます。矢印の端点は、5つの座標系 – `first` か `second` の軸、あるいは `graph`, `screen`, `character` – のうちの1つを選択して指定できます。詳細は `coordinates` を参照して下さい。"from" の場所の座標系指定子は、"to" の場所に影響を及ぼすことはありません。グラフの枠をはみ出る矢印を書くこともできますが、出力端末によってはエラーを生ずることがあります。終点が "to" の代わりに "rto" で指定されている場合、始点からの相対的な位置に描かれます。この場合、線形軸(非対数軸)、および `graph`, `screen` 座標に対しては、始点と終点の距離が与えられた相対的な値に対応します。一方、対数軸に対しては、与えられた相対的な値は、始点から終点への積因子(比)に対応します。よって、対数軸の場合、相対的な値として `0` や負の値を与えることは許されません。

`nohead` を指定することで、矢先のない矢 – すなわち線分を書くこともできます。これは描画の上に線分を描く別な方法を与えます。デフォルトでは1つの矢は1つの矢先をその終端に持っています。`backhead` は始点に矢先を描き、`heads` は線分の両端に矢先を描きます。全ての出力形式が両端の矢先の描画をサポートしているわけではありません。

矢先の大きさは `size <length>,<angle>` または `size <length>,<angle>,<backangle>` で制御できます。`<length>` は矢先の各枝の長さで、`<angle>` は矢先の枝と矢軸がなす角度(単位は度)です。`<length>` の単位は `x` 軸と同じですが、それは `<length>` の前に `first`, `second`, `graph`, `screen`, `character` をつけることで変更できます。詳細は `coordinates` を参照してください。`<backangle>` は、`filled` かまたは `empty` とともに使われた場合のみ効力を持ち、その場合、`<backangle>` は矢先の後ろの部分の矢軸との切り角(`<angle>` と同じ方向; 単位は度)になります。出力形式 `fig` は、制限された切り角関数を持っていて、それは3つの異なる形をサポートしていて、それは2つの閾値で決定します: 70度未満の場合、矢先はへこんだ切り角を持ち、110度を超える場合、後ろの部分に尖った角を持ち、その間の角では、矢先の後ろは直線になります。

`filled` を指定すると、塗りつぶされた矢先を作ります(`heads` が使われている場合)。塗りつぶしは、多角形の塗りつぶしが行えるような出力形式でサポートされていて、そのリストについては `pm3d` を参照してください。他の出力形式では矢先は閉じられますが塗りつぶされません。それと同じ効果(閉じらるが塗られない)は、`empty` を指定しても得られます。また、`metafont`, `metapost`, `latex`, `tgif` のように、矢をそれら自身の独自のルーチンで描くような出力形式では、矢先の塗りつぶしや矢先の枠線描きはもちろんサポートされません。

線種はユーザの定義したラインスタイルのリストから選ぶこともできますし(`set style line` 参照)、用意されている `<line_type>` の値(デフォルトのラインスタイルのリストの番号)そして `<linewidth>`(デフォルトの幅の倍数)を使ってここで定義することもできます。

しかし、ユーザー定義済のラインスタイルが選択された場合、その属性(線種、幅)は、単に他の `set arrow` コマンドで適当な番号や `lt`, `lw`などを指定しても、変更はできないことに注意して下さい。

`front` を指定すると、矢はグラフのデータの上に描かれます。`back` が指定された場合(デフォルト)は矢はグラフのデータの下に描かれます。`front` を使えば、密集したデータで矢が見えなくなることを防ぐことができます。

例:

原点から (1,2) への矢印をユーザ定義済のラインスタイル 5 で描くには:

```
set arrow to 1,2 ls 5
```

描画領域の左下角から (-5,5,3) へタグ番号 3 の矢印を描くには:

```
set arrow 3 from graph 0,0 to -5,5,3
```

矢印の端を 1,1,1 に変更し、矢先を外して幅を 2 にするには:

```
set arrow 3 to 1,1,1 nohead lw 2
```

x=3 の所へグラフの下から上まで鉛直線を描くには:

```
set arrow from 3, graph 0 to 3, graph 1 nohead
```

T 字型の矢先を両端に持つ鉛直方向の矢を描くには:

```
set arrow 3 from 0,-5 to 0,5 heads size screen 0.1,90
```

始点からの相対的な距離をグラフ座標で与えて矢を描くには:

```
set arrow from 0,-5 rto graph 0.1,0.1
```

x の対数軸に相対的な終点を指定して矢を描く場合:

```
set logscale x
set arrow from 100,-5 rto 10,10
```

これは 100,-5 から 1000,5 までの矢を描きます。線形軸 (y) に対しては相対的な座標 10 が "差 10" を意味するのに対し、対数軸 (x) に対しては相対的な座標 10 は "積因子 10" として働きます。

2 番の矢印を消すには:

```
unset arrow 2
```

全ての矢印を消すには:

```
unset arrow
```

全ての矢印の情報を (タグの順に) 見るには:

```
show arrow
```

矢印のデモ

### 43.3 Autoscale

自動縮尺機能 (autoscale) は x, y, z の各軸に対して独立に、または一括して指定できます。デフォルトでは全ての軸に対して自動縮尺設定を行います。

書式:

```
set autoscale {<axes>{|min|max|fixmin|fixmax|fix} | fix | keepfix}
unset autoscale {<axes>}
show autoscale
```

ここで、<axes> (軸) は x, y, z, cb, x2, y2, xy のいずれかです。min または max を軸に追加指定すると (xy ではありませんが) それは gnuplot にその軸の最小値、または最大値のみを自動縮尺させることになります。軸も何も指定されていない場合は全ての軸が対象となります。

fixmin や fixmax や fix を追加指定すると、等間隔の目盛りの自動縮尺時の、次の目盛り位置までの範囲の自動拡大を gnuplot に行わせないようにします。set autoscale fix はこれを全ての軸に対して設定します。コマンド set autoscale keepfix は、直前の fix の設定を維持したまま全ての軸を自動縮尺にします。

自動縮尺機能を使うときは、描画範囲は自動的に割り出され、従属変数軸 (plot のときは y 軸、splot のときは z 軸) は、関数やデータの値域が収まるように設定されます。

従属変数軸 (y または z) の自動縮尺機能が指定されていない場合は、現在の y や z の描画範囲がそのまま使われます。

独立変数軸 (plot のときは x 軸、splot のときは x,y 軸) の自動縮尺機能が指定されている場合は、描画される全てのデータファイルの点が収まるように定義域をとるようになります。データファイルが 1 つも指定されていない場合は、自動縮尺機能はなんの効果もありません。つまり、関数のみが指定されていて

データーファイルを使わない場合は、 $x$  軸の描画範囲 ( $z = f(x,y)$  を描画しているときは  $y$  軸も) は影響をうけません。

範囲に関するより詳しい情報に関しては `set xrange` を見てください。

媒介変数モード (parametric) でも自動縮尺機能は有効です (`set parametric` 参照)。この場合、より多くの従属変数があるので、 $x$ ,  $y$ ,  $z$  各軸に関して、より多くの制御が行われます。媒介変数モードでの独立変数 (仮変数) は `plot` では  $t$  で `splot` では  $u$ ,  $v$  です。そして媒介変数モードでは、自動縮尺機能は ( $t$ ,  $u$ ,  $v$ ,  $x$ ,  $y$ ,  $z$ ) の全ての描画範囲を制御し、 $x$ ,  $y$ ,  $z$  の範囲の自動設定を完全に行います。

自動縮尺機能は、極座標モード (polar mode) でも `plot` の媒介変数モードと同様に機能しますが、極座標モードでは `set dummy` で独立変数を  $t$  から変更するできる (`set dummy` 参照) という拡張があります。

目盛りが第二の軸に表示され、しかもこれらの軸に対する描画が行われなかつた場合には、`x2range` と `y2range` は `xrange` と `yrange` の値を受け継ぎます。これは、`xrange` と `yrange` が整数個の目盛り幅に自動縮尺される「前」に行われますので、場合によって予期しない結果をもたらす可能性があります。それを避けるために `fixmin`, `fixmax` オプションを使うことができます。

例:

以下は  $y$  軸の自動縮尺機能を指定します (他の軸には影響を与えません):

```
set autoscale y
```

以下は  $y$  軸の最小値に対してのみ自動縮尺機能を指定します ( $y$  軸の最大値、および他の軸には影響を与えません):

```
set autoscale ymin
```

以下は  $x2$  軸の隣の目盛りへの自動範囲拡大機能を無効にし、よって描画データ内、または関数に対する丁度の描画範囲を維持します:

```
set autoscale x2fixmin
set autoscale x2fixmax
```

以下は  $x$ ,  $y$  両軸の自動縮尺機能を指定します:

```
set autoscale xy
```

以下は  $x$ ,  $y$ ,  $z$ ,  $x2$ ,  $y2$  全軸の自動縮尺機能を指定します:

```
set autoscale
```

以下は  $x$ ,  $y$ ,  $z$ ,  $x2$ ,  $y2$  全軸の自動縮尺機能を禁止します:

```
unset autoscale
```

以下は  $z$  軸のみについて自動縮尺機能を禁止します:

```
unset autoscale z
```

### 43.3.1 Parametric mode

媒介変数表示モード (`set parametric`) においては、`xrange` も `yrange` と同様に縮尺を変えることができます。つまり、媒介変数モードにおいては、 $x$  軸方向も自動的に縮尺が調整され、描こうとしている媒介変数表示の関数が収まるようになります。もちろん、 $y$  軸方向も媒介変数モードでない時同様に自動的に縮尺を変えます。 $x$  軸について自動縮尺機能が設定されていない場合は、現在の  $x$  の範囲が使われます。

データファイルは媒介変数モードでもそうでない状態でも同様に描画されます。しかし、データファイルと関数が混在している場合には、違いがあります: 媒介変数モードでなければ、 $x$  の自動縮尺機能は、関数の範囲をデータの描画範囲に合わせます。しかし媒介変数モードではデータの範囲は関数の範囲に影響しません。

それには、片手落ちにならないように `set autoscale t` というコマンドも用意されています。しかしその効果は非常に小さいものです。自動縮尺機能が設定されていると、`gnuplot` が  $t$  の範囲が無くなってしまうと判断した場合に範囲を少し調整します。自動縮尺機能が設定されていないとこのようなときにはエラーとなります。このような動作は実はあまり意味がなく、よって `set autoscale t` というコマンドは存在意義に疑問があります。

`splot` では上記の発想の元に拡張されています。自動縮尺機能が設定されている場合、 $x$ ,  $y$ ,  $z$  の各描画範囲は計算結果が収まるように設定され縮尺調整されます。

### 43.3.2 Polar mode

極座標モード (`set polar`) では、`xrange` と `yrange` は極座標から求められ、それによって自動的に範囲設定がなされます。言いかえると、極座標モードでは描こうとしている極座標関数が収まるように x 軸、y 軸が自動的に縮尺が調整されます。

極座標モードで関数を描画する場合、`rrange` も自動範囲設定されます。データファイルを描画する場合はさらに `trange` も自動範囲設定がなされます。もし、`trange` がある象限 (四分円) に収まるならば、自動縮尺機能によりその象限のみの描画が行われることに注意してください。

1 つ、あるいは 2 つの範囲は明示的に設定してその他のものを指定しない場合は予期しない結果を引き起こすかも知れません。以下も参照

極座標のデモ。

## 43.4 Bars

コマンド `set bars` は誤差グラフ (errorbar) の両端のマークの幅、および描画スタイル `candlesticks`, `financebars` の箱の幅を制御します。

書式:

```
set bars {small | large | fullwidth | <size>}
unset bars
show bars
```

`small` は 0.0, `large` は 1.0 と同じです。サイズを指定しなければデフォルトの値は 1.0 です。

キーワード `fullwidth` は `errorbar` を伴うヒストグラム (histograms) にのみ関連します。これは `errorbar` の両端の幅を、関連するヒストグラムの箱の幅と同じに設定しますが、箱の幅自体を変更することはありません。

## 43.5 Bmargin

コマンド `set bmargin` は、下部の余白のサイズを設定します。詳細は `set margin` を参照してください。

## 43.6 Border

`set border` と `unset border` は `plot` や `splot` でのグラフの枠の表示を制御します。枠は必ずしも軸とは一致しないことに注意してください。`plot` では大抵一致しますが、`splot` では大抵一致していません。

書式:

```
set border {<integer>} {front | back} {linewidth | lw <line_width>}
            {{linestyle | ls <line_style>} | {linetype | lt <line_type>}}
unset border
show border
```

`set view 56,103` のように任意の方向で表示される `splot` では、x-y 平面上の 4 つの角は手前 (front), 後ろ (back), 左 (left), 右 (right) のように参照されます。もちろんこの同じ 4 つの角は天井の面にもあります。よって、例えば x-y 平面上の後ろと右の角をつなぐ境界を "底の右後ろ (bottom right back)" と言い、底と天井の手前の角をつなぐ境界を "鉛直手前 (front vertical)" と呼ぶことにします (この命名法は、読者が下の表を理解するためだけに使われます)。

枠は、12 ビットの整数に符号化されています: 下位 4 ビットは `plot` に対する外枠、`splot` に対しては底面の外枠、次の 4 ビットは `splot` の鉛直な外枠、そして上位 4 ビットは `splot` の天井面の外枠を制御します。その < 整数 > 値は次の表の対応する項目の数字の和になります:

グラフ境界の符号化		
ビット	plot	splot
1	下	底の左手前
2	左	底の左後ろ
4	上	底の右手前
8	右	底の右後ろ
16	効果なし	鉛直左
32	効果なし	鉛直後ろ
64	効果なし	鉛直右
128	効果なし	鉛直の手前
256	効果なし	天井の左後ろ
512	効果なし	天井の右後ろ
1024	効果なし	天井の左手前
2048	効果なし	天井の右手前

ビットは一つ一つ、あるいはいくつかをまとめて一緒にこのコマンドで付加できます。

デフォルトの値は 31 で、これは plot では 4 方向の外枠全て、splot では底面の枠線全部と z 軸を描くことを意味します。

2 次元描画では境界はすべての描画要素の一番上に描かれます (front)。もし境界を描画要素の下に描かせたい場合は、set border back としてください。

<line\_style>, <line\_type>, <line\_width> を指定して、枠線の描画にそれらを反映させることができます (現在の出力装置がサポートするものに限定されます)。

plot では、第 2 軸を有効にすることで、下と左以外の境界に目盛りを描くことができます。詳細は xtics を参照してください。

"unset surface; set contour base" などによって splot で底面にのみ描画する場合、鉛直線や天井はそれらが指定されていても描画されません。

set grid のオプション 'back', 'front', 'layerdefault' でも、描画出力の境界線を書く順番を制御できます。

例:

以下は、デフォルトの枠線を描きます:

```
set border
```

以下は、plot では左と下、splot では底面の左手前と左後ろの枠線を描きます:

```
set border 3
```

以下は、splot で周りに完全な箱を描きます:

```
set border 4095
```

以下は、手前の鉛直面と天井のない箱を描きます:

```
set border 127+256+512 # または set border 1023-128
```

以下は、plot に対して上と右枠線のみを描き、それらを軸として目盛りづけします:

```
unset xtics; unset ytics; set x2tics; set y2tics; set border 12
```

## 43.7 Boxwidth

コマンド set boxwidth は boxes, boxerrorbars, candlesticks, histograms スタイルにおける棒のデフォルトの幅を設定するために使います。

書式:

```
set boxwidth {<width>} {absolute|relative}
show boxwidth
```

デフォルトでは、隣り合う棒が接するように各々の棒の幅が広げられます。それとは異なるデフォルトの幅を設定するには `set boxwidth` コマンドを使用します。`relative` の場合の幅は、デフォルトの幅に対する比であると解釈されます。

修飾子 `relative` を指定しなかった場合、棒の幅 (`boxwidth`) として指定された明示的な値は、現在の `x` 軸の単位での数字 (`absolute`) であると解釈されます。`x` 軸が対数軸 (`set log` 参照) である場合、`boxwidth` の値は実際には  $x=1$  でのみ "絶対的" となり、その物理的な長さが軸全体を通じて保持されます(すなわち、棒は `x` 座標の増加にともなって狭くなったりはしません)。対数軸の `x` 軸の範囲が  $x=1$  から離れている場合は、適切な幅を見出すには何度か試してみる必要があるかも知れません。

デフォルトの値は、`boxes` や `boxerrorbars` スタイルの幅指定用の追加のデータ列の明示的な値があればそれによって置き換えられます。4列のデータの場合、第4列目の値が棒の幅として使われます。ただし、その幅が -2.0 の場合には棒の幅は自動計算されます。詳細は `style boxes` と `style boxerrorbars` を参照してください。

棒の幅を自動的にセットするには

```
set boxwidth
```

とする、あるいは4列のデータに対しては以下のようにします。

```
set boxwidth -2
```

`plot` のキーワード `using` を使っても同じ効果を得ることができます:

```
plot 'file' using 1:2:3:4:(-2)
```

棒の幅を自動的な値の半分にするには

```
set boxwidth 0.5 relative
```

棒の幅を絶対的な値 2 にするには

```
set boxwidth 2 absolute
```

### 43.8 Clabel

`gnuplot` は、`clabel` が設定されている時には、各々の等高線のレベルに対して使う線種を変化させます。このオプションが有効である場合(デフォルト)、凡例によって各々の線種を、それが表す `z` のレベルとともに表示されます。

書式:

```
set clabel {'<format>'}
unset clabel
show clabel
```

書式文字列のデフォルトは `%8.3g` で、小数部分は3桁表示されます。もし `key` がそのデフォルトの値から変更されていれば、その配置は不十分なものになるかもしれません。

最初の等高線の線種、または `clabel` が無効である場合の唯一つの等高線の線種は、(曲面の線種 +1) になります。等高線の点は曲面の点と同じものになります。

`set contour` も参照してください。

### 43.9 Clip

`gnuplot` はグラフの端の辺りのデータ点や線をクリッピングすることができます。

書式:

```
set clip <clip-type>
unset clip <clip-type>
show clip
```

gnuplot は点や線に対するクリップ型 (clip-type) として、points, one, two の 3 種類をサポートしています。ある描画に対して、これらのクリップ型は任意の組み合せで設定することができます。pm3d の色地図やカラー曲面で塗りつぶされた四辺形はこのコマンドでは制御できませんが、set pm3d clip1in や set pm3d clip4in によって可能であることに注意してください。

クリップ型 points を設定すると、描画領域内にはあるけれど境界線に非常に近いような点をクリップする（実際には描画しないだけですが）ように gnuplot に指示します。これは点として大きなマークを使用したときに、そのマークが境界線からはみ出さないようにする効果があります。points をクリップしない場合、境界線の辺りの点が汚く見えるかもしれません。その場合、x や y の描画範囲 (xrange, xrange) を調整してみて下さい。

クリップ型 one を設定すると、一端のみが描画領域にあるような線分も描画するように gnuplot に指示します。この際、描画領域内にある部分のみが実際に描画される範囲です。設定しなかった場合、このような線分は描画対象とならず、どの部分も描画されません。

両端は共に描画範囲に無いが描画領域を通過するという線分もあります。クリップ型 two を設定することによって、このような線分の描画領域の部分を描画することができます。

どのような状況でも、描画範囲の外に線が引かれることはありません。

デフォルトでは、noclip points, clip one, noclip two となっています。

全てのクリップ型の設定状況を見るには以下のようにします：

```
show clip
```

過去のバージョンとの互換性のため以下の書式も使用可能です：

```
set clip
unset clip
```

set clip は set clip points と同義です。unset clip は 3 種のクリップ型全てを無効にします。

### 43.10 Cntrparam

set cntrparam は等高線の生成方法、およびそれを滑らかに描画する方法を制御します。show contour は現在の contour の設定だけでなく cntrparam の設定をも表示します。

書式:

```
set cntrparam { { linear
    | cubicspline
    | bspline
    | points <n>
    | order <n>
    | levels { auto {<n>} | <n>
        | discrete <z1> {,<z2>{,<z3>...}}
        | incremental <start>, <incr> {,<end>}
    }
}
show contour
```

このコマンドは 2 つの機能を持っています。一つは等高線上の点（データ点の線形補間、あるいは関数の標本化 (isosample) による点）での z の値の設定で、もう一つは、そのように決定された z が等しい点同士を等高線で結ぶ方法の制御です。<n> は整数型の定数式、<z1>, <z2> ... は任意の定数式です。各オプション変数の意味は次の通りです：

linear, cubicspline, bspline — 近似（補間）方法を指定します。linear ならば、等高線は曲面から得られた値を区別的に直線で結びます。cubicspline (3 次スプライン) ならば、区別的な直線はいくぶんならかな等高線が得られるように補間されますが、多少波打つ可能性があります。bspline (B-spline) は、より滑らかな曲線を描くことが保証されますが、これは z の等しい点の位置を近似しているだけです。

points — 最終的には、全ての描画は、区別的な直線で行われます。ここで指定する数は、bspline または cubicspline での近似に使われる線分の数を制御します。実際には cubicspline と bspline の区間（曲線線分）の数は points と線分の数の積に等しくなります。

**order** — bspline 近似の次数です。この次数が大きくなるにつれて、等高線はなめらかになります（もちろん、高次の bspline 曲線になるほど、元の区分的直線からは離れていきます）。このオプションは bspline モードでのみ有効です。指定できる値は、2（直線）から 10 までの整数です。

**levels** — 等高線のレベルの数は、auto（デフォルト）、discrete, incremental と等高線のレベル数 <n> で制御します。

auto では、<n> は仮のレベルの数であり、実際のレベルの数は、簡単なラベルを生成するように調節されます。曲面の z 座標が zmin から zmax の範囲にあるとき、等高線はその間の dz の整数倍になるよう生成されます。ここで、dz は 10 のあるべき乗の 1, 2, 5 倍、のいずれかです（2 つの目盛りの間を丁度割り切るように）。

levels discrete では、等高線は指定された  $z = <z_1>, <z_2> \dots$  に対して生成されます。指定した個数が等高線のレベルの個数となります。discrete モードでは、set cntrparams levels <n> という指定は常に無視されます。

incremental では、等高線は  $z = <\text{start}>$  から始まり、<increment> ずつ増えて行き限界の個数に達するまで書かれます。<end> はその等高線の数を決定するのに使われますが、これは後の set cntrparam levels <n> によって常に変更されます。z 軸が対数軸の場合、set ztics の場合と同様に、<increment> は積因子として解釈されます。

コマンド set cntrparam が引数なしに呼ばれた場合は、次のデフォルトの値が使われます: linear, 5 points, order 4, 5 auto levels

例:

```
set cntrparam bspline
set cntrparam points 7
set cntrparam order 10
```

以下はレベルの基準が合えば 5 個のレベルが自動的に選択されます:

```
set cntrparam levels auto 5
```

以下は .1, .37, .9 にレベルを設定します:

```
set cntrparam levels discrete .1,1/exp(1),.9
```

以下は 0 から 4 まで、1 ずつ増やすレベルを設定します:

```
set cntrparam levels incremental 0,1,4
```

以下はレベルの数を 10 に設定します（増加の最後の値（end）または自動で設定されるレベルの数は変更されます）:

```
set cntrparam levels 10
```

以下はレベルの数は保持したままレベルの開始値と増分値を設定します:

```
set cntrparam levels incremental 100,50
```

等高線を描く場所の制御に関しては set contour を、等高線のラベルの書式と線種の制御に関しては set clabel を参照してください。

以下も参照してください。

等高線のデモ（contours.dem）

および

ユーザ定義レベルの等高線のデモ（discrete.dem）。

### 43.11 Color box

色の一覧表、すなわち pm3d の palette の min\_z から max\_z までの滑らかな色の勾配は、unset colorbox が使われていない限りカラー・ボックス（colorbox）に描かれます。

```
set colorbox
set colorbox {
    { vertical | horizontal }
```

```

{ default | user }
{ origin x, y }
{ size x, y }
{ noborder | bdefault | border [line style] }
}
show colorbox
unset colorbox

```

カラーボックスの位置は、`default` または `user` で指定でき、後者の場合その位置や大きさを `origin` や `size` コマンドで設定します。

`vertical` と `horizontal` は色勾配の方向を切替えます。

`origin x, y` と `size x, y` は `user` オプションとの組でのみ使用されます。`x, y` の値は、デフォルトではスクリーン座標と解釈されますが、これは 3 次元描画用のかしこまったオプションに過ぎません。`set view map` による `splot` を含む 2 次元描画では、任意の座標系での指定が可能です。例えば以下を試してください:

```
set colorbox horiz user origin .1,.02 size .8,.04
```

これは水平方向の色勾配をグラフの下の辺りに描画します。

`border` は境界描画を ON にします（デフォルト）し、`noborder` は境界描画を OFF にします。`border` の後に正の整数を与えると、それを境界を描画する時の `line style` のタグとして使います。例えば:

```
set style line 2604 linetype -1 linewidth .4
set colorbox border 2604
```

は `line style 2604`、すなわち細い線のデフォルトの境界色 (-1) で境界を描画します。`bdefault` (デフォルト) は、カラー ボックスの境界の描画にデフォルトの境界の `line style` を使います。

カラー ボックスの軸は `cb` と呼ばれ、通常の軸のコマンドで制御されます。すなわち `set/unset/show` で `cbrange`, `[m]cbtics`, `format cb`, `grid [m]cb`, `cblabel` などが、そして多分 `cbdata`, `[no]cbdtics`, `[no]cbmtics` なども使えるでしょう。

パラメータ無しの `set colorbox` はデフォルトの位置へ切替えます。`unset colorbox` はカラー ボックスのパラメータをデフォルト値にリセットし、その上でカラー ボックスを OFF にします。

`set pm3d`, `set palette`, `x11 pm3d`, `set style line` の各項目も参照してください。

### 43.12 Contour

コマンド `set contour` は曲面の等高線を引くことを指示します。このオプションは `splot` でのみ有効です。これは、格子状データ (grid data) を必要とします。詳細は、`grid data` を参照してください。非格子状データで等高線を描きたい場合は、格子を生成するために `set dgrid3d` を使用します。

書式:

```

set contour {base | surface | both}
unset contour
show contour

```

これらの 3 つのオプションは等高線をどこに引くかを指定します。`base` では等高線を `x/y` 軸の刻みのある底面に描かれ、`surface` では等高線はその曲面自身の上に描かれ、`both` では底面と曲面上の両方に描かれます。オプションが指定されていない場合は `base` であると仮定されます。

等高線の描画に影響を与えるパラメータについては `set cntrparam` を、等高線のラベルの制御に関しては `set clabel` を参照してください。

等高線のみのグラフが得るために、曲面自身の描画をしないようにすることもできます (`set surface` 参照)。`set size` を使って、グラフを画面一杯に描画することも可能ですが、そういう出力形式よりも、等高線のデータをファイルに書き出し、それを再び 2 次元データとして読み込んで描画すればよりよい制御が可能になります:

```

unset surface
set contour

```

```
set cntrparam ...
set table 'filename'
splot ...
unset table
# contour info now in filename
set term <whatever>
plot 'filename'
```

等高線を描くためには、データは格子状データ ("grid data") である必要があります。そのようなファイルでは、一つの y-孤立線上の全ての点が順に並べられています。そして隣の y-孤立線上の点が順に並べられ、そして隣、と続いていきます。y-孤立線同士を分離するには一行の空行 (空白、復帰、改行以外の文字を含まない行) を挟みます。splot datafile も参照してください。

以下も参照してください。

等高線のデモ (contours.dem)

および

ユーザ定義レベルの等高線のデモ (discrete.dem).

### 43.13 Data style

このコマンドの形式は現在は推奨されていません。set style data を参照してください。

### 43.14 Datafile

コマンド set datafile は、plot, splot, fit コマンドで入力データを読む場合に、その列 (field) の解釈の仕方を制御するオプションを持ちます。現在は、4 つのそのようなオプションが実装されています。

#### 43.14.1 Set datafile fortran

コマンド set datafile fortran は、入力ファイルの Fortran D 型、Q 型の定数値の特別なチェックを可能にします。この特別なチェックは入力処理を遅くしますので、実際にそのデータファイルが Fortran D 型、Q 型の定数を持っている場合にのみこれを選択すべきです。このオプションは、その後で unset datafile fortran を行えば無効にできます。

#### 43.14.2 Set datafile missing

コマンド set datafile missing は gnuplot に、どのような文字列がデータファイル中の欠けたデータ (missing data) を意味するのかを指示します。実際にその文字列がどのように取り扱われるかは、plot や splot コマンドの using 指定子にも依存します。

書式:

```
set datafile missing {"<string>"}
show datafile missing
unset datafile
```

例:

```
# IEEE NaN ("Not a Number") コードを含むエントリを無視する
set datafile missing "NaN"
```

例:

```
set datafile missing "?"
set style data lines
plot '-'>
    1 10
    2 20
```

```

3 ?
4 40
5 50
e
plot '-' using 1:2
1 10
2 20
3 ?
4 40
5 50
e
plot '-' using 1:(\$2)
1 10
2 20
3 ?
4 40
5 50
e

```

最初の plot は、"3 ?" の行では最初のデータのみを認識し、そこでは 1 行に 1 つしかデータがなかった場合の規則が適用され、行番号が "x" でデータが "y" と見なされ、その点は (2,3) に (この場合は誤って) 描画されます。

2 番目の plot は、真中の行を正しく無視します。描画される曲線は、(2,20) と (4,40) が結ばれます。

3 番目の plot も真中の行を正しく無視しますが、この場合は (2,20) と (4,40) は結ばれません。

デフォルトの missing 用の文字は何も定義されていませんが、多くの場合数値が期待されるべき所にそれと認識できない文字が見つかったらそれは欠けたデータ (missing data) であると見なされます。

#### 43.14.3 Set datafile separator

コマンド set datafile separator "<char>" は、gnuplot に、この後の入力ファイルのデータ列の分離文字が、空白 (whitespace) でなくて <char> であると指示します。このコマンドの最も一般的な使用例は、表計算ソフトやデータベースソフトによって作られた csv (コンマ区切り) ファイルを読む時でしょう。デフォルトのデータ列の分離文字は空白 (whitespace) です。

書式:

```
set datafile separator {"<char>" | whitespace}
```

例:

```
# タブ区切りのファイルを入力
set datafile separator "\t"

# コンマ区切りのファイルを入力
set datafile separator ","
```

#### 43.14.4 Set datafile commentschars

コマンド set datafile commentschars は gnuplot に、どの文字がデータファイル中のコメントを意味するのかを指示します。gnuplot は、指定された文字の中の一つがデータ行の最初の非空白文字として現われた場合、その文字以下を無視します。

書式:

```
set datafile commentschars {"<string>"}
show datafile commentschars
unset commentschars
```

デフォルトの string は、VMS では "#!" で、それ以外では "#" です。

よって、データファイルの以下の行は完全に無視されます

```
# 1 2 3 4
```

が、以下の行

```
1 # 3 4
```

は、もし

```
set datafile missing '#'
```

が指定されていなければ、予期せぬ結果を生じます。

例:

```
set datafile commentschars "#!%"
```

#### 43.14.5 Set datafile binary

コマンド `set datafile binary` は、データファイルの読み込み時にバイナリファイルをデフォルトと設定するのに使われます。書式は、それが `plot` または `splot` コマンドで使われるのと正確に同じです。<binary list> に関しては、詳しくは `binary` を参照してください。

書式:

```
set datafile binary <binary list>
show datafile binary
show datafile
unset datafile
```

例:

```
set datafile binary filetype=auto
set datafile binary array=512x512 format="%uchar"
```

#### 43.15 Decimalsign

コマンド `set decimalsign` は、目盛りの見出し、あるいは `set label` 文字列に書かれる数の小数点記号を選択します。

書式:

```
set decimalsign {<value> | locale {"<locale>"}}
unset decimalsign
show decimalsign
```

引数 `<value>` は、通常の小数点記号に置き換えて使う文字列です。典型的なものはピリオド '.' やコンマ ',' ですが他にも有用なものがあるでしょう。引数 `<value>` を省略すると、小数点の区切りはデフォルト(ピリオド)から変更されません。`unset decimalsign` も `<value>` を省略するのと同じ効果を持ちます。

例:

多くのヨーロッパ諸国での正しい出力形式を得るには:

```
set decimalsign ',','
```

次のことに注意してください: 明示的な文字列を設定した場合、これは軸の目盛りなどの gnuplot の `gprintf()` 書式関数で出力される数値のみに影響し、入力データの書式指定や `sprintf()` 書式関数で出力される数値には影響しません。それらの入力や出力の形式の挙動も変更したい場合は、代わりに以下を使用してください:

```
set decimalsign locale
```

これは、gnuplot に、入力と出力の書式を、環境変数 `LC_ALL`, `LC_NUMERIC`, `LANG` の現在の設定に従ったものを使わせるようにします。

```
set decimalsign locale "foo"
```

これは、gnuplot に、入力と出力の書式を、ロケール "foo" に従ったものにしますが、そのロケールがインストールされている必要があります。もしロケール "foo" が見つからなかった場合、エラーメッセージが出力され、小数点の設定は変更されません。linux システム上では、そこにインストールされているロケールの一覧は "locale -a" で見ることができます。linux のロケール文字列はだいたい "sl\_SI.UTF-8" のような形式をしていますが、Windows のロケール文字列は "Slovenian\_Slovenia.1250"、または "slovenian" のような形式です。ロケール文字列の解釈は、C のランタイムライブラリが行うことに注意してください。古い C ライブラリでは、ロケール設定のサポート(例えば数字の 3 行毎の区切り文字など)を部分的にしか提供していないかもしれません。

```
set decimalsign locale; set decimalsign "."
```

これは、現在のロケールに合ったどんな小数点でも、全ての入出力に対して使用するように設定しますが、gnuplot の内部関数 `gprintf()` を使って書式化する数値は明示的に指定された `'` になります(上書き)。

### 43.16 Dgrid3d

コマンド `set dgrid3d` は、非格子状データから格子状データへの写像機能を有効にし、そのためのパラメータを設定します。格子状データの構造についての詳細は、`splot grid_data` を参照してください。

書式:

```
set dgrid3d {<row_size>} {,{<col_size>} {,<norm>}}
unset dgrid3d
show dgrid3d
```

デフォルトでは `dgrid3d` は無効になっています。有効になると、ファイルから読み込まれる 3 次元のデータは「散在した」データ(非格子状データ)であると見なされます。格子は、グラフと等高線の描画のために、散在したデータを囲む矩形から得られる寸法と、`row_size/col_size` で指定される数の行と列を持つように生成されます。格子は x 方向(行)と y 方向(列)に等間隔です。z の値は散在するデータの z の値の重み付きの平均として計算されます。

3 番目のパラメータであるノルム(norm) は、重み付けを制御するもので、各点は格子点からの距離の norm 乗の逆数で重み付けされます。(実際には、`dx, dy` を各データ点と格子点との差の成分であるとすると、重みは  $dx^{norm} + dy^{norm}$  で与えられます。2 のべきのノルム、特に 4, 8, 16 に関しては、その重みの計算はユークリッド距離を使うことで  $(dx^2+dy^2)^{norm/2}$  のように最適化されていますが、任意の負でない整数を使うことも可能です。)

格子点に近いデータ点程それはその格子点により大きい影響を与え、ノルムの値が大きい程格子点から離れた点の影響は小さくなります。

`dgrid3d` オプションは散在するデータから格子状データに変換する簡単なローパスフィルタです。この問題に対するより洗練された手法が存在しますので、この単純な方法が不十分であれば、gnuplot の外でそのような方法でデータを前処理すべきでしょう。

(z の値は全てのデータ点の重み付けとして求められ、隣接するデータ点を補間しているではありません。よって、端の影響が予期しない、望ましくない結果を引き起こす可能性があります。ある場合には、小さいノルムの値により局所的な平均ではなくデータ点の距離の平均を反映したような格子点が生成され、一方、大きなノルムの値により、隣接するデータ点を滑らかに変化させるのではなく最も近くのデータ点と同じ値を持つ格子点による階段("steps")を作ってしまいます。格子の領域内部は任意の境界条件に関する補外法により埋めることができますが、変数は正規化されず、よって x と y の単位が x, y 各方向に関して、点の相対的な重みに影響をおよぼすことになります。)

例:

```
set dgrid3d 10,10,1      # デフォルト
set dgrid3d ,,4
```

最初のものは、構成する格子を 10x10 にし、重みの計算のノルムは 1 にします。2 番目の例はノルムのみ 4 に変更します。以下も参照

```
scatter.dem: dgrid3d のデモ
```

### 43.17 Dummy

コマンド `set dummy` はデフォルトの仮変数名を変更します。

書式:

```
set dummy {<dummy-var>} {,<dummy-var>}
show dummy
```

デフォルトでは、gnuplot は plot では、媒介変数モード、あるいは極座標モードでは "t", そうでなければ "x" を独立変数（仮変数）とし、同様に splot では、媒介変数モードでは (splot は極座標モードでは使えません) "u" と "v", そうでなければ "x" と "y" を独立変数とします。

仮変数は、物理的に意味のある名前、あるいはより便利な名前として使う方が便利でしょう。例えば、時間の関数を描画する場合:

```
set dummy t
plot sin(t), cos(t)
```

このコマンドでは、少なくとも一つの仮変数が設定される必要があります。set dummyだけだとエラーメッセージが表示されます。

例:

```
set dummy u,v
set dummy ,s
```

第二の例は、2番目の変数を s とします。

### 43.18 Encoding

コマンド set encoding は文字のコード化 (encoding) を選択します。

書式:

```
set encoding {<value>}
show encoding
```

有効な値 (value) は以下の通りです。

- default - 出力形式にデフォルトのコード化の使用を命令
- iso\_8859\_1 - 多くの Unix ワークステーションや MS-Windows で使用可能な最も一般的な西ヨーロッパフォント。このコード化は PostScript の世界で 'ISO-Latin1' として知られているものです。
- iso\_8859\_2 - 中央/東ヨーロッパで使用されるフォント
- iso\_8859\_15 - ユーロ記号を含む iso\_8859\_1 の亜種
- koi8r - 良く使われる Unix のキリル文字コード化
- koi8u - Unix のウクライナ地方のキリル文字コード化
- cp437 - MS-DOS のコードページ
- cp850 - 西ヨーロッパの OS/2 のコードページ
- cp852 - 中央/東ヨーロッパの OS/2 のコードページ
- cp1250 - 中央/東ヨーロッパの MS Windows のコードページ

一般に、コード化の設定は出力形式の設定の前に行なう必要があります。このコード化はどんな出力形式でもサポートされているとは限らず、そして出力形式は要求されたどんな非標準文字も生成できなければいけません。PostScript と X11 と wxt 出力形式はすべてのコード化をサポートしています。OS/2 Presentation Manager は iso\_8859\_2 に対しては自動的にコードページ 912 に変更します。

### 43.19 Fit

fit の設定は、コマンド fit がその出力を行うファイルを定義します。使用している gnuplot が使えるように作られていれば、このコマンドは当てはめのパラメータ誤差を変数に書き出すかをも制御できます。

書式:

```
set fit {logfile {"<filename>"}}
unset fit
show fit
```

引数 <filename> は、単一引用符か二重引用符で囲む必要があります。

ファイル名を指定しなかった場合、または unset fit を使用した場合は、ログファイルはデフォルトの値である "fit.log"、または環境変数 FIT\_LOG の値にリセットされます。

DOS と同様の OS のユーザは、以下のこと注意してください。文字 \ (バックスラッシュ) は 2 重引用符内では特別な意味を持つので、別なディレクトリのファイルを指定する場合は単一引用符で囲むか、またはそれぞれの \ を \\ と書く必要があります。または、DOS 上であっても、\ の代わりに / (フォワードスラッシュ) を使うこともできます。

与えられたログファイル名が / か \ で終わっている場合、それはディレクトリ名と解釈され、ログファイルはそのディレクトリの "fit.log" となります。

オプション errorvariables が ON にされると、fit コマンドで計算されたこの当てはめパラメータの誤差が、そのパラメータの名前に "\_err" をつけた名前のユーザ定義変数にコピーされます。これは主に、当てはめ関数とデータの描画グラフの上にパラメータとその誤差を参照用に出力するのに使われます。例えば:

```
set fit errorvariables
fit f(x) 'datafile' using 1:2 via a, b
print "error of a is:", a_err
set label 'a=%6.2f', a, '+/- %6.2f', a_err
plot 'datafile' using 1:2, f(x)
```

### 43.20 Fontpath

fontpath の設定は、フォントファイルを読み込む場合のファイルの検索パスを追加定義します。今のところ、postscript 出力形式のみが fontpath をサポートしています。ファイルが現在のディレクトリに見つからなかった場合、fontpath のディレクトリが検索されます。サポートしているフォントファイルの形式に関するより詳しい説明は terminal postscript セクションの文書中にあります。

書式:

```
set fontpath {"pathlist1" {"pathlist2" ...}}
show fontpath
```

パス名は単一のディレクトリ名、または複数のパス名のリストとして入力します。複数のパスからなるパスリストは OS 固有のパス区切り、例えば Unix ではコロン (:), DOS//Windows/OS/2/Amiga ではセミコロン (;) 等で区切れます。show fontpath, save, save set コマンドは、可搬性のために OS 固有のパス区切りをスペース ( ) で置き換えます。ディレクトリ名がエクスクラメーションマーク (!) で終わっている場合、そのディレクトリのサブディレクトリも検索されます。

環境変数 GNUMPLOT\_FONTPATH が設定されている場合、その内容は fontpath に追加されますが、それが設定されていない場合システムに依存したデフォルトの値が使用されます。最初にフォントパスを使ったときに、その幾つかのディレクトリが存在するかテストされ、セットされます。よって、一番最初の set fontpath, show fontpath, save fontpath や、埋め込みフォントを使用した場合の plot, splot は、少し時間がかかります。それを少しでも短くしたければ、環境変数 GNUMPLOT\_FONTPATH を設定してください。そうすればディレクトリのチェックは OFF になります。デフォルトのフォントパスが何であるかは、show fontpath で見ることができます。

しかし、show fontpath は、ユーザ定義の fontpath と環境変数によるシステム fontpath を別々に表示しますし、save, save set コマンドは、可搬性の目的のためにユーザ定義の fontpath のみを保存します。

gd ライブライアリを通じて TrueType フォントにアクセスできる出力ドライバもたくさんあります。これらのドライバに対するフォントの検索パスは、環境変数 GDFONTPATH で制御されます。

### 43.21 Format

座標軸の刻みの見出しが、コマンド set format で書式を設定できます。

書式:

```
set format {<axes>} {"<format-string>"}
set format {<axes>} {'<format-string>'}
show format
```

ここで、<axes> (軸) は x, y, xy, x2, y2, z, cb、または何も指定しない (その場合、一度にすべての軸を参照します) のいずれかです。刻みの見出しの文字列の長さ ('printf' で整形された後の) は 100 文字まで、と制限されています。書式文字列 (<format-string>) を省略した場合、それはデフォルトの "%g" になります。LaTeX ユーザにはよく "%g" が好まれます。空の文字列 "" を指定した場合、刻み自身は表示されますが見出しありません。すべての刻みを消すには、unset xtics や unset ytics を使用してください。

改行文字 (\n) も書式文字列で使えます。それを解釈させるには、單一引用符 ('') ではなく ("") を使ってください。syntax の項も参照してください。

デフォルトの書式文字列は両軸とも "%g" ですが、"%2f" や "%3.0em" などの書式が好まれることも多いでしょう。倍精度小数に対して 'printf' と出力形式が受けつけることができる書式であればそれは正しく動作するでしょう。他にもいくつかのオプションが追加されています。書式文字列が浮動小数に対するもののようにあれば gnuplot は妥当な文字列に変換しようとします。

"%" が頭につかない文字はそのまま表示されます。よって、書式文字列内にスペースや文字列などを入れることができます。例えば "%g m" とすれば、数値の後に " m" が表示されます。 "%" 自身を表示する場合には "% %%" のように 2 つ重ねます。

刻みに関するより詳しい情報については set xtics を参照してください。また、この方法で出力される数字にデフォルト以外の小数分離文字を使うやり方については set decimalsign を参照してください。以下も参照。

エレクトロン（電子）デモ (electron.dem).

### 43.21.1 Gprintf

文字列関数 gprintf("format",x) は、gnuplot コマンドの set format, set timestamp などと同様の、gnuplot 独自の書式指定子を使います。これらの書式指定子は、標準的な C 言語の関数である sprintf() のものと全く同じではありません。そのために、gnuplot には sprintf("format",x,...) 関数も用意されています。gnuplot の書式オプションの一覧については、format specifiers を参照してください。

### 43.21.2 Format specifiers

使用可能な書式 (時間/日付モードでない場合) は以下の通りです:

目盛りラベルの数値書式表記	
書式	説明
%f	固定小数点表記
%e, %E	指数表記; 指数の前に "e", "E" をつける
%g, %G	%e (または %E) と %f の略記
%x, %X	16 進表記
%o, %O	8 進表記
%t	10 進の仮数部
%l	現在の対数尺の底を基數とする仮数部
%s	現在の対数尺の底を基數とする仮数部; 補助単位 (scientific power)
%T	10 進の指数部
%L	現在の対数尺の底を基數とする指数部
%S	補助単位の指数部 (scientific power)
%c	補助単位文字
%P	の倍数

補助単位 ('scientific' power) は、指数が 3 の倍数であるようなものです。補助単位指数 ("%c") の文字への変換は -18 から +18 までの指数に対してサポートされています。この範囲外の指数の場合、書式は通常の指数形式に戻ります。

ほかに使うことのできる修飾詞 ("% と書式指定子の間に書くもの) には、次のいくつかがあります: "-" は数字を左詰めにし、"+" は正の数にも符号をつけ、" " (空白一つ) は負の数に "-" をつけるべき場所に正の数の場合に空白を一つつけ、"#" は小数点以下の数字が 0 だけであっても小数点をつけ、正の整数は出力幅を定め、出力幅指定の直前の "0" (文字でなく数字) は先頭に空いた部分を空白で埋める代わりに 0

で埋め、小数点の後に非負の整数を書いたものは精度を意味します（整数の場合は最小桁、小数の場合は小数点以下の桁数）。

これらの全ての修飾語をサポートしていない OS もあるでしょうし、逆にこれ以外のものをもサポートする OS もあるでしょう。疑わしい場合は、適切な資料を調べ、そして実験してみてください。

例:

```
set format y "%t"; set ytics (5,10)      # "5.0" と "1.0"
set format y "%s"; set ytics (500,1000)    # "500" と "1.0"
set format y "%+12.3f"; set ytics(12345)    # "+12345.000  "
set format y "%.2t*10^%+03T"; set ytic(12345) # "1.23*10^+04"
set format y "%s*10^{\$S}"; set ytic(12345)   # "12.345*10^{\{3\}}"
set format y "%s %cg"; set ytic(12345)      # "12.345 kg"
set format y "%.0P pi"; set ytic(6.283185)   # "2 pi"
set format y "%.0f%%"; set ytic(50)          # "50%"
```

set log y 2; set format y '%l'; set ytics (1,2,3)  
#"1.0", "1.0", "1.5" と表示される（3 は  $1.5 * 2^1$  なので）

丸めと指数が必要となるような書式で 9.999 の様な数字が書かれる場合は問題が起ることがあります。軸のデータ型が日時データ (time/date) の場合、書式文字列は 'strftime' 関数 ('gnuplot' 外。"man strftime" としてみてください) に関する有効な指定を行う必要があります。使える入力書式指定の一覧に関しては set timefmt を参照してください。

### 43.21.3 Time/date specifiers

日時データモード (time/date mode) では、次の書式が使用できます:

Tic-mark label Date/Time Format Specifiers	
書式	説明
%a	曜日名の省略形 (Sun,Mon,...)
%A	曜日名 (Sunday,Monday,...)
%b, %h	月名の省略形 (Jan,Feb,...)
%B	月名 (January,February,...)
%d	日 (1-31)
%D	"%m/%d/%y" の簡略形
%k	時 (0-23; 1 桁または 2 桁)
%H	時 (00-23; 常に 2 桁)
%l	時 (1-12; 1 桁または 2 桁)
%I	時 (01-12; 常に 2 桁)
%j	その年の通算日 (1-366)
%m	月 (1-12)
%M	分 (0-60)
%p	"am" または "pm"
%r	"%I:%M:%S %p" の簡略形
%R	"%H:%M" の簡略形
%S	秒 (0-60)
%T	"%H:%M:%S" の簡略形
%U	その年の通算週 (週は日曜日からと数える)
%w	曜日番号 (0-6, 日曜 = 0)
%W	その年の通算週 (週は月曜日からと数える)
%y	西暦 (0-99)
%Y	西暦 (4 桁)

数字を表す書式では、これらの指定子（% の後ろ、指定子の前）に "0" ("オー" でなく "ゼロ") をつけることで、先頭に空白ができる場合に空白の代わりに 0 で埋めることができ、また最小の出力幅を正の整数で指定することもできます（出力される数字を表示するのに指定した幅が足りない場合は無視されます）。表示する文字の長さは 24 文字まで、という制限があり、長すぎた部分は切り捨てられます。

例:

日時のデータが "76/12/25 23:11:11" の場合

```
set format x          # デフォルトでは "12/25/76" \n "23:11"
set format x "%A, %d %b %Y" # "Saturday, 25 Dec 1976"
set format x "%r %D"       # "11:11:11 pm 12/25/76"
```

日時のデータが "98/07/06 05:04:03" の場合

```
set format x "%y/%m/%d %H:%M:%S" # "98/ 7/ 6 5:04:003"
```

## 43.22 Function style

このコマンドの形式は現在は推奨されていません。set style function を参照してください。

## 43.23 Functions

show functions コマンドはユーザーが定義した関数とその定義内容を表示します。

書式:

```
show functions
```

gnuplot における関数の定義とその使い方については expressions の項を参照してください。以下も参照

ユーザ定義関数でのスプライン (spline.dem)

および

関数と複素変数を翼に使用 (airfoil.dem)。

## 43.24 Grid

コマンド set grid は格子線を描きます。

書式:

```
set grid {{no}{m}xtics} {{no}{m}ytics} {{no}{m}ztics}
      {{no}{m}x2tics} {{no}{m}y2tics}
      {{no}{m}cbtics}
      {polar {<angle>}}
      {layerdefault | front | back}
      { {linestyle <major_linestyle>}
        | {linetype | lt <major_linetype>}
        {linewidth | lw <major_linewidth>}
        { , {linestyle | ls <minor_linestyle>}
          | {linetype | lt <minor_linetype>}
          {linewidth | lw <minor_linewidth>} } }
```

unset grid  
show grid

格子線は任意の軸の任意の主目盛/副目盛に対して有効/無効にでき、その主目盛りと副目盛りに対する線種、線幅も指定でき、現在の出力装置がサポートする範囲で、あらかじめ定義したラインスタイルを使用することもできます。

さらに、2 次元の描画では極座標格子も使うことができます — 定義可能な区間にに対して、選択された目盛りを通る同心円と中心からの放射状の線が描かれます (その区間は set angles の設定にしたがって度、またはラジアンで指定します)。極座標格子は現在は極座標モードでは自動的には生成されないことに注意してください。

set grid が描く前に、必要な目盛りは有効になっていなければなりません。gnuplot は、存在しない目盛りに対する格子の描画の命令は単に無視します。しかし、後でその目盛りが有効になればそれに対する格子も描きます。

副格子線に対する線種を何も指定しなければ、主格子線と同じ線種が使われます。デフォルトの極座標の角度は 30 度です。

`front` を指定すると、格子線はグラフのデータの上に描かれます。`back` が指定された場合は格子線はグラフのデータの下に描かれます。`front` を使えば、密集したデータで格子線が見えなくなることを防ぐことができます。デフォルトでは `layerdefault` で、これは 2D 描画では `back` と同じです。3D 描画のデフォルトは、格子とグラフの枠を 2 つの描画単位に分離し、格子は後ろに、枠は描画データまたは関数の前に書きます。ただし、`hidden3d` モードでは、それがそれ自身の並び換えをしていますので、格子線の順番のオプションは全て無視され、格子線も隠線処理にかけられます。これらのオプションは、実際には格子線だけでなく、`set border` による境界線とその目盛りの刻み (`set xtics` 参照) にも影響を及ぼします。

`z` の格子線は描画の底面に描かれます。これは描画の周りに部分的な箱が描画されている場合にはいいでしょう — `set border` を参照してください。

### 43.25 Hidden3d

`set hidden3d` コマンドは曲面描画 (`splot` 参照) で隠線処理を行なうように指示します。その処理の内部アルゴリズムに関する追加機能もこのコマンドで制御できます。

書式:

```
set hidden3d {defaults} |
  {{offset <offset>} | {nooffset}}
  {trianglepattern <bitpattern>}
  {{undefined <level>} | {noundefined}}
  {{no}altdiagonal}
  {{no}bentover} }

unset hidden3d
show hidden3d
```

`gnuplot` の通常の表示とは異なり、隠線処理では与えられた関数、またはデータの格子線を、実際の曲面がその曲面の背後にあって隠されている部分は見せないと同じように処理します。これを可能にするためには、曲面は '格子状' (`splot datafile` 参照) である必要があり、またそれらは `with lines` か `with linespoints` で描かれていないければいけません。

`hidden3d` が有効なときは、格子線だけでなく、面部分や土台の上の等高線 (`set contour` 参照) も隠されます。複数の面を描画している場合は、各曲面は自分自身と他の曲面で隠される部分も持ちます。曲面上への等高線の表示 (`set contour surface`) は機能しません。

見出しと矢印は常に表示され、影響を受けません。グラフの説明 (`key`) も曲面に隠されることはありません。`gnuplot` バージョン 4.2 ではグラフ上に曲面が一つもない状態でも、`set hidden3d` は `with points`, `with labels with vectors` の 3 次元の描画スタイルに影響を与えます。グラフ内の各々の描画をこの処理から明示的に除外したいときは、`with` 指定に特別のオプション `nohidden3d` を追加してください。

`hidden3d` は、`pm3d` モードで描画された、塗り潰された曲面には影響を与えません。`pm3d` の曲面に対して同様の効果を得るには、これの代わりに `set pm3d depthorder` を使ってください。

関数値は格子孤立線の交点で評価されます。見ることの出来る線分を求めるときは個々の関数値、あるいはデータ点の間はそのアルゴリズムによって線形補間されます。これは、`hidden3d` で描画する場合と `nohidden3d` で描画する場合で関数の見かけが異なることを意味します。なぜならば、後者の場合関数値は各標本点で評価されるからです。この違いに関する議論については、`set samples` と `set isosamples` を参照してください。

曲面の隠される部分を消去するのに使われるアルゴリズムは、このコマンドで制御されるいくつかの追加オプションを持っています。`defaults` を指定すればそれらはすべて、以下で述べるようなデフォルトの値に設定されます。`defaults` が指定されなかった場合には、明示的に指定されたオプションのみが影響を受け、それ以外のものは以前の値が引き継がれます。よって、それらのオプションの値をいちいち修正することなく、単に `set {no}hidden3d` のみで隠線処理をオン/オフできることになります。


最初のオプション `offset` は '裏側' の線を描画する線の線種に影響を与えます。通常は曲面の表裏を区別するために、裏側の線種は、表側の線種より一つ大きい番号の線種が使われます。`offset <offset>` によって、その追加する値を、デフォルトの 1 とは異なる増分値に変更できます。`nooffset` オプションは `offset 0` を意味し、これは表裏で同じ線種を使うことになります。

次のオプションは `trianglepattern <bitpattern>` です。`<bitpattern>` は 0 から 7 までの数字で、ビットパターンと解釈されます。各曲面は三角形に分割されますが、このビットパターンの各ビットはそれらの三角形の各辺の表示を決定します。ビット 0 は格子の水平辺、ビット 1 は格子の垂直辺、ビット 2 は、元々の格子が 2 つの三角形に分割されるときの対角辺です。デフォルトのビットパターンは 3 で、これは全ての水平辺と垂直辺を表示し、対角辺は表示しないことを意味します。対角辺も表示する場合は 7 を指定します。

オプション `undefined <level>` は、定義されていない（欠けているデータまたは未定義の関数値）か、または与えられた `x,y,z` の範囲を超えているデータ点に適用させるアルゴリズムを指示します。そのような点は、それでも表示されてしまうか、または入力データから取り除かれます。取り除かれてしまう点に接する全ての曲面要素は同様に取り除かれ、よって曲面に穴が生じます。`<level> = 3` の場合、これは `noundefined` と同じで、どんな点も捨てられません。これは他の場所であらゆる種類の問題を引き起こし得るので使わないべきです。`<level> = 2` では未定義の点は捨てられますが、範囲を超えた点は捨てられません。`<level> = 1` では、これがデフォルトですが、範囲を超えた点も捨てられます。

`noalldiagonal` を指定すると、`undefined` が有効のとき（すなわち `<level>` が 3 でない場合）に起こる以下の場合のデフォルトでの取扱いを変更できます。入力曲面の各格子状の部分は一方の対角線によって 2 つの三角形に分割されます。通常はそれらの対角線の全てが格子に対して同じ方向を向いています。もし、ある格子の 4 つの角のうち一つが `undefined` 処理によりとり除かれていて、その角が通常の方向の対角線に乗っている場合は、その両方の三角形が取り除かれてしまいます。しかし、もしデフォルトの設定である `alldiagonal` が有効になっている場合、その格子については他方向の対角線が代わりに選択され、曲面の穴の大きさが最小になるようにします。

`bentover` オプションは今度は `trianglepattern` とともに起こる別のこととを制御します。かなりしづくちゃの曲面では、下の ASCII 文字絵に書いたように、曲面の 1 つの格子が 2 つに分けられた三角形の表と裏の反対側が見えてしまう場合（すなわち、元の四角形が折り曲げられている（'bent over'）場合）があります：



曲面の格子の対角辺が `<bitpattern>` の 2 bit によって見えるようにはなってはいない場合、上の対角辺 CB はどこにも書かれないことになります。それが結果の表示を理解しにくいものになります。デフォルトで定義される `bentover` オプションは、このような場合それを表示するようにします。もしそうしたくないなら、`nobentover` を選択してください。以下も参照

隠線処理のデモ (`hidden.dem`)

および

複雑な隠線のデモ (`singulr.dem`) .

### 43.26 Historysize

注意: コマンド `set historysize` は、gnuplot が GNU readline ライブライアリを使うように configure された場合のみ有効です。

書式:

```
set historysize <int>
unset historysize
```

`historysize` の値は、gnuplot の終了時に `history` の行を高々その行数までに切り捨てるために使われます。デフォルトは 500 です。`unset historysize` は `history` の切捨てを無効にしますので、`history` ファイルに書き出す行数の制限をしません。

### 43.27 Isosamples

関数を面として描画する場合の孤立線（格子）の密度はコマンド `set isosamples` で変更できます。

書式:

```
set isosamples <iso_1> {,<iso_2>}
show isosamples
```

各曲面グラフは  $<\text{iso\_1}>$  個の u-孤立線と  $<\text{iso\_2}>$  個の v-孤立線を持ちます。 $<\text{iso\_1}>$  のみ指定すれば、 $<\text{iso\_2}>$  は  $<\text{iso\_1}>$  と同じ値に設定されます。デフォルトでは、u, v それぞれ 10 本の標本化が行われます。標本数をもっと多くすればより正確なグラフが作られますか、時間がかかります。これらのパラメータは、データファイルの描画には何も影響を与えません。

孤立線とは、曲面の一つの媒介変数を固定して、もう一つの媒介変数によって描かれる曲線のことです。孤立線は、曲面を表示する単純な方法を与えます。曲面  $s(u,v)$  の媒介変数 u を固定することで u-孤立線  $c(v) = s(u_0, v)$  が作られ、媒介変数 v を固定することで v-孤立線  $c(u) = s(u, v_0)$  ができます。

関数の曲面グラフが隠線処理なしで描かれている場合、set samples は各孤立線上で標本化される点の数を制御します。set samples と set hidden3d も参照してください。等高線描画ルーチンは、関数の点の標本化は各孤立線の交点で行われると仮定しているので、関数の曲面と等高線の解像度を変更するときは、isosamples と同じように samples を変更するのが望ましいでしょう。

### 43.28 Key

コマンド set key は描画された曲線の説明や表題を表示することを可能にします。

説明 (key) の内容、すなわち描画される個々のデータ集合や関数につける名前、およびそれらグラフの曲線とグラフ上の点を表す記号からなるサンプルは、plot (または splot) コマンドの title, with オプションにより決定されます。より詳しい情報については plot title, plot with を参照してください。

書式:

```
set key {on|off} {default}
    {{inside | outside} | {lmargin | rmargin | tmargin | bmargin}
     | {at <position>}}
    {left | right | center} {top | bottom | center}
    {vertical | horizontal} {Left | Right}
    {{no}reverse} {{no}invert}
    {samplen <sample_length>} {spacing <vertical_spacing>}
    {width <width_increment>}
    {height <height_increment>}
    {{no}autotitles {columnheader}}
    {title "<text>"} {{no}enhanced}
    {{no}box { {linestyle | ls <line_style>}
              | {linetype | lt <line_type>}
              {linewidth | lw <line_width>}}}
unset key
show key
```

set key off か unset key によって説明表示なしに描画することができます。

キー内の各要素は vertical (縦) または horizontal (横) に従って重ねられます。vertical の場合、key は可能ならば 2,3 個の縦の列を使います。すなわち、各要素は垂直スペースがなくなるまでは 1 つの列に整列されますが、そこから新しい列が開始されます。horizontal の場合は、key は可能ならば 2,3 個の横の行を使用します。

デフォルトでは、key はグラフ領域の内側の右上の角に置かれます。キーワード left, right, top, bottom, center, inside, outside, lmargin, rmargin, tmargin, bmargin (, above, over, below, under) は、グラフ領域の他の場所への自動的な配置のために使用します。key の描画をどこに置くかをより詳しく指示するための at <position> もあります。この場合、キーワード left, right, top, bottom, center が同様の基準点合わせの設定の目的で使われます。

配置の仕組みを理解ための最も重要な概念は、グラフ領域、すなわち内か外かということと、グラフ領域の境界との間の余白 (margin) を考えることです。グラフ領域に沿って、キーワード left/center/right (l/c/r) と top/center/bottom (t/c/b) は、key をその領域の内側のどこに置くかを制御します。

モード inside では、key はキーワード left (l), right (r), top (t), bottom (b), center (c) によって以下の図のように描画領域の境界に向かって出力されます:

t/l	t/c	t/r
c/l	c	c/r
b/l	b/c	b/r

モード `outside` でも上と同様に自動的に配置されますが、グラフ領域の境界に対して、というよりもむしろ見た目に対して、というべきでしょう。すなわち、グラフの境界は、グラフ領域の外の `key` の場所を作るために、内側に移動することになります。しかし、これは他のラベルの邪魔をしますし、もしかしたら出力デバイスによってはエラーを引き起こすかもしれません。`key` の出力に合わせてどの描画境界が移動するかは、上に述べた `key` の位置、および重ね上げの方向に依存します。4 方向の中心揃えのオプション (`center`) に関しては、どの境界が動くのかに関するあいまいさはありませんが、角への出力のオプションについては、重ね上げ方向が `vertical` の場合は左または右の境界が、`horizontal` の場合は上または下の境界が、それぞれ内側に適切に移動します。

余白 (margin) の書き方は、重ね上げの方向にかかるない自動的な配置を可能にしています。`lmargin` (`lm`), `rmargin` (`rm`), `tmargin` (`tm`), `bmargin` (`bm`) のうちの一つを、矛盾しない 1 方向のキーワードと組み合わせて使用した場合、以下の図に示した場所に `key` が配置されます:

`l/tm c/tm r/tm`

t/lm	t/rm
c/lm	c/rm
b/lm	b/rm

`l/bm c/bm r/bm`

キーワード `above` と `over` は `tmargin` と同じ意味です。以前のバージョンとの互換性のために、`above` と `over` は `l/c/r` や重ね上げ方向のキーワードなしで使用すると、`center` で `horizontal` を使います。キーワード `below` と `under` は `bmargin` と同じ意味です。互換性のために、`below` と `under` は `l/c/r` や重ね上げ方向のキーワードなしで使用すると `center` で `horizontal` を使います。さらに、`outside` も互換性のために `t/b/c` や重ね上げ方向のキーワードがなければ、`top`, `right`, `vertical` (つまり上の `t/rm` と同じ) を使用します。

説明の位置 (`<position>`) は、以前のバージョンと同様単に `x,y,z` を指定してもいいですが、その最初のサンプル行の座標の座表系を選択するための 5 つのキーワード (`first`, `second`, `graph`, `screen`, `character`) を頭につけることもできます。詳細は `coordinates` を参照してください。`<position>` が与えられた場合の `left`, `right`, `top`, `bottom`, `center` の効果は、`label` コマンドで配置される文字列の場合と同じように基準位置の位置合わせに使用されます。すなわち、`left` は `key` が `<position>` の右に置かれて左合わせで出力されます。他の場合も同様です。

説明中のラベル (名前) の行揃えは `Left`, `Right` (デフォルト) で指示します。ラベル文字列と曲線のサンプルは左右入れ替えることができます (`reverse`) し、全体を枠で囲むこともできます (`box {...}`)。その枠の線は、線種 (`linetype`), 線幅 (`linewidth`)、あるいは定義済のラインスタイル (`linestyle`) を指定することができます。ただ、全ての出力装置が線幅の選択をサポートしているとは限らないことに注意してください。

デフォルトでは最初の描画のラベルが説明の一番上に現われ、それに続くラベルがその下に並んで行きます。オプション `invert` は、最初のラベルを説明の一番下に置き、それに続くラベルをその上に並べて行きます。このオプションは、説明のラベルの縦の並びの順番を、積み上げ形式のヒストグラム (`histograms`) の箱の順番に合わせるときに便利でしょう。

グラフ曲線のサンプルの線分の長さは `sampelen` で指定できます。その長さは目盛りの長さと、`<sample_length>*(文字幅)` の和として計算されます。`sampelen` は、グラフ上の点のサンプルの位置にも (もしサンプル線分自身が書かれなくても) 影響を与えています。それは、点の記号はサンプル線分の中央に書かれるためです。

行間の垂直スペースは、`spacing` で指定できます。その幅は、点のサイズ (`pointsize`) と垂直な目盛りのサイズと `<vertical_spacing>` の積になります。この垂直スペースは、文字の高さよりも小さくはならないことが保証されています。

`<width_increment>` は、文字列の長さに加えたり減らしたりする幅（何文字分か）を表す数値です。これは、説明に外枠を書き、文字列に制御文字を使う場合にだけ有用でしょう。gnuplot は外枠の幅を計算するときは、ラベル文字列の文字数を単純に数えるだけなので、それを修正するのに使えます。

`<height_increment>` は、説明の箱の高さに加えたり減らしたりする高さ（何文字分か）を表す数値です。これは主に、説明の回りに箱を描く場合のためのものですが、他に、自動的に設定される説明の位置を `<height_increment>/2` だけ垂直方向にずらすのにも使えます。

`plot` や `splot` で描画される全ての曲線は、デフォルトのオプション `autotitles` に従って表題（title）がつけられます。表題の自動生成は、`noautotitle` で抑制できますがその場合、`(s)plot ... title ...` で明示的に指定された表題のみが描かれることになります。

`set key autotitles columnheader` オプションは、gnuplot が `configure` オプション `-enable-datastrings` 付きでインストールされた場合に有効です。このコマンドは各描画データの先頭行の各列のエントリをテキスト文字列と解釈し、対応する描画グラフのタイトルとして使用します。描画される量が、複数の列データの関数である場合は、gnuplot はどの列をタイトルの描画に使えばいいのかわからなくなりますので、そのような場合、`plot` コマンド上で、例えば `plot "datafile" using (($2+$3)/$4) title 3 with lines` のように明示的にタイトルの列を指定する必要があります。

表題は説明の上につけることもできます (`title "<text>"`) – 単一引用符 ('') と二重引用符 ("") の違いについては `syntax` を参照してください。説明の表題の行揃えは、グラフの表題の行揃えと同じものが使われます。

明示的に与えられた表題は、出力形式がサポートする拡張文字列属性を使って出力されます。詳細は `enhanced text` を参照してください。そのデフォルトの挙動は `noenhanced` オプションで無効にできます。

`set key` のデフォルトは、`on, right, top, vertical, Right, noreverse, noinvert, samplen 4, spacing 1.25, title "", nobox` です。説明の枠の線種はデフォルトではグラフ描画の外枠と同じものが使われます。`set key default` とするとデフォルトの設定に戻ります。

説明は、1 行に 1 曲線分ずつの数行のまとまりとして書かれます。各行の右側には (`reverse` を使っていれば左側には) その曲線と同じ種類の直線のサンプルが引かれ、他の側には `plot` コマンドから得られる文字列（title）が置かれます。これらの行は、架空の直線が説明の左側と右側を分けるかのように垂直に整列されます。コマンド `set key` で指定する座標はこの架空の線分の上の端の座標です。`plot` では直線の位置を指定するために `x` と `y` だけが使われ、`splot` では、`x, y, z` の値全てを使い、グラフを 2 次元面へ投影するのと同じ方法を使って、架空の直線の 2 次元画面での位置を生成します。

`TeX`, `PostScript`, またはこれらと同等の、整形情報が文字列に埋め込まれる出力を使う場合は、gnuplot は説明の位置合わせのための文字列の幅を正しく計算できません。よって説明を左に置く場合は `set key left Left reverse` という組合せを使うのがいいでしょう。そうすれば説明の枠と枠内のすき間は文字列そのままの幅に合わせられます。

`splot` で等高線を書く場合、説明には等高線のラベルも表示されます。これらのラベルの並び具合がうまくいかない、または別な位置に小数点を置きたい場合はそのラベルの書式を指定できます。詳細は `set clabel` を参照してください。

例:

以下はデフォルトの位置に説明を表示します:

```
set key default
```

以下は説明を表示しなくします:

```
unset key
```

以下はデフォルトの（第一の）座標系での (2,3.5,2) の位置に説明を表示します:

```
set key at 2,3.5,2
```

以下は説明をグラフの下に表示します:

```
set key below
```

以下は説明を左下角に表示し、テキストは左に行揃えで、タイトルをつけ、線種 3 の外枠を書きます:

```
set key left bottom Left title 'Legend' box 3
```

### 43.29 Label

`set label` コマンドをすることによって任意の見出し (label) をグラフ中に表示することができます。

書式:

```
set label {<tag>} {"<label text>"} {at <position>}
    {left | center | right}
    {norotate | rotate {by <degrees>}}
    {font "<name>{,<size>}"}
    {noenhanced}
    {front | back}
    {textcolor <colorspect>}
    {point <pointstyle> | nopoint}
    {offset <offset>}
unset label {<tag>}
show label
```

位置 (<position>) は x,y か x,y,z のどちらかで指定し、座標系を選択するにはその座標の前に `first`, `second`, `graph`, `screen`, `character` をつけます。詳細は `coordinates` の項を参照してください。

タグ (<tag>) は見出しを識別するための整数値です。タグを指定しなかった場合未使用のもので最も小さい値が自動的に割り当てられます。現在の見出しを変更するときはそのタグと変更したい項目を指定して `set label` コマンドを使います。

<label text> は文字列定数でも構いませんし、文字列変数、または文字列の値を持つ式でも構いません。`strings`, `sprintf`, `gprintf` を参照。

デフォルトでは、指定した点 x,y,z に見出しの文章の左端が来るよう配配置されます。x,y,z を見出しのどこに揃えるかを変更するには変数 <justification> を指定します。これには、`left`, `right`, `center` のいずれかが指定でき、それぞれ文章の左、右、真中が指定した点に来るよう配配置されるようになります。描画範囲の外にはみ出るような指定も許されますが、座標軸の見出しや他の文字列と重なる場合があります。

`rotate` を指定するとラベルは縦書きになります (もちろん出力ドライバが対応していれば、ですが)。`rotate by <degrees>` が与えられた場合は、それに適合している出力ドライバは指定された角度で文字列を書こうとしますがそうでない出力形式では、垂直な文字列として扱われます。

フォントとそのサイズは、出力形式がフォントの設定をサポートしていれば `font "<name>{,<size>}"` で明示的に選択できます。そうでない出力形式では、デフォルトのフォントが使われます。

通常は、現在の出力形式がサポートしていれば、ラベル文字列の全ての文字列に拡張文字列処理モード (enhanced text mode) が使用されます。`noenhanced` を使用することで、特定のラベルを拡張文字列処理から外すことができます。これは、ラベルが例えばアンダースコア (\_) を含んでいる場合などに有用です。`enhanced text` を参照してください。

`front` が与えられると、見出しがデータのグラフの上に書かれます。`back` が与えられると (デフォルト)、見出しがグラフの下に書かれます。`front` を使うことで、密なデータによって見出しが隠されてしまうことを避けることが出来ます。

`textcolor <colorspect>` は文字列の色を変更します。`<colorspect>` は線種か、あるいは `pm3d` カラーパレット (`splot` でのみ有効) への割当のいずれかです。`set palette` を参照してください。

'`textcolor`' は、'`tc`' と省略可能です。

'`tc default`' は、文字色をデフォルトにします。

'`tc lt <n>`' は、文字色を線種 <n> (line type) と同じものにします。

'`tc ls <n>`' は、文字色を line style <n> と同じものにします。

'`tc palette z`' は、見出しの z の位置に対応したパレット色になります。

'`tc palette cb <val>`' は、色見本 (colorbar) の <val> の色になります。

'`tc palette fraction <val>`' (0<=val<=1) は、[0:1] から 'palette' の  
灰色階調/カラーへの写像に対応した色になります。

<pointstyle> がキーワード `lt`, `pt`, `ps` とともに与えられると (style 参照)、与えられたスタイルと、与えられた線種の色で見出し位置に点 (point) が描画され、見出し文字列は少し移動されます。このオプションは `mouse` 拡張された出力形式でのラベルの配置に、デフォルトで使用されています。見出し文字列近くの点の描画機能を `off` (これがデフォルト) にするには、`nopoint` を使用してください。

その移動は、デフォルトでは、<pointstyle> が与えられれば pointsize の単位で 1,1 で、<pointstyle> が与えられていなければ 0,0 です。移動は、追加の offset <offset> でも制御できます。ここで、<offset> は x,y かまたは x,y,z の形式ですが、それに座標系を選択して、その前に first, second, graph, screen, character のいずれかをつけることもできます。詳細は coordinates を参照してください。

もし一つ(あるいはそれ以上の)軸が時間軸である場合、座標は timefmt の書式にしたがって引用符で囲まれた文字列で与える必要があります。set xdata と set timefmt を参照してください。

EEPIC, Imagen, LaTeX, TPIC で出力する場合は、\\"を使うことで見出しを改行させることができます。  
例:

(1,2) の位置に "y=x" と書く場合:

```
set label "y=x" at 1,2
```

Symbol フォントのサイズ 24 の "シグマ" ( ) をグラフの真中に書く場合:

```
set label "S" at graph 0.5,0.5 center font "Symbol,24"
```

見出し "y=x^2" の右端が (2,3,4) に来るようにし、タグ番号として 3 を使う場合:

```
set label 3 "y=x^2" at 2,3,4 right
```

その見出しを中央揃えにする場合:

```
set label 3 center
```

タグ番号 2 の見出しを削除する場合:

```
unset label 2
```

全ての見出しを削除する場合:

```
unset label
```

全ての見出しをタグ番号順に表示する場合:

```
show label
```

x 軸が時間軸であるグラフに見出しを設定する例:

```
set timefmt "%d/%m/%y,%H:%M"
set label "Harvest" at "25/8/93",1
```

データと、新たに当てはめられたパラメータによる当てはめ関数を描画したい場合、fit の後でかつ plot の前に以下を実行します:

```
set label sprintf("a = %3.5g",par_a) at 30,15
bfit = gprintf("b = %s*10^%S",par_b)
set label bfit at 30,20
```

見出し文字列を小さい点から少しだけ移動する場合:

```
set label 'origin' at 0,0 point lt 1 pt 2 ps 3 offset 1,-1
```

pm3d を使った 3 次元のカラー曲面上のある点の位置に、その z の値(この場合 5.5)に対応した色を見出し文字列につける場合:

```
set label 'text' at 0,0,5.5 tc palette z
```

### 43.30 Lmargin

コマンド set lmargin は左の余白のサイズをセットします。詳細は set margin を参照してください。

### 43.31 Loadpath

`loadpath` の設定は、`call`, `load`, `plot`, `splot` コマンドのデータファイル、コマンドファイルの検索パスを追加定義します。ファイルが現在のディレクトリに見つからなかった場合、`loadpath` のディレクトリが検索されます。

書式:

```
set loadpath {"pathlist1" {"pathlist2" ...}}
show loadpath
```

パス名は単一のディレクトリ名、または複数のパス名のリストとして入力します。複数のパスからなるパスリストは OS 固有のパス区切り、例えば Unix ではコロン (':'), DOS//Windows/OS/2/Amiga ではセミコロン (';') 等で区切れます。`show loadpath`, `save`, `save set` コマンドは、可搬性のために OS 固有のパス区切りをスペース (' ') で置き換えます。

環境変数 `GNUPlot_LIB` が設定されている場合、その内容は `loadpath` に追加されますが、`show loadpath` は、ユーザ定義の `loadpath` と環境変数によるシステム `loadpath` を別々に表示しますし、`save`, `save set` コマンドは、可搬性の目的のためにユーザ定義の `loadpath` のみを保存します。

### 43.32 Locale

`locale` の設定は `{x,y,z}{d,m}tics` が書く日付の言語を決定します。

書式:

```
set locale {"<locale>"}
```

`<locale>` にはインストールされたシステムで使うことの出来る任意の言語を指定できます。可能なオプションについてはシステムのドキュメントを参照してください。デフォルトの値は環境変数 `LC_TIME`, `LC_ALL`, または `LANG` から決定されます。

小数点に関する `locale` を変更したい場合は `set decimalsign` を参照してください。

### 43.33 Logscale

書式:

```
set logscale <axes> <base>
unset logscale <axes>
show logscale
```

ここで、`<axes>` (軸) は、`x`, `x2`, `y`, `y2`, `z`, `cb` の任意の順序の組み合せが可能です。また、`<base>` は、対数スケールの底です。`<base>` が指定されなかった場合は 10 になります。もし、`<axes>` が指定されなかった場合は、全部が指定されたことになります。`unset logscale` は、指定した軸の対数スケールを解除します。

例:

`x, z` 両軸について対数スケールを設定する:

```
set logscale xz
```

`y` 軸について底 2 とする対数スケールを設定する:

```
set logscale y 2
```

`pm3d plot` 用に `z` と色の軸に対数スケールを設定する:

```
set logscale zcb
```

`z` 軸の対数スケールを解除する:

```
unset logscale z
```

### 43.34 Macros

これによりコマンドラインのマクロ置換機能を有効にすると、コマンドライン内の `@<stringvariablename>` の形式の部分文字列は、文字列変数 `<stringvariablename>` に含まれるテキスト文字列に置き換えられます。`substitution` 参照。

書式:

```
set macros
```

### 43.35 Mapping

データが `splot` に球面座標や円柱座標で与えられた場合、`set mapping` コマンドは `gnuplot` にそれをどのように扱うかを指定するのに使われます。

書式:

```
set mapping {cartesian | spherical | cylindrical}
```

デフォルトではカーテシアン座標（通常の `x,y,z` 座標）が使われます。

球面座標では、データは 2 つか 3 つの列（またはその個数の `using` エントリ）として与えられます。最初の 2 つは、`set angles` で設定された単位での方位角 (`theta`) と仰角 (`phi`)（すなわち "経度" と "緯度"）とみなされます。半径 `r` は、もし 3 列目のデータがあればそれが使われ、もしなければ 1 に設定されます。各変数の `x,y,z` との対応は以下の通りです:

```
x = r * cos(theta) * cos(phi)
y = r * sin(theta) * cos(phi)
z = r * sin(phi)
```

これは、"極座標系" というより、むしろ "地学上の座標系"（緯度、経度）に相当することに注意してください（すなわち、`phi` は `z` 軸となす角、というより赤道から計った仰角、になります）。

円柱座標では、データはやはり 2 つか 3 つの列で与えられ、最初の 2 つは `theta`（`set angle` で指定された単位の）と `z` と見なされます。半径 `r` は球面座標の場合と同様、3 列目のデータがあればそれが、なければ 1 と設定されます。各変数の `x,y,z` との対応は以下の通りです:

```
x = r * cos(theta)
y = r * sin(theta)
z = z
```

`mapping` の効果は、`splot` コマンド上の `using` によるフィルタで実現することも可能ですが、多くのデータファイルが処理される場合は `mapping` の方が便利でしょう。しかし、`mapping` を使っていても、もしファイルのデータの順番が適切でなかったら結局 `using` が必要になってしまいます。

`mapping` は `plot` では何もしません。

`world.dem`: `mapping` のデモ。

### 43.36 Margin

自動的に計算される周囲の余白（margin）はコマンド `set margin` で変更できます。`show margin` は現在の設定を表示します。

書式:

```
set bmargin {<margin>}
set lmargin {<margin>}
set rmargin {<margin>}
set tmargin {<margin>}
show margin
```

`<margin>` の単位には、適切と思われる、文字の高さと幅が使われます。正の値は余白の絶対的な大きさを定義し、負の値（または無指定）は `gnuplot` によって自動計算される値を使うことになります。3 次元描画では、今のところ、左の余白（`lmargin`）の設定のみが効力を持ちます。

描画の余白は通常目盛り、目盛りの見出し、軸の見出し、描画のタイトル、日付、そして境界の外にある場合の key (グラフ見出し) のサイズ等を元に計算されます。しかし、目盛りの刻みが境界でなく軸の方についている場合 (例えば set xtics axis によって)、目盛りの刻み自身とその見出しある余白の計算には含まれませんし、余白に書かれる他の文字列の位置の計算にも含まれません。これは、軸と境界が非常に近い場合、軸の見出しが他の文字列を上書きする可能性を示唆します。

### 43.37 Mouse

コマンド `set mouse` はマウス機能を有効にします。現在、pm, x11, ggi, windows, wxt の各出力ドライバがマウス拡張されています。2 種類のマウスマードがあります。2 次元グラフモードは、2 次元グラフ、または地図型表示 (すなわち、z の回転角が 0, 90, 180, 270, 360 度の `set view` での `splot`、および `set view map`) で動作し、グラフ上の位置の追跡、拡大表示、グラフの注釈づけ等を可能にします。`splot` による 3 次元グラフに対しては、グラフの視方向 (view) と伸縮の変更が、マウスボタン 1 と 2 (によるドラッグ) で行えます。これらのボタンにさらに `<ctrl>` キーを押すと、座標系のみが回転、拡大されますが、これは大きなデータに対して有用でしょう。ボタン 2 の垂直方向のドラッグを `shift` キーと同時に使うと、z 軸の一番下の位置 (`ticslevel`) を上下します。

マウスは多重描画 (`multiplot`) モードでは無効ですが、`unset multiplot` で多重描画が終了すれば、マウス機能は ON になり、直前の描画 (ほぼ `replot` で描画されるもの) に対して作用します。

書式:

```
set mouse {doubleclick <ms>} {nodoubleclick} \
    {{no}zoomcoordinates} \
    {noruler | ruler {at x,y}} \
    {polardistance{deg|tan} | nopolardistance} \
    {format <string>} \
    {clipboardformat <int>/<string>} \
    {mouseformat <int>/<string>} \
    {{no}labels} {labeloptions <string>} \
    {{no}zoomjump} {{no}verbose}

unset mouse
```

ダブルクリックの解像度はミリ秒 (ms) 単位で与えます。これは、ボタン 1 用のもので、現在のマウス位置を `clipboard` にコピーするのに使います。シングルクリックでそれを行うようにするには、0 ms を指定してください。デフォルトの値は 300 ms です。

オプション `zoomcoordinates` は、拡大 (zoom) の時に、ズームの枠の端にその座標を書くかどうかを決定し、デフォルトでは ON になっています。

オプション `noruler` と `ruler` は、定規 (ruler) 機能を off, on にします。`ruler` には座標を与えることもできます。これは、デフォルトのキー割り当て 'r' に対応します。

オプション `polardistance` は、マウスカーソルから定規 (ruler) までの距離を極座標でも表示 (距離、および角度または傾き) するかどうかを決定します。これはデフォルトのキー割り当て '5' に対応します。

オプション `format` は、`fprintf` と似た書式文字列で、ドライバウィンドウとクリップボード上で実数をどのように表示するかを決定します。デフォルトは "% #g" です。

`clipboardformat` と `mouseformat` は、ボタン 1 とボタン 2 の作用時

-- 座標をクリップボードへコピー、マウス位置に一時的に注釈をつける --

の文字列の書式用に使われます。これは、キー割り当て '1', '2', '3', '4' に対応します (ドライバのヘルプウィンドウ参照)。引数が文字列の場合、その文字列は C 言語の書式で、2 つの実数の指定子を含んでいる必要があります。例: `set mouse mouseformat "mouse = %5.2g, %10.2f"`。この文字列を再び OFF にするには `set mouse mouseformat ""` としてください。

以下の書式が組み込まれています (書式 6 は書式文字列が既に指定されている場合のみ使用可):

0	カッコ内に実際の座標	例: [1.23, 2.45]
1	カッコ無に実際の座標	例: 1.23, 2.45
2	x == 時間書式 (timefmt)	[('set timefmt' の設定), 2.45]
3	x == 日付	[31. 12. 1999, 2.45]

```

4  x == 時刻          [23:59, 2.45]
5  x == 日付 / 時刻    [31. 12. 1999 23:59, 2.45]
6  文字列で与えられた代替書式    ""

```

ボタン 2 の gnuplot のラベルを実際に得るには、オプション `labels` を使用します（デフォルトでは `nolabels` で、ボタン 2 は単に一時的な注釈をマウス位置に描画します）。ラベルは現在の `mouseformat` の設定に従って書かれます。`labeloptions` は、コマンド `set label` にどんなオプションを渡すかを制御します。デフォルトは "pointstyle 1" で、これはラベル位置に小さいプラス (+) を描画します。その点の大きさは、`set pointsize` コマンドで与えられることに注意してください。ラベルは、ラベルの点の上で Ctrl キーを押してボタン 2 をクリックすることで消すことができます。実際のラベルの位置にどれ位近くでクリックしなければいけないかの閾値も `pointsize` で決定されます。

オプション `zoomjump` が ON の場合、ボタン 3 によるズーム領域の選択を開始すると、マウスポインタは自動的に少しだけずれた位置に移動します。これは、ごく小さい（または空でさえある）ズーム領域を選択してしまうことを避けるのに便利でしょう。デフォルトでは `zoomjump` は OFF です。

オプション `verbose` が ON の場合、実行時の報告コマンドが表示されます。このオプションはドライバウィンドウ上で 6 を打つことで ON/OFF がスイッチできます。デフォルトでは `verbose` は OFF になっています。

ドライバウィンドウ上で 'h' を打つと、マウスとキー割当の短い説明が表示されます。これは、ユーザ定義のキー割当、すなわち `bind` コマンドによる `hotkeys` (`bind` 参照) も表示されます。ユーザ定義の `hotkey` はデフォルトのキー割当を無効にします。

ドライバウィンドウ上で 'q' を打つと、ウィンドウを閉じます。このキーは `bind` コマンドでは無効にできません。

`bind` や `label` も参照してください。

#### 43.37.1 X11 mouse

`x11` の出力形式のオプション `set term x11 <n>` を使って複数の X11 描画ウィンドウが開いている場合、マウスコマンドとホットキーの機能をちゃんと使えるのは現在の描画ウィンドウのみです。しかし、他のウィンドウも左下にマウスの座標を表示位はしてくれるでしょう。

他のスクリーン出力形式との統一性も考え、X11 でのマウスサポートは、標準入力がどこからであっても、デフォルトで ON になります。しかし、UNIX 風の OS の中には、`/dev/null` のような特殊な入力デバイスを選択してはいけないものがあり、そのようなデバイスでマウスを ON にすると gnuplot がハングアップしてしまうことがあります。このような状況では、`unset mouse` でマウス機能を OFF にしてください。

#### 43.38 Multiplot

コマンド `set multiplot` は gnuplot を多重描画モードにします。これは複数の描画を同じページ、ウィンドウ、スクリーンに表示するものです。

書式:

```

set multiplot { layout <rows>,<cols>
                {rowsfirst|columnsfirst} {downwards|upwards}
                {title <page title>}
                {scale <xscale>{,<yscale>}} {offset <xoff>{,<yoff>}}
            }
unset multiplot

```

出力形式 (terminal) によっては、コマンド `unset multiplot` が与えられるまで何の描画も表示されないことがあります。この場合このコマンドによりページ全体の描画が行なわれ、gnuplot は標準の単一描画モードになります。それ以外の出力形式では、各 `plot` コマンドがそれぞれ表示を更新しますが、それは、これまで出力したものすべてを再描画し、それに新しいものを追加する方法、あるいは、現在の表示に単に新しいものを追加するだけのやり方、のいずれかです。

次の描画で使われる領域は、新しい描画が行われる前に消されることはありませんから、描画を "挿入" するような場合などでそうしたいなら、`clear` コマンドを使用してください。

定義済の見出しやベクトルは、各描画において、毎回現在のサイズと原点に従って書かれます（それらが screen 座表系で定義されていない場合）。それ以外の全ての set で定義されるものも各描画すべてに適用されます。もし 1 度の描画にだけ現われて欲しいものを作りたいなら、それが例えば日付 (timestamp) だとしたら、set multiplot と unset multiplot で囲まれたブロック内の plot (または splot, replot) 命令の一つを set time と unset time ではさんでください。

multiplot のタイトルは、個々の描画タイトルがあったとしても、それとは別のもので、ページの上部にそのためのキャンバス全体の幅にわたるスペースが確保されます。

layout が指定されていない場合、あるいはより良い位置合わせをしたい場合は、コマンド set origin と set size 各描画で正しい位置に設定する必要があります。詳細は set origin と set size の項目を参照してください。

例:

```
set multiplot
set size 0.4,0.4
set origin 0.1,0.1
plot sin(x)
set size 0.2,0.2
set origin 0.5,0.5
plot cos(x)
unset multiplot
```

これは、cos(x) のグラフを、sin(x) の上に積み重ねて表示します。

set size と set origin は全体の描画領域を参照しそれは各描画で利用されます。set term size も参照してください。描画境界を一列に揃えたいならば、set margin コマンドで、境界の外の余白サイズを同じサイズに揃えることが出来ます。その使用に関しては set margin を参照してください。余白サイズは文字サイズ単位の絶対的な数値単位を使用することに注意してください。よって残ったスペースに描かれるグラフは表示するデバイスの表示サイズに依存します。例えば、プリンタとディスプレイの表示は多分違ったものになるでしょう。

オプション layout により、各描画の前にそれぞれ与えていた set size や set origin コマンドなしに、単純な複数グラフの描画を作成できます。それらの設定は自動的に行なわれ、いつでもその設定を変更できます。layout では表示は <rows> 行と <cols> 列の格子に分割され、各格子は、その後に続く対応する名前のオプションによって列 (rowsfirst)、あるいは行 (columnsfirst) が先に埋められて行きます。描画グラフの積み上げは下方向 (downwards) に、または上方向 (upwards) に伸びるようにできます。デフォルトは rowsfirst で downwards です。

各描画は scale で伸縮を、offset で位置の平行移動を行なうことができます。scale や offset の y の値が省略された場合は、x の値がそれに使用されます。unset multiplot により自動配置機能はオフになり、そして set size と set origin の値は set multiplot layout の前の状態に復帰されます。

例:

```
set size 1,1
set origin 0,0
set multiplot layout 3,2 columnsfirst scale 1.1,0.9
[ ここには 6 つまでの描画コマンド ]
unset multiplot
```

上の例では 6 つの描画が 2 列の中に上から下へ、左から右へと埋められて行きます。各描画は水平サイズが 1.1/2、垂直サイズが 0.9/3 となります。

以下も参照

[multiplot のデモ \(multiplt.dem\)](#)

### 43.39 Mx2tics

x2 (上) 軸の小目盛り刻みの印は set mx2tics で制御されます。set mxtics を参照してください。

### 43.40 Mxtics

x 軸の小目盛り刻みの印は set mxtics で制御されます。unset mxtics によってそれを表示させなくする事が出来ます。同様の制御コマンドが各軸毎に用意されています。

書式:

```
set mxtics {<freq> | default}
unset mxtics
show mxtics
```

これらの書式は mytics, mztics, mx2tics, my2tics, mcbtics に対しても同じです。

<freq> は大目盛り間の、小目盛りによって分割される区間の数（小目盛りの数ではありません）です。通常の線形軸に対してはデフォルトの値は 2 か 5 で、これは大目盛りによって変化します。よって大目盛り間に 1 つ、または 4 つの小目盛りが入ることになります。default を指定することによって小目盛りの数はデフォルトの値に戻ります。

軸が対数軸である場合、分割区間の数はデフォルトでは有意な数にセットされます（10 個の長さを元にして）。<freq> が与えられていればそちらが優先されます。しかし、対数軸では通常の小目盛り（例えば 1 から 10 までの 2, 3, ..., 8, 9 の刻み）は、9 つの部分区間しかありませんが、<freq> の設定は 10 とすることでそうなります。

小目盛りを任意の位置に設定するには、("label" <pos> <level>, ...) の形式を set {x|x2|y|y2|z}tics で使用してください。ただし、<label> は空（""）で、<level> を 1 にします。

コマンド set m{x|x2|y|y2|z}tics は、大目盛りが一様の間隔の場合にのみ働きます。もし全ての大目盛りが set {x|x2|y|y2|z}tics によって手動で配置された場合は、この小目盛りのコマンドは無視されます。自動的な大目盛りの配置と手動の小目盛りの配置は、set {x|x2|y|y2|z}tics と set {x|x2|y|y2|z}tics add と使うことで共存できます。

例:

```
set xtics 0, 5, 10
set xtics add (7.5)
set mxtics 5
```

この場合、大目盛りは 0,5,7.5,10、小目盛は 1,2,3,4,6,7,8,9 の場所

```
set logscale y
set format y ""
set ytics 1e-6, 10, 1
set ytics add ("1" 1, ".1" 0.1, ".01" 0.01, "10^-3" 0.001, \
               "10^-4" 0.0001)
set mytics 10
```

この場合、大目盛りは指定された書式で、小目盛は対数的に配置

デフォルトでは小目盛りの表示は、線形軸ではオフで、対数軸ではオンになっています。その設定は、大目盛りに対する axis|border と {no}mirror の指定を継承します。これらに関する情報については set xtics を参照してください。

### 43.41 My2tics

y2 (右) 軸の小目盛り刻みの印は set my2tics で制御されます。set mxtics を参照してください。

### 43.42 Mytics

y 軸の小目盛り刻みの印は set mytics で制御されます。set mxtics を参照してください。

### 43.43 Mztics

z 軸の小目盛り刻みの印は set mztics で制御されます。set mxtics を参照してください。

### 43.44 Offsets

オフセットは、自動縮尺されたグラフの中のデータの周りに境界を置く仕組みを提供します。

書式:

```
set offsets <left>, <right>, <top>, <bottom>
unset offsets
show offsets
```

各オフセットは定数、または数式が使え、それらのデフォルトの値は 0 です。左右のオフセットは x 軸と同じ単位で指定し、上下のオフセットは y 軸と同じ単位で指定します。正のオフセットの値はグラフを指定された方向へ伸ばします。例えば正の下方向のオフセットは y の最小値をより小さな値にします。許されている範囲での負のオフセットは、自動縮尺、あるいはクリッピングとの思いもよらぬ結果を生む可能性があります。

オフセットは `splot` では無視されます。

例:

```
set offsets 0, 0, 2, 2
plot sin(x)
```

この `sin(x)` のグラフの y の範囲は [-3:3] になります。それは、関数の y の範囲は [-1:1] に自動縮尺されますが、垂直方向のオフセットがそれぞれ 2 であるためです。

### 43.45 Origin

コマンド `set origin` はスクリーン上で曲面描画の原点を指定（すなわち、グラフとその余白）するのに使用します。その座標系はスクリーン座標系 (screen) で与えます。この座標系に関する情報については `coordinates` を参照してください。

書式:

```
set origin <x-origin>, <y-origin>
```

### 43.46 Output

デフォルトでは、グラフは標準出力に表示されます。コマンド `set output` はその出力を指定されたファイルやデバイスにリダイレクトします。

書式:

```
set output {"<filename>"}
show output
```

ファイル名は引用符で囲まなければなりません。ファイル名が省略された場合は、直前の `set output` で開かれたファイルがクローズされ、新たな出力が標準出力 (STDOUT) に送られます。(もし、`set output "STDOUT"` とすると出力は "STDOUT" という名前のファイルに送られるかもしれません！["かもしれない" というのは、例えば `x11` や `wxt` などの terminal (出力形式) では `set output` が無視されるからです。])

MSDOS のユーザは次のことに注意すべきです: 文字 '\\' は 2 重引用符の中では特別な意味を持ちます。よって、別のディレクトリにあるファイル名を指定する場合は单一引用符を用いるべきでしょう。

`set terminal` と `set output` の両方を指定する場合、`set terminal` を先に指定する方が安全です。それは、ある種の terminal では、OS が必要とするフラグをセットすることがあるからです。例えば、OS がファイルを開くときに (礼儀良く) ファイルがフォーマットされているかどうかを知る必要があるような OS などがそれ該当します。

`popen` 関数を持つようなマシン (Unix 等) では、ファイル名の最初を '|' とすることにより、出力をシェルコマンドにパイプで渡すことが可能です。例えば以下の通りです:

```
set output "|lpr -Plaser filename"
set output "|lp -dlaser filename"
```

MSDOS では、`set output "PRN"` とすると標準のプリンタに出力されます。VMS では出力は任意のスプール可能なデバイスに送ることが出来ます。出力を DECnet 透過なタスクに送ることも可能で、それはある種の柔軟性を与えてくれます。

### 43.47 Parametric

`set parametric` コマンドは `plot` および `splot` の意味を通常の関数描画から媒介変数表示 (parametric) 関数描画に変更します。`unset parametric` を使えば元の描画モードに戻ります。

書式:

```
set parametric
unset parametric
show parametric
```

2 次元グラフにおいては、媒介変数表示関数はひとつの媒介変数に対する 2 つの関数で定められます。例としては `plot sin(t),cos(t)` とすることによって円が描けます (アスペクト比が正しく設定されていれば – `set size` 参照)。gnuplot は、両方の関数が媒介変数による `plot` のために与えられていなければエラーメッセージを出します。

3 次元グラフにおいては面は  $x = f(u,v)$ ,  $y = g(u,v)$ ,  $z = h(u,v)$  で定められます。よって 3 つの関数を組で指定する必要があります。例としては、`cos(u)*cos(v),cos(u)*sin(v),sin(u)` とすることによって球面が描けます。gnuplot は、3 つ全部の関数が媒介変数による `splot` のために与えられていなければエラーメッセージを出します。

これによって表現できる関数群は、単純な  $f(x)$  型の関数群の内包することになります。なぜならば、2 つ (3 つ) の関数は  $x$ ,  $y$  (,  $z$ ) の値を独立に計算する記述ができるからです。実際、 $t,f(t)$  のグラフは、一番目の関数のような恒等関数を用いて  $x$  の値が計算される場合に  $f(x)$  によって生成されるグラフと等価です。同様に、3 次元での  $u,v,f(u,v)$  の描画は  $f(x,y)$  と等価です。

媒介変数表示関数は、 $x$  の関数、 $y$  の関数 (,  $z$  の関数) の順に指定し、それらは共通の媒介変数およびその変域で定義されることに留意して下さい。

さらに、`set parametric` の指定は、新しい変数変域を使用することを暗に宣言します。通常の  $f(x)$  や  $f(x,y)$  が `xrange`, `yrange` (, `zrange`) を使用するのに対して、媒介変数モードではそれに加えて、`trange`, `urange`, `vrangle` を使用します。これらの変域は `set trange`, `set urange`, `set vrangle` によって直接指定することも、`plot` や `splot` で指定することもできます。現時点では、これらの媒介変数のデフォルトの変域は `[-5:5]` となっています。将来的にはこれらのデフォルト値をもっと有意なものに変更する予定です。

### 43.48 Plot

コマンド `show plot` は現在の描画コマンド、すなわち `replot` コマンドで再現される、直前に行われた `plot` や `splot` コマンドを表示します。

さらにコマンド `show plot add2history` は、この現在の描画コマンドを `history` に書き出します。これは、`replot` を使って直前の描画コマンドに曲線を追加した場合、そしてコマンド行全体をすぐに編集したい場合に便利です。

### 43.49 Pm3d

`pm3d` は `splot` の一つのスタイルで、パレットに割り付けられた 3 次元、4 次元データを、カラー/灰色の色地図/曲面として描画します。これは `pm3d` アルゴリズムを用いていて、これはデータが格子状であっても、データ走査毎に点の数が違っているような非格子状のデータであっても、前処理することなく描画できます。

カラー曲面の描画は、`palette` で指定した色割当による多角形の塗りつぶしをサポートしている出力形式で行えます。現在サポートしている出力形式には以下のものが含まれます。

画像出力ドライバ:

```
OS/2 Presentation Manager
X11
Linux VGA (vgagl)
```

```

GGI
Windows
AquaTerm (Mac OS X)
wxWidgets (wxt)
画像ファイル出力ドライバ:
PostScript
pslatex, pstex, epslatex
gif, png, jpeg
(x)fig
tgif
cgm
pdf
svg

```

まず、地図/曲面がどのように描かれるのかについて記述します。入力データは、関数を評価して得られるかまたは splot data file から得られます。曲面は、走査(孤立線)の繰り返しで構成されます。pm3d アルゴリズムでは、最初の走査で検出された隣り合う 2 点と、次の走査で検出された他の 2 点の間の領域が、これら 4 点の z の値(または追加された 'color' 用の列の値、using 参照)に従って灰色で(またはカラーで)塗られます。デフォルトでは 4 つの角の値の平均値が使われますが、それはオプション corners2color で変更できます。それなりの曲面を描くためには、隣り合う 2 点の走査が交差してはいけなくて、近接点走査毎の点の数が違いすぎてはいけません。もちろん、最も良いのは走査の点の数が同じことです。他には何も必要ではありません(例えばデータは格子状である必要もない)。他にもこの pm3d アルゴリズムは、入力された(計測された、あるいは計算された)領域の外には何も描かない、という長所があります。

曲面の色づけは、以下のような入力データに関して行われます:

1. 関数、または 1 つか 3 つのデータ列からなるデータの splot: 上に述べた四辺形の 4 つの角の z 座標の平均値(または corners2color)から、灰色の範囲 [0:1] を与える zrange または cbrange の範囲 [min\_color\_z,max\_color\_z] への対応により、灰色/カラーの値が得られます。この値は、直接灰色の色地図用の灰色の値として使うことができます。正規化された灰色の値をカラーに対応させることもできます - 完全な説明は set palette 参照。
2. 2 つか 4 つのデータ列からなるデータの splot: 灰色/カラーの値は、z の値の代わりに最後の列の座標を使って得られますので、色と z 座標が独立なものになります。これは 4 次元データの描画に使うことができます。

他の注意:

1. 物理学者の間では、gnuplot の文書やソースに現われる 'iso\_curve'(孤立線)という言葉よりも、上で言及した '走査(scan)' という言葉の方が使われています。1 度の走査と他の走査の記録により色地図を評価する、というのはそういう意味です。
2. 'gray' や 'color' の値(scale)は、滑らかに変化するカラーパレットへの、連続な変数の線形写像です。その写像の様子は描画グラフの隣に長方形で表示されます。この文書ではそれを "カラー ボックス(colorbox)" と呼び、その変数をカラー ボックス軸の変数と呼びます。set colorbox, set cbrange を参照してください。
3. pm3d の色づけを 3 次元曲面ではなく 2 次元描画に使うには、set view map か set pm3d map を使用してください。

書式(オプションは任意の順で与えることができます):

```

set pm3d
set pm3d {
    { at <bst combination> }
    { interpolate <steps in scan>,<steps between scans> }
    { scansautomatic | scansforward | scansbackward | depthorder }
    { flush { begin | center | end } }
    { ftriangles |noftriangles }
    { clip1in | clip4in }
    { corners2color { mean|geomean|median|min|max|c1|c2|c3|c4 } }
    { hidden3d <linestyle> | nohidden3d }
    { implicit | explicit }
}

```

```

    { map }
}
show pm3d
unset pm3d

```

データまたは関数の描画スタイル (style) がグローバルに、または with オプションで pm3d に設定されている場合、色付きの曲面が描画されます。オプション implicit が有効になっている場合、pm3d の曲面は線分による曲面の網目による表示も一緒に行なわれます。詳しいことは、この節の下の方をご覧ください。

色の曲面は底面か天井 (この場合は灰色/カラーの平面地図) か曲面上の点の z 座標 (灰色/カラー曲面) に描くことができます。その選択は、オプション at に、b, t, s の 6 つまでの組合せの文字列をつけて指定することで行えます。例えば at b は底面のみに描画しますし、at st は最初に曲面に描いて次に天井面に色地図を描きますし、at bstbst は … 真面目な話、こんなものは使いません。

塗られた四辺形は、次から次へと描画されて行きます。曲面を描画する場合 (at s)、後の四辺形が前のものに重なり (上書き) ます (gnuplot は塗られた多角形の網の重なりの相互作用を計算するような仮想現実ツールではありません)。

最初に走査されるデータが最初に描くか最後に描くかを切替えるスイッチオプション scansforward と scansbackward を試してみてください。デフォルトは scansautomatic で、これは gnuplot に走査の順を推測させます。

2 回の連続する走査で点の数が同じでなかった場合、四辺形の点の取り始めを、両方の走査の最初から (flush begin) にするか、最後から (flush end) にするか、真中から (flush center) にするかを決定しなければいけません。flush (center|end) は scansautomatic とは両立せず、よって flush center または flush end を指定して scansautomatic が設定された場合、それは無言で scansforward に変更されます。

2 回の連続する走査で点の数が同じでなかった場合、個々の走査で点が足りない場合に、走査の最後に色三角形を描くかどうかをオプション ftriangles は指示します。これは滑らかな色地図の境界を描くのに使われます。

四辺形の x,y 座標に関するクリッピングは 2 つの方法で行われます。clip1in: 各四辺形の全ての 4 点が定義されなければならず、少なくともそのうちの 1 点が x, y の範囲におさまっていなければなりません。clip4in: 各四辺形の全ての 4 点が x, y の範囲におさまっていなければなりません。

描画される各 pm3d 四辺形には一つの灰色/カラー値が対応します (4 頂点間で滑らかなカラー変化は起こりません)。その値は、corners2color <option> に従って周囲の角の z 座標から計算されます。<option> は 'mean' (デフォルト), 'geomean', 'median' で、曲面のカラーの平滑化に幾つかの種類を与え、「min」, 「max」はそれぞれ最小値、最大値を選択します。これらは鋭敏な、あるいは急激なピーク値を持つようなピクセルイメージや色地図を作るときには必要ありません。そのような場合には、むしろオプション 'c1', 'c2', 'c3', 'c4' を使って、四辺形の色の割当にただ一つの角の z 座標を使うようにすればいいでしょう。どの角が 'c1' に対応するのかを知るために何回か実験してみる必要があるでしょう。その向きは描画の方向に依存しています。pm3d アルゴリズムは、カラー曲面を入力データ点の範囲の外には描かないで、オプション 'c<j>' は、格子の 2 つのヘリに沿ったピクセルが、どの四辺形の色にも寄与しない、という結果をもたらします。例えば、pm3d アルゴリズムを 4x4 のデータ点の格子に適用するスクリプト demo/pm3d.dem (是非見てください) では、(4-1)x(4-1)=9 色しかない長方形が生成されます。

与えられた節点に対して、その周りの 4 つの節点の平均化された (x,y) 座標から角を得て四辺形を作り、その四辺形を節点の色で塗る、といったような他の描画アルゴリズムが将来実装されるかもしれません。これは、イメージの描画 (2 次元の格子) に対しては image と rgbimage スタイルによって既に行なわれています。

z の値の範囲と曲面の色の値の範囲は、z と cb に関する set log 同様、set zrange と set cbrange によって独立に調整し得ることに注意してください。色地図は cb 軸のみで調節されます。set view map と set colorbox を参照してください。

オプション hidden3d は、線種 (linestyle) を引数に取りますが、それは set style line ... で生成しなければなりません (その線種は pm3d の設定時には存在している必要はありませんが、描画時には必要です)。これが設定されると、線は隠線処理を考慮に入れながら、指定された線種で描画されます。これは、set hidden3d コマンドを使うよりもはるかに効果的で、これは実際に隠線処理を計算することはしませんが、塗りつぶされた多角形を正しい順序で描いて行きます。よって、pm3d を使う場合のお勧めの選択は以下の通りです:

```
set pm3d at s hidden3d 100
set style line 100 lt 5 lw 0.5
unset hidden3d
unset surf
splot x*x+y*y
```

従来、このコマンドに `{transparent|solid}` のオプションが用意されていましたが、現在はそれらはそれぞれ `set grid {front|layerdefault}` で行なうことができます。

`set pm3d map` は `set pm3d at b; set view map; set style data pm3d; set style func pm3d;` を省略したものです。これは、`set view map` がなかったころの旧バージョンへの互換性のためのものです。入力データ点をフィルタするための `zrange`、および色の範囲の変更用の `cbrange` を注意して適切に使用してください。`set (no)surface` も効果（副作用？）があるようです。

オプション `interpolate` は格子点をより細かな網目に補間し、色四角形も近似的に補間します。データ描画に対しては、これは曲面の色の変化を滑らかにし、曲面の色の尖りを補正します。関数描画に対しては、この補間は細かさの代わりにメモリを消費してしまう、といったことくらいの意味しかありませんから、関数描画の場合は普通 `samples` や `isosamples` を使うべきでしょう。

色づけの設定はカラー ボックスの描画と同様に `set palette` で決定されます。一つの描画では一つのパレットのみが存在し得ます。いくつもの曲面を異なるパレットで描画するには、`origin` と `size` を固定して `multiplot` を使うことで行えます。出力ドライバが利用できる色を使い尽くしてしまう場合には `set palette maxcolors` を使うことを忘れずに。

gnuplot の起動時はモードは `explicit` になっています。歴史的な、よって互換性のために、コマンド `set pm3d;` (すなわちオプションを指定しない場合) と `set pm3d at X ...` (すなわち `at` が最初のオプションの場合) はモードを `implicit` に設定します。`set pm3d;` はさらにその他のオプションをデフォルトの値に設定します。

オプション `implicit` が `ON` の場合、全ての曲面の描画が追加的にデフォルトの型で行われます。例えば

```
splot 'fred.dat' with lines, 'lola.dat' with lines
```

は、両方の描画（網目の曲面）を追加的に pm3d 曲面に描きます。`set pm3d;` の後はこちらの方が慣れているでしょう。

オプション `explicit` が `ON` (または `implicit` が `OFF`) の場合、属性 `with pm3d` が指定された描画のみが pm3d 曲面として描画されます。例えば

```
splot 'fred.dat' with lines, 'lola.dat' with pm3d
```

は、「freq.dat」は線で（線のみで）描画され、「lola.dat」は pm3d 曲面として描かれます。

デフォルトのデータ/関数の描画スタイルを pm3d にしたい場合は、例えば

```
set style data pm3d
```

とします。この場合、オプション `implicit` と `explicit` は効力を持たません。

いくつかの描画においては、それらはコマンドラインで与えられた順に描画されることに注意してください。これは特に、以前の描画を上書きしてそれで一部を隠してしまう可能性がある、曲面の塗りつぶしに關して関心を持たれることです。

`splot` コマンドライン上で `with pm3d` が指定されている場合はオプション `at` も使えます。以下の描画は、異なった高さで 3 つのカラー曲面を描きます:

```
set border 4095
set pm3d at s
splot 10*x with pm3d at b, x*x-y*y, x*x+y*y with pm3d at t
```

`set palette`, `set cbrange`, `set colorbox`, `x11 pm3d`, そしてもちろんデモファイル `demo/pm3d.dem` も参考になるでしょう。

### 43.50 Palette

パレットは、pm3d で、カラー等高線や多角形、カラーヒストグラム、色勾配の背景、その他実装されている、あるいは実装されるものの塗りつぶしで使われる、色の記憶場所です。ここではそれは滑らかで“連續的な” カラーや灰色階調のパレットを意味しますが、それを単にパレットと呼ぶことにします。

カラーパレットは、多角形の色の塗りつぶしと滑らかな色のパレットをサポートした出力形式を必要とし、それは現在、pm3d で一覧表示される出力形式で使用可能です。色の値の範囲は、set cbrange と set log cb で独立に調整可能です。カラーパレット全体は colorbox 中に表示されます。

書式:

```
set palette
set palette {
    { gray | color }
    { gamma <gamma> }
    { rgbformulae <r>,<g>,<b>
        | defined { ( <gray1> <color1> {, <grayN> <colorN>}... ) }
        | file '<filename>' {datafile-modifiers}
        | functions <R>,<G>,<B>
    }
    { model { RGB | HSV | CMY | YIQ | XYZ } }
    { positive | negative }
    { nops_allcF | ps_allcF }
    { maxcolors <maxcolors> }
}
show palette
show palette palette <n> {{float | int}}
show palette gradient
show palette fit2rgbformulae
show palette rgbformulae
show palette colordnames
```

set palette は (すなわちオプションなしでは) デフォルトの値を設定します。それ以外の場合、オプションは任意の順に与えることができます。show palette は、現在のパレットの属性を表示します。

show palette gradient は、パレットの勾配 (gradient) の定義が (それが適切であれば) 表示されます。show palette rgbformulae は、定義済で利用できる、灰色値からカラーへの変換公式が表示されます。show palette colordnames は実装されている色名を表示します。

show palette palette <n> は、<n> 個の離散的な色を持つパレットの、現在のパレットの設定によって計算される RGB の値の組とパレットの表を、画面、または set output で指定されたファイルに書き出します。デフォルトの広い表は、追加のオプション float または int によって、3 列の [0..1] の実数値だけにするか [0..255] の整数値だけにするかをそれぞれ指定できます。この方法で gnuplot のカラーパレットを、Octave のような他の画像アプリケーションに渡すことができます。このようなテキスト形式の RGB の一覧表に加え、test palette コマンドにより、グラフィック的に現在のパレットの R,G,B の状態を鑑賞することもできます。

以下のオプションは、色付けの属性を決定します。

このパレットを使用する図は、gray か color になります。例えば、pm3d カラー曲面では、範囲 [min\_z,max\_z] が灰色の範囲 [0:1] に対応していて、微小曲面四辺形の 4 つの角の z 座標の平均値をこの範囲の中に対応させることで各微小部分の灰色の値 (gray) が得られます。この値は、灰色階調の色地図での灰色の値として直接使うことができますし、カラーの色地図では、その灰色の値から (R,G,B) への変換、すなわち [0:1] から ([0:1],[0:1],[0:1]) への写像が使われます。

基本的に、2 種類の異なる写像方式が利用可能です: 1 つは灰色からカラーへの解析的な公式、もう一つは離散的な対応表の補間によるものです。palette rgbformulae と palette functions が解析的な公式用で、palette defined と palette file が補間表用です。palette rgbformulae は postscript 出力のサイズを小さくすることができます。

コマンド show palette fit2rgbformulae は、現在の set palette に最も良く対応する set palette rgbformulae を見つけ出します。当然、それは rgbformulae パレット以外に対しても意味を持ちます。このコマンドは主に、パレットの rgbformulae 定義が gnuplot と同じ物を使っている外部プログラム、例えば zimg などにとって有用です (

```
http://zimg.sourceforge.net
)。
set palette gray は、灰色階調のみのパレットにし、set palette rgbformulae, set palette defined,
```

`set palette file`, `set palette functions` はカラー・パレットにします。灰色パレットから直前のカラー・パレットへ、`set palette color` で簡単に復帰できます。

`set palette gamma <gamma>` による自動的なガンマ補正は、灰色のパレット (`set palette gray`) のみに行われます。灰色階調への線形写像は `gamma=1` に相当します (`test palette` 参照)。カラー・パレットに対してはガンマ値は無視されます。

ほとんどの出力形式は、有限個の色数しかサポートしていません (例えば gif では 256 個)。デフォルトでは、パレットは、gnuplot のデフォルトの線種の色の宣言の後に残ったものが全て pm3d 用に割り当てられます。よって、出力形式が利用できる色の場所がなくなってしまった場合、`multiplot` は失敗してしまうでしょう。その場合、`set palette maxcolors <maxcolors>` で適当に小さい値を設定すべきです。このオプションは、z=定数の高低を離散個に分割し、よって等高線の疑似的な塗りつぶしを行うことにも使えます。デフォルトの値は 0 で、これは出力形式のパレットの残りの全てを割り当てる、あるいは正確な RGB への対応を使用することを意味します。

RGB 色空間が作業を行うのに常にもっとも有用な色空間であるとは限らない、という理由で、色空間は `model` を使うことで、RGB, HSV, CMY, YIQ, XYZ のいずれかに変更できます。RGB 以外の色空間では `set palette defined` の表で色名を使うと、それはおかしな色になります。全ての説明は RGB 色空間用に書いてありますが、それぞれの色空間で、例えば R は H, C, Y, X のことを意味することに注意してください (G, B も同様)。

全ての色空間で、全ての値は [0,1] に制限されています。

RGB は赤、緑、青を、CMY は水色 (Cyan)、紫 (Magenta)、黄 (Yellow) を、HSV は色相 (Hue)、彩度 (Saturation)、明度 (Value) をそれぞれ意味します。YIQ は全米商業カラーテレビ放送協会 (the U.S. Commercial Color Television Broadcasting) の使ったカラー・モデルで、RGB 記録方式を元にしていますが、白黒テレビに対する後方互換性を持っています。XYZ は CIE ('Commission Internationale de l'Eclairage'; 国際照明委員会) が定義した色モデルの 3 つの原刺激値です。色モデルのより詳しい情報については以下を参照してください:

<http://www.cs.rit.edu/~ncs/color/glossary.htm>

および

<http://cs.fit.edu/wds/classes/cse5255/cse5255/davis/index.html>

#### 43.50.1 Rgbformulae

`rgbformulae` 用には 3 つの適切な割り当て関数が選ばれる必要があります。この選択は `rgbformulae <r>,<g>,<b>` を通して行われます。使うことができる割り当て関数の一覧は `show palette rgbformulae` で見ることができます。デフォルトは 7,5,15 で、他の例としては 3,11,6, 21,23,3, 3,23,21 などがあります。3,-11,-6 のような負の値は、逆のカラーを意味します (すなわち、1-gray をその関数に代入します。下記の `positive` と `negative` のオプションも参照)。

RGB の色空間では、いくつかの良い割り当て公式があります:

7,5,15	... 伝統的 pm3d (黒-青-赤-黄)
3,11,6	... 緑-赤-紫
23,28,3	... 海 (緑-青-白); 他の組み合わせも試してみてください
21,22,23	... 温度色 (黒-赤-黄-白)
30,31,32	... 白黒のカラー表示化 (黒-青-紫-黄-白)
33,13,10	... 虹 (青-緑-黄-赤)
34,35,36	... AFM 温度色 (黒-赤-黄-白)

HSV 色空間でのフルカラー・パレット:

3,2,2	... 赤-黄-緑-水色-青-紫-赤
-------	--------------------

`rgbformulae` という名前で呼ばれていても、例の通り、それらの関数は実際には `<H>,<S>,<V>` または `<X>,<Y>,<Z>, ...` といった色の成分を決定するかもしれないということに注意してください。

図の色を反転させるには `positive` や `negative` を使ってください。

他の色体系に対する最も良い `rgbformulae` の集合は、以下のコマンドで見つけることができることを覚えておいてください。

`show palette fit2rgbformulae`

### 43.50.2 Defined

灰色から RGB への対応は palette defined を使うことで手動で設定できます: 色勾配 (gradient) は RGB の値を与えるために定義され使用されます。勾配は、 $[0,1]$  の灰色値から  $[0,1] \times [0,1] \times [0,1]$  の RGB 空間への、区分的に線形な写像です。その線形補間に使われる灰色値と RGB 値の組を指定する必要があります:

書式:

```
set palette defined { ( <gray1> <color1> {, <grayN> <colorN>}... ) }
```

$<\text{grayX}>$  は  $[0,1]$  に割り当てられるべき灰色値で、 $<\text{colorX}>$  はそれに対応する RGB 色です。カラー値は 3 種類の方法で指定することができます:

```
<color> := { <r> <g> <b> | '<color-name>' | '#rrggbb' }
```

赤、緑、青に対応する空白で区切られた 3 つの値 (それぞれ  $[0,1]$  内)、引用符でくくられた色名、または引用符でくくられた X 形式の指定方式、のいずれかです。勾配の定義では、これらの 3 種の型を自由に組み合わせることができますが、色空間として RGB でないものが選択された場合色名 "red" は少し違ったものになるでしょう。使用できる色名は show palette colormenames でその一覧を見ることができます。

$<\text{r}>$  と書いても、HSV 色空間ではそれは  $<\text{H}>$  成分を、CIE-XYZ 空間では  $<\text{X}>$  を、といったように選択されたカラー モデルに依存して意味が違うことに注意してください。

$<\text{gray}>$  の値は実数の昇順に並べる必要があります。その列の値は自動的に  $[0,1]$  に変換されます。

カッコつきの勾配の定義なしで set palette defined とした場合、RGB 色空間にし、あらかじめ設定されたフルスペクトル色勾配を使用します。勾配を表示するには show palette gradient を使用してください。

例:

灰色のパレット (役に立たないが教訓的な) を生成するには:

```
set palette model RGB
set palette defined ( 0 "black", 1 "white" )
```

青黄赤のパレット (全てが等価の) を生成するには:

```
set palette defined ( 0 "blue", 1 "yellow", 2 "red" )
set palette defined ( 0 0 0 1, 1 1 1 0, 2 1 0 0 )
set palette defined ( 0 "#0000ff", 1 "#ffff00", 2 "#ff0000" )
```

虹のようなパレットを生成するには:

```
set palette defined ( 0 "blue", 3 "green", 6 "yellow", 10 "red" )
```

HSV 色空間でのフルカラースペクトル:

```
set palette model HSV
set palette defined ( 0 0 1 1, 1 1 1 1 )
set palette defined ( 0 0 1 0, 1 0 1 1, 6 0.8333 1 1, 7 0.8333 0 1 )
```

あまり色を使わないパレットを生成するには:

```
set palette model RGB maxcolors 4
set palette defined ( 0 "blue", 1 "green", 2 "yellow", 3 "red" )
```

'交通信号' (滑らかではなく gray = 1/3, 2/3 で飛びを持つ):

```
set palette model RGB
set palette defined (0 "dark-green", 1 "green", 1 "yellow", \
                    2 "dark-yellow", 2 "red", 3 "dark-red" )
```

### 43.50.3 Functions

色の割り当ての R(gray), G(gray), B(gray) の 3 つの関数を与えるには set palette functions  $<\text{Rexpr}>$ ,  $<\text{Gexpr}>$ ,  $<\text{Bexpr}>$  を使ってください。それらの 3 つの関数の変数は、 $[0,1]$  の値を取る変数 gray であり、その値も  $[0,1]$  の中に取る必要があります。 $<\text{Rexpr}>$  は、HSV 色空間が選択されている場合は、H の値を表す式でなければいけないことに注意してください (他の式、または他の色空間でも同様です)。

例:

フルカラーパレットを生成するには:

```
set palette model HSV functions gray, 1, 1
```

黒から金色への良いパレット:

```
set palette model XYZ functions gray**0.35, gray**0.5, gray**0.8
```

ガンマ補正の白黒のパレット:

```
gamma = 2.2
color(gray) = gray**(1./gamma)
set palette model RGB functions color(gray), color(gray), color(gray)
```

#### 43.50.4 File

`set palette file` は基本的に `set palette defined (<gradient>)` と同じで、この `<gradient>` がデータファイルから読み込まれます。4 列 (gray, R,G,B) かまたは 3 列 (R,G,B) のデータが `using` データファイル修飾子によって選択される必要があります。3 列の場合、行番号が `gray` の値として使われますが、その `gray` の範囲は自動的に [0,1] にスケール変換されます。ファイルは通常のデータファイルとして読み込まれるので、全てのデータファイル修飾子が使えます。例えば HSV 色空間が選択されている場合には、R は実際には H を指すということに注意してください。

例によって、`<filename>` が '-' の場合は、データがインライン形式で引き続いて与えられ、一つの e のみの行でそれが終了することを意味します。

勾配 (gradient) を表示するには `show palette gradient` を使用してください。

例:

RGB のパレットを [0,255] の範囲で読み込む:

```
set palette file 'some-palette' using ($1/255):($2/255):($3/255)
```

等距離の虹色 (青-緑-黄-赤) パレット:

```
set palette model RGB file "--"
0 0 1
0 1 0
1 1 0
1 0 0
e
```

バイナリパレットファイルも同様にサポートされています。binary general を参照してください。例: R,G,B の double のデータの 64 個の 3 つ組をファイル palette.bin に出力し、それを読み込む:

```
set palette file "palette.bin" binary record=64 using 1:2:3
```

#### 43.50.5 Gamma correction

灰色の配色に対するガンマ補正は `set palette gamma <gamma>` で ON できます。`<gamma>` のデフォルトは 1.5 で、これは多くの出力形式に適切な値です。

カラーの配色に対しては gnuplot では自動的なガンマ補正是行いません。しかしガンマ補正是簡単に実装できます。ここに例として、ガンマの値の少し異なる赤、緑、青の成分に、明示的な関数を与えた灰色階調画像の例を紹介します。

例:

```
set palette model RGB
set palette functions gray**0.64, gray**0.67, gray**0.70
```

補間された勾配を使ってガンマ補正を行うには、適当なカラーに中間の値を指定します。

```
set palette defined ( 0 0 0 0, 1 1 1 1 )
```

の代わりに例えば以下を指定してください:

```
set palette defined ( 0 0 0 0, 0.5 .73 .73 .73, 1 1 1 1 )
```

または、線形補間が "ガンマ補正" の補間に十分良く適合するまでより良い中間の点を探してください。

### 43.50.6 Postscript

postscript ファイルのサイズを小さくする目的で、灰色の輝度値、そして全てではないいくつかの計算された RGB の輝度値がそのファイル中に書かれます。成分関数は postscript 言語で直接コード化され、pm3d の描画の直前にヘッダとしておかれます。/g や /cF の定義を参照してください。通常その定義をその中に書くことは、3 つの式のみが使われる場合に意味を持ちます。しかし、multiplot やその他の理由で postscript ファイル中のその変換関数を直接手で編集したいと思うかも知れません。これがデフォルトのオプション nops\_allcF です。オプション ps\_allcF を使うと、全ての公式の定義が postscript ファイル中に書かれます。一つのグラフ中で、異なる曲面に異なるパレットを持たせたいという目的で postscript ファイルを編集したい場合に、このオプションに関心を持つでしょう。その機能は、origin と size を固定して multiplot を使うことで実現できるでしょう。

pm3d 色地図が、格子状、あるいはほぼ規則正しいデータから postscript ファイルとして描画された場合、gnuplot と一緒に配布される awk スクリプト pm3dCompress.awk を使うことで、この postscript ファイルのサイズをほぼ 50% 位まで小さくすることができます。このファイルを文書に取り込む場合、あるいは低速のプリンタに大きなファイルを印刷する前などに、この機能に関心を持つことでしょう。使用法は以下の通りです：

```
awk -f pm3dCompress.awk thefile.ps >smallerfile.ps
```

pm3d 色地図が四角形の格子状データから postscript ファイルとして描画された場合、同時に配布される awk スクリプト pm3dConvertToImage.awk を使うことで、そのサイズを更に小さくすることができます。使用法：

```
awk -f pm3dConvertToImage.awk <thefile.ps >smallerfile.ps
```

postscript 出力の灰色階調をカラーへ、またはその逆、そして <maxcolors> の定義などを、手動で変更することができます。

### 43.50.7 Colornames

gnuplot は限定された個数の色の名前を持っています。これらは、pm3d パレットでつながれる色の範囲を定義するのに、あるいは個々の線種やラインスタイルの色を出力形式に依存しない形で定義したりするのに使えます。gnuplot の持つ色名の一覧を見るには、コマンド show palette colornames を使用してください。set palette, linestyle も参照。

## 43.51 Pointsize

コマンド set pointsize は描画で使われる点の大きさを変更します。

書式:

```
set pointsize <multiplier>
show pointsize
```

デフォルトは 1.0 倍です。画像データ出力では、大きいポイントサイズの方が見やすいでしょう。

一つの描画に対するポイントサイズは plot コマンドの上でも変更できます。詳細は plot with を参照してください。

ポイントサイズの設定は、必ずしも全ての出力形式でサポートされているわけではないことに注意してください。

## 43.52 Polar

コマンド set polar はグラフの描画方法を xy 直交座標系から極座標系に変更します。

書式:

```
set polar
unset polar
show polar
```

version 3.7 では、極座標モードにおいていくつか変更がなされ、よって、version 3.5 やそれ以前の版用のスクリプトには修正が必要になるでしょう。主な変更は、仮変数  $t$  が角度として使われるようになったことで、それにより  $x$  と  $y$  の値の範囲が独立に制御できるようになりました。その他の変更は以下の通りです：1) 目盛りはもう  $0$  軸に自動的にはつきません— set xtics axis nomirror; set ytics axis nomirror を使ってください 2) 格子が選択されてもそれは自動的に極座標には従いません— set grid polar を使ってください 3) 格子は角度に関してはラベル付けされません— 必要なら set label を使ってください

極座標モードでは、仮変数 ( $t$ ) は角度を表します。 $t$  のデフォルトの範囲は  $[0:2*pi]$  ですが、単位として度が選択されれば  $[0:360]$  となります (set angles 参照)。

コマンド unset polar は描画方法をデフォルトの xy 直交座標系に戻します。

set polar コマンドは splot ではサポートされていません。splot に対する同様の機能に関しては set mapping を参照してください。

極座標モードでは  $t$  の数式の意味は  $r=f(t)$  となり、 $t$  は回転角となります。trange は関数の定義域 (角度) を制御し、xrange と xrange はそれぞれグラフの  $x,y$  方向の範囲を制御することになります。これらの範囲と rrangle は自動的に設定されるか、または明示的に設定できます。これらすべての set range コマンドの詳細に関しては set xrange の項を参照してください。

例：

```
set polar
plot t*sin(t)
plot [-2*pi:2*pi] [-3:3] t*sin(t)
```

最初の plot はデフォルトの角度の範囲の  $0$  から  $2*pi$  を使用します。半径とグラフのサイズは自動的に縮尺されます。2 番目の plot は角度の定義域を拡張し、グラフのサイズを  $x,y$  のいずれの方向にも  $[-3:3]$  に制限します。

set size square とすると gnuplot はアスペクト比 (縦横の比) を  $1$  にするので円が (橙円でなく) 円に見えるようになります。以下も参照

極座標のデモ (polar.dem)

および

極座標データの描画 (poldat.dem)。

### 43.53 Print

コマンド set print は print コマンドの出力をファイルにリダイレクトします。

書式:

```
set print
set print "-"
set print "<filename>"
set print "<filename>" append
set print "|<shell_command>"
```

"<filename>" がない場合は出力は <STDERR> になります。"-" という <filename> は <STDOUT> を意味します。append フラグはファイルを追加 (append) モードで開くことを意味します。パイプをサポートするプラットホーム上では、<filename> が "!" で始まっていたら、<shell\_command> へのパイプが開かれます。

### 43.54 Object

このコマンドはその後の 2 次元描画すべてに表われる单一のオブジェクトを定義します。オブジェクトはいくつでも定義できます。オブジェクトの型は、現在は rectangle (長方形) のみをサポートしています。個々の長方形は、対角に向かい合う 2 つの頂点の組を定める点の組で指定します。デフォルトのスタイルの属性の組 (塗り潰し、色、境界) は、コマンド set style rectangle の設定を受け継ぎますが、個々の長方形を別々のスタイル属性で描画することももちろん可能です。

書式:

```
set object <index> rectangle
  {from <position> {to|rto} <position> |
   center <position> size <w>,<h> |
   at <position> size <w>,<h>}
  {front|back|behind} {fc|fillcolor <colorspec>} {fs <fillstyle>}
  {default} {lw|linewidth <width>}
```

長方形の位置は、対角に向かい合う 2 つの頂点（左下と右上）の位置、あるいは中心点の位置と横幅（<w>）と縦幅（<h>）で指定できます。いずれの場合も、点の位置は、軸の座標（first, second）、グラフ領域内の相対座標（graph）、またはスクリーン座標（screen）のいずれかを使用できます（coordinates 参照）。オプション at と center は同じ意味です。

front を指定すると、長方形はすべての描画要素の前（上）に描画されますが、front と指定されたラベルよりは後ろ（下）になります。back を指定するとすべての描画要素、すべてのラベルの後ろに配置されます。behind は軸や back の長方形を含むすべてのものその後ろに配置されます。これは、グラフやページ全体の背景に色をつけるのに利用できます。

長方形の塗り潰しの色は <colorspec> で指定します。fillcolor は fc と省略できます。塗り潰しパターンは <fillstyle> で指定します。詳細は colorspec と fillstyle を参照してください。キーワード default を指定すると、これらの属性は描画が実際に行われるときのデフォルトの設定を受け継ぎます。set style rectangle 参照。

例:

```
# 座標軸で囲まれた領域全体の背景を水色に
set object 1 rect from graph 0, graph 0 to graph 1, graph 1 back
set object 1 rect fc rgb "cyan" fillstyle solid 1.0

# 左下角が 0,0, 右上角が 2,3 の赤い四角を一つ置く
set object 2 rect from 0,0 to 2,3 fc lt 1

# 青い境界の空（塗り潰さない）長方形を置く
set object 3 rect from 0,0 to 2,3 fs empty border 3

# 頂点は移動しないまま、塗り潰しと色をデフォルトに変更
set object 2 rect default
```

### 43.55 Rmargin

コマンド set rmargin は右の余白のサイズをセットします。詳細は set margin を参照してください。

### 43.56 Rrange

コマンド set rrange は極座標モードのグラフの半径方向の範囲を設定します。詳細は set xrange を参照してください。

### 43.57 Samples

関数、またはデータの補間にに関するサンプリング数はコマンド set samples で変更できます。

書式:

```
set samples <samples_1> {,<samples_2>}
show samples
```

デフォルトではサンプル数は 100 点と設定されています。この値を増やすとより正確な描画が出来ますが遅くなります。このパラメータはデータファイルの描画には何の影響も与えませんが、補間/近似のオプションが使われている場合はその限りではありません。2 次元描画については plot smooth を、3 次元描画に関しては set dgrid3d を参照してください。

2 次元のグラフ描画が行なわれるときは `<samples_1>` の値のみが関係します。

隠線処理なしで曲面描画が行なわれるときは、`samples` の値は孤立線毎に評価されるサンプル数の指定になります。各 v-孤立線は `<samples_1>` 個のサンプル点を持ち、u-孤立線は `<samples_2>` 個のサンプル数を持ちます。`<samples_1>` のみ指定すると、`<samples_2>` の値は `<samples_1>` と同じ値に設定されます。set `isosamples` の項も参照してください。

### 43.58 Size

書式:

```
set size {{no}square | ratio <r> | noratio} {<xscale>,<yscale>}
show size
```

`<xscale>` と `<yscale>` は描画全体の拡大の倍率で、描画全体とはグラフとラベルと余白の部分を含みます。

重要な注意:

gnuplot の以前の版では、「set size」の値を、出力する描画領域（キャンバス）のサイズを制御するのにも使っていた出力形式もありましたが、すべての出力形式がそうだったわけではありませんでした。version 4.2 では、ほとんどの出力形式が以下に述べるルールに従うようになりましたが、png/gif/jpeg 出力形式については、「backwards compatibility」を参照してください。

`set term <terminal_type> size <XX>, <YY>` は、出力ファイルのサイズ、または "キャンバス" のサイズを制御します。サイズパラメータの有効な値については、個々の出力形式のヘルプを参照してください。デフォルトでは、グラフはそのキャンバス全体に描画されます。

`set size <XX>, <YY>` は、描画自体をキャンバスのサイズに対して相対的に伸縮させます。1 より小さい伸縮値を指定すると、グラフはキャンバス全体を埋めず、1 より大きい伸縮値を指定すると、グラフの一部分のみがキャンバス全体に合うように描画されます。1 より大きい伸縮値を指定すると、ある出力形式では問題が起こるかもしれないことに注意してください。set term size 参照。

`ratio` は、指定した `<xscale>`, `<yscale>` の描画範囲内で、グラフのアスペクト比（縦横比）を `<r>` にします（`<r>` は x 方向の長さに対する y 方向の長さの比）。

`<r>` の値を負にするとその意味は違って来ます。`<r>=-1` のとき、x 軸、y 軸の双方の単位（つまり 1）の目盛の長さが同一になるよう設定されます。（例えば地理データ表示に向く）。`<r>= -2` のとき、y 軸の単位目盛の長さは x 軸の単位目盛の長さの 2 倍に設定されます。`<r>` が負の値に関して以下同様です。

gnuplot が指定されたアスペクト比のグラフをちゃんと書けるかは選択される出力形式に依存します。グラフの領域は出力の指定された部分にちゃんと收まり、アスペクト比が `<r>` であるような最大の長方形となります（もちろん適当な余白も残しますが）。

`square` は `ratio 1` と同じ意味です。

`noratio` と `nosquare` はいずれもグラフをその出力形式（terminal）でのデフォルトのアスペクト比に戻しますが、`<xscale>` と `<yscale>` はそのデフォルトの値（1.0）には戻しません。

`ratio` と `square` は 3 次元描画では意味を持ちません。

例:

グラフが有効な描画領域全体を埋めるような大きさに設定します:

```
set size 1,1
```

グラフを通常の半分の大きさで正方形にします:

```
set size square 0.5,0.5
```

グラフの高さを横幅の 2 倍にします:

```
set size ratio 2
```

以下も参照

翼のデモ。

### 43.59 Style

デフォルトの描画スタイルは、`set style data` と `set style function` で設定できます。関数やデータのデフォルトの描画スタイルを個々に変更する方法については `plot with` を参照してください。スタイルの一覧全体は `plotting styles` を参照してください。

書式:

```
set style function <style>
set style data <style>
show style function
show style data
```

指定できる描画要素のデフォルトスタイルも設定できます。

```
set style arrow <n> <arrowstyle>
set style fill <fillstyle>
set style histogram <histogram style options>
set style line <n> <linestyle>
```

#### 43.59.1 Set style arrow

各出力形式は矢や点の形のデフォルトの集合を持っていて、それはコマンド `test` で参照できます。`set style arrow` は矢の形、幅、点の形、サイズを定義し、それらを後で使うときにいちいち同じ情報を繰り返して指定しなくてもインデックスで参照できるようにします。

書式:

```
set style arrow <index> default
set style arrow <index> {nohead | head | heads}
    {size <length>,<angle>{,<backangle>}}
    {filled | empty |nofilled}
    {front | back}
    { {linestyle | ls <line_style>}
      | {linetype | lt <line_type>}
      {linewidth | lw <line_width>} }
```

```
unset style arrow
show style arrow
```

`<index>` は整数で、それで矢のスタイル (`arrowstyle`) を特定します。

`default` を指定すると、全ての `arrow` スタイルパラメータはそのデフォルトの値になります。

`<index>` の `arrowstyle` が既に存在する場合、他の全ては保存されたまま、与えられたパラメータのみが変更されます。`<index>` が存在しなければ、指定されなかった値はデフォルトの値になります。

`nohead` を指定することで、矢先のない矢 – すなわち線分を書くこともできます。これは描画の上に線分を描く別な方法を与えます。デフォルトでは 1 つの矢先がついています。`heads` の指定で線分の両端に矢先が描かれます。

矢先の大きさは `size <length>,<angle>` または `size <length>,<angle>,<backangle>` で制御できます。`<length>` は矢先の各枝の長さで、`<angle>` は矢先の枝と矢軸がなす角度 (単位は度) です。`<length>` の単位は `x` 軸と同じですが、それは `<length>` の前に `first`, `second`, `graph`, `screen`, `character` をつけることで変更できます。詳細は `coordinates` を参照してください。`<backangle>` は、`filled` かまたは `empty` がともに使われた場合のみ効力を持ち、その場合、`<backangle>` は矢先の後ろの部分の矢軸との切り角 (`<angle>` と同じ方向; 単位は度) になります。出力形式 `fig` は、制限された切り角関数を持っています、それは 3 つの異なる形をサポートしていて、それは 2 つの閾値で決定します: 70 度未満の場合、矢先はへこんだ切り角を持ち、110 度を超える場合、後ろの部分に尖った角を持ち、その間の角では、矢先の後ろは直線になります。

`filled` を指定すると、塗りつぶされた矢先を作ります (`heads` が使われている場合)。塗りつぶしは、多角形の塗りつぶしが行えるような出力形式でサポートされていて、そのリストについては `pm3d` を参照してください。他の出力形式では矢先は閉じられますが塗りつぶされません。それと同じ効果 (閉じらるが

塗られない) は、empty を指定しても得られます。また、metafont, metapost, latex, tgif のように、矢をそれら自身の独自のルーチンで描くような出力形式では、矢先の塗りつぶしや矢先の枠線描きはもちろんサポートされません。

線種はユーザの定義したラインスタイルのリストから選ぶこともできますし (set style line 参照)、用意されている <line\_type> の値 (デフォルトのラインスタイルのリストの番号) そして <linewidth> (デフォルトの幅の倍数) を使ってここで定義することもできます。

しかし、ユーザー定義済のラインスタイルが選択された場合、その属性 (線種、幅) は、単に他の set style arrow コマンドで適当な番号や lt, lw などを指定しても、変更はできないことに注意して下さい。

front を指定すると、矢はグラフのデータの上に描かれます。back が指定された場合 (デフォルト) は矢はグラフのデータの下に描かれます。front を使えば、密集したデータで矢が見えなくなることを防ぐことができます。

例:

矢先がなく、倍の幅が矢を描くには:

```
set style arrow 1 nohead lw 2
set arrow arrowstyle 1
```

その他の例については set arrow を参照してください。

### 43.59.2 Set style data

コマンド set style data はデータ描画に対するデフォルトの描画スタイルを変更します。

書式:

```
set style data <plotting-style>
show style data
```

選択項目については plotting styles を参照してください。項目を指定しなかった場合、その一覧が表示されます。show style data は現在のデフォルトのデータ描画スタイルを表示します。

### 43.59.3 Set style fill

コマンド set style fill は boxes, histograms, candlesticks, filledcurves のスタイルの設定に使われます。

書式:

```
set style fill {empty | solid {<density>} | pattern {<n>}}
               {border {<linetype>} | noborder}
```

デフォルトの塗りつぶしスタイル (fillstyle) は empty です。

オプション solid は、出力形式がサポートしている場合、その色でのベタ塗りを行います。パラメータ <density> は塗りつぶし色の強さを表していて <density> が 0.0 なら箱は空、<density> が 1.0 なら箱はその内部は現在の線種と完全に同じ色で塗られます。出力形式によっては、この強さを連続的に変化させられるものもありますが、その他のものは、部分的な塗りつぶしの幾つかのレベルを実装しているに過ぎません。パラメータ <density> が与えられなかった場合はデフォルトの 1 になります。

オプション pattern は、出力ドライバによって与えられるパターンでの塗りつぶしを行います。利用できる塗りつぶしパターンの種類と数は出力ドライバに依存します。塗りつぶしの boxes スタイルで複数のデータ集合を描画する場合そのパターンは、複数の曲線の描画における線種の周期と同様、有効なパターンを、パターン <n> から始めて周期的に利用します。

オプション empty は、箱を塗りつぶしませんが、これがデフォルトです。これは、オプション solid でパラメータ <density> を 0 に設定した場合と同じです。

デフォルトの border は、現在の線の種類の実線で箱の境界を描きます。border <lt> はその境界が線の種類 (linetype) <lt> で描かれるようにします。noborder は境界の線が描かれないようにします。

### 43.59.4 Set style function

コマンド set style function は関数描画に対するデフォルトの描画スタイルを変更します。

書式:

```
set style function <plotting-style>
show style function
```

選択項目については **plotting styles** を参照してください。項目を指定しなかった場合、その一覧が表示されます。**show style function** は現在のデフォルトの関数描画スタイルを表示します。

#### 43.59.5 Set style increment

書式:

```
set style increment {default|userstyles}
show style increment
```

デフォルトでは、同じグラフ上の次の描画は、現在の出力形式でデフォルトで定義されている線種の次のもので行われます。しかし、**set style increment user** を選択すると、デフォルトの線種ではなく、ユーザ定義ラインスタイル番号のものを使用させることができます。

例:

```
set style line 1 lw 2 lc rgb "gold"
set style line 2 lw 2 lc rgb "purple"
set style line 4 lw 1 lc rgb "sea-green"
set style increment user

plot f1(x), f2(x), f3(x), f4(x)
```

これは、関数  $f_1, f_2, f_4$  は新たにユーザ定義されたラインスタイルで描画されます。ユーザ定義ラインスタイルが見つからない場合は、代わりにそれに対応するデフォルトの線種が利用されます。例えば、上の例では、 $f_3(x)$  はデフォルトの線種 3 で描画されます。

#### 43.59.6 Set style line

出力装置にはおのののデフォルトの線種と点種の集合があり、それらはコマンド **test** で見ることができます。**set style line** は線種と線幅、点種と点の大きさを、個々の呼び出しで、それらの情報を全部指定する代わりに、単なる番号で参照できるようにあらかじめ定義するものです。

書式:

```
set style line <index> default
set style line <index> {{linetype | lt} <line_type> | <colorspec>}
{{linecolor | lc} <colorspec>}
{{linewidth | lw} <line_width>}
{{pointtype | pt} <point_type>}
{{pointscale | ps} <point_size>}
{palette}

unset style line
show style line
```

**default** を指定すると、全ての **line** スタイルパラメータはそのデフォルトの値になります。

**<index>** の **linestyle** が既に存在する場合、他の全ては保存されたまま、与えられたパラメータのみが変更されます。**<index>** が存在しなければ、指定されなかった値はデフォルトの値になります。

線種と点種現在の出力装置が持つデフォルトの種類から選ばれます。線幅と点の大きさはデフォルトの幅、大きさに対する乗数です（しかし、ここでの **<point\_size>** は、**set pointscale** で与えられる乗数には影響を受けないことに注意してください）。

線種と線幅のデフォルトの値はそのラインスタイル番号 (**index**) です。幅と大きさのデフォルトの大きさはどちらも 1 です。

このようにつくられるラインスタイルは、デフォルトの型（線種、点種）を別なものに置き換えることはしないので、ラインスタイル、デフォルトの型、どちらも使えます。もし、定義したスタイルをデフォルトの線種の代わりに使いたい場合は、**set style increment** を参照してください。

全ての出力装置が `linewidth` や `pointsize` をサポートしているわけではありません。もしサポートされていない場合はそれらのオプションは無視されます。

出力形式に依存しない色を `linecolor <colors>` か `linetype <colors>` (省略形は `lc`, `lt`) のいずれかを使って割り当てることができます。この場合、色は RGB の 3 つ組で与えるか、gnuplot の持つパレットの色名、現在のパレットに対する小数指定、または `cbrange` への現在のパレットの対応に対する定数値、のいずれかで与えます。`colors`, `colorspec`, `set palette`, `colormaps`, `cbrange` も参照してください。

`set style line <n> linetype <lt>` は、出力形式に依存した点線/破線のパターンと色の両方をセットします。`set style line <n> linecolor <colors>` や `set style line <n> linetype <colors>` は、現在の点線/破線のパターンを変更せずに新しい線色を設定します。

3 次元モード (`splot` コマンド) では、"linetype palette z" の省略形として特別にキーワード `palette` を使うことも許されています。その色の値は、`splot` の `z` 座標 (高さ) に対応し、曲線、あるいは曲面に沿って滑らかに変化します。

例: 以下では、番号 1, 2, 3 に対するデフォルトの線種をそれぞれ赤、緑、青とし、デフォルトの点の形をそれぞれ正方形、十字、三角形であるとします。このとき以下のコマンド

```
set style line 1 lt 2 lw 2 pt 3 ps 0.5
```

は、新しいラインスタイルとして、緑でデフォルトの 2 倍の幅の線、および三角形で半分の幅の点を定義します。また、以下のコマンド

```
set style function lines
plot f(x) lt 3, g(x) ls 1
```

は、`f(x)` はデフォルトの青線で、`g(x)` はユーザの定義した緑の線で描画します。同様に、コマンド

```
set style function linespoints
plot p(x) lt 1 pt 3, q(x) ls 1
```

は、`p(x)` を赤い線で結ばれたデフォルトの三角形で、`q(x)` は緑の線で結ばれた小さい三角形で描画します。

```
splot sin(sqrt(x*x+y*y))/sqrt(x*x+y*y) w 1 pal
```

は、`palette` に従って滑らかな色を使って曲面を描画します。これはそれをサポートした出力形式でしかちゃんと動作しないことに注意してください。`set palette`, `set pm3d` も参照してください。

```
set style line 10 linetype 1 linecolor rgb "cyan"
```

は、RGB カラーをサポートするすべての出力形式で、ラインスタイル 10 に実線の水色を割り当てます。

#### 43.59.7 Plotting styles

コマンド `set style data` と `set style function` は、その後の `plot` や `splot` コマンドのデフォルトの描画スタイルを変更します。

線 (line) や点 (point) スタイルで使われる種類 (線で言えば実線、点線、色など、点で言えば丸、四角、バツなど) は、`plot` や `splot` コマンドで明示的に指定されたものになるか、または使用している出力形式が利用可能なものから順番に選ばれたものになります。どんなものが使えるかはコマンド `test` でチェックしてください。

2 列より多くの情報を必要とするスタイル (例えば `errorbars` や `errorlines`) はいずれも `splot` や 関数の `plot` には使うことはできません。`boxes`, `filledcurves`, およびいくつかある `steps` スタイルは `splot` で使うことはできません。適切でないスタイルが指定された場合、それは `points` に変更されます。

上の警告は、`plot with labels` には適用されず、この場合 3 列目のデータは座標の情報ではなくデータの出典を指します。`set style labels` を参照。

2 列より多い 2 次元データに対しては、gnuplot は `errobars` か `errorlines` スタイルを選択しようとしますが、`plot` コマンドのオプション `using` を使えば、使いたいスタイルにあわせて、適切な列のデータを使うように設定できます (この議論の中で、"列 (column)" は、データファイルの各行の列と、`using` リスト中の項目の 2 つの意味で使われています)。

3列のデータは、xerrorbars, yerrorbars (または errorbars), xerrorlines, yerrorlines (または errorlines), boxes, boxerrorbars のみが扱えます。他の描画スタイルが指定された場合、それは yerrorbars に変更されます。スタイル boxerrorbars の箱の横幅は自動的に計算されます。

4列のデータは、xerrorbars, yerrorbars (または errorbars), xyerrorbars, xerrorlines, yerrorlines (または errorlines), xyerrorlines, boxyerrorbars, boxerrorbars のみが扱えます。不適切な描画スタイルは yerrorbars に変更されます。

5列のデータは、boxerrorbars, financebars, candlesticks スタイルのみが扱えます。不適切な描画スタイルは、描画の前に boxerrorbars に変更されます。

6列のデータは、xyerrorbars, xyerrorlines, boxyerrorbars スタイルのみが扱えます。不適切な描画スタイルは、描画の前に xyerrorbars に変更されます。

線を含む/含まない誤差線表示 (error bar) についてのより詳しい情報は、plot errorlines と plot errorbars を参照してください。

#### 43.59.8 Set style rectangle

コマンド set object rectangle で定義された長方形には別々のスタイルを設定できます。しかし、特別なスタイルの指定をしなければ、その長方形はコマンド set style rectangle によるデフォルトを受け継ぎます。

書式:

```
set style rectangle {front|back} {fillcolor <colorspec>} {fs <fillstyle>}
                    {lw|linewidth <lw>}
```

colorspec と fillstyle を参照してください。fillcolor は fc と省略できます。

例:

```
set style rectangle back fc rgb "white" fs solid 1.0 border -1
set style rectangle fc linestyle 3 fs pattern 2 noborder
```

デフォルトの設定は、背景色での塗り潰しで、境界は黒になっています。

**43.59.8.1 Boxerrorbars** 描画スタイル boxerrorbars は 2 次元のデータ描画でのみ利用可能です。これは boxes と yerrorbars スタイルの組合せです。y の誤差が "ydelta" の形式で与えられて、箱の横幅があらかじめ -2.0 に設定されて (set boxwidth -2.0) なければ、箱の横幅は 4 列目の値で与えられます。y の誤差が "ylow yhigh" の形式で与えられる場合は箱の横幅は 5 列目の値で与えられます。特別な場合として、"ylow yhigh" の誤差形式の 4 列のデータに対する boxwidth = -2.0 という設定があります。この場合箱の横幅は、隣接する箱にくっつくように自動的に計算されます。3 列のデータの場合も、横幅は自動的に計算されます。

箱の高さは、yerrorbars スタイル同様に y の誤差の値から決定されます – y-ydelta から y+ydelta まで、あるいは ylow から yhigh まで、これらは何列のデータが与えられているかによって決まります。以下も参照

errorbar デモ。

**43.59.8.2 Boxes** boxes スタイルは 2 次元描画でのみ利用可能です。これは与えられた x 座標を中心とし、x 軸から (グラフの境界から、ではありません) 与えられた y 座標までの箱を書きます。箱の幅は 3 つのうち一つの方法で決定されます。それがデータの描画で、かつファイルが 3 列目のデータを持っている場合はそれが箱の幅にセットされます。そうでなくて set boxwidth コマンドで箱の幅がセットされていた場合それが使われます。そのどちらでもない場合、箱の幅は、隣接する箱がくっつくように自動的に計算されます。

箱の中身は現在の塗りつぶしスタイル (fillstyle) に従って塗りつぶされます。詳細は set style fill を参照してください。新しい塗りつぶしスタイルを plot コマンド上で指定することもできます。

塗りつぶしスタイルが empty の場合は、箱は背景色で塗りつぶされます。

塗りつぶしスタイルが solid の場合は、箱は現在の描画色でベタ塗りされます。これには追加オプション <density> があり、それは塗りつぶし密度を意味し、0 は背景色、1 は描画色そのものになります。

塗りつぶしスタイルが `pattern` の場合は、箱は現在の描画色であるパターンで塗りつぶされますが、出力ドライバがサポートしている必要があります。

例:

データファイルを塗りつぶした箱で描画し、箱同士を少し垂直方向にスペースを空ける (棒グラフ):

```
set boxwidth 0.9 relative
set style fill solid 1.0
plot 'file.dat' with boxes
```

パターンでの塗りつぶしスタイルの箱で `sin` と `cos` のグラフを描画:

```
set style fill pattern
plot sin(x) with boxes, cos(x) with boxes
```

`sin` はパターン 0 で、`cos` はパターン 1 で描画されます。追加される描画は出力ドライバがサポートするパターンを循環的に使用します。

それぞれのデータ集合で明示的に塗りつぶしスタイルを指定:

```
plot 'file1' with boxes fs solid 0.25,
      'file2' with boxes fs solid 0.50,
      'file3' with boxes fs solid 0.75,
      'file4' with boxes fill pattern 1,
      'file5' with boxes fill empty
```

現在、以下の出力ドライバのみが `empty` 以外の塗りつぶしスタイルをサポートしています: `x11`, `windows`, `pm`, `wxt`, `postscript`, `fig`, `pbm`, `png`, `gif`, `hpdj`, `hppj`, `hpljii`, `hp500c`, `jpeg`, `nec_cp6`, `epson_180dpi`, `epson_60dpi`, `epson_lx800`, `okidata`, `starc`, `tandy_60dpi`。BeOS ドライバ (`be`) はまだテストされていません。

**43.59.8.3 Boxxyerrorbars** `boxxyerrorbars` スタイルは 2 次元のデータ描画でのみ利用可能です。これは `boxes` と `xyerrorbars` スタイルの組合せです。

箱の幅と高さは `xyerrorbars` スタイル同様 `x`, `y` の誤差から決定されます – `xlow` から `xhigh` までと `ylow` から `yhigh` まで、または `x-xdelta` から `x+xdelta` までと `y-ydelta` から `y+ydelta` まで。これらは何列のデータが与えられているかによって決まります。

箱の塗りつぶしが機能する場合、箱の内部は現在の塗りつぶしスタイル (`fillstyle`) に従って塗られます。詳細は `set style fill` と `boxes` を参照してください。`plot` コマンド上で新しい塗りつぶしスタイルを指定することもできます。

**43.59.8.4 Candlesticks** `candlesticks` スタイルは、金融データの 2 次元のデータ描画、および統計データのひげ付きの棒グラフを生成するのに使えます。5 列のデータが必要で、順に `x` 座標 (多分日付)、開始値、最安値、最高値、終値、となります。記号は、水平方向には `x` を中心とし、垂直方向には開始値と終値を境界とする長方形が使われます。そして、その `x` 座標のところに、長方形のてっぺんから最高値までと、長方形の底から最安値までの垂直線が引かれますが、この垂直線は最高値と最安値が入れ替わっても変更されません。

長方形の幅はコマンド `set boxwidth` で制御できますが、以前の `gnuplot` への後方互換性として、`boxwidth` パラメータが設定されていない場合は `set bars <width>` で制御されるようになっています。

デフォルトでは、鉛直線分のてっぺんと底には垂直に交わる水平線は引かれません。それを引きたい場合、例えば典型的な例は箱ひげ図 (box-and-whisker plot) での使用ですが、描画コマンドにキーワード `whiskerbars` を追加してください。デフォルトでは、水平線は箱 (candlestick) の水平幅一杯に引かれますが、それは全体の幅に対する割合を指定することで変更できます。

デフォルトでは長方形は `(開始値)>(終値)` の場合は空で、`(開始値)<(終値)` の場合は 3 本の垂直線が入ります。箱の塗りつぶしが有効である場合、その長方形は `set style fill <fillstyle>` にしたがって塗りつぶされます。`set bars, financebars` を参照してください。また、以下も参照してください。

#### 金融データのデモ

注意: 中央値を表すための記号などを追加したい場合、以下の例のように、ひげ付きの棒グラフに他の描画コマンドを追加する必要があります:

```
# データ列: X, 最小値, '1/4 位の値', 中央値, '3/4 位の値', 最大値,
set bars 4.0
set style fill empty
plot 'stat.dat' using 1:3:2:6:5 with candlesticks title 'Quartiles', \
     ''           using 1:4:4:4:4 with candlesticks lt -1 notitle
# ひげの上に水平線を伴う描画で、水平線の幅を全体幅の 50% にする
plot 'stat.dat' using 1:3:2:6:5 with candlesticks whiskerbars 0.5
set boxwidth と set bars, set style fill を参照してください。
```

43.59.8.5 Dots dots スタイル filledcurves は各点に小さなドットを描画します。これはたくさんの点からなる散布図の描画に便利でしょう。出力形式によっては (post, pdf など) ドットの大きさは linewidth を変更することで制御できることもあります。

43.59.8.6 Filledcurves スタイル filledcurves は 2 次元描画でのみ利用可能です。これは 3 種類の異なる指定が可能です。最初の 2 種類は関数描画、あるいは 2 列の入力データ用のもので、後で紹介するようにオプションで更なる指定ができます。最初のものは closed で、これは曲線それ自身を閉多角形と見なし、これがデフォルトです。

2 種類目は指定された軸、あるいは水平線、垂直線、与えられた点などと、曲線との間に作られる領域を塗りつぶします。

3 種類目は 3 列の入力データを必要とし、それらは x 座標と、それに対する 2 つの y 座標からなり、それらは同じ x 座標の集合に対する 2 つの曲線の y 座標に対応します。そしてその 2 つの曲線の間の領域が塗りつぶされます。

書式:

```
set style [data | function] filledcurves [option]
plot ... with filledcurves [option]
```

ここで、オプションは以下の通りです:

```
[closed | {above | below} {x1 | x2 | y1 | y2} [= <a>] | xy=<x>,<y>]
```

最初の 2 種類の塗りつぶしでは以下のようなオプションで更なる指定が可能です。

```
filledcurves closed    ... 丁度閉曲線で囲まれる領域
filledcurves x1        ... x1 軸
filledcurves x2        ... x2 軸 (y1, y2 軸も同様)
filledcurves y1=0      ... (y1 軸での) 直線 y=0 i.e. x1 軸と平行
filledcurves y2=42      ... (y2 軸での) 直線 y=42 i.e. x2 軸と平行
filledcurves xy=10,20   ... x1,y1 軸での点 10,20 (扇型のような形状)
```

入力された 2 つの曲線の間の領域の塗りつぶしの例:

曲線間の塗りつぶしデモ。

```
plot 'data' using 1:2:3 with filledcurves
```

above と below オプションは

```
... filledcurves above {x1|x2|y1|y2}=<val>
```

および

```
... using 1:2:3 with filledcurves below
```

の形のコマンドに適用可能です。どちらの場合でも、これらのオプションは塗りつぶし領域を、境界線、または境界曲線の片側に制限します。

注意: この描画モードは全ての出力形式でサポートされるとは限りません。

データファイルから描かれた曲線の塗りつぶしを拡大すると、何もなくなったり正しくない領域になることがあります、それは gnuplot が、領域ではなく点や線をクリッピングしているからです。

<a>, <x>, <y> が描画領域の外にある場合、それらはグラフの境界へ移動されます。よって、オプション xy=<x>,<y> を指定した場合の実際の塗りつぶし領域は、 xrange や yrange に依存します。

**43.59.8.7 Financebars** financebars スタイルは金融データの 2 次元のデータ描画でのみ利用可能です。5 列のデータが必要で、順に x 座標(多分日付)、開始値、最安値、最高値、終値、となります。記号は、水平方向にはその x 座標に置かれ、垂直方向には最高値と最安値を端とする線分が使われます。そして、その線分に水平左側の刻みが開始値の所に、水平右側の刻みが終り値の所につきます。その刻みの長さは set bars で変更できます。記号は最高値と最安値が入れ替わっても変わりません。set bars と candlesticks、および以下も参照してください。

金融データデモ。

**43.59.8.8 Fsteps** fsteps スタイルは 2 次元描画でのみ利用可能です。これは 2 本の線分で隣り合う点をつなぎます: 1 本目は  $(x_1, y_1)$  から  $(x_1, y_2)$  まで、2 本目は  $(x_1, y_2)$  から  $(x_2, y_2)$  まで。以下も参照 steps デモ。

**43.59.8.9 Histeps** histeps スタイルは 2 次元描画でのみ利用可能です。これはヒストグラムの描画での利用を意図しています。 $y$  の値は、 $x$  の値を中心に置くと考え、 $x_1$  での点は  $((x_0+x_1)/2, y_1)$  から  $((x_1+x_2)/2, y_1)$  までの水平線として表現されます。端の点では、その線はその x 座標が中心になるように延長されます。隣り合う点同士の水平線の端は、その両者の平均値のところでの鉛直線、すなわち  $((x_1+x_2)/2, y_1)$  から  $((x_1+x_2)/2, y_2)$  の線分で結ばれます。

autoscale が有効である場合、 $x$  の範囲は、その延長された水平線の範囲ではなく、データ点の範囲が選択されます。よって、端の点に関してはその水平線は半分しか描かれないことになります。以下も参照 steps デモ。

histeps は単なる描画スタイルにすぎず、gnuplot には、ヒストグラムの箱を生成する能力や、データ集合から母集団を決定する能力などはありません。

**43.59.8.10 Histograms** スタイル histograms は 2 次元描画でのみ有効です。これは、データの各列の並びから平行な棒グラフを作ります。plot コマンドの各要素は、それに関する目盛りの値や説明のタイトルが付属するかも知れませんが、単一の入力データを指定する必要があります(例えば入力ファイルの 1 つの列)。現在は、4 種類のヒストグラム形式のスタイルがサポートされています。

```
set style histogram clustered {gap <gapsize>}
set style histogram errorbars {gap <gapsize>} {<linewidth>}
set style histogram rowstacked
set style histogram columnstacked
```

デフォルトのスタイルは set style histogram clustered gap 2 に対応しています。このスタイルでは、並列に指定されたデータの値の集合は、選択されたデータ列のそのそれぞれの序列(行番号)に対応する x 座標の場所に、各々箱のグループとして固められて置かれます。よって、 $<n>$  個のデータ列を並列に指定した場合、最初の固まりは  $x=1$  を中心とする  $<n>$  個の箱の固まりからなり、その各々の高さは、その  $<n>$  データ列各々の最初(1 行目)の値が取られます。その後に少し空白(gap)が空けられ、次に各データ列の次(2 行目)の値に対応する箱の固まりが  $x=2$  を中心として置かれます。以下同様です。デフォルトの空白(gap)幅の 2 は、箱の固まり同士の間の空白が、箱 2 つの幅に等しいことを意味します。同じ列に対する箱は全て同じ色または同じパターンで与えられます(set style fill 参照)。

箱の固まりそれぞれは、データファイルの 1 つの行から得られます。そのような入力ファイルの各行の最初の項目が見出し(ラベル)であることは良くあることです。その列にある見出し(ラベル)は、using に xticlabels オプションをつけることで、それに応する箱の固まりの真下の x 軸に沿ったところに置くことができます。

errorbars スタイルは、各エントリに対して 2 列の入力を必要とする以外は clustered スタイルにとても良く似ています。最初の列は clustered スタイルの場合と全く同様に箱の高さ( $y$  の値)として使われます。2 つ目の列は誤差の大きさとして扱われ、これは箱の天井に縦の誤差線を生成するのに使われます。誤差線の見た目は、現在の set basrs の値と  $<linewidth>$  オプション指定で制御できます。

積み上げ型のヒストグラムも 2 つの形式がサポートされています。それらはコマンド set style histogram {rowstacked|columnstacked} で選択できます。これらのスタイルにおいて、選択された列のデータの値は積み上げられた箱として集められます。デフォルトの積み上げモードは rowstacked です。

スタイル rowstacked は、まず最初に選択された列の各行の値を x 軸のそれぞれの位置に配置します: 1 行目の値は  $x=1$  の箱、2 行目のは  $x=2$ 、以下同様となります。2 番目以降に選択された列に対応する箱

は、それらの上に積み重ねられて行きます。そして結果として、 $x=1$  にできる箱の積み重ねは、各列の最初の値（1 行目の値）からなり、 $x=2$  の箱の積み重ねは各列の 2 行目の値、などになります。同じ列に対する箱は全て同じ色または同じパターンで与えられます（`set style fill` 参照）。

スタイル `columnstacked` も同様ですが、こちらは各箱の積み上げは（各行のデータからではなく）各列のデータからなります。最初に指定された列の各行のデータが  $x=1$  の箱の積み上げを生成し、2 番目に指定した列の各行のデータが  $x=2$  の箱の積み上げ、などになります。このスタイルでは、各箱の色は、各データ項目の（列番号ではなく）行番号から決定されます。

箱の幅はコマンド `set boxwidth` で変更できます。箱の塗りつぶしスタイルはコマンド `set style fill` で設定できます。

`histograms` は  $x$  軸は常に  $x1$  軸を使いますが、 $y$  軸に関しては  $y1$  軸か  $y2$  軸かを選択できます。`plot` 命令が、`histograms` と他のスタイルの描画の両方を含む場合、`histogram` でない方は、 $x1$  軸を使うか  $x2$  軸を使うかを選択できます。

例：

複数の列を持つデータファイルを箱の固まり型（clustered）のヒストグラム（デフォルトスタイル）で描画：

```
set boxwidth 0.9 relative
set style data histograms
set style fill solid 1.0 border -1
plot 'file.dat' using 2, '' using 4, '' using 6
```

これは、 $x$  軸上の各整数値を中心とするそれぞれ 3 つの箱（鉛直な棒）毎の固まりによる描画を生成します。入力ファイルの最初の列にラベルが含まれているならそれを、以下の少し変更したコマンドで  $x$  軸に沿って配置できます。

```
plot 'file.dat' using 2, '' using 4, '' using 6:xticlabels(1)
```

ファイルが、各データの測定値と誤差評価の両方を含んでいる場合、描画に誤差線を追加することができます。以下のコマンドは誤差線を  $(y-<\text{error}>)$  から  $(y+<\text{error}>)$  に引き、その頭に箱と同じ幅の水平線をつけます。誤差線と誤差線の端の線は、黒で線幅 2 で描画されます。

```
set bars fullwidth
set style histogram errorbars gap 2 lt -1 lw 2
plot 'file.dat' using 2:3, '' using 4:5, '' using 6:7:xticlabels(1)
```

同じデータを行毎の積み上げ型（rowstacked）のヒストグラムで描画：

```
set style histogram rows
plot 'file.dat' using 2, '' using 4, '' using 6:xtic(1)
```

これは、データファイルの 2,4,6 列目の値を高さに持つ 3 つの部分の積み上げによる鉛直な棒からなる描画を生成します。

最後に以下のコマンド

```
set style histogram columnstacked
plot 'file.dat' using 2, '' using 4, '' using 6
```

は、3 つの鉛直な積み重ねの棒を生成します。 $x=1$  にある棒は、データファイルの 2 列目の各行の値に対応する箱からなります。 $x=2$  にある棒は、データファイルの 4 列目の各行の値に対応する箱から、 $x=3$  には 6 列目の各行の値に対する箱からなります。これは、gnuplot の通常の入力の縦、横の解釈を入れ換えることになりますので、key タイトルや  $x$  軸の目盛りの見出しの指定も変更する必要があります。

```
set style histogram columnstacked
plot '' u 5:key(1)          # 1 列目を key タイトルに使用
plot '' u 5 title columnhead #
```

**43.59.8.10.1 Newhistogram** 一回の描画に 1つより多くのヒストグラムの組を作ることもできます。この場合、描画コマンド newhistogram { "title" } { <linestyle> } を使うことで、それらを強制的に分離し、またそれぞれのラベルを分離することができます。例:

```
set style histogram cluster
plot newhistogram "Set A", 'a' using 1, '' using 2, '' using 3, \
    newhistogram "Set B", 'b' using 1, '' using 2, '' using 3
```

ラベル "Set A" と "Set B" は、それぞれのヒストグラムの組の下、x 軸の全てのラベルの下の位置に現われます。

コマンド newhistogram は、ヒストグラムの色付けを強制的に指定した色 (linetype) で始めるのにも使えます。デフォルトでは、色の番号はヒストグラムの境界をまたいでさえも連続的に増加し続けます。次の例は、複数のヒストグラムに同じ色付けを施します。

```
plot newhistogram "Set A" lt 4, 'a' using 1, '' using 2, '' using 3, \
    newhistogram "Set B" lt 4, 'b' using 1, '' using 2, '' using 3
```

**43.59.8.11 Image** スタイル image は 2 次元画像の描画用のものです。これは、3 次元データ (x,y,value) に対する plot、または射影された 4 次元データ (x,y,z,value) に対する splot の両方で使用できます。このとき、表示平面では、画像データは表示平面内の 2 方向 (必ずしも垂直でなくても良い) に関して等間隔のサンプリング格子を形成していると仮定されます。それは言い換えれば、4 つの隣接点の組が常に同じ大きさの平行四辺形を作ることを意味します。変数 value は、現在のパレットに対するパレット色 (灰色値) を表します。

パレットと画像描画をサポートしている出力形式では、image スタイルは描画領域にぴったり合うように適切に配置され伸縮されたデータ配列 (matrix) を作ろうとします。そのような出力は効率的ですし、描画も速く行われます。しかし、パレットと画像描画をサポートしない、あるいはまだそれらが実装されていない出力形式では、image スタイルは長方形をピクセル毎に塗りつぶそうとしますので、あまり効率はよくありません。また、一般的の平行四辺形の画像に対しては、現在は常に平行四辺形をピクセル毎に塗りつぶします。

画像の各データ点の座標は 1 ピクセルの中心になります。すなわち、M x N 個のデータ集合は M x N ピクセルの画像を生成します。これは、M x N 個のデータ集合が (M-1) x (N-1) 要素を作成する pm3d の構造とはやや異なります。画像データの格子の走査方向は 8 つの組合せの任意の一つが可能です。

以下はいくつかの出力形式に関する特有の情報です:

x11 と wxt - ピクセルはディスプレイの解像度に合うように繰り返されるか

一定の割合で潰されるかされ、フィルタ用の処理は他に持っています。よって、かなり高い面積の割合の画像が潰される場合、エイリアシングが起こる可能性があります。

postscript (pslatex, epslatex, pstex) - 標本化補間機能は無効になり、画

像は元の解像度のままコピーされます。

png - 出力はインストールされている libgd のバージョンに依存します。

gd 1.8.4 truecolor はサポートされていませんがドライバの関数は動作します。

gd 2.0.4 truecolor は動作しますが、truecolor を選択しないと画像は空白になってしまいます。

gd 2.0.9 truecolor も動作しますし、非 truecolor も動作します。

rgbimage も参照してください。

**43.59.8.12 Impulses** impulses スタイルは、x 軸 (グラフの境界ではなく) から、splot では格子の土台からの垂直な線分を各点に対して表示します。

**43.59.8.13 Labels** スタイル labels は、gnuplot が configure 時に –enable-datastrings オプションをつけてインストールされている場合にのみ有効です。2 次元描画では labels スタイルは 3 つの入力データ列の指定が必要です。3 番目の列として見つかるテキスト文字列を、最初の 2 列の指定子によって生成される X,Y 座標の位置に表示します。フォント、色、回転角やその他の描画テキストの属性は追加オプションとして指定可能です (set label 参照)。次の例は、入力ファイルの 4 列目をテキストラベルとして取り出して 2 次元描画を行ないます (tc lt 2 は textcolor linetype 2 の省略形で、これは緑です)。

```
plot 'datafile' using 1:(0.5 * $2):4 with labels font "arial,11" tc lt 2
```

スタイル labels は 3 次元描画でも使用できます。この場合 X,Y,Z と文字列、の 4 つの入力列指定が必要です。

```
splot 'datafile' using 1:2:3:4 with labels
```

datastrings, set style data も参照してください。

**43.59.8.14 Lines** lines スタイルは隣接する点を真直な線分で結びます。linetype, linewidth, linestyle も参照してください。

**43.59.8.15 Linespoints** linespoints スタイルは lines と points の両方を行ないます。すなわち、各点に小さな記号をつけ、そして隣接する点を真直な線分で結びます。コマンド set pointsize を使って点 (point) の記号の大きさを変更できます。その使い方については set pointsize を参照してください。linespoints は lp と略すことが出来ます。

**43.59.8.16 Points** points スタイルは各点に小さな記号を表示します。その記号の大きさを変更するにはコマンド set pointsize が使えます。その使用法については set pointsize の項を参照してください。

**43.59.8.17 Steps** steps スタイルは 2 次元描画でのみ利用可能です。これは 2 本の線分で隣り合う点をつなぎます: 1 本目は (x1,y1) から (x2,y1) まで、2 本目は (x2,y1) から (x2,y2) まで。以下も参照 steps デモ。

**43.59.8.18 Rgbimage** スタイル rgbimage は 2 次元画像の描画用で、概念上は image に似ています。詳細は image を参照してください。その違いは、plot では (x,y,r,g,b) の 5 次元データが、splot では (x,y,z,r,g,b) の 6 次元データが必要で、それらが座標と画像のカラー成分を意味します。現在は、カラー画像は標本化格子が長方形でかつその面が表示面方向に向いている場合にのみ機能します。一般的な方向の場合は多角形の塗りつぶしでなされるべきですが、それはパレット表から取り出すのみで、原色の配合はサポートしていません。

image も参照してください。

**43.59.8.19 Vectors** 2 次元の vectors スタイルは (x,y) から (x+xdelta,y+ydelta) までのベクトルを書きます。よって 4 列のデータが必要です。ベクトルの先端には小さな矢先も書きます。3 次元の vectors スタイルも同様ですが、データは 6 列必要です。vectors スタイルを使っての splot は set mapping cartesian のみでサポートされています。キーワード "with vectors" は arrow スタイルの指定子を後ろに伴うことができます。詳細は arrowstyle を参照してください。

例:

```
plot 'file.dat' using 1:2:3:4 with vectors head filled lt 2
splot 'file.dat' using 1:2:3:(1):(1):(1) with vectors filled head lw 2
```

set clip one と set clip two は 2 次元のベクトルの描画に影響を与えます。詳細は set clip と arrowstyle を参照してください。

**43.59.8.20 Xerrorbars** xerrorbars スタイルは 2 次元のデータ描画のみで利用可能です。xerrorbars は、水平の誤差指示線 (error bar) が表示される以外は dots と同じです。各点 (x,y) において (xlow,y) から (xhigh,y) まで、または (x-xdelta,y) から (x+xdelta,y) までの線分が引かれますが、これらはいくつのデータ列が与えられるかによって変わります。誤差指示線の端には刻みの印が付けられます (set bars が使われていなければ。詳細に関しては set bars を参照してください)。

**43.59.8.21 Xyerrorbars** `xyerrorbars` スタイルは 2 次元のデータ描画のみで利用可能です。`xyerrorbars` は、水平、垂直の誤差指示線 (error bar) も表示される以外は `dots` と同じです。各点  $(x,y)$  において  $(x,y-y\delta)$  から  $(x,y+y\delta)$  までと  $(x-x\delta,y)$  から  $(x+x\delta,y)$  まで、または  $(x,y-l)$  から  $(x,y+h)$  までと  $(x-l,y)$  から  $(x+h,y)$  までの線分が引かれますが、これらはいくつのデータ列が与えられるかによって変わります。誤差指示線の端には刻みの印が付けられます (`set bars` が使われていなければ。詳細に関しては `set bars` を参照してください)。

データが、サポートされていない混合型の形式で与えられた場合、`plot` コマンドの `using` フィルタを使って適切な形に直さないといけません。例えばデータが  $(x,y,x\delta,y\delta)$  という形式である場合、以下のようにします:

```
plot 'data' using 1:2:($1-$3):($1+$3):4:5 with xyerrorbars
```

**43.59.8.22 Yerrorbars** `yerrorbars` (または `errorbars`) スタイルは 2 次元のデータ描画のみで利用可能です。`yerrorbars` は、垂直の誤差指示線 (error bar) が表示される以外は `points` に似ています。各点  $(x,y)$  において  $(x,y-y\delta)$  から  $(x,y+y\delta)$  まで、または  $(x,y-l)$  から  $(x,y+h)$  までの線分が引かれますが、これらはいくつのデータ列が与えられるかによって変わります。誤差指示線の端には刻みの印が付けられます (`set bars` が使われていなければ。詳細に関しては `set bars` を参照してください)。以下も参照

`errorbar` デモ。

**43.59.8.23 Xerrorlines** `xerrorlines` スタイルは 2 次元のデータ描画のみで利用可能です。`xerrorlines` は `linespoints` に似ていますが、水平の誤差線が描かれることが違います。各点  $(x,y)$  で、データ列の個数に応じて  $(x-l,y)$  から  $(x+h,y)$  まで、または  $(x-x\delta,y)$  から  $(x+x\delta,y)$  までの線分が描かれ、そして刻の印が誤差線の端に置かれます (`set bars` が使われていない場合 – 詳細は `set bars` 参照)。

**43.59.8.24 Xyerrorlines** `xyerrorlines` スタイルは 2 次元のデータ描画のみで利用可能です。`xyerrorlines` は `linespoints` に似ていますが、水平と垂直の誤差線も描かれることが違います。各点  $(x,y)$  で、データ列の個数に応じて、 $(x,y-y\delta)$  から  $(x,y+y\delta)$  までと  $(x-x\delta,y)$  から  $(x+x\delta,y)$  まで、あるいは  $(x,y-l)$  から  $(x,y+h)$  までと  $(x-l,y)$  から  $(x+h,y)$  までの線分が描かれ、そして刻の印が誤差線の端に置かれます (`set bars` が使われていない場合 – 詳細は `set bars` 参照)。

データが、サポートされていない混合型の形式で与えられた場合、`plot` コマンドの `using` フィルタを使って適切な形に直さないといけません。例えばデータが  $(x,y,x\delta,y\delta)$  という形式である場合、以下のようにします:

```
plot 'data' using 1:2:($1-$3):($1+$3):4:5 with xyerrorlines
```

**43.59.8.25 Yerrorlines** `yerrorlines` (または `errorlines`) スタイルは 2 次元のデータ描画のみで利用可能です。`yerrorlines` は `linespoints` に似ていますが、垂直の誤差線が描かれることが違います。各点  $(x,y)$  で、データ列の個数に応じて  $(x,y-y\delta)$  から  $(x,y+y\delta)$  まで、または  $(x,y-l)$  から  $(x,y+h)$  までの線分が描かれ、そして刻の印が誤差線の端に置かれます (`set bars` が使われていない場合 – 詳細は `set bars` 参照)。以下も参照。

`エラーバー` のデモ

## 43.60 Surface

コマンド `set surface` は `splot` による曲面の表示を制御します。

書式:

```
set surface
unset surface
show surface
```

曲面はデータや関数に対して、`with` で指定されたスタイル、あるいは他の適切なスタイルで書かれます。`unset surface` が実行されれば `splot` は関数やデータファイルの点に対する点や線を書きません。その場合でも `set contour` の設定により曲面の等高線は書かれます。`unset surface; set contour base` は等高線を格子の土台に表示する際に便利です。`set contour` も参照してください。

### 43.61 Table

table モードが有効な場合、plot と splot コマンドは、現在の出力形式に対する実際の描画を生成する替わりに X Y {Z} R の値の複数列からなる表形式のアスキー出力を行ないます。文字 R は、次の 3 種類のうちの一つです：その点が有効な範囲内にある場合は "i"、範囲外の場合は "o"、未定義値 (undefined) の場合は "u" です。データの書式は、軸のラベルの書式 (set format 参照) によって決まり、列は一つの空白で区切られます。これは、等高線を生成し、例えば plot で描画するなど、それを再利用するために保存したいときに便利です。例については set contour を参照してください。この方法は、補間されたデータを保存するのにも使うことができます (set samples, set dgrid3d 参照)。

書式:

```
set table {"outfile"}
plot <whatever>
unset table
```

表形式の出力は、指定されたファイルに書き出されますが、指定がない場合は現在 set output で指定されている値に出力されます。現在の出力形式の標準的な描画に戻すには、unset table を明示的に行なう必要があります。

### 43.62 Terminal

gnuplot は数多くのグラフィック形式をサポートしています。コマンド set terminal を使って gnuplot の出力の対象となる形式の種類を選んでください。出力先をファイル、または出力装置にリダイレクトするには set output を使ってください。

書式:

```
set terminal <terminal-type> | push | pop
show terminal
```

<terminal-type> が省略されると gnuplot は利用可能な出力形式の一覧を表示します。<terminal-type> の指定には短縮形が使えます。

set terminal と set output の両方を使う場合、set terminal を最初にする方が安全です。それは、OS によっては、それが必要とするフラグをセットする出力形式があるからです。

いくつかの出力形式はたくさんの追加オプションを持ちます。例えば png, postscript などの項を参照してください。各 <term> に対し、直前の set term <term> <options> で使用されたオプションは記憶され、その後の set term <term> がそれをリセットすることはありません。これは例えば印刷時に有用です。幾つかの異なる出力形式を切替える場合、前のオプションを繰り返し唱える必要はありません。

コマンド set term push は、現在の出力形式とその設定を set term pop によって復帰するまで記憶しています。これは save term, load term とほぼ同等ですが、ファイルシステムへのアクセスは行わず、よって例えばこれは、印刷後にプラットホームに依存しない形で出力形式を復帰する目的に使えます。gnuplot の起動後、デフォルト、または startup ファイルに書かれた出力形式が自動的に記憶 (push) されます。よって、明示的に出力形式を記憶させることなく、任意のプラットホーム上でデフォルトの出力形式を set term pop によって復帰させる、という動作を期待したスクリプトを可搬性を失わずに書くことが出来ます。

有効な出力形式の一覧全体については terminal を参照してください。

### 43.63 Termoption

コマンド set termoption は、現在使用している出力形式の振舞いを、新たな set terminal コマンドの発行なしに変更することを可能にします。このコマンド一つに対して一つのオプションのみが変更できます。そしてこの方法で変更できるオプションはそう多くはありません。現在使用可能なオプションは以下のもののみです。

```
set termoption {no}enhanced
set termoption font "<fontname>{,<fontsize>}"
```

### 43.64 Tics

全ての軸の(見出しのつく)大目盛りの制御を一度に行なうことは、コマンド `set tics` で可能です。

全ての軸の(見出しのつく)大目盛りはコマンド `set tics` で一度に制御できます。目盛りは `unset tics` で消え、`set tics` で(デフォルトの状態の)目盛りがつきます。各単一軸の大目盛りの制御を行なう同様のコマンドがあります(`tics` の前に軸名をつけたもの)。

書式

```
set tics {axis | border} {{no}mirror}
    {in | out} {scale {default | <major> {,<minor>}{}}
    {{no}rotate {by <ang>}} {offset <offset> | nooffset}
    { font "name{,<size>}" }
    { textcolor <colors> }
```

```
unset tics
show tics
```

指定されたオプションは、全ての軸、すなわち `x, y, z, x2, y2, cb` 軸に適用されます。

`axis` と `border` は `gnuplot` に目盛り(目盛りの刻自身とその見出し)を、それぞれ軸につけるのか、境界につけるのかを指示します。軸が境界にとても近い場合、`axis` を使用すると境界が表示されなければ(`set border` 参照)目盛りの見出し文字を境界の外に出してしまうでしょう。この場合自動的なレイアウトアルゴリズムによる余白設定は大抵よくないものとなってしまいます。

`mirror` は `gnuplot` に反対側の境界の同じ位置に、見出しのない目盛りを出力するよう指示します。`nomirror` は、あなたが想像している通りのことを行ないます。

`in` と `out` は目盛りの刻みを内側に描くか外側に描くかを切り替えます。

目盛りの刻みのサイズは `scale` で調整できます。`<minor>` の指定が省略された場合は、それは  $0.5 * <\text{major}>$  になります。デフォルトのサイズは、大目盛りが 1.0 で小目盛りが 0.5 で、これは `scale default` で呼びだせます。

`rotate` は、文字列を 90 度回転させて出力させようとします。これは、文字列の回転をサポートしている出力ドライバ(terminal)では実行されます。`norotate` はこれをキャンセルします。`rotate by <ang>` は角度 `<ang>` の回転を行ないますが、これはいくつかの出力形式(terminal)でサポートされています。

`x` と `y` 軸の大目盛りのデフォルトは `border mirror norotate` で、`x2, y2` 軸は `border nomirror norotate` がデフォルトです。`z` 軸のデフォルトは `nomirror` です。

`<offset>` は `x,y` または `x,y,z` の形式ですが、それに座標系を選択して、その前に `first`, `second`, `graph`, `screen`, `character` のいずれかをつけることもできます。`<offset>` は、目盛りの見出し文字列のデフォルトの位置からのずらし位置で、そのデフォルトの単位系は `character` です。詳細は `coordinates` を参照してください。`nooffset` は `offset` を OFF にします。

オプションなしの `set tics` は、目盛りの刻みを内側にしますが、その他の全てのオプションは直前の値を保持します。

大目盛り(ラベルのつく)他の制御に関しては `set xtics` を、小目盛りの制御に関しては `set mxtics` もそれぞれ参照してください。これらのコマンドは、各軸毎にその一つの軸の制御を提供します。

### 43.65 Ticslevel

`set xyplane` 参照。

### 43.66 Ticscale

コマンド `set ticscale` は現在は推奨されていません。代わりに `set tics scale` を使ってください。

### 43.67 Timestamp

コマンド `set timestamp` は描画の日付と時刻を左の余白に表示します。

書式:

```
set timestamp {"<format>"} {top|bottom} {{no}rotate}
    {offset {<xoff>} {<yoff>}} {font "<fontspec>"}
unset timestamp
show timestamp
```

書式文字列 (format) を使って、書かれる日付と時刻の書式を選択することができます。デフォルトは `asctime()` が使用する "%a %b %d %H:%M:%S %Y" です (曜日、月名、日、時、分、秒、4 枠の西暦)。`top` と `bottom` を使って日付を左の余白の上に配置するか、下に配置するかを選択できます (デフォルトは下)。`rotate` は、もし出力形式がサポートしていればですが、日付を垂直方向の文字列にします。定数 `<xoff>`、`<yoff>` はずれ (offset) を意味し、これによってより適切な位置合わせが行えます。`<font>` は日付が書かれるフォントを指定します。

`timestamp` の代わりに省略名 `time` を使っても構いません。

例:

```
set timestamp "%d/%m/%y %H:%M" offset 80,-2 font "Helvetica"
```

日付の書式文字列に関する詳しい情報については `set timefmt` を参照してください。

### 43.68 Timefmt

このコマンドは、データが日時の形式になっている場合に、その時系列データに適用されます。これはコマンド `set xdata time` も与えられていないと意味がありません。

書式:

```
set timefmt "<format string>"
show timefmt
```

文字列引数 (`<format string>`) は `gnuplot` に日時データをデータファイルからどのように読むかを指示します。有効な書式は以下の通りです:

Time Series timedata Format Specifiers	
書式	説明
%d	何日, 1-31
%m	何月, 1-12
%y	何年, 0-99
%Y	何年, 4 枠
%j	1 年の何日目, 1-365
%H	何時, 0-24
%M	何分, 0-60
%s	Unix epoch (1970-01-01, 00:00 UTC) からの秒数
%S	何秒, 0-60
%b	月名 (英語) の 3 文字省略形
%B	月名 (英語)

任意の文字を文字列中で使用できますが、規則に従っている必要があります。`\t` (タブ) は認識されます。バックスラッシュ + 8 進数列 (`\nnn`) はそれが示す文字に変換されます。日時要素の中に分離文字がない場合、`%d`, `%m`, `%y`, `%H`, `%M`, `%S` はそれぞれ 2 枠の数字を読み込み、`%Y` は 4 枠、`%j` は 3 枠の数字を読み込みます。`%b` は 3 文字を、`%B` は必要な分だけの文字を要求します。

空白 (スペース) の扱いはやや違います。書式文字列中の 1 つの空白は、ファイル中の 0 個、あるいは 1 つ以上の空白文字列を表します。すなわち、"`%H %M`" は "1220" や "12 20" を "12 20" と同じように読みます。

データ中の非空白文字の集まりそれぞれは、`using n:n` 指定の一つ一つの列とカウントされます。よって `11:11 25/12/76 21.0` は 3 列のデータと認識されます。混乱を避けるために、日時データが含まれる場合 `gnuplot` は、あなたの `using` 指定が完璧なものであると仮定します。

`gnuplot` は数字でない文字列を読めないので、日付データが曜日、月の名前を含んでいる場合、書式文字列でそれを排除しなければいけません。しかし、"`%a`", "`%A`", "`%b`", "`%B`" でそれらを表示することは

できます: これら、及び日時データの出力の他のオプションの詳細に関しては `set format` を参照してください (`gnuplot` は数値から月や曜日を正しく求めます)。

他の情報については `set xdata` と `Time/date` の項も参照してください。

例:

```
set timefmt "%d/%m/%Y\t%H:%M"
```

は、`gnuplot` に日付と時間がタブで分離していることを教えます (ただし、あなたのデータをよく見てください。タブだったものがどこかで複数のスペースに変換されませんか? 書式文字列はファイル中に実際にある物と一致していなければなりません)。以下も参照

時系列データ (time data) デモ。

### 43.69 Title

コマンド `set title` は、描画の上の真中に書かれる描画タイトルを生成します。`set title` は `set label` の特殊なもの、とみなせます。

書式:

```
set title {"<title-text>"} {offset <offset>} {font "<font>{,<size>}"}
           {{textcolor | tc} {lt <line_type> | default}} {{no}enhanced}
show title
```

`<offset>` を `x,y` または `x,y,z` の形式で指定した場合は、タイトルは与えられた値だけ移動されます。それに座標系を選択して、その前に `first`, `second`, `graph`, `screen`, `character` のいずれかをつけることもできます。詳細は `coordinates` を参照してください。デフォルトでは `character` 座標系が使われます。例えば、"set title offset 0,-1" はタイトルの `y` 方向の位置のみ変更し、大ざっぱに言って 1 文字分の高さだけタイトルを下に下げます。1 文字の大きさは、フォントと出力形式の両方に依存します。

`<font>` はタイトルが書かれるフォントを指定するのに使われます。`<size>` の単位は、どの出力形式 (terminal) を使っているかによって変わります。

`textcolor lt <n>` は文字の色を線種 (line type) `<n>` と同じものにします。

`noenhanced` は、拡張文字列処理 (enhanced text) モードが有効になっている場合でも、タイトルを拡張文字列処理させないようにします。

`set title` をパラメータなしで使うとタイトルを消去します。

バックスラッシュ文字列の作用、及び文字列を囲む單一引用符と二重引用符の違いについては `syntax` を参照してください。

### 43.70 Tmargin

コマンド `set tmargin` は上の余白のサイズをセットします。詳細は `set margin` を参照してください。

### 43.71 Trange

コマンド `set trange` は、媒介変数モード、あるいは極座標モードでの `x, y` の値を計算するのに使われる媒介変数の範囲を設定します。詳細は `set xrange` を参照してください。

### 43.72 Urange

`set urange` と `set vrangle` は、`splot` の媒介変数モードで `x, y, z` の値を計算するのに使われる媒介変数の範囲を設定します。詳細は `set xrange` を参照してください。

### 43.73 Variables

`show variables` コマンドは全てのユーザ定義変数とその値の一覧を表示します。

書式:

```
show variables {all}
```

追加キーワードの "all" は、GPVAL\_ で始まる gnuplot 定義変数 の一覧も表示します。

### 43.74 Version

コマンド `show version` は現在起動している gnuplot のバージョン、最終修正日、著作権者と、FAQ や info-gnuplot メーリングリスト、バグレポート先のメールアドレスを表示します。対話的にプログラムが呼ばれているときはスクリーン上にその情報を表示します。

書式:

```
show version {long}
```

`long` オプションを与えると、さらにオペレーティングシステム、gnuplot インストール時のコンパイルオプション、ヘルプファイルの置き場所、そして(再び) 有用なメールアドレスを表示します。

### 43.75 View

コマンド `set view` は `splot` の視線の角度を設定します。これは、グラフ描画の 3 次元座標をどのように 2 次元の画面 (screen) に投影するかを制御します。これは、描画されたデータの回転と伸縮の制御を与えてくれますが正射影しかサポートしていません。3 次元射影、および 2 次元描画的地図上への 2 次元直交射影がサポートされています。

書式:

```
set view { <rot_x>{,{<rot_z>}{{,<scale>}{{,<scale_z>}}}} | map }
show view
```

ここで `<rot_x>` と `<rot_z>` は、画面に投影される仮想的な 3 次元座標系の回転角 (単位は度) の制御で、最初は (すなわち回転が行なわれる前は) 画面内の水平軸は `x`、画面内の垂直軸は `y`、画面自身に垂直な軸が `z` となっています。最初は `x` 軸の周りに `<rot_x>` だけ回転されます。次に、新しい `z` 軸の周りに `<rot_z>` だけ回転されます。

コマンド `set view map` は、描画を地図として表示するのに使います。これは等高線 (contour) の描画、およびカラー表示 (pm3d) の色地図に使えます。後者に関しては、入力データ点のフィルタ用の `zrange` の設定、および色の範囲の縮尺に関する `cbrange` の設定を適切に行うことについて注意してください。

`<rot_x>` は [0:180] の範囲に制限されていて、デフォルトでは 60 度です。`<rot_z>` は [0:360] の範囲に制限されていて、デフォルトでは 30 度です。`<scale>` は `splot` 全体の伸縮率を制御し、`<scale_z>` は `z` 軸の伸縮のみを行ないます。伸縮率のデフォルトはどちらも 1.0 です。

例:

```
set view 60, 30, 1, 1
set view , , 0.5
```

最初の例は 4 つの全てをデフォルトの値にしています。2 つめの例は縮小率のみを 0.5 に変更しています。`set ticslevel` も参照してください。

### 43.76 Vrange

コマンド `set urange` と `set vrangle` は、`splot` の媒介変数 (パラメータ) モードで `x`, `y`, `z` の値を計算するのに使われる媒介変数の範囲を設定します。詳細は `set xrange` を参照してください。

### 43.77 X2data

コマンド `set x2data` は `x2` (上) 軸のデータを時系列 (日時) 形式に設定します。詳細は `set xdata` を参照してください。

### 43.78 X2dtics

コマンド `set x2dtics` は `x2` (上) 軸の目盛りを曜日に変更します。詳細は `set xdtics` を参照してください。

### 43.79 X2label

コマンド `set x2label` は `x2` (上) 軸の見出しを設定します。詳細は `set xlabel` を参照してください。

### 43.80 X2mtics

コマンド `set x2mtics` は、`x2` (上) 軸を 1 年の各月に設定します。詳細は `set xmtics` を参照してください。

### 43.81 X2range

コマンド `set x2range` は `x2` (上) 軸の表示される水平範囲を設定します。詳細は `set xrange` を参照してください。

### 43.82 X2tics

コマンド `set x2tics` は `x2` (上) 軸の、見出し付けされる大目盛りの制御を行ないます。詳細は `set xtics` を参照してください。

### 43.83 X2zeroaxis

コマンド `set x2zeroaxis` は、原点を通る `x2` (上) 軸 ( $y2 = 0$ ) を描きます。詳細は `set zeroaxis` を参照してください。

### 43.84 Xdata

このコマンドは `x` 軸のデータ形式を日時データにセットします。同様のコマンドが他の軸それぞれに用意されています。

書式:

```
set xdata {time}
show xdata
```

`ydata`, `zdata`, `x2data`, `y2data`, `cbdata` にも同じ書式が当てはまります。

`time` オプションはデータが日時データであることを伝えます。オプションをつけない場合、データ型は通常のものに戻ります。

gnuplot にどのように日時データを読みこませるかについては、`set timefmt` を参照してください。日時データは今世紀の始まり（訳注：厳密には 2000 年 1 月 1 日の始まり）からの秒数に変換されます。時間書式 (`timefmt`) は現在はただ一つだけしか使えません。それは、全ての日時データ項目がこの書式に一致しなければならないことを意味します。また、範囲の指定は、日時指定が数式と解釈されるのを避けるために、その書式に従った文字列を引用符で囲んで指定すべきです。

目盛り刻みの見出し (label) を表示するのには関数 `'strftime'` (unix でそれを調べるには "man strftime" とタイプしてください) が使われます。`set format x "string"` で、10 進数の書式ではなさそうなもの (2 つ以上の '%'、または %f でも %g でもないもの) が与えられていなければ、gnuplot はこれを適当に意味のある書式で計算して表示します。

他の情報については `Time/date` も参照してください。

### 43.85 Xdtics

コマンド `set xdtics` は x 軸の目盛りの刻みを曜日に変換します (0=Sun, 6=Sat)。6 を越える場合は 7 による余りが使われます。`unset xdtics` はその見出しをデフォルトの形式に戻します。他の軸にも同じことを行なう同様のコマンドが用意されています。

書式:

```
set xdtics
unset xdtics
show xdtics
```

`ydtics`, `zdtics`, `x2dtics`, `y2dtics`, `cbdtics` にも同じ書式が当てはまります。

`set format` コマンドも参照してください。

### 43.86 Xlabel

コマンド `set xlabel` は x 軸の見出しを設定します。他の軸にも見出しを設定する同様のコマンドがあります。

書式:

```
set xlabel {"<label>"} {offset <offset>} {font "<font>{,<size>}"}
    {{textcolor | tc} {lt <line_type> | default}} {{no}enhanced}
    {rotate by <degrees>}
show xlabel
```

同じ書式が `x2label`, `ylabel`, `y2label`, `zlabel`, `cblabel` にも適用されます。

`<offset>` を `x,y` かまたは `x,y,z` の形式で指定した場合は、見出しが与えられた値だけ移動されます。それに座標系を選択して、その前に `first`, `second`, `graph`, `screen`, `character` のいずれかをつけることもできます。詳細は `coordinates` を参照してください。デフォルトでは `character` 座標系が使われます。例えば、"set xlabel offset -1,0" は見出しの x 方向の位置のみ変更し、大ざっぱに言って 1 文字分の幅だけ見出しど左にずらします。1 文字の大きさは、フォントと出力形式の両方に依存します。

`<font>` は見出しが書かれるフォントを指定するのに使われます。フォントの `<size>` (大きさ) の単位は、どんな出力形式を使うかに依存します。

`textcolor lt <n>` は、見出し文字列を線種 `<n>` の色にします。

`noenhanced` は、拡張文字列処理 (enhanced text) モードが有効になっている場合でも、ラベル文字列を拡張文字列処理させないようにします。

見出しを消去するには、オプションをつけずに実行します。例: "set y2label"

軸の見出しのデフォルトの位置は以下の通りです:

`xlabel`: x 軸の見出しが下の軸の下の真中

`ylabel`: y 軸の見出しが出力形式依存で、以下の 3 つのいずれか:

1. 水平方向の文字列で描画の左上に左端に合わせて配置されます。文字列の回転を行なえない出力形式では多分これが選択されます。`set x2tics` が同時に使われている場合、`ylabel` は `x2` 軸の見出しの左端と重なるかも知れません。これは `ylabel` の位置か左の余白を調整することで対処できるでしょう。

2. 垂直方向の文字列で、描画の左で垂直方向に中央揃えされます。文字列を回転できる出力形式では多分これが選択されます。

3. 水平方向の文字列で、描画の左で垂直方向に中央揃えされます。LaTeX, TPIC ドライバではこれが選択されます。EPPIC ドライバでは、描画に重ならないような、文字の積み重ねを生成しますが、その他のドライバ (LaTeX や TPIC のような) では、`ylabel` が描画に上書きするのを避けるために、`\backslash` を使って改行を入れる必要があるかもしれません。

`zlabel`: z 軸の見出しが軸の表示範囲より上で、見出しの真中が z 軸の真上

`cblabel`: 色見本 (color box) の軸の見出しが箱に沿って中央揃えされ、箱の向きが水平なら下に、垂直なら右

`y2label`: y2 軸の見出しが y2 軸の右。その位置は、出力形式依存で y 軸と同様の規則で決定。

x2label: x2 軸の見出しは上の軸の上で、描画タイトルよりは下。これは、改行文字を使えば、それによる複数の行からなる描画タイトルで x2 軸の見出しを生成することも可能。例:

```
set title "This is the title\n\nThis is the x2label"
```

これは二重引用符を使うべきであることに注意してください。この場合、もちろん 2 つの行で同じフォントが使われます。

y 軸と y2 軸のラベルは、デフォルトの方向に対して明示的に回転させることができます、これは 2 次元描画で、かつ文字列の回転をサポートする出力形式のみで有効です。

もし軸の位置のデフォルトの位置が気に入らないならば、代わりに set label を使ってください。このコマンドは文字列をどこに配置するかをもっと自由に制御できます。

バックスラッシュ文字列の作用、及び文字列を囲む単一引用符と二重引用符の違いに関するより詳しい情報については syntax を参照してください。

### 43.87 Xmtics

コマンド set xmtics は x 軸の目盛りの見出しを月に変換します。1=Jan (1 月)、12=Dec (12 月) となります。12 を越えた数字は、12 で割ったあまりの月に変換されます。unset xmtics で目盛りはデフォルトの見出しに戻ります。他の軸に対しても同じ役割をする同様のコマンドが用意されています。

書式:

```
set xmtics
unset xmtics
show xmtics
```

x2mtics, ymtics, y2mtics, zmtics, cbmtics にも同じ書式が適用されます。

コマンド set format も参照してください。

### 43.88 Xrange

コマンド set xrange は表示される水平方向の範囲を指定します。他の軸にも同様のコマンドが存在しますし、極座標での半径 r, 媒介変数 t, u, v にも存在します。

書式:

```
set xrange { [{<min>}:{<max>}] } {{no}reverse} {{no}writeback} }
           | restore
show xrange
```

ここで <min> と <max> には定数、数式、または '\*' で、'\*' は自動縮尺機能を意味します。日時データの場合、範囲は set timefmt の書式に従った文字列を引用符で囲む必要があります。省略された値は変更されません。

yrange, zrange, x2range, y2range, cbrange, rrange, trange, urange, vrangle は同じ書式を使用します。

オプション reverse は軸の方向を逆にします。例えば set xrange [0:1] reverse は、1 が左、0 が右であるような軸にします。これは、もちろん set xrange [1:0] と同じですが、reverse は主に自動縮尺 (autoscale) で用いられることを意図しています。

オプション writeback は、set xrange で占められているバッファの中に自動縮尺機能により作られた範囲を保存します。これは、いくつかの関数を同時に表示し、しかしその範囲はそのうちのいくつかのものから決定させたい場合に便利です。writeback の作用は、plot の実行中に機能するので、そのコマンドの前に指定する必要があります。最後に保存した水平方向の範囲は set xrange restore で復元できます。例を上げます。

```
set xrange [-10:10]
set yrange [] writeback
plot sin(x)
set yrange restore
replot x/2
```

この場合、y の範囲 (yrange) は  $\sin(x)$  の値域として作られた [-1:1] の方になり、 $x/2$  の値域 [-5:5] は無視されます。上記のそれぞれのコマンドの後に show yrange を実行すれば、上で何が行なわれているかを理解する助けになるでしょう。

2 次元描画において、xrange と yrange は軸の範囲を決定し、trange は、媒介変数モードの媒介変数の範囲、あるいは極座標モードの角度の範囲を決定します。同様に 3 次元媒介変数モードでは、xrange, yrange, zrange が軸の範囲を管理し、urange と yrange が媒介変数の範囲を管理します。

極座標モードでは、rrange は描画される半径の範囲を決定します。 $\langle rmin \rangle$  は半径への追加の定数として作用し、一方  $\langle rmax \rangle$  は半径を切り捨てる (clip) ように作用し、 $\langle rmax \rangle$  を越えた半径に対する点は描画されません。xrange と yrange は影響されます。これらの範囲は、グラフが  $r(t)-rmin$  のグラフで、目盛りの見出しにはそれぞれ  $rmin$  を加えたようなものであるかのようにセットされます。

全ての範囲は部分的に、または全体的に自動縮尺されますが、データの描画でなければ、パラメータ変数の自動縮尺機能は意味がないでしょう。

範囲は plot のコマンドライン上でも指定できます。コマンドライン上で与えられた範囲は単にその plot コマンドでだけ使われ、set コマンドで設定された範囲はその後の描画で、コマンドラインで範囲を指定していないもの全てで使われます。これは splot も同じです。

例:

x の範囲をデフォルトの値にします:

```
set xrange [-10:10]
```

y の範囲が下方へ増加するようにします:

```
set yrange [10:-10]
```

z の最小値には影響を与えずに (自動縮尺されたまま)、最大値のみ 10 に設定します:

```
set zrange [:10]
```

x の最小値は自動縮尺とし、最大値は変更しません:

```
set xrange [*:]
```

### 43.89 Xtics

x 軸の (見出しのつく) 大目盛りは コマンド set xtics で制御できます。目盛りは unset xtics で消え、set xtics で (デフォルトの状態の) 目盛りがつきます。y,z,x2,y2 軸の大目盛りの制御を行なう同様のコマンドがあります。

書式:

```
set xtics {axis | border} {{no}mirror}
    {in | out} {scale {default | <major> {,<minor>}}}
    {{no}rotate {by <ang>}} {offset <offset> | nooffset}
    {add}
    { autofreq
        | <incr>
        | <start>, <incr> {,<end>}
        | {"<label>"} <pos> {<level>} {,"<label>"}... }
    { font "name{,<size>}" }
    { textcolor <colors> }
```

```
unset xtics
show xtics
```

同じ書式が ytics, ztics, x2tics, y2tics, cbtics にも適用されます。

axis と border は gnuplot に目盛り (目盛りの刻自身とその見出し) を、それぞれ軸につけるのか、境界につけるのかを指示します。軸が境界にとても近い場合、axis を使用すると目盛りの見出し文字を境界の外に出してしまうでしょう。この場合自動的なレイアウトアルゴリズムによる余白設定は大抵よくないものとなってしまいます。

**mirror** は gnuplot に反対側の境界の同じ位置に、見出しのない目盛りを出力するよう指示します。**nomirror** は、あなたが想像している通りのことを行ないます。

**in** と **out** は目盛りの刻みを内側に描くか外側に描くかを切り替えます。

目盛りの刻みのサイズは **scale** で調整できます。**<minor>** の指定が省略された場合は、それは  $0.5^*<\text{major}>$  になります。デフォルトのサイズは、大目盛りが 1.0 で小目盛りが 0.5 で、これは **scale default** で呼びだせます。

**rotate** は、文字列を 90 度回転させて出力させようとします。これは、文字列の回転をサポートしている出力ドライバ (terminal) では実行されます。**norotate** はこれをキャンセルします。**rotate by <ang>** は角度 **<ang>** の回転を行ないますが、これはいくつかの出力形式 (terminal) でサポートされています。

x と y 軸の大目盛りのデフォルトは **border mirror norotate** で、x2, y2 軸は **border nomirror norotate** がデフォルトです。z 軸には、{axis | border} オプションは無効で、デフォルトは **nomirror** です。z 軸の目盛りをミラー化したいなら、多分 **set border** でそのための空間をあける必要があるでしょう。

**<offset>** は x,y かまたは x,y,z の形式で指定しますが、それに座標系を選択して、その前に **first**, **second**, **graph**, **screen**, **character** のいずれかをつけることもできます。**<offset>** は刻み文字のデフォルトの位置からのずれを表し、デフォルトの座標系は **character** です。詳細は **coordinates** を参照してください。**nooffset** はずらしを無効にします。

例:

**xtics** をより描画に近づける:

```
set xtics offset 0,graph 0.05
```

オプションなしで **set xtics** を実行すると、目盛りが表示される状態であれば、それはデフォルトの境界、または軸を復元し、そうでなければ何もしません。その前に指定した目盛りの間隔、位置 (と見出し) は保持されます。

目盛りの位置は、デフォルト、またはオプション **autofreq** が指定されていれば自動的に計算されます。そうでなければ、次の 2 つの形式で指定されます:

暗示的な **<start>**, **<incr>**, **<end>** 形式は、目盛りの列を **<start>** から **<end>** の間を **<incr>** の間隔で表示します。**<end>** を指定しなければ、それは無限大とみなされます。**<incr>** は負の値も可能です。**<start>** と **<end>** の両方が指定されていない場合、**<start>** は - 、**<end>** は + とみなされ、目盛りは **<incr>** の整数倍の位置に表示されます。軸が対数軸の場合、目盛りの間隔 (増分) は、積因子として使用されます。

負の **<start>** や **<incr>** を、数値の後ろに指定すると (例えば **rotate by <angle>** とか **offset <offset>** の後ろ)、gnuplot の構文解析器は、その値からその負の **<start>** や **<incr>** の値の引き算を行おうとする間違いを犯します。これを回避するには、そのような場合は、**0-<start>** や **0-<incr>** のように指定してください。

例:

```
set xtics border offset 0,0.5 -5,1,5
```

最後の ',' のところで失敗します。

```
set xtics border offset 0,0.5 0-5,1,5
```

か

```
set xtics offset 0,0.5 border -5,1,5
```

ならば、ちゃんと指示通りに、目盛りを境界に、目盛り見出し文字列を 0,0.5 文字分だけずらして、**start**, **increment**, **end** をそれぞれ -5,1,5 に設定します。

**set grid** のオプション 'front', 'back', 'layerdefault' も、x 軸の目盛りの描画の順序に影響します。

例:

目盛りを 0, 0.5, 1, 1.5, ..., 9.5, 10 の位置に生成

```
set xtics 0,.5,10
```

目盛りを ..., -10, -5, 0, 5, 10, ... に生成

```
set xtics 5
```

目盛りを 1, 100, 1e4, 1e6, 1e8 に生成

```
set logscale x; set xtics 1,100,1e8
```

明示的な ("<label>" <pos> <level>, ...) の形式は、任意の目盛りの位置、あるいは数字でない見出しの生成も可能にします。この形式では、目盛りは位置の数字の順に与える必要はありません。各目盛りは位置 (pos) と見出し (label) を持りますが、見出しあは必須ではありません。見出しあは二重引用符で囲まれた文字列であることに注意してください。それは、"hello" のような固定文字列でも構いませんし、"%3f clients" のようにその位置を数字に変換する書式文字列を含んでも構いませんし、空文字列 "" でも構いません。より詳しい情報については set format を参照してください。もし、文字列が与えられなければ、デフォルトの数字の見出しが使用されます。

明示的な形式では 3 つ目のパラメータとして "レベル" (level) を与えることができます。デフォルトのレベルは 0 で、これは大目盛りを意味し、レベルが 1 の場合小目盛りが生成されます。レベルを指定する場合はラベルも必ず指定する必要があります。

例:

```
set xtics ("low" 0, "medium" 50, "high" 100)
set xtics (1,2,4,8,16,32,64,128,256,512,1024)
set ytics ("bottom" 0, "" 10, "top" 20)
set ytics ("bottom" 0, "" 10 1, "top" 20)
```

2 番目の例では、全ての目盛りが見出しが付けられます。3 番目の例では、端のものだけが見出しが付けられます。4 番目の例の、見出しあの目盛りは小目盛りになります。

通常明示的な（手動の）目盛り位置が与えられた場合、自動的に生成される目盛りは使われません。逆に、set xtics auto のようなものが指定された場合は、以前に手動で設定した目盛りは消されてしまいます。この手動の目盛りと自動的な目盛りを共存させるにはキーワード add を使用してください。これは追加する目盛りのスタイルの前に書かなければいけません。

例:

```
set xtics 0,.5,10
set xtics add ("Pi" 3.14159)
```

これは自動的に目盛りの刻みを x 軸に 0.5 間隔でつけますが、 のところに明示的な見出しがも追加します。しかし指定しても、表示されるのはあくまで描画範囲のものだけです。

目盛りの見出しの書式（または省略）は set format で制御されます。ただしそれは set xtics (<label>) の形式の明示的な見出し文字列が含まれていない場合だけです。

（見出しが付かない）小目盛りは、set mxtics コマンドで自動的に追加するか、または位置を手動で set xtics ("<pos> 1, ...) の形式で与えることもできます。

時系列データの場合、位置の値は timefmt の書式にしたがった日付、または時刻を引用符で囲んで与えなければいけません。<start>, <incr>, <end> 形式を使う場合、<start> と <end> は timefmt に従って与えますが、<incr> は秒単位で与える必要があります。その時刻は実際には set format で与えた書式に従って表示されます。

例:

```
set xdata time
set timefmt "%d/%m"
set format x "%b %d"
set xrange ["01/12":"06/12"]
set xtics "01/12", 172800, "05/12"

set xdata time
set timefmt "%d/%m"
set format x "%b %d"
set xrange ["01/12":"06/12"]
set xtics ("01/12", "" "03/12", "05/12")
```

これらは両方とも "Dec 1", "Dec 3", "Dec 5", の目盛りを生成しますが、2 番目の例 "Dec 3" の目盛りは見出しが付けられません。

### 43.90 Xyplane

`set xyplane` コマンドは 3D 描画で描かれる `xy` 平面の位置を調整するのに使われます。後方互換性のために、"set ticslevel" も同じ意味のコマンドとして使うことができます。

書式:

```
set ticslevel <frac>
set xyplane <frac>
set xyplane at <zvalue>
show xyplane
```

`set ticslevel <frac>` は、`xy` 平面を `Z` 軸の範囲のどこに置くかを決定します。`<frac>` には、`xy` 平面と `z` の一番下の位置との差、`z` 軸の範囲全体に対する割合を与えます。デフォルトの値は 0.5 です。負の値も許されていますが、そうすると 3 つの軸の目盛りの見出しが重なる可能性があります。

`xy` 平面を `z` 軸の '`pos`' の位置に置くには、`ticslevel` の値を  $(\text{pos} - \text{zmin}) / (\text{zmin} - \text{zmax})$  としてください。しかし、この位置は `z` の範囲 (`zrange`) を変更した場合は変わってしまいます。

もう一つの形式である `set xyplane at <zvalue>` は、現在の `z` の範囲を気にすることなく、指定した `z` の値の位置に `xy` 平面を固定します。よって、`x,y,z` 軸を共通の原点を通るようにするには、`set xyplane at 0` とすればいいことになります。

`set view`, `set zeroaxis` も参照してください。

### 43.91 Xzeroaxis

コマンド `set xzeroaxis` は  $y = 0$  の直線を描きます。詳細に関しては、`set zeroaxis` を参照してください。

### 43.92 Y2data

コマンド `set y2data` は `y2` (右) 軸のデータを時系列 (日時) 形式に設定します。詳細は `set xdata` を参照してください。

### 43.93 Y2dtics

コマンド `set y2dtics` は `y2` (右) 軸の目盛りを曜日に変更します。詳細は `set xdtics` を参照してください。

### 43.94 Y2label

コマンド `set y2label` は `y2` (右) 軸の見出しを設定します。詳細は `set xlabel` を参照してください。

### 43.95 Y2mtics

コマンド `set y2mtics` は `y2` (右) 軸の目盛りを 1 年の各月に変更します。詳細は `set xmtics` を参照してください。

### 43.96 Y2range

コマンド `set y2range` は `y2` (右) 軸の表示される垂直範囲を設定します。詳細は `set xrange` を参照してください。

### 43.97 Y2tics

コマンド `set y2tics` は `y2` (右) 軸の、見出し付けされる大目盛りの制御を行ないます。詳細は `set xtics` を参照してください。

### 43.98 Y2zeroaxis

コマンド `set y2zeroaxis` は、原点を通る  $y_2$  (右) 軸 ( $x_2 = 0$ ) を描きます。詳細は `set zeroaxis` を参照してください。

### 43.99 Ydata

コマンド `set ydata` は  $y$  軸のデータを時系列 (日時) 形式に設定します。`set xdata` を参照してください。

### 43.100 Ydtics

コマンド `set ydtics` は  $y$  軸の目盛りを曜日に変更します。詳細は `set xdtics` を参照してください。

### 43.101 Ylabel

このコマンドは  $y$  軸の見出しを設定します。`set xlabel` を参照してください。

### 43.102 Ymtics

コマンド `set ymtics` は、 $y$  軸の目盛りを月に変更します。詳細は `set xmtics` を参照してください。

### 43.103 Yrange

コマンド `set yrangle` は、 $y$  方向の垂直範囲を設定します。詳細は `set xrange` を参照してください。

### 43.104 Ytics

コマンド `set ytics` は  $y$  軸の (見出し付けされる) 大目盛りを制御します。詳細は `set xtics` を参照してください。

### 43.105 Yzeroaxis

コマンド `set yzeroaxis` は  $x = 0$  の直線 ( $y$  軸) を書きます。詳細は `set zeroaxis` を参照してください。

### 43.106 Zdata

コマンド `set zdata` は  $z$  軸のデータを時系列 (日時) 形式に設定します。`set xdata` を参照してください。

### 43.107 Zdtics

コマンド `set zdtics` は  $z$  軸の目盛りを曜日に変更します。詳細は `set xdtics` を参照してください。

### 43.108 Zzeroaxis

コマンド `set zzeroaxis` は  $(x=0,y=0)$  を通る直線を描きます。これは、2D 描画、および `set view map` での `splot` では効力を持たせん。詳細は、`set zeroaxis`, `set xyplane` を参照してください。

### 43.109 Cbdata

このコマンドはカラー ボックス軸のデータを時系列 (日時) 形式に式に設定します。`set xdata` を参照してください。

### 43.110 Cbdtics

コマンド `cbdtics` はカラー ボックス 軸の目盛りの刻みを曜日に変換します。詳細は `set xdtics` を参照してください。

### 43.111 Zero

`zero` の値は、0.0 に近いデフォルトの閾値を表します。

書式:

```
set zero <expression>
show zero
```

`gnuplot` は、(複素数値を持つ点の描画においては) その値の虚数部分の絶対値が `zero` 閾値より大きい場合 (つまり実数でない値を持つ点) は、その点を描画しません。この閾値は `gnuplot` の他の様々な部分においてその(大まかな) 数値誤差の閾値としても使われています。デフォルトの `zero` の値は `1e-8` です。`1e-3` (= 典型的なビットマップディスプレイの解像度の逆数) より大きい `zero` の値は設定すべきではないでしょうが、`zero` を `0.0` と設定するのは意味のないことではありません。

### 43.112 Zeroaxis

`x` 軸は `set xzeroaxis` によって描かれ、`unset xzeroaxis` によって削除されます。同様の `y`, `x2`, `y2`, `z` 軸用のコマンドが同様の働きをします。

書式:

```
set {x|x2|y|y2|z}zeroaxis { {linestyle | ls <line_style>}
    | { linetype | lt <line_type>}
    { linewidth | lw <line_width>}}
unset {x|x2|y|y2|z}zeroaxis
show {x|y|z}zeroaxis
```

デフォルトでは、これらのオプションはオフになっています。選択された 0 の軸は `<line_type>` の線の型と `<line_width>` の線の幅 (現在使用している出力形式がサポートしていれば) で、あるいはあらかじめ定義された `<line_style>` のスタイルで描かれます。

線の型を指定しなければ、軸は通常の軸の線の型 (型 0) で描かれます。

`set zeroaxis` は `set xzeroaxis`; `set yzeroaxis` と同等です。`z` 軸は、`set zzeroaxis` を使って独立に設定する必要があることに注意してください。

例:

`y=0` の軸を見るように簡単に書く場合:

```
set xzeroaxis
```

太い線にして、違った色、または点線パターンにしたい場合:

```
set xzeroaxis linetype 3 linewidth 2.5
```

### 43.113 Zlabel

このコマンドは `z` 軸の見出しを設定します。`set xlabel` を参照してください。

### 43.114 Zmtics

コマンド `set zmtics` は `z` 軸の目盛りを月に変更します。詳細は `set xmtics` を参照してください。

### 43.115 Zrange

コマンド `set zrange` は  $z$  軸方向に表示される範囲を設定します。このコマンドは `splot` にのみ有効で、`plot` では無視されます。詳細は `set xrange` を参照してください。

### 43.116 Ztics

コマンド `set ztics` は  $z$  軸の（見出し付けされる）大目盛りを制御します。詳細は `set xtics` を参照してください。

### 43.117 Clabel

このコマンドはカラー ボックス軸の見出しを設定します。`set xlabel` を参照してください。

### 43.118 Cbmtics

コマンド `set cbmtics` はカラー ボックス軸の目盛りの見出しを月に変換します。詳細は `set xmtics` を参照してください。

### 43.119 Cbrange

コマンド `set cbrange` は、スタイル `with pm3d`, `with image` や `with palette` などによって現在のパレット (palette) を使って色付けされる値の範囲を設定します。その範囲外の値に対しては、最も近い限界の値の色が使用されます。

カラー ボックス軸 (cb-軸) が `splot` で自動縮尺されている場合は、そのカラー ボックスの範囲は `zrange` が使われます。`splot ... pm3d|palette` で描画される点は、異なる `zrange` と `cbrange` を使うことでフィルタリングできます。

`set cbrange` の書式に関する詳細は `set xrange` を参照してください。`set palette` と `set colorbox` も参照してください。

### 43.120 Cbtics

コマンド `set cbtics` はカラー ボックス軸の（見出し付けされる）大目盛りを制御します。詳細は `set xtics` を参照してください。

## 44 Shell

`shell` コマンドは対話的なシェルを起動します。`gnuplot` に戻るには、VMS では `logout` を、Unix ならば `exit` もしくは END-OF-FILE 文字を、AmigaOS では `endcli` を、MS-DOS か OS/2 ならば `exit` を入力して下さい。

シェルコマンドを実行する方法は 2 つあります：コマンド `system` を使うか！（VMS では `$`）を使うか。前者は、コマンド文字列をパラメータとして取るので、他の `gnuplot` コマンドのどこでも使うことができますが、後者の書式は、その行にそのコマンドただ一つであることを要求します。これらの場合コマンドが終了するとすぐに制御は `gnuplot` に戻ってきます。例えば AmigaDOS, MS-DOS, OS/2 では、

```
! dir
```

または

```
system "dir"
```

とするとディレクトリの一覧を表示して `gnuplot` に戻ってきます。

`system` を使う他の例：

```
system "date"; set time; plot "a.dat"
print=1; if (print) replot; set out; system "lpr x.ps"
```

Atari では、! コマンドは、最初にシェルが既にロードされているか調べて、有効ならばそれを使います。例えば、gnuplot が gulam から起動されている場合に、これは実用的です。

## 45 Splot

splot は 3 次元描画のためのコマンドです（もちろんご存知でしょうが、実際にはその 2 次元への射影です）。それは関数、またはデータファイルから、plot コマンドとともに良く似た方法でその描画を作ります。

plot コマンドと共に仕様については plot を参照して下さい。ここではそれと異なるものだけ詳細に取り上げます。なお、plot の axes オプションは splot では使えないことに特に注意して下さい。

書式:

```
splot {<ranges>}
      <function> | "<datafile>" {datafile-modifiers}
      {<title-spec>} {with <style>}
      {, {definitions,} <function> ...}
```

ここで、関数 <function>、またはクオートでくくられたデータファイル名のどちらかが必要です。関数は、一本の数式、あるいは媒介変数モードでは 3 つの数式の組です。

デフォルトでは、splot は描画されるデータの下に完全な xy 面を描きます。z の一番下の目盛りと xy 平面の位置関係は set ticslevel で変更できます。splot の射影の向きは set view で制御できます。詳細は set view, set ticslevel を参照して下さい。

splot コマンドの範囲の指定の書式は plot の場合と同じです。媒介変数モードでなければ範囲は xrange, yrange, zrange の順であり、媒介変数モードでは urange, vrangle, xrange, yrangle, zrange の順です。

title オプションも plot と同じです。with も plot とほぼ同じですが、splot では利用可能な描画スタイルは lines, points, linespoints, dots, impulses に限られています。plot で使えるエラーバーの機能は splot にはありません。

datafile オプションにはさらに違いがあります。

show plot も参照してください。

### 45.1 Data-file

plot と同じように、ファイルに含まれる離散的なデータは、そのファイル名をクオートで囲んで指定することで描画できます。

書式:

```
splot '<file_name>' {binary <binary list>}
      {matrix}
      {index <index list>}
      {every <every list>}
      {using <using list>}
```

"" や "-" といった特別なファイル名も plot のときと同様に許されます。

手短にいうと、binary や matrix はそのデータが特別な形であることを、index は多重データ集合ファイルからどのデータ集合を選んで描画するかを、every は各データ集合からどのデータ行（部分集合）を選んで描画するかを、using は各データ行からどのように列を選ぶかを指定します。

index と every オプションは plot の場合と同じように振舞います。using も、using のリストが 2 つでなく 3 つ必要であるということを除いては同様です。

thru や smooth といった plot のオプションは splot では利用できません。しかし、cntrparam や dgrid3d が、制限されてはいますが平滑化のために用意されています。

データファイルの形式は、各点が  $(x,y,z)$  の 3 つ組である以外は、本質的に `plot` と同じです。もし一つの値だけが与えられれば、それは  $z$  として使われ、データブロック番号が  $y$  として、そして  $x$  はそのデータブロック内での番号が使われます。もし 2 つ、あるいは 4 つの値が与えられれば、`gnuplot` はその最後の値を `pm3d plot` での色の計算に使います。3 つの値は  $(x,y,z)$  の組と見なされます。他に値があれば、それは一般に誤差と見なされます。それは `fit` で使うことが可能です。

`splot` のデータファイルでは、1 行の空行はデータブロックの分離子です。`splot` は個々のデータブロックを、関数の  $y$ -孤立線と同じものとして扱います。1 行の空行で分離されている点同士は線分で結ばれることはありません。全てのデータブロックが全く同じ点の数を持つ場合、`gnuplot` はデータブロックを横断し、対応する点同士を結ぶ孤立線を描きます。これは "grid data" と呼ばれ、曲面の描画、等高線の描画 (`set contour`)、陰線処理 (`set hidden3d`) では、この形のデータであることが必要となります。`splot grid_data` も参照して下さい。

3 列の `splot` データにおいては、媒介変数モード (`parametric`) を指定することはもはや不要です。

#### 45.1.1 Binary matrix

Gnuplot は、`array`, `record`, `format`, `filetype` などの general バイナリ形式を特定するようなキーワードをつけてオプション `binary` を使うことで、matrix バイナリファイルを読み込むことができます。他の変換用の general バイナリキーワードは、matrix バイナリファイルにも適用できるでしょう。(詳細は `binary general` を参照。)

以前のバージョンでは、`gnuplot` は動的にバイナリデータかどうかを判断していましたが、現在は、ファイル名の後に `binary` キーワードを直接指定することが必要です。

単精度浮動小数の数値が次のように保存されています:

```
<N+1> <y0> <y1> <y2> ... <yN>
<x0> <z0,0> <z0,1> <z0,2> ... <z0,N>
<x1> <z1,0> <z1,1> <z1,2> ... <z1,N>
: : : : ... :
```

これらは以下のよう 3 つの数字の組に変換されます:

```
<x0> <y0> <z0,0>
<x0> <y1> <z0,1>
<x0> <y2> <z0,2>
:
<x0> <yN> <z0,N>

<x1> <y0> <z1,0>
<x1> <y1> <z1,1>
:

```

そして、これらの 3 つの数字の組は `gnuplot` の孤立線に変換され、その後 `gnuplot` が通常の方法で描画の残りを行います。

行列やベクトルの操作のサブルーチン (C による) が `binary.c` に用意されています。バイナリデータを書くルーチンは

```
int fwrite_matrix(file,m,nrl,nrl,ncl,nch,row_title,column_title)
```

です。これらのサブルーチンを使う例が `bf_test.c` として用意されていて、これはデモファイル `demo/binary.dem` 用に複数のバイナリファイルを生成します。

`index` キーワードは、ファイルフォーマットが 1 つのファイルにつき 1 つの曲面しか許さないため、サポートされません。`every` や `using` フィルタはサポートされます。`using` は、データがあたかも上の 3 つ組の形で読まれたかのように働きます。

`binary general` の項や以下も参照。

バイナリファイルの `splot` のデモ。

### 45.1.2 Example datafile

以下は 3 次元データファイルの描画の単純な一つの例です。

```
splot 'datafile.dat'
```

ここで、"datafile.dat" は以下を含むとします:

```
# The valley of the Gnu.
```

```
0 0 10
0 1 10
0 2 10
```

```
1 0 10
1 1 5
1 2 10
```

```
2 0 10
2 1 1
2 2 10
```

```
3 0 10
3 1 0
3 2 10
```

この "datafile.dat" は  $4 \times 3$  の格子 (それぞれ 3 点からなるブロックの 4 つの行) を定義することに注意して下さい。行 (データブロック) は 1 行の空行で区切られます。

$x$  の値はそれぞれのデータブロックの中で定数になっていることに注意して下さい。もし  $y$  を定数の値とし、隠線処理が有効な状態で描画すると、その曲面は裏返しで書かれることになります。

格子状データ (grid data) に対して、個々のデータブロック内で  $x$  の値を定数としておく必要はありませんし、同じ場所の  $y$  の値を同じ値に揃えておく必要もありません。gnuplot は個々のデータブロック内の点の数が等しいということを必要としているだけです。

しかし、等高線を導くのに用いられる曲面の網目は、対応する点を列的に選んで結ぶため、不揃いの格子データに対する曲面の描画への影響は予想できません。それはケースバイケースの原理でテストすべきでしょう。

### 45.1.3 Matrix\_ascii

```
{s}plot 'a.dat' matrix
```

のキーワード `matrix` (キーワード `binary` がその後ろにつかない場合) は、データが ascii 数字の配列形式で保存されていることを意味します。各ブロックの  $z$  の値は一行で一度に読みれます。すなわち、

```
z11 z12 z13 z14 ...
z21 z22 z23 z24 ...
z31 z32 z33 z34 ...
```

など。

3 次元では、配列の曲面描画の  $x$ 、 $y$  番号は、それぞれ配列の列番号と行番号 (0 から始まる) に対応します。軸の変換や伸縮も、 $x=\$1$ ,  $y=\$2$ ,  $z=\$3$  による 3 列のデータファイルに対する通常の方法と同様に行えます。例:

```
splot 'a.dat' matrix using (1+$1/100):(1+$2*10):3
```

空行、またはコメント行は配列データを終了させ、新たな曲面の網 (mesh) を開始します。いつものことですが、`splot` コマンドの `index` オプションを使ってファイル内の網を選択できます。

2 次元描画における配列データの縦、および横の行の描画の例については、`matrix` を参照してください。

#### 45.1.4 Matrix

データファイルは、ascii 形式かバイナリ形式の matrix (配列) フォーマットを使用することも可能で、binary または matrix binary がバイナリ形式を意味します。詳しくは、matrix ascii と matrix binary を参照してください。

splot での基本的な使用法:

```
splot 'a.dat' matrix
splot 'a.gpbin' {matrix} binary
```

splot での進んだ使用法:

```
splot 'a.dat' matrix using 1:2:3
splot 'a.gpbin' {matrix} binary using 1:2:3
```

これにより軸の座標と z の値を独立に変換できるようになります。

plot での使用法:

```
plot 'a.dat' matrix
plot 'a.dat' matrix using 1:3
plot 'a.gpbin' {matrix} binary using 1:3
```

これらは配列の行を描画し、using 2:3 とすれば配列の列を描画、using 1:2 は、点の座標を描画します (多分無意味です)。オプション every を適用することで明示的に行や列を指定できます。

例 – ASCII データファイルの配列の軸の拡大:

```
splot 'a.dat' matrix using (1+$1):(1+$2*10):3
```

例 – ASCII データファイルの配列の第 3 行の描画:

```
plot 'a.dat' matrix using 1:3 every 1:999:1:2
```

(行は 0 から数えられるので、3 ではなくて 2 を指定します)。

## 45.2 Grid data

3 次元描画のためのルーチンは、個々の網目の格子においては一つの標本点と一つのデータ点がある、という形の格子状データ用に設計されています。各データ点は、関数の値を評価すること (set isosample 参照)、またはデータファイルを読み込むこと (splot datafile 参照) によって生成されます。“孤立線”という言葉は関数に対しても、データに対してもその網目の線を表すものとして用いられます。網目は、必ずしも x, y に関する長方形でなくともよく、u, v で媒介変数表示されても構わないことに注意して下さい。set isosamples を参照して下さい。

しかし、gnuplot はそのような形式を必ずしも必要とはしません。例えば関数の場合は、samples は isosamples と違っていても構いません。すなわち、x-孤立線のうち、1 本の y-孤立線と交わらないものがいくつかあることがあります。データファイルの場合は、個々のデータブロックのばらついた点の個数が全て同じであれば、“孤立線は”はデータブロックの点を結び、“横断孤立線”は各データブロックの対応する点同士を結び、“曲面”を作ろうとします。どちらの場合でも、等高線、および隠線処理モードは点が意図したフォーマットであった場合とは違った描画を与えることになります。ばらつきのあるデータは set dgrid3d によって {異なる} 格子状データに変換することができます。

等高線に関するコードは、y-孤立線の点と、それに対応する隣の y-孤立線上の点の間の線分に沿っての z の張力を計測します。よって、x-孤立線に、y-孤立線との交点とはならないような標本点があるような曲面に対しては、splot の等高線はそのような標本点を無視することになります。以下を試してみて下さい:

```
set xrange [-pi/2:pi/2]; set yrang [-pi/2:pi/2]
set style function lp
set contour
set isosamples 10,10; set samples 10,10;
splot cos(x)*cos(y)
set samples 4,10; replot
set samples 10,4; replot
```

### 45.3 Splot overview

`splot` は点の集まりとして、あるいは、それらの点を結ぶことによって曲面を表示することができます。`plot` と同様に、点はデータファイルから読むこともできますし、指定された区間で関数の値を評価して得ることもできます。`set isosamples` を参照して下さい。曲面は、各点を線分で結ぶことで近似的に作られます。`set surface` を参照して下さい。そしてその場合曲面は `set hidden3d` で不透明にもできます。3次元曲面を眺める向きは、`set view` で変更できます。

さらに、格子上のデータ点に対しては、`splot` は同じ高さを持つ点を補間することができ (`set contour` 参照)、そしてそれを結んで等高線を描くことができます。さらに、その結び方には真直な線分や滑らかな線を使うことができます (`set cntrparam` 参照)。関数は、常に `set isosamples` と `set samples` で決定される格子状データとして評価されます。一方、ファイルのデータは、`data-file` に書かれているような格子状データフォーマットにするか、あるいは格子データを生成する (`set dgrid3d` 参照) ということをしなければそろはなりません。

等高線は曲面の上に表示することもできますし、底面に射影することもできます。底面への射影は、ファイルに書き出すこともでき、そしてそれを `plot` で再び読み込んで `plot` のより強い整形能力を生かすこともできます。

## 46 System

`system` はコマンドを実行するためにシェルを起動します。詳細は `help shell` を参照してください。

## 47 Test

このコマンドは、出力形式やパレットでどのような出力が行なえるかを画像でテストし表示します。

書式:

```
test {terminal | palette [rgb|rbg|grb|gbr|brg|bgr]}
```

`test` または `test terminal` は、現在使用中の出力形式 (`terminal`) で使える線の種類、点の種類、または有用なそして利用可能な他の描画を生成します。

`test palette` は、現在のカラーパレット (`palette`) で計算した  $R(z)$ ,  $G(z)$ ,  $B(z)$  ( $0 \leq z \leq 1$ ) の状態を画像で描画します。それは実際には、本来あなたが `show palette palette 256 float` として得るべき綺麗な描画です。追加のパラメータは、`r,g,b` の文字の組合せで、それにより描画する順番が決定されます。`set palette gray` で色々試してみてください。デフォルトの順番は `rgb` です。

## 48 Unset

コマンド `set` で設定したオプションは、それに対応した `unset` コマンドを使うことで、そのデフォルトの値に戻すことが可能です。

例:

```
set xtics mirror rotate by -45 0,10,100
...
unset xtics
```

## 49 Update

このコマンドは当てはめ (`fit`) のパラメータの現在の値を、初期値のファイルの形式で (`fit` の項で説明されている)、与えられたファイルに書き出します。これは、現在の値を、後で使うために、あるいは終了/中断した当てはめを再実行するために保存しておくのに有用です。

書式:

```
update <filename> {<filename>}
```

2番目のファイル名を指定すると、元のパラメータファイルは変更せずに2番目のファイルの方に更新された値を書き出します。

そうでなければ、指定したファイルが存在すれば gnuplot はそのファイル名に .old をつけてファイル名を変更し、指定したファイル名のファイルを新たに開き直します。つまり、"update 'fred'" とすると、それは "!rename fred fred.old; update 'fred.old' 'fred'" したことと同じことになります。["filename.ext" の 12 文字しか使えない DOS や他のシステムでは、"ext" が "old" になって "filename" は最初のファイルに関係するもの (多分どれかがすぐに分かるでしょう) が使われます。VMS では、ファイルのバージョン管理システムが使われるため、名前の変更は行なわれません。]

より詳しい情報に関しては fit を参照してください。

## Part III

# Terminal types

## 50 Terminal

gnuplot はとても多くの出力形式をサポートしています。これらは、適切な出力形式を、必要なら機能を変更する追加オプションをつけて選択することにより設定されます。set terminal を参照してください。この文書は、あなたのシステム上で初期設定およびインストールがなされなかっただけに利用できない出力形式についても記述されているかも知れません。インストールされた個々の gnuplot で、どの出力形式が有効なのかの一覧を見るには、オプションを何もつけずに 'set terminal' と打ってください。

### 50.1 Aed767

出力ドライバ aed512 と aed767 は AED グラフィック端末をサポートします。この 2 つのドライバは、水平方向の範囲のみが違っていて、それぞれ 512 ピクセルと 767 ピクセルです。垂直方向はいずれも 575 ピクセルです。これらのドライバにはオプションはありません。

### 50.2 Aifm

注意: この出力ドライバは時代遅れです。Adobe Illustrator はレベル 1 の PostScript ファイルを直接認識するので、set terminal post level1 を代わりに使うべきです。

Adobe Illustrator 3.0+ ドライバ aifm には、いくつかのオプションを設定できます。

書式:

```
set terminal aifm {<color>} {"<fontname>"} {<fontsize>}
```

<color> は color か monochrome、"<fontname>" は正式な PostScript フォント名、<fontsize> はフォントサイズを PostScript のポイント数で与えます。これらは set size コマンドでスケール変換される前のものです。default を選択すると、全てのオプションがデフォルトの値: monochrome, "Times-Roman", 14pt, に設定されます。

AI は複数のページをサポートしていないので、複数のグラフは直接他のグラフの上に書かれます。しかし、各グラフは個別にグループ化されていて、AI 上で簡単にそれらを分離することが出来ます (単にそれらを拾い上げて移動させるだけです)。

例:

```
set term aifm
set term aifm 22
set size 0.7,1.4; set term aifm color "Times-Roman" 14
```

### 50.3 Amiga

Commodore (コモドール) Amiga コンピュータ用の amiga 出力ドライバは、スクリーン (default)、または、Kickstart 3.0 またはそれ以上のものがインストールされていればカレントパブリックスクリーン上のウィンドウのどちらでも描画できます。フォントとそのサイズも選択できます。

書式:

```
set terminal amiga {screen | window} {"<fontname>"} {<fontsize>}
```

デフォルトのフォントは 8 ポイントの "topaz" です。

screen オプションは、仮想スクリーンを使うので、スクリーンより大きいグラフを作ることも可能です。

### 50.4 Apollo

apollo 出力ドライバは、Apollo グラフィックプリミティブリソースをサポートしウィンドウのリサイズ後のリスケーリングもサポートしています。オプションはありません。

固定サイズのウィンドウが必要ならば、gpr 出力を代わりに使用してください。

### 50.5 Aqua

この出力形式は Mac OS X 上の表示に関する AquaTerm.app に依存しています。

書式:

```
set terminal aqua {<n>} {title "<wintitle>"} {size <x> <y>}
    {font "<fontname>{,<fontsize>}"}
    {{no}enhanced} {solid|dashed} {dl <dashlength>}}
```

<n> は描画するウィンドウの番号 (デフォルトでは 0) <wintitle> はタイトルバーに表示される名前 (デフォルトは "Figure <n>"), <x> <y> は描画サイズ (デフォルトは 846x594 pt = 11.75x8.25 インチ) です。

使用されるフォントは <fontname> で指定し (デフォルトは "Times-Roman")、フォントサイズは <fontsize> で設定します (デフォルトは 14.0 pt)。古い書式 {fname "<fontname>"} {fsiz <fontsize>} もとりあえずサポートしています。

aqua 出力形式は、拡張テキスト出力モード (enhanced text mode; enhanced 参照) を、重ね書き以外はサポートしています。フォントの使用はシステムで有効なフォントに制限されています。文字エンコーディングは set encoding で選択できますが、現在は iso\_latin\_1, iso\_latin\_2, cp1250 と default (= UTF8) がサポートされています。

曲線は、実線か点線 (デフォルトは実線) のいずれかで描画でき、点線の間隔は倍率 <dashlength> (>0) で変更できます。

### 50.6 Atari ST (via AES)

atari 出力形式には、画面の色と文字のサイズを設定するオプションがあります。

書式:

```
set terminal atari {<fontsize>} {<col0> <col1> ... <col15>}
```

色を指定する場合、必ず文字サイズを指定しなければいけません。個々の (16 色までの) 色は、3 枚の 16 進数で与え、各枚は赤、緑、青 (RGB) の強度を (この順で) あらわします。0-15 の範囲は、実際にスクリーンが持つカラーの範囲に調整されます。例えば通常の ST スクリーンでは、奇数と偶数の強度は同じになります。

例:

```
set terminal atari 4      # 小さい (6x6) フォント
set terminal atari 6 0    # 黒背景に白の白黒画面
set terminal atari 13 0 fff f0 f ff f0f
    # 最初の 7 色を黒、白、赤、緑、青、シアン、紫とし、
    # 大きいフォント (8x16) を使用
```

また、環境変数 GNCOLORS が設定されていれば、それもオプション文字列として解釈されますが、明示的なターミナルオプションの方が優先されます。

## 50.7 Be

gnuplot は X サーバを利用する be 出力ドライバを提供します。この出力形式は、環境変数 DISPLAY がセットされているか、環境変数 TERM が xterm にセットされているか、またはコマンドラインオプションとして -display が使われていれば起動時に自動的に選択されます。

書式:

```
set terminal be {reset} {<n>}
```

複数のグラフ描画ウィンドウをサポートしています。set terminal be <n> は番号 n のウィンドウに出力します。n>0 の場合、その番号はウィンドウタイトルとアイコン名に gplt <n> として付けられます。現在のウィンドウはカーソル記号の変化で区別できます（デフォルトカーソルから十字カーソルへ）。

gnuplot ドライバが別な出力ドライバに変更されても、描画ウィンドウは開いたままになります。描画ウィンドウは、そのウィンドウにカーソルを置いて文字 q を押すか、ウィンドウマネージャのメニューの close を選択すれば閉じることができます。reset を実行すれば全てのウィンドウを一度に閉じれます。それは実際にウィンドウを管理している子プロセスを終了します（もし -persist が指定されていなければ）。描画ウィンドウは -persist オプションが与えられていなければ、対話の終了時に自動的に閉じられます。描画サイズとアスペクト比は、gnuplot のウィンドウをリサイズすることでも変更できます。

線の幅と点のサイズは gnuplot の set linestyle で変更可能です。

出力ドライバ be に関しては、gnuplot は（起動時に）、コマンドライン、または設定ファイルから、geometry や font, name などの通常の X Toolkit オプションやリソースの指定を受け付けます。それらのオプションについては X(1) マニュアルページ（やそれと同等のもの）を参照してください。

他にも be 出力形式用の多くの gnuplot のオプションがあります。これらは gnuplot を呼ぶときにコマンドラインオプションとして指定するか、または設定ファイル ".Xdefaults" のリソースとして指定できます。これらは起動時に設定されるので、gnuplot 実行時には変更できません。

### 50.7.1 Command-line\_options

X Toolkit オプションに加え、以下のオプションが gnuplot の立ち上げ時のコマンドラインで、またはユーザのファイル ".Xdefaults" 内のリソースとして指定できます:

'-mono'	カラーディスプレイ上で強制的に白黒描画
'-gray'	グレイスケールまたはカラーディスプレイ上でのグレイスケール描画 (デフォルトではグレイスケールディスプレイは白黒描画を受け付ける)
'-clear'	新しい描画を表示する前に（瞬間的に）画面を消去
'-raise'	各描画後に描画ウィンドウを最前面へ出す
'-noraise'	各描画後に描画ウィンドウを最前面へ出すことはしない
'-persist'	gnuplot プログラム終了後も描画ウィンドウを残す

上記のオプション、はコマンドラインでの指定書式で、".Xdefaults" にリソースとして指定するときは異なる書式を使います。

例:

```
gnuplot*gray: on
```

gnuplot は描画スタイル points で描画する点のサイズの制御にも、コマンドラインオプション (-pointsize <v>) とリソース (gnuplot\*pointsize: <v>) を提供しています。値 v は点のサイズの拡大率として使われる実数値 ( $0 < v \leq 10$ ) で、例えば -pointsize 2 はデフォルトのサイズの 2 倍、-pointsize 0.5 は普通のサイズの半分の点が使われます。

### 50.7.2 Monochrome\_options

白黒ディスプレイに対しては gnuplot は描画色 (foreground) も背景色 (background) も与えません。デフォルトでは背景は白、描画は黒です。-rv や gnuplot\*reverseVideo: on の場合には背景が黒で描画は白になります。

### 50.7.3 Color\_resources

カラーディスプレイに対しては、gnuplot は以下のリソース (ここではそのデフォルトの値を示します)、または白黒階調 (greyscale) のリソースを参照します。リソースの値はシステム上の BE rgb.txt ファイルに書かれている色名、または 16 進の色指定 (BE のマニュアルを参照) か、色名と強度 (0 から 1 の間の値) をコンマで区切った値を使用できます。例えば blue, 0.5 は半分の強度の青、を意味します。

```
gnuplot*background: white
gnuplot*textColor: black
gnuplot*borderColor: black
gnuplot*axisColor: black
gnuplot*line1Color: red
gnuplot*line2Color: green
gnuplot*line3Color: blue
gnuplot*line4Color: magenta
gnuplot*line5Color: cyan
gnuplot*line6Color: sienna
gnuplot*line7Color: orange
gnuplot*line8Color: coral
```

これらに関するコマンドラインの書式は、例えば以下の通りです。

例:

```
gnuplot -background coral
```

### 50.7.4 Grayscale\_resources

-gray を選択すると、gnuplot は、グレイスケールまたはカラーディスプレイに対して、以下のリソースを参照します (ここではそのデフォルトの値を示します)。デフォルトの背景色は黒であることに注意してください。

```
gnuplot*background: black
gnuplot{textGray: white
gnuplot*borderGray: gray50
gnuplot*axisGray: gray50
gnuplot*line1Gray: gray100
gnuplot*line2Gray: gray60
gnuplot*line3Gray: gray80
gnuplot*line4Gray: gray40
gnuplot*line5Gray: gray90
gnuplot*line6Gray: gray50
gnuplot*line7Gray: gray70
gnuplot*line8Gray: gray30}
```

### 50.7.5 Line\_resources

gnuplot は描画の線の幅 (ピクセル単位) の設定のために以下のリソースを参照します (ここではそのデフォルトの値を示します)。0 または 1 は最小の線幅の 1 ピクセル幅を意味します。2 または 3 の値によってグラフの外観を改善できる場合もあるでしょう。

```
gnuplot*borderWidth: 2
gnuplot*axisWidth: 0
gnuplot*line1Width: 0
gnuplot*line2Width: 0
gnuplot*line3Width: 0
gnuplot*line4Width: 0
gnuplot*line5Width: 0
gnuplot*line6Width: 0
gnuplot*line7Width: 0
gnuplot*line8Width: 0
```

`gnuplot` は線描画で使用する点線の形式の設定用に以下のリソースを参照します。0 は実線を意味します。2 行の 10 進数 `jk` (`j` と `k` は 1 から 9 までの値) は、`j` 個のピクセルの描画に `k` 個の空白のピクセルが続くパターンの繰り返しからなる点線を意味します。例えば '`16`' は 1 個のピクセルの後に 6 つの空白が続くパターンの点線になります。さらに、4 行の 10 進数でより詳細なピクセルと空白の列のパターンを指定できます。例えば、'`4441`' は 4 つのピクセル、4 つの空白、4 つのピクセル、1 つの空白のパターンを意味します。以下のデフォルトのリソース値は、白黒ディスプレイ、あるいはカラー／白黒階調 (grayscale) ディスプレイ上の白黒描画における値です。カラーディスプレイではそれらのデフォルトの値はほとんど 0 (実線) で、`axisDashes` のみがデフォルトで '`16`' の点線となっています。

```
gnuplot*borderDashes: 0
gnuplot*axisDashes: 16
gnuplot*line1Dashes: 0
gnuplot*line2Dashes: 42
gnuplot*line3Dashes: 13
gnuplot*line4Dashes: 44
gnuplot*line5Dashes: 15
gnuplot*line6Dashes: 4441
gnuplot*line7Dashes: 42
gnuplot*line8Dashes: 13
```

## 50.8 Cgi

`cgi` と `hcgi` 出力ドライバは SCO CGI ドライバをサポートします。`hcgi` はプリント用で、環境変数 `CGIPRNT` が設定されている必要があります。`cgi` はディスプレイ、またはハードコピーに使え、環境変数 `CGIDISP` が設定されていればディスプレイを使用し、そうでなければ `CGIPRNT` を使用します。

これらの出力ドライバにはオプションはありません。

## 50.9 Cgm

`cgm` ドライバは CGM 出力 (Computer Graphics Metafile Version 1) を生成します。このファイルフォーマットは ANSI 規格書 X3.122-1986 "Computer Graphics - Metafile for the Storage and Transfer of Picture Description Information" で定義されているものの一部分です。`cgm` には多くのオプションがあります。

書式:

```
set terminal cgm {<mode>} {<color>} {<rotation>} {solid | dashed}
    {width <plot_width>} {linewidth <line_width>}
    {"<font>"} {<fontsize>}
    {<color0> <color1> <color2> ...}
```

ここで、`<mode>` は `landscape`, `portrait`, `default` のいずれか; `<color>` は `color`, `monochrome` のいずれか; `<rotation>` は `rotate`, `norotate` のいずれか; `solid` は全ての曲線を実線で描き、どんな点線パターンも塗りつぶします; `<plot_width>` はポイント単位でのグラフの仮定されている幅; `<line_width>` はポイント単位での線幅 (デフォルトは 1); `<font>` はフォントの名前; そして `<fontsize>` はポイント単位でのフォントのサイズ (デフォルトは 12) です。

デフォルトでは cgm は Y 軸の見出しに 90 度回転した文字を使用します。

最初の 6 つのオプションはどの順番で指定しても構いません。default を選択すると、全てのオプションをそのデフォルトの値にします。

各カラー (color) は 'xrrggb' の形式です。ここで、x は文字 'x' そのもの、'rrggb' は 16 進数での赤、緑、青の成分です。例えば 'x00ff00' は緑になります。背景色が最初に設定され、以下は描画色となります。

例:

```
set terminal cgm landscape color rotate dashed width 432 \
    linewidth 1 'Helvetica Bold' 12      # デフォルト
set terminal cgm linewidth 2 14 # やや広い線とやや大きいフォント
set terminal cgm portrait "Times Italic" 12
set terminal cgm color solid    # 面倒な点線など消えてしまえ !
```

### 50.9.1 Font

CGM (Computer Graphics Metafile) ファイルの最初の部分、メタファイルの記述部分には、フォントリスト (font table) が含まれています。画像の本体部では、フォントはこのリストにある番号で指定されます。デフォルトではこのドライバは以下の 35 個のフォントリストを生成し、さらにこのリストの Helvetica, Times, Courier の各フォントの italic を oblique で置き換えたもの、およびその逆による 6 つの追加のフォントが含まれます (Microsoft Office と Corel Draw CGM の import フィルタは italic と oblique を同じものとして扱うからです)。

CGM fonts
Helvetica
Helvetica Bold
Helvetica Oblique
Helvetica Bold Oblique
Times Roman
Times Bold
Times Italic
Times Bold Italic
Courier
Courier Bold
Courier Oblique
Courier Bold Oblique
Symbol
Hershey/Cartographic_Roman
Hershey/Cartographic_Greek
Hershey/Simplex_Roman
Hershey/Simplex_Greek
Hershey/Simplex_Script
Hershey/Complex_Roman
Hershey/Complex_Greek
Hershey/Complex_Script
Hershey/Complex_Italic
Hershey/Complex_Cyrillic
Hershey/Duplex_Roman
Hershey/Triplex_Roman
Hershey/Triplex_Italic
Hershey/Gothic_German
Hershey/Gothic_English
Hershey/Gothic_Italian
Hershey/Symbol_Set_1
Hershey/Symbol_Set_2
Hershey/Symbol_Math
ZapfDingbats
Script

15

これらのフォントの最初の 13 個は WebCGM で要求されているものです。Microsoft Office の CGM import フィルタはその 13 個の標準フォントと 'ZapfDingbats' と 'Script' をサポートしています。しかし、そのスクリプト (script) フォントは '15' という名前でしかアクセスできません。Microsoft の import フィルタの font の置き換えに関するより詳しい情報については、

```
C:\Program Files\Microsoft Office\Office\Cgmimp32.hlp
```

のヘルプファイル、または

```
C:\Program Files\Common Files\Microsoft Shared\Grphflt\Cgmimp32.cfg
```

の設定ファイルなどをチェックしてください。

set term コマンドでデフォルトのフォントリストにないフォント名を指定することも可能です。その場合、その指定したフォントが最初に現われる新しいフォントリストが作られます。そのフォント名に関して、スペル、単語の先頭の大文字化やどこにスペースが入るかなどが、作られる CGM ファイルを読むアプリケーションにとって適切なものであるかをちゃんと確認する必要があります。(gnuplot と任意の MIL-D-28003A 準拠アプリケーションは、フォント名の大文字小文字の違いは無視します。) 新しいフォントをいくつも追加したい場合は、set term コマンドを繰り返し使用してください。

例:

```
set terminal cgm 'Old English'
set terminal cgm 'Tengwar'
```

```
set terminal cgm 'Arabic'
set output 'myfile.cgm'
plot ...
set output
```

set label コマンドでは新しいフォントを導入することはできません。

### 50.9.2 Fontsize

フォントは、ページが 6 インチの幅であると仮定して縮尺されます。size コマンドでページの縦横比が変更されていたり、CGM ファイルが異なる幅に変換されている場合、結果としてフォントのサイズも拡大されたり縮小されたりすることになります。仮定されている幅を変更するには、width オプションを使用してください。

### 50.9.3 Linewidth

linewidth オプションは線の幅をポイント単位 (pt) で設定します。デフォルトの幅は 1 pt です。fontsize や width オプションのところで説明されているように、ページの実際の幅によってその縮尺は影響を受けます。

### 50.9.4 Rotate

norotate オプションはテキストの回転をしないようにします。例えば Word for Windows 6.0c 用の CGM 入力フィルタは回転された文字列を受け付けますが、Word に付属する DRAW エディタはそれを受け付けることができず、グラフを編集すると (例えば曲線に見出しがつける)、全ての回転された文字列は水平方向になって保存されてしまい、Y 軸の見出しがクリップされる境界線を越えてしまうでしょう。norotate オプションを使えば、見栄えの良くない場所から Y 軸の見出しが始まっていますが、編集によってダメージを受けることはなくなります。rotate オプションはデフォルトの挙動を保証します。

### 50.9.5 Solid

solid オプションは描画の点線の線描画スタイルを無効するのに使います。これは、カラーが有効である場合、また点線にすることでグラフが見にくくなる場合に有用でしょう。dashed オプションはデフォルトの挙動を保証し、この場合個々のグラフに異なる点線のパターンが与えられます。

### 50.9.6 Size

CGM グラフのデフォルトのサイズは、横置き (landscape) では幅 32599、縦 23457、縦置き (portrait) では幅 23457、縦 32599 です。

### 50.9.7 Width

CGM ファイルの全ての長さは抽象的な単位を持ち、そのファイルを読むアプリケーションが最終的なグラフのサイズを決定します。デフォルトでは最終的なグラフの幅は 6 インチ (15.24 cm) であると仮定されています。この幅は正しいフォントサイズを計算するのに使われ、width オプションで変更できます。キーワード width の後に幅をポイント単位で指定します。(ここで、ポイントは PostScript と同様 1/72 インチを意味します。この単位は TeX では"big point" と呼ばれています。) 他の単位から変換するには、gnuplot の数式が使えます。

例:

```
set terminal cgm width 432          # デフォルト
set terminal cgm width 6*72         # 上と同じ値
set terminal cgm width 10/2.54*72   # 10 cm の幅
```

### 50.9.8 Nofontlist

デフォルトのフォントリスト (font table) は WebCGM で勧告されているフォントを含んでいて、これは Microsoft Office と Corel Draw の CGM (Computer Graphics Metafile) 入力フィルタに適合しています。他のアプリケーションは異なるフォント、あるいは異なるフォント名を使用するかも知れませんが、それはマニュアルには書かれていません。そのための回避策として、オプション `nofontlist` を使用して CGM ファイルからフォントリストを削除するという方法があります。この場合、読み込んだアプリケーションはデフォルトのフォントリストを使用するでしょう。gnuplot はその場合でもフォント番号の選択のために自分のデフォルトのフォントリストを使用します。よって、'Helvetica' が 1 番になり、それがあなたの使用的するアプリケーションのデフォルトフォントリストの最初のものになります。'Helvetica Bold' がそのフォントリストの 2 番目のフォントに対応し、他も同様となります。

以前の `winword6` オプションは今は `nofontlist` の同意語に成り下がりました。`winword6` オプションが回避策として使われていた、色やフォントリストに関する問題は、結局 gnuplot のバグであるとわかり、それは今では修正されています。

### 50.10 Corel

`corel` 出力形式は CorelDraw 用の出力です。

書式:

```
set terminal corel { default
    | {monochrome | color
        {"<font>" {<fontsize>
            {<xsize> <ysize> {<linewidth> }}}}}
```

ここで、フォントサイズ (`fontsize`) と線幅 (`linewidth`) はポイント単位、横幅 (`xsize`) と縦幅 (`ysize`) はインチ単位です。デフォルトの値はそれぞれ、`monochrome`, "SwitzerlandLight", 22, 8.2, 10, 1.2 です。

### 50.11 Debug

このドライバは gnuplot のデバッグのために提供されているものです。おそらくソースコードを修正するユーザのみが使用するものでしょう。

### 50.12 Dospc

`dospc` ドライバは PC の任意のグラフィックボードをサポートし、それらを自動検出します。これは、`gcc`, または Zortec C/C++ コンパイラを使っていないときのみ利用可能です。

### 50.13 Dumb

ダム端末 (`dumb`) ドライバにはサイズの指定と改行制御用のオプションがあります。

書式:

```
set terminal dumb {[no]feed} {<xsize> <ysize>}
    {[no]enhanced}
```

`<xsize>`, `<ysize>` はダム端末 のサイズを設定し、デフォルトはそれぞれ 79, 24 となっています。最後の改行は、`feed` オプションが設定されている場合のみ出力されます。

例:

```
set term dumb nofeed
set term dumb 79 49 # VGA screen---何故こうしたい人がいるんだろう ?
```

## 50.14 Dxf

dxf ドライバは、AutoCad (リリース 10.x) に取り込むことができる画像を生成します。このドライバ自身にはオプションはありませんが、描画に関するいくつかの特徴は他の方法で変更できます。デフォルトの大きさは AutoCad の単位での 120x80 で、これは set size で変更できます。dxf は 7 色(白、赤、黄、緑、水色、青、赤紫)を使いますが、これを変更するにはドライバソースファイルを修正する必要があります。白黒の出力装置を使う場合、それらの色は線の太さの違いで表現されます。詳細は AutoCad の印刷/プロッタ出力コマンドに関する記述を参照してください。

## 50.15 Dxy800a

このドライバは Roland DXY800A プロッタをサポートします。オプションはありません。

## 50.16 Eepic

eepic ドライバは LaTeX picture 環境を拡張するものをサポートします。これは latex ドライバに代わる別な選択肢です。

このドライバによる出力は、LaTeX 用の "eepic.sty" マクロパッケージと共に使われることを仮定しています。それを使うには、"eepic.sty" と "epic.sty"、および "tpic" \special 命令群をサポートするプリントドライバが必要です。もし、あなたの使うプリントドライバがそれらの "tpic" \special 命令をサポートしていない場合でも、"eepicemu.sty" を使うことでそれらのうちのいくつかを使えるようになります。dvips と dvipdfm は "tpic" \special をサポートしています。

書式:

```
set terminal eepic {color, dashed, rotate, small, tiny, default, <fontsize>}
```

オプション: オプションは任意の順番で与えることができます。'color' は gnuplot に \color{...} コマンドを生成させ、それによりグラフをカラーにします。このオプションを使用する場合は、latex 文書のプリアンブルに \usepackage{color} を入れる必要があります。'dashed' は線種に点線を使用することを許可します。このオプションを指定しないと、色々な太さの実線のみが使われます。'dashed' と 'color' は一方のみが意味を持ち、'color' が指定された場合、'dashed' は無視されます。'rotate' は本当に回転 (90 度) された文字列を使用するようになります。指定しない場合は、1 文字 1 文字、上に積み上げていく方法で回転された文字列を作ります。このオプションを使う場合は、\usepackage{graphicx} をプリアンブルに入れる必要があります。'small' は point スタイルでのグラフ描画の印として \scriptsize の記号を使用します (多分これは TeX ではダメで、LaTeX2e でしか使えないでしょう)。デフォルトでは標準の数式のサイズを使用します。'tiny' は、それに \scriptscriptstyle の記号を使用します。'default' は全てのオプションをデフォルトの値にリセットします。デフォルトは、color はなし、dashed line はなし、疑似回転 (積み上げ) 文字列の使用、大きなサイズの記号の使用、です。<fontsize> は picture 環境内でのフォントサイズを指定する数字です。単位は pt (ポイント) で、10 pt はほぼ 3.5 mm です。フォントサイズを指定しない場合、全てのグラフ内の文字は \footnotesize に設定されます。

注意: 文字 # (およびその他 (La)TeX で特別な意味を持つその他の文字) を \\ (バックスラッシュ 2 つ) でエスケープすることを忘れないでください。グラフの角が近すぎると点線は実線のようになります。(これが tpic specials の一般的な問題なのか、eepic.sty や dvips/dvipdfm のバグが原因なのかは私にはわかりません。) デフォルトの eepic グラフの大きさは 5x3 インチで、これは 'set size a,b' で縮尺可能です。数ある中で、点 (point) は、LaTeX のコマンド "\Diamond", "\Box" などを使って描かれます。これらのコマンドは現在は LaTeX2e のコアには存在せず latexsym パッケージに含まれていますが、このパッケージ基本配布の一部であり、よって多くの LaTeX のシステムの一部になっています。このパッケージを使うことを忘れないでください。 latexsym の代わりに amssymb パッケージを使うことも可能です。LaTeX に関する全てのドライバは文字列の配置の制御に特別な方法を提供します: '{' で始まる文字列は、'}' で閉じる必要がありますが、その文字列全体が水平方向にも垂直方向にもセンタリングされます。 '[' で始まる文字列の場合は、位置の指定をする文字列 (t,b,l,r のうち 2 つまで) が続き、次に ']'、文字列本体、で最後に '}' としますが、この文字列は LaTeX が LR-box として整形します。'\rule{}{}' を使えばさらに良い位置合わせが可能でしょう。

例: set term eepic

は、グラフを picture 環境に含まれた eepic マクロとして出力します。  
そのファイルを LaTeX 文書に \input で取り込んでください。

```
set term eepic color tiny rotate 8
```

eepic マクロを \color マクロ、point 印は \scriptsize の大きさ、本当に回転された文字の使用、および全ての文字を 8pt にセットして出力します。

見出しの位置合わせに関して: gnuplot のデフォルト (大抵それなりになるが、そうでないこともある):

```
set title '\LaTeX\ -- $ \gamma'
```

水平方向にも垂直方向にもセンタリング:

```
set label '\LaTeX\ -- $ \gamma' at 0,0
```

位置を明示的に指定 (上に合わせる):

```
set xlabel '[t]\LaTeX\ -- $ \gamma'
```

他の見出し - 目盛りの長い見出しに対する見積り:

```
set ylabel '[r]\LaTeX\ -- $ \gamma $\rule{7mm}{0pt}'
```

## 50.17 Emf

emf ドライバは EMF (Enhanced Metafile Format) ファイルを生成します。このファイルフォーマットは MS Win32 システムの標準メタファイルです。

書式:

```
set terminal emf {<color>} {solid | dashed} {linewidth <X>}
               {size XX,YY}
               {"<fontname>"} {<fontsize>}      #古い書式
               {font "<fontname>,<fontsize>"} #新しい書式
```

<color> は color (カラー) か monochrome (白黒)、solid は全ての線を実線で書き、点線のパターンは塗りつぶします。linewidth <factor> は全ての線幅をここで指定する値倍にします。<font> はフォント名、<fontsize> はポイント単位でのフォントの大きさです。

出力画像の形式的な (名ばかりの) サイズは、デフォルトでは適当な単位での 1024x768 になっています。オプション size を使って別な形式的なサイズを指定できます。

最初の 2 つのオプションはどちらが先でも構いません。default を選択すると全てのオプションがそのデフォルトの値になります。

例:

```
set terminal emf 'Times Roman Italic' 12
set terminal emf color solid    # 面倒な点線など消えてしまえ !
```

## 50.18 Emxvga

emxvga, emxvesa, vgal の各ドライバはそれぞれ SVGA, vesa SVGA, VGA グラフィックボードの PC をサポートします。これらは DOS、または OS/2 上で、"emx-gcc" でコンパイルされたることを意図しています。これらにはさらに VESA パッケージと SVGAKIT が必要です。これらは Johannes Martin (JMARTIN@GOOFY.ZDV.UNI-MAINZ.DE) が保守し David J. Liu (liu@phri.nyu.edu) が拡張しているライブラリです。

書式:

```
set terminal emxvga
set terminal emxvesa {vesa-mode}
set terminal vgal
```

唯一のオプションは emxvesa に対する vesa のモードで、デフォルトではそれは G640x480x256 となっています。

## 50.19 Epslatex

`epslatex` ドライバは LaTeX で処理すべき出力を生成します。

書式:

```
set terminal epslatex {default}
set terminal epslatex {standalone | input}
{oldstyle | newstyle}
{level1 | leveldefault}
{color | colour | monochrome}
{solid | dashed}
{dashlength | dl <DL>}
{linewidth | lw <LW>}
{rounded | butt}
{palfuncparam <samples>{,<maxdeviation>}}
{size <XX>{unit},<YY>{unit}}
{header <header> | noheader}
{blacktext | colortext | colourtext}
{{font} "fontname{,fontsize}" <fontsize>}}
```

`epslatex` 出力形式は、文字列を PostScript コードに含ませる代わりに LaTeX ファイルに移すことを除けば `terminal postscript eps` 同様に描画します。よって、`postscript terminal` と多くのオプションが共通です。

version 4.0 から 4.2 にかけて、`postscript` 出力形式とのより良い互換性のために `epslatex` 出力形式のデフォルトの状態にいくつかの変更が行なわれて来ました。描画サイズは  $5 \times 3$  インチから  $5 \times 3.5$  インチへと変更され、文字幅は従来はフォントサイズの 50% と評価していましたが、現在は 60% と見なしています。より多くの Postscript の線種や記号も使われます。以前の状態にほぼ等しい状態にするにはオプション `oldstyle` を指定してください。(実際にはごくわずかな違いが残ります: 記号のサイズがわずかに違い、目盛り刻 (tics) は従来の半分になっていますがそれは `set tics scale` で変更できます。そして矢 (arrow) に関しては `postscript` 出力形式で使える全ての機能が利用できます。)

以下のようなエラーメッセージが出た場合:

```
"Can't find PostScript prologue file ... "
```

`postscript prologue` の項目を見て、その指示に従ってください。

オプション `color` はカラーを有効にし、`monochrome` は各要素を黒と白描画します。さらに、`monochrome` は灰色の palette も使用しますが、これは、明示的に `colorspec` で指定された部品の色を変更しません。

`solid` は全てのグラフの点線のパターンを実線で上書きします。`dashlength` または `dl` は点線の線分の長さを `<DL>` (0 より大きい実数) に設定し、`linewidth` または `lw` は全ての線の幅を `<LW>` に設定します。

デフォルトでは、生成される PostScript コードは、特にフィルタリングや `filledcurves` のようなでこぼこな領域のパターン塗りつぶしにおいて、PostScript Level 2 として紹介されている言語機能を使います。PostScript Level 2 の機能は条件的に保護されていて、PostScript Level 1 のインタープリタがエラーを出さず、むしろメッセージか PostScript Level 1 による近似であることを表示するようになっています。`level1` オプションは、これらの機能を近似する PostScript Level 1 で代用し、PostScript Level 2 コードを一切使用しません。これは古いプリンタや、Adobe Illustrator の古いバージョンなどで必要になるかもしれません。このフラグ `level1` は出力された PostScript ファイルのある一行を手で編集することで、後から強制的に PostScript Level 1 機能を ON/OFF にすることもできます。`level2` のコードが含まれている場合、上の機能は現われないか、このフラグがセットされた場合、あるいは PostScript インタプリタプログラムが `level2` 以上の PostScript を解釈するとは言わなかった場合に警告文に置き換わります。

`rounded` は、線の端や接合部を丸くし、デフォルトの `butt` は尖った端と角張った接合部を使用します。`palfuncparam` は `set palette functions` から出力の傾きをどのようにコード化するかを制御します。解析的な色の成分関数 (`set palatte functions` で設定される) は、`postscript` 出力では傾きの線形補完を用いてコード化されます: まず色の成分関数が `<samples>` 個の点で標本化され、そしてそれらの点は、結果として線形補完との偏差が `<maxdeviation>` 以内に収まるように削除されます。ほとんど全ての有効な

パレットで、デフォルトの `<samples> =2000` と `<maxdeviation>=0.003` の値をそのまま使うのが良いでしょう。

PostScript 出力のデフォルトの大きさは 10 インチ x 7 インチです。EPS 出力のデフォルトの大きさは 5 x 3.5 インチです。オプション `size` はこれらをユーザが指定したものに変更します。デフォルトでは X と Y のサイズの単位はインチとみなされますが、他の単位(現在は cm のみ)も使うことはできます。描画の BoundingBox (PostScript ファイルの外枠) は、サイズが変更された画像を丁度含むように正しく設定されます。スクリーン座標は、オプション `size` で指定された描画枠の全体が 0.0 から 1.0 になります。注意: これは、以前は、出力形式での設定よりも、コマンド `set size` で設定した方がいい、と言っていたことの変更を意味します。以前の方法では BoundingBox は変更されずに残ってしまい、スクリーン座標が実際の描画の限界に対応していませんでした。

`blacktext` は、たとえカラー モードでも全ての文字列を黒で書きます。

`epslatex` ドライバは文字列の配置の制御に特別な方法を提供します: (a) '{' で始まる文字列は、'}' で閉じる必要がありますが、その文字列全体が LaTeX によって水平方向にも垂直方向にもセンタリングされます。 (b) '[' で始まる文字列の場合は、位置の指定をする文字列 (t,b,l,r,c のうち 2 つまで) が続き、次に ']{'、文字列本体、で最後に '}' としますが、この文字列は LaTeX が LR-box として整形します。`\rule{}{}` を使えばさらに良い位置合わせが可能でしょう。`pslatex` ドライバに関する説明も参照してください。複数行の見出しを作成するには `\shortstack` を使用してください。例えば、

```
set ylabel '[r]{\shortstack{first line \\ second line}}'
```

`set label` コマンドのオプション `back` は使えますが、他の出力形式のものとは少し違っています。`front` の場合の見出しが他の全ての要素の上に出力されるのに対して、`back` を使った見出しが他の全ての要素の下に出力されます。

このドライバは 2 つの別のファイルを作ります。1 つは図の eps の部分で、もう一つは LaTeX の部分です。LaTeX ファイルの名前は、`set output` コマンドのものが使われ、eps ファイルの名前はその拡張子(通常 `.tex`)を `.eps` に置き換えたものになります。出力ファイルを指定しなければ LaTeX 出力は行なわれません! `multiplot` モード以外では、次の描画を行なう前にその出力ファイルをクローズするのを忘れないでください。

LaTeX の文書で図を取り込むには '`\input{filename}`' してください。`.eps` ファイルは `\includegraphics{...}` コマンドで取り込むので、よって LaTeX のプリアンブルに `\usepackage{graphicx}` も入れる必要があります。`textcolour` オプションで色付きの文字列を使用している場合は、LaTeX のプリアンブルに `\usepackage{color}` も入れる必要があります。

この eps ファイルから '`epstopdf`' を使って pdf ファイルを作ることもできます。graphics パッケージが適切に設定されている場合、その LaTeX のファイルは、変更なしに `pdflatex` によっても処理でき、その場合 eps ファイルの代わりに pdf ファイルが取り込まれます。

フォントの選択に関する挙動はヘッダーモードに依存します。どの場合でも、与えられたフォントサイズはスペースの計算にちゃんと使用されます。`standalone` モードが使われなかった場合は、`include` される場所での実際の LaTeX フォントとフォントサイズが使われる所以、よってフォントの変更には LaTeX コマンドを使ってください。例えばフォントサイズとして LaTeX 文書内で 12pt を使う場合は、オプション '"" 12' を使います。この場合フォント名は無視されます。`standalone` を使う場合は、与えられたフォントとフォントサイズが使われます。詳細は下記を参照してください。

文字列がカラーで表示されるかどうかは TeX の Bool 値変数 `\ifGPcolor` と `\ifGPblacktext` で制御します。`\ifGPcolor` が true で `\ifGPblacktext` が false の場合のみ文字列はカラーで表示されます。それらは生成される TeX ファイルを変更するか、またはあなたの TeX ファイルで大域的に与えてください。例えば

```
\newif\ifGPblacktext
\GPblacktexttrue
```

をあなたのファイルのプリアンブルに書きます。部分的な指定は大域的な値が与えられていないときのみ働きます。

`epslatex` 出力形式を使う場合、`set output` コマンドで TeX ファイルの名前を拡張子付き(通常 "`.tex`")で与えてください。`eps` ファイルの名前はその拡張子を "`.eps`" に置き換えた名前になります。

`standalone` モードを使う場合、その LaTeX ファイルに完全な LaTeX のヘッダが付加され、eps ファイルのファイル名には "`-inc`" が追加されます。`standalone` モードは、`dvips`, `pdfTeX`, `VTeX` を使う場合に正しいサイズで出力されるような TeX ファイルを作ります。デフォルトは `input` モードで、これは `\input` コマンドを使って、別の

LaTeX ファイルから読み込まれるようなファイルを生成します。

"" か "default" 以外のフォント名が与えられた場合、それは LaTeX のフォント名と解釈されます。それは、'fontname,fontseries,fontshape' の、コンマで区切られた 3 つ以下の部分からなります。デフォルトのフォントシェイプ、フォントシリーズが使いたい場合はそれらは省略できます。つまり、フォント名に対する正式な書式は、'[fontname][,fontseries][,fontshape]' となります。そのいずれの部分も名前に関しては LaTeX のフォント体系の慣習に従います。fontname は 3 から 4 文字の長さで、次のような規則で作られています: 1 つ目がフォントの製造元を表し、次の 2 つがフォント名、オプションで追加される 1 つは特別なフォントを意味し、例えば 'j' は旧式の数字を持つフォント、'x' は expert フォント等となっています。以下には、多くのフォントの名前について書かれています。

<http://www.tug.org/fontname/fontname.pdf>

例えば 'cmr' は Computer Modern Roman フォント、'ptm' は Times-Roman, 'phv' は Helvetica 等を表します。フォントシリーズは文字の線の太さを意味し、大半は 'm' で普通 ("medium")、'bx' または 'b' が太字 (bold) フォントを意味します。フォントシェイプは一般に 'n' が立体 (upright)、'it' はイタリック、'sl' は斜体 (slanted)、'sc' は小さい大文字 (small caps) となります。これらとは異なるフォントシリーズやフォントシェイプで与えられるフォントも存在します。

例:

Times-Roman で太字 (シェイプは周りの文字列と同じもの) を使う場合:

```
set terminal epslatex 'ptm,bx'
```

Helvetica で太字でイタリックを使う場合:

```
set terminal epslatex 'phv,bx,it'
```

斜体のシェイプで周りのフォントを使い続ける場合:

```
set terminal epslatex ',,sl'
```

小型の大文字 (small caps) を使う場合:

```
set terminal epslatex ',,sc'
```

この方法では文字列のフォントのみが変更されます。数式のフォントも変更したい場合は、"gnuplot.cfg" ファイルかまたは header オプションを使う必要がありますが、これについては以下に書きます。

standalone モードでは、フォントサイズは set terminal コマンドで与えられたフォントサイズが使われます。指定したフォントサイズが使えるためには LaTeX の検索パスに "size<size>.clo" というファイルが存在しなければなりません。デフォルトでは 10pt, 11pt, 12pt がサポートされています。パッケージ "extsizes" がインストールされていれば、8pt, 9pt, 14pt, 17pt, 20pt も追加されます。

オプション header は文字列引数を取ります。その文字列は、生成される LaTeX ファイルに書き込まれます。standalone モードを使う場合、その文字列はプリアンブルの \begin{document} コマンドの直前に書き込まれます。input モードでは、その文字列は \begingroup コマンドの直後に置かれ、描画への設定がすべて局所的になるようにします。

例:

T1 フォントエンコーディングを使い、文字列、数式フォントを Times-Roman とし、サンセリフフォントを Helvetica と変更する場合:

```
set terminal epslatex standalone header \
"\usepackage[T1]{fontenc}\n\usepackage{mathptmx}\n\usepackage{helvet}"
```

描画の外の文字列には影響を与えないように描画内で太字 (boldface) フォントを使う場合:

```
set terminal epslatex input header "\bfseries"
```

ファイル "gnuplot.cfg" が LaTeX によって見つけられると、standalone モードを使っている場合は、それは生成される LaTeX 文書のプリアンブルに取り込まれます。それは追加の設定を行なうのに使えます。例えば、文書のフォントを TImes-Roman, Helvetica, Courier と変更し、("mathptmx.sty" で扱われている) 数式フォントを入れる場合:

```
\usepackage{mathptmx}
\usepackage[scaled=0.92]{helvet}
\usepackage{courier}
```

ファイル "gnuplot.cfg" は header コマンドで与えられるヘッダ情報の前に読み込まれます。よって、"gnuplot.cfg" で行なわれる設定のいくつかを header を使って上書きすることができます。

## 50.20 Epson-180dpi

このドライバはエプソンプリンタのいくつかとそれに類似するものをサポートします。

**epson-180dpi** と **epson-60dpi** はそれぞれ 180dpi (ドット/インチ), 60dpi の解像度の Epson LQ 型 24 ピンプリンタ用のドライバです。

**epson-lx800** は Epson LX-800, Star の NL-10 や NX-1000, PROPRINTER などの適当なプリンタに通用できる、一般的な 9 ピンプリンタドライバです。

**nec-cp6** は NEC CP6 や Epson LQ-800 などのプリンタで使える、一般的な 24 ピンプリンタ用のドライバです。

**okidata** ドライバは 9 ピンの OKIDATA 320/321 標準プリンタをサポートします。

**starc** ドライバは Star カラープリンタ用です。

**tandy-60dpi** ドライバは 9 ピン 60dpi の Tandy DMP-130 シリーズ用です。

**nec-cp6** にのみオプションがあります。

書式:

```
set terminal nec-cp6 {monochrome | colour | draft}
```

デフォルトでは白黒 (monochrome) です。

これらのドライバのそれぞれで PC 上で印刷する場合はバイナリコピーが必要です。print を使ってはいけません。その代わりに copy file /b lpt1: としてください。

## 50.21 Excl

**excl** ドライバは EXCL レーザープリンターや 1590 のような Talaris プリンタをサポートします。オプションはありません。

## 50.22 Fig

**fig** ドライバは Fig グラフィック言語での出力を生成します。

書式:

```
set terminal fig {monochrome | color}
    {landscape | portrait}
    {small | big | size <xsize> <ysize>}
    {metric | inches}
    {pointsmax <max_points>}
    {solid | dashed}
    {fontsize <fsiz>e}
    {textnormal | {textspecial texthidden textrigid}}
    {{thickness|linewidth} <units>}
    {depth <layer>}
    {version <number>}
```

**monochrome** と **color** は 画像を白黒にするか **color** にするかを決定します。**small** と **big** は、デフォルトの **landscape** モードではグラフを 5x3 インチにするか 8x5 インチにするか、**portrait** モードでは 3x5 インチにするか 5x8 インチにするかを決定します。**size** は描画範囲を **<xsize>\*<ysize>** に設定 (変更) します。この場合の単位は、**inches** か **metric** かの設定によってそれぞれインチかセンチメートルになります。この設定は "xfig" での編集に対するデフォルトの単位としても使われます。

**pointsmax <max\_points>** は折れ線の最大点数を設定します。

**solid** は、実線 (solid) の **linestyle** が全部使われてしまった後で自動的に使われる点線の使用を抑制し、別な形で表示します。

**fontsize** はテキストフォントのサイズを **<fsiz>e** ポイントに設定します。**textnormal** はテキストフラグの設定をリセットして **postscript** フォントを選択し、**textspecial** はテキストフラグを **LaTeX special**

に設定し、`texthidden`, `textrigid` はそれぞれ無表示のテキスト、スケーリングされないテキスト用のフラグを設定します。

`depth` は全ての線と文字列に対する、重なりに関するデフォルトの深さ (depth layer) を設定します。デフォルトの深さは 10 で、"xfig" でグラフの上に何かを上書きするための余地を残しています。

`version` は生成される `fig` 出力の書式バージョンを設定します。現在は、バージョン 3.1 と 3.2 のみがサポートされています。

`thickness` はデフォルトの線の太さを設定し、指定されなければ 1 になります。太さの変更は、`plot` コマンドの `linetype` の値に 100 倍の数を加えることで実現できます。同様に、(デフォルトの深さに対する) 各描画要素の深さの値は `<linetype>` に 1000 倍の数を加えることで制御できます。よってその深さは `<layer>+<linetype>/1000` になり、線の太さは `(<linetype>%1000)/100` になりますが、その値が 0 の場合はデフォルトの線の太さになります。`linewidth` は `thickness` と同義です。

`plot` コマンドの `point` スタイルによる描画の際の記号を `fig` ドライバで追加することもできます。記号の指定は (`pointtype` の値) % 100 の 50 以上の値が使われ、その塗りつぶしの濃さは `<pointtype>%5` の値で制御し、その輪郭は黒 (`<pointtype>%10 <5` の場合) または現在の色で書かれます。利用可能な記号は以下の通りです。

50 - 59:	円
60 - 69:	正方形
70 - 79:	ひし形
80 - 89:	上向きの三角形
90 - 99:	下向きの三角形

これらの記号の大きさはフォントの大きさと関係しています。デフォルトでは記号の深さは、良いエラーバーを実現するために、線の深さより 1だけ小さい値になっています。`<pointtype>` が 1000 より大きい場合、深さは `<layer>+<pointtype>/1000-1` になります。`<pointtype>%1000` が 100 より大きい場合塗りつぶし色は `(<pointtype>%1000)/100-1` になります。

有効な塗りつぶし色 (1 から 9) は、黒、青、緑、水色、赤、紫、黄、白、暗い青 (白黒モードでは 1 から 6 までは黒で 7 から 9 までは白) です。

`<linetype>` と `<pointtype>` の詳細については `plot with` を参照してください。

`big` オプションは以前のバージョンの `bfig` ドライバの代用品ですが、このドライバは今はもうサポートされていません。

例:

```
set terminal fig monochrome small pointsmax 1000 # デフォルト
plot 'file.dat' with points linetype 102 pointtype 759
```

は、黄色で塗りつぶされた円を生成し、それら輪郭は幅 1 の青い線です。

```
plot 'file.dat' using 1:2:3 with err linetype 1 pointtype 554
```

は黒い線によるエラーバーと赤で塗りつぶされた円を生成します。この円は線よりも 1 層だけ上になります (デフォルトでは深さは 9)。

円の上にエラーバーを書くには以下のようにしてください。

```
plot 'file.dat' using 1:2:3 with err linetype 1 pointtype 2554
```

## 50.23 Ggi

`ggi` ドライバは X や `svgalib` のような異なるターゲット上で動作します。

書式:

```
set terminal ggi [acceleration <integer>] [[mode] {mode}]
```

X では、ウィンドウマネージャの機能を使ってウィンドウのサイズを変更することはできませんが、モードを `mode` オプションを使って、例えば以下のように変更することができます:

- V1024x768
- V800x600
- V640x480
- V320x200

他のモードについては、ggi (libggi) のドキュメントを参照してください。キーワード mode は追加してもしなくても結構です。libggi のマニュアルページで紹介されているように、環境変数でターゲットを選択することをお勧めします。X上でDGAを使うなら、例えば以下のようにしてください。

```
bash> export GGI_DISPLAY=DGA
csh> setenv GGI_DISPLAY DGA
```

acceleration は、相対的なポインタ動作イベントを発生するターゲット(例えば DGA)でのみ使用され、正の整数で相対的な距離に対する倍率(積因子)を表します。デフォルトの acceleration は 7 です。

例:

```
set term ggi acc 10
set term ggi acc 1 mode V1024x768
set term ggi V1024x768
```

## 50.24 Gif

書式:

```
set terminal gif
    {tiny | small | medium | large | giant}
    {{no}transparent} {{no}enhanced}
    {font <face> {<pointsize>}}
    {animate {delay <time>} {{no}optimize}}
    {size <x>,<y>} {{no}crop}
    {<color0> <color1> <color2> ...}
```

GIF 画像は libgd によって生成されますが、オプションで libfreetype を使って TrueType フォントを使うこともできます。

GIF の描画を、ImageMagick パッケージのプログラム 'display' を使って、出力をパイプに渡すことで以下のようにうまく見ることができます:

```
set term gif
set output '| display gif:-'
```

連続する plot コマンドの出力は、display のウィンドウで対話的に <space> を打つことで見てください。その中から特定の一つをディスクに保存するには display ウィンドウ上でマウス左ボタンをクリックして save を選択してください。gd library によって次の 5 つの基本フォントが直接サポートされています: tiny (5x8 ピクセル), small (6x12 ピクセル), medium, (7x13 Bold), large (8x16), giant (9x15 ピクセル)。これらのフォントは大きさを変更したり回転したり(純粋に水平か垂直な文字列しか使えません)することはできません。

transparent はドライバに透明化(transparent) GIF を生成するよう指示します。この場合最初の指定色が透明化される色になります。デフォルトでは notransparent です。

enhanced は拡張テキスト処理機能(上付き文字、下付き文字、および複数のフォントの利用)を有効にします。詳細は enhanced を参照してください。PNG/GIF 出力ドライバは、それ自身拡張処理モードの命令をすべてサポートしていますが、これらの機能のうちいくつかは、それが使用している libgd インターフェースのバージョンやどのようなフォントが利用可能になっているかに依存します。

もし gnuplot が使用するローカルの gd ライブラリが TrueType フォントと Adobe Type 1 フォントをサポートするように構築されている場合、オプション font <face> {<pointsize>} でそれを選択できます。<face> は、そのフォントファイルのフルパス名かフォントフェース名(font face name)ですが、後者は、環境変数 GDFONTPATH で指示されるディレクトリ中のファイル名の先頭部分であると見なされるものです。よって、'set term gif font "Face"' は、<あるディレクトリ>/Face.ttf か <あるディレクトリ>/Face.pfa を探そうとします。TrueType も Adobe Type 1 フォントも自由に大きさを変更できます

し、任意の角度で回転できます。フォントを指定しなかった場合、gnuplot はより良いデフォルトフォントを見つけるために環境変数 `GNUPLOT_DEFAULT_GDFONT` を参照します。

オプション `animate` はあなたの手元にある `gd` ライブラリがアニメーション gif の作成をサポートする場合にのみ有効です。作成される画像の表示間隔は、1/100 秒単位で指定できます（デフォルトは 5）。ただし実際の表示間隔は、使用する表示ソフトによって変化します。アニメーション画像列は、次の `set output` か `set term` コマンドによって終了します。オプション `optimize` は、アニメーションに関する 2 つの効果を持ちます。

1) アニメーション全体を通じて单一のカラーマップが使用されます。これはアニメーションの全てのフレームで使用される全ての色が最初のフレームで定義されている必要があります。

2) 可能ならば、個々のフレームで一つ前のフレームと違う部分のみがアニメーションファイルに保存されます。これはファイルサイズを小さくしてくれますが、透明化機能を使用している場合には働かないかもしれません。

これら両方の最適化はより小さいサイズの出力ファイルを作ろうとするものですが、多分その減少量は、長いアニメーションかまたはフレームサイズがとても小さな場合にのみ意味がある程度でしょう。オプション `nooptimize` はこれらの効果をいずれも無効にします。各フレームは、個々のカラーマップ（プライベートカラーマップ）を使い、丸ごと保存されていきます。一つ注意しておきますが、最適化されていないアニメーションファイルは外部ユーティリティを使って後処理することができますし、その後処理によって gnuplot の最適化よりも小さなファイルが作られるかもしれません。デフォルトでは `nooptimize` です。

サイズ `<x,y>` はピクセル単位で与えます。デフォルトは 640x480 です。ピクセル数は `set size` コマンドで縮尺することでも変更できます。`crop` は完成した描画グラフの端の空白領域を取り除いて、結果として画像サイズを小さくします。デフォルトは `nocrop` です。

個々の色は `'rrggbb'` の形式で指定します。ここで、`x` はそのまま文字 'x' を意味し、`'rrggbb'` は赤、緑、青の 16 進数での成分です。例えば `'x00ff00'` は緑を表します。背景色を最初に指定し、その後ろに境界色、X 軸と Y 軸の色、グラフの描画色、と続きます。色数は最大 256 色まで設定できます。

例:

```
set terminal gif medium size 640,480 \
    xffffff x000000 x404040 \
    xff0000 xffa500 x66cdaa xcdb5cd \
    xadd8e6 x0000ff xdda0dd x9500d3      # デフォルト
```

この例では透明化されない背景色が白で、境界は黒、軸は灰色、そして描画色として赤、橙色、青緑色（medium aquamarine）、アザミ色（thistle 3）、明青色（light blue）、青、かきつばた色（plum）、暗いすみれ色（dark violet）の 8 色を使います。

```
set terminal gif font 'arial' 14 size 800,600
```

これは、環境変数 `GDFONTPATH` で指定されたディレクトリにある 'arial' という名前で 14pt のフォントサイズの TrueType フォントを検索します。

## 50.25 Gnugraph(GNU plotutils)

`gnugraph` ドライバは、デバイスに依存しない GNU の `plot` グラフィック言語の出力を生成します。それを "plot2ps" で PostScript 形式の出力にした大きさのデフォルトは 5 インチ x 3 インチです。これは `set size` で約 8.25 x 8.25 まで増加させることができます。

書式:

```
set terminal gnugraph {"<fontname>"} {<fontsize>}
    {type <pt>} {size "<size>"}
```

デフォルトは 10 ポイントの "Courier" です。

`type` には以下のオプションが追加できます: X, pnm, gif, ai, ps, cgm, fig, pcl5, hpgl, tek, meta (default)。`size` オプション (default では a4) は、そのまま直接 `plotutils` に渡されますので、ユーザは正しい値を設定する義務を負います。詳細は `plotutils` のドキュメントを参照してください。

例:

```
set terminal gnugraph type hpgl size "a4"
set terminal gnugraph size "a4,xoffset=-5mm,yoffset=2.0cm" type pnm
```

非 GNU 版の gnugraph ドライバもありますが、それは GNU 版のを取り除かないとコンパイルはされません。

## 50.26 Gpic

gpic ドライバは FSF (the Free Software Foundations) の "groff" パッケージの中の GPIC 形式のグラフを生成します。デフォルトの大きさは 5 x 3 インチです。オプションは原点に関するもののみで、デフォルトでは (0,0) です。

書式:

```
set terminal gpic {<x> <y>}
```

ここで x と y の単位はインチです。

単純なグラフを整形するには以下のようにします。

```
groff -p -mpic -Tps file.pic > file.ps
```

pic からの出力はパイプで eqn に渡すこともできるので、'set label' と set {x/y}label コマンドでグラフに複雑な関数の式を入れることも可能ですが。例えば、

```
set ylab '@space 0 int from 0 to x alpha ( t ) roman d t@'
```

とすれば、以下のコマンドによって y 軸に綺麗な積分が見出し付けされます。

```
gpic filename.pic | geqn -d@@ -Tps | groff -m[macro-package] -Tps
> filename.ps
```

このようにして作られた図は文書に綺麗に当てはまるように縮尺することができます。pic 言語は簡単に理解できるので、必要なら容易にグラフを直接編集できます。gnuplot で作られる pic ファイルの全ての座標は x+gnuplotx, y+gnuploty の形で与えられます。デフォルトでは x, y の値は 0 です。いくつかのファイルに対してその x,y を 0 と設定している行を削除すれば、以下のようにして複数のグラフを一つの図の中に入れてしまうことができます (デフォルトの大きさは 5.0x3.0 インチ):

```
.PS 8.0
x=0;y=3
copy "figa.pic"
x=5;y=3
copy "figb.pic"
x=0;y=0
copy "figc.pic"
x=5;y=0
copy "figd.pic"
.PE
```

これは、横に 2 つ、縦に 2 つずつ並んだ 4 つのグラフからなる、8 インチの広さの図を生成します。

以下のコマンドでも同じことができます。

```
set terminal gpic x y
```

これを例えば以下のように使います。

```
.PS 6.0
copy "trig.pic"
.PE
```

## 50.27 Gpr

gpr ドライバは Apollo Graphics Primitive Resource の固定サイズウィンドウをサポートします。オプションはありません。

色々なサイズのウィンドウが必要なら、apollo ドライバを代わりに使用してください。

## 50.28 Grass

grass ドライバは GRASS 地理情報システムのユーザが gnuplot を利用することを可能にします。詳しい情報については grassp-list@moon.celer.army.mil に連絡を取ってください。ページは GRASS グラフウィンドウの現在のフレームに書かれます。オプションはありません。

## 50.29 Hercules

これらのドライバは PC のモニタをサポートし、グラフィックボードを自動検出します。これらは、gcc、または Zortec C/C++ でコンパイルされた場合のみ利用可能です。

## 50.30 Hp2623a

hp2623a ドライバはヒューレットパッカード (Hewlett Packard) HP2623A をサポートします。オプションはありません。

## 50.31 Hp2648

hp2648 ドライバはヒューレットパッカード (Hewlett Packard) HP2647 と HP2648 をサポートします。オプションはありません。

## 50.32 Hp500c

hp500c ドライバはヒューレットパッカード (Hewlett Packard) 社の HP DeskJet 500c をサポートします。これには解像度と圧縮に関するオプションがあります。

書式:

```
set terminal hp500c {<res>} {<comp>}
```

ここで res は 75, 100, 150, 300 のいずれかの解像度 (DPI; dots per inch) で、comp は "rle" か "tiff" です。他の設定をするとそれはデフォルトの値になります。デフォルトは 75 dpi で圧縮はなしです。高解像度でのラスタライズはたくさんのメモリを必要とします。

## 50.33 Hpgl

hpgl ドライバは HP7475A プロッタのような装置用の HPGL 出力を行ないます。これは 2 つの設定可能なオプションを持ちます: それはペンの数と eject オプションで、"eject" は描画後にプロッタにページを排出させるよう指示しデフォルトでは 6 つのペンを使い、描画後のページの排出は行ないません。

国際的文書セット ISO-8859-1 と CP850 を set encoding iso\_8859\_1 や set encoding cp850 で認識させることができます (詳細は set encoding を参照してください)。

書式:

```
set terminal hpgl {<number_of_pens>} {eject}
```

以下の設定

```
set terminal hpgl 8 eject
```

は、以前の hp7550 ドライバと同等で、設定

```
set terminal hpgl 4
```

は、以前の hp7580b ドライバと同等です。

pcl5 ドライバは、Hewlett-Packard Designjet 750C、Hewlett-Packard Laserjet III, Hewlett-Packard Laserjet IV のようなプロッタをサポートします。これは実際には HPGL-2 を使用しているのですが、装置間で名前の衝突があります。このドライバにはいくつかのオプションがありますが、それらは以下に示した順序で指定しなければいけません:

書式:

```
set terminal pcl5 {mode <mode>} {<plotsize>}
  {{color {<number_of_pens>}} | monochrome} {solid | dashed}
  {font <font>} {size <fontsize>} {pspoints | nospoints}
```

<mode> は `landscape` か `portrait` です。<plotsize> はグラフの物理的な描画サイズで、それは以下のうちのいずれかです: `letter` は標準の(8 1/2" X 11")出力、`legal` は(8 1/2" X 14")出力、`noextended` は(36" X 48")出力(`letter` サイズ比)、`extended` は(36" X 55")出力(ほぼ `legal` サイズ比)。`color` は複数のペン(すなわちカラー)描画用で <number\_of\_pens> はカラー出力で使用されるペンの本数(すなわち色数)です。`monochrome` は 1 本のペン(例えば黒)の描画です。`solid` は全ての線を実線で描き、`dashed` は異なる点線や鎖線パターンで線を描き分けます。`<font>` は `stick`, `univers`, `cg_times`, `zapf_dingbats`, `antique_olive`, `arial`, `courier`, `garamond_antigua`, `letter_gothic`, `cg_omega`, `albertus`, `times_new_roman`, `clarendon`, `coronet`, `marigold`, `truetype_symbols`, `wingdings` のいずれかです。`<fontsize>` はポイント単位でのフォントの大きさです。点の種類(point type)は、`nospoints` を指定することで標準的なデフォルトの設定から選択できるようになりますが、`pspoints` を指定すると `postscript terminal` と同じ点の種類の設定から選択できるようになります。

これらのオプションのいくつかの組み込まれたサポートは、プリンタに依存することに注意してください。例えば全てのフォントは恐らく HP Laserjet IV ではサポートされているでしょうが、HP Laserjet III と Designjet 750C では 2,3(例えば `univers`, `stick`) がサポートされているのみでしょう。また、`laserjet` は白黒の出力装置なので、それらではカラーも明らかに使えません。

**デフォルト:** `landscape`, `noextended`, `color (6 pens)`, `solid`, `univers`, 12  
point, nospoints

`pcl5` では国際的文字セットはプリンタで扱われますので、テキスト文字列に適切な 8-bit 文字コードを入れるだけで、わざわざ `set encoding` で邪魔をする必要はありません。

HPGL グラフィックは多くのソフトウェアパッケージで取り込むことが可能です。

### 50.34 Hpljii

`hpljii` ドライバは HP Laserjet Sries II プリンタを、`hpdj` ドライバは HP DeskJet 500 プリンタをサポートします。これらのドライバでは、解像度の選択が可能です。

**書式:**

```
set terminal hpljii | hpdj {<res>}
```

ここで `res` は 75, 100, 150, 300 のいずれかの解像度(DPI; dots per inch)で、デフォルトは 75 です。高解像度でのラスタライズはたくさんのメモリを必要とします。

`hp500c` ドライバは `hpdj` とほぼ同じですが、`hp500c` は加えてカラーと圧縮もサポートしています。

### 50.35 Hppj

`hppj` ドライバは HP PaintJet と HP3630 プリンタをサポートします。オプションはフォントを選択するものがあるのみです。

**書式:**

```
set terminal hppj {FNT5X9 | FNT9X17 | FNT13X25}
```

中間サイズフォント(FNT9X17)がデフォルトです。

### 50.36 Imagen

`imagen` ドライバは Imagen レーザプリンタをサポートします。これは 1 ページに複数のグラフを配置することも可能です。

**書式:**

```
set terminal imagen {<fontsize>} {portrait | landscape}
  {[<horiz>,<vert>]}
```

`fontsize` はデフォルトでは 12 ポイントで、レイアウトのデフォルトは `landscape` です。`<horiz>` と `<vert>` はグラフを横方向と縦方向に何列置くかを指定します。これらのデフォルトは 1 です。

例:

```
set terminal imagen portrait [2,3]
```

これは、1 ページに 6 つのグラフを横に 2 列、縦に 3 列、縦置き (portrait) で配置します。

### 50.37 Iris4d

`iris4d` ドライバは Silicon Graphics IRIS 4D コンピュータをサポートします。唯一のオプションは 8-bit か 24-bit かの色の深さの指定で、デフォルトでは 8 です。

書式:

```
set terminal iris4d {8 | 24}
```

色の深さは自由に選べるものではなく、ハードウェアに適切な値を選択すべきです。

24-bit モードを使う場合、その色は直接ファイル `.gnuplot_iris4d` で指定することができますが、このファイルは現在のディレクトリ、`HOME` 環境変数に設定されるホームディレクトリの順に検索されます。このファイルには、背景、境界、見出し、そして 9 個の描画色の RGB 値を、この順に書きます。例えば以下にデフォルトの色を持つファイルを記します:

85	85	85	Background	(暗い灰色; dark gray)
0	0	0	Boundary	(黒; black)
170	0	170	Labeling	(赤紫; magenta)
85	255	255	Plot Color 1	(明るい水色; light cyan)
170	0	0	Plot Color 2	(赤; red)
0	170	0	Plot Color 3	(緑; green)
255	85	255	Plot Color 4	(明るい赤紫; light magenta)
255	255	85	Plot Color 5	(黄色; yellow)
255	85	85	Plot Color 6	(明るい赤; light red)
85	255	85	Plot Color 7	(明るい緑; light green)
0	170	170	Plot Color 8	(水色; cyan)
170	170	0	Plot Color 9	(茶色; brown)

このファイルは RGB の 3 つの組が丁度 12 行なくてはいけません。空行も許されませんが、各行の 3 つ目以降の文字は全て無視されます。

### 50.38 Jpeg

書式:

```
set terminal jpeg
    {{no}interlace}
    {{tiny | small | medium | large | giant}}
    {font <face> {<pointsize>}}
    {size <x>,<y>} {{no}crop}
    {{no}enhanced}
    {<color0> <color1> <color2> ...}
```

JPEG 画像は libgd によって生成されますが、オプションで libfreetype を使って TrueType フォントを使うこともできます。

オプション `interlace` はプログレッシブ JPEG 画像を生成します。デフォルトは `nointerlace` です。

`gd library` によって次の 5 つの基本フォントが直接サポートされています: `tiny` (5x8 ピクセル), `small` (6x12 ピクセル), `medium`, (`7x13 Bold`), `large` (8x16), `giant` (9x15 ピクセル)。これらのフォントは大きさを変更したり回転したり (純粹に水平か垂直な文字列しか使えません) することはできません。

`gnuplot` が TrueType (\*.ttf) フォントか Adobe Type 1 (\*.pfa) フォントをサポートするように構築されている場合、オプション `font <face> {<pointsize>}` でそれを選択できます。`<face>` は、そのフォント

ファイルのフルパス名かフォントフェース名 (font face name) ですが、後者は、環境変数 GDFONTPATH で指示されるディレクトリ中のファイル名の先頭部分であると見なされるものです。よって、'set term jpeg font "Face"' は、<あるディレクトリ>/Face.ttf か <あるディレクトリ>/Face.pfa を探そうとします。TrueType も Adobe Type 1 フォントも自由に大きさを変更できますし、任意の角度で回転できます。フォントを指定しなかった場合、gnuplot はより良いデフォルトフォントを見つけるために環境変数 GNUPLOT\_DEFAULT\_GDFONT を参照します。

**enhanced** は拡張テキスト処理機能 (上付き文字、下付き文字、および複数のフォントの利用) を有効にします。詳細は **enhanced** を参照してください。PNG/JPEG 出力ドライバは、それ自身拡張処理モードの命令をすべてサポートしていますが、これらの機能のうちいくつかは、それが使用している libgd イブリのバージョンやどのようなフォントが利用可能になっているかに依存します。

サイズ <x,y> はピクセル単位で与えます。デフォルトは 640x480 です。ピクセル数は **set size** コマンドで縮尺することでも変更できます。**crop** は完成した描画グラフの端の空白領域を取り除いて、結果として画像サイズを小さくします。デフォルトは **nocrop** です。

個々の色は 'xrrggb' の形式で指定します。ここで、x はそのまま文字 'x' を意味し、'rrggb' は赤、緑、青の 16 進数での成分です。例えば 'x00ff00' は緑を表します。背景色を最初に指定し、その後ろに境界色、X 軸と Y 軸の色、グラフの描画色、と続きます。色数は最大 256 色まで設定できます。

例:

```
set terminal jpeg medium size 640,480 \
    xffffff x000000 x404040 \
    xff0000 xffa500 x66cdaa xcdb5cd \
    xadd8e6 x0000ff xdda0dd x9500d3      # デフォルト
```

この例では透明化されない背景色が白で、境界は黒、軸は灰色、そして描画色として赤、橙色、青緑色 (midium aquamarine)、アザミ色 (thistle 3)、明青色 (light blue)、青、かきつばた色 (plum)、暗いすみれ色 (dark violet) の 8 色を使います。

```
set terminal jpeg large font arial size 800,600
```

これは、環境変数 GDFONTPATH で指定されたディレクトリにある 'arial' という名前で大きいサイズ (14pt) TrueType フォントを検索します。

### 50.39 Kyo

**kyo** と **prescribe** のドライバは Kyocera (京セラ) レーザープリンタをサポートします。この両者の唯一の違いは、**kyo** が "Helvetica" を使うのに対して **prescribe** が "Courier" を使うことだけです。オプションはありません。

### 50.40 Latex

**latex** と **emtex** ドライバは 2 つのオプションを持ちます。

書式:

```
set terminal latex | emtex {courier | roman | default} {<fontsize>}
```

**fontsize** には任意のサイズを指定します。デフォルトは、それを埋め込む文書のフォントの設定を引き継ぎます。

あなたの DVI ドライバが任意のサイズのフォントを作り出すことができない場合 (例えば dvips)、標準的な 10, 11, 12 ポイントサイズでなんとかしのいでください。

METAFONT ユーザへの警告: METAFONT は妙なサイズは好みません。

LaTeX に関する全てのドライバは文字列の配置の制御に特別な方法を提供します: '{' で始まる文字列は、'}' で閉じる必要がありますが、その文字列全体が水平方向にも垂直方向にもセンタリングされます。 '[' で始まる文字列の場合は、位置の指定をする文字列 (t,b,l,r のうち 2 つまで) が続き、次に ']' 、文字列本体、で最後に '}' としますが、この文字列は LaTeX が LR-box として整形します。'\rule{}{}' を使えばさらに良い位置合わせが可能でしょう。

数ある中で、点 (point) は、LaTeX のコマンド "\Diamond" と "\Box" を使って描かれます。これらのコマンドは現在は LaTeX2e のコアには存在せず、latexsym パッケージに含まれていますが、このパッケージ基本配布の一部であり、よって多くの LaTeX のシステムの一部になっています。このパッケージを使うことを忘れないでください。

例: 見出しの位置合わせに関して: gnuplot のデフォルト (大抵それなりになるが、そうでないこともある):

```
set title '\LaTeX\ -- $ \gamma'
```

水平方向にも垂直方向にもセンタリング:

```
set label '{\LaTeX\ -- $ \gamma}' at 0,0
```

位置を明示的に指定 (上に合わせる):

```
set xlabel '[t]{\LaTeX\ -- $ \gamma}',
```

他の見出し - 目盛りの長い見出しに対する見積り:

```
set ylabel '[r]{\LaTeX\ -- $ \gamma $\rule{7mm}{0pt}}'
```

## 50.41 Linux

linux ドライバには指定するオプションは何もありません。それはデフォルトのモード用に環境変数 GSVG-AMODE を参照します。もしそれが設定されていなければ 1024x768x256 をデフォルトとして使用しますが、それができなければ 640x480x16 (標準の VGA) とします。

## 50.42 Macintosh

macintosh ドライバではいくつかのオプションを設定できます。

書式:

```
set terminal macintosh {singlewin | multiwin} {vertical | novertical}
    {size <width>, <height> | default}
```

'singlewin' は単一ウィンドウへの制限であり、アニメーション用に有用です。'multiwin' は複数のウィンドウが使えるようになります。'vertical' は gx オプションの元でのみ有効です。このオプションでは回転した文字列を垂直方向に書きます。'novertical' はこのオプションを無効にします。

'size <width>, <height>' は preferences dialog で設定されたグラフサイズを書き換え、'set term mac size default' か 'set term mac default' が指定されるまでその値は保持されます。

'set term mac size default' はウィンドウサイズの設定を preferences dialog で設定したものにします。

'set term mac default' は全てのオプションをデフォルトの値に設定します。  
デフォルト値は nogx, multiwin, novertical です。

multiwin オプションの元でグラフを作成し、その後 singlewin にスイッチした場合、次の plot コマンドでは更に一つウィンドウが作られ、この新しいウィンドウでは singlewin が有効な間使用されます。そして multiwin に戻り、いくつかのグラフを生成し、また再び singlewin にスイッチすると、もし元の 'singlewin' ウィンドウがまだ開いていればそれが使われ、そうでなければ新しい 'singlewin' ウィンドウが作られます。'singlewin' ウィンドウは番号付けはされません。

## 50.43 Mf

`mf` ドライバは METAFONT プログラムへの入力ファイルを作ります。よってその図は TeX の文書中では文字と同じように使うことができます。

文書中で図を使うには、`gnuplot` の出力するファイルを入力として METAFONT プログラムを実行する必要があります。よって、ユーザはフォントが作られるプロセスと新しく作ったフォントをドキュメントに取り込むための基礎知識が必要となります。しかし、使用するサイトで METAFONT プログラムが適切に設定されていれば、経験のないユーザでもそう問題なく操作はできるでしょう。

グラフ中の文字は METAFONT の文字セットに基づいてサポートされます。現状では Computer Modern Roman フォントセットが入力ですが、ユーザは必要なフォントを何でも自由に選ぶことができます。ただし、その選んだフォントの METAFONT ソースファイルが使える状態になっている必要があります。個々の文字は METAFONT の中に別々のピクチャー変数に保存され、文字が必要になったときにこれらの変数が操作（回転、伸縮等）されます。欠点は、METAFONT プログラムが解釈に要する時間です。ある計算機（つまり PC）では、ピクチャー変数をたくさん使用しすぎることで、使えるメモリの量の限界が問題を起こすこともあります。

`mf` ドライバにはオプションはありません。

### 50.43.1 METAFONT Instructions

- 出力形式 (terminal) を METAFONT にセット:

```
set terminal mf
```

- 出力ファイル名を設定。例えば:

```
set output "myfigures.mf"
```

- グラフの描画。各グラフは別々の文字を生成し、そのデフォルトの大きさは 5x3 インチですが、この大きさは `set size 0.5,0.5` のようにしてどんなサイズにでも自由に変更できます。

- `gnuplot` を終了

- `gnuplot` の出力ファイルに対して METAFONT を実行し、TFM ファイルと GF ファイルを作ります。グラフは割と大きい（5x3 インチ）ので、memmax の値が少なくとも 150000 である METAFONT を使う必要があるでしょう。Unix では、それは通常 `bigrm` という名前でインストールされているでしょう。以下では、`virmf` コマンドが big 版の METAFONT であると仮定し、実行例を示します:

- METAFONT の立ち上げ:

```
virmf '&plain'
```

- 出力装置の選択: METAFONT プロンプト ('\*) 上で次のように打ちます:

```
\mode:=CanonCX; % あなたの使用するプリンタを指定
```

- 拡大率 (magnification) の選択 (オプション):

```
mag:=1; % あなたの好みの値を指定
```

- `gnuplot` で作ったファイルを入力:

```
input myfigures.mf
```

典型的な Unix マシンでは、`virmf '&plain'` を実行するスクリプト "mf" があるので、`virmf &plain` の代わりに `mf` を使えます。これにより `mfput.tfm` と `mfput.$$$gf` (`$$$` は出力装置の解像度) の 2 つのファイルが作られます。上の作業は、すべてをコマンドライン上で簡単に実行することもできます: `virmf '&plain' '\mode:=CanonCX; mag:=1; input myfigures.mf'` この場合、作られるファイル名は `myfigures.tfm` と `myfigures.300gf` という名前になります。

- `gftopk` を使って GF ファイルから PK ファイルを生成:

```
gftopk myfigures.300gf myfigures.300pk
```

`gftopk` が作るファイルの名前はあなたが使用する DVI ドライバに依存しますので、サイトの TeX の管理者にフォント名の規則について聞いてください。次に TFM ファイルと PK ファイルを適当なディレク

トリにインストールするかまたは環境変数を適切な値に設定します。通常それは、TEXFONTS にカレントディレクトリを含めることと、あなたが使用する DVI ドライバが使用している環境変数（標準的な名前はありませんが ...）に対して同じことをやれば済みます。これは TeX がフォントメトリック (TFM) ファイルを見つけ、DVI ドライバが PK ファイルを見つけられるようにするために必要な作業です。

- 文書にそのグラフを入れるために TeX にそのフォント名を指示:

```
\font\gnufigs=myfigures
```

各グラフは、最初のグラフが文字 0、2 番目のグラフが文字 1 というように、それぞれ一つの文字として保存されています。上記の作業を行なうと、グラフはその他の文字と同じように使うことができ、例えばグラフ 1 と 2 を文書中にセンタリングして置くために plain TeX ファイル中ですべきことは:

```
\centerline{\gnufigs\char0}
\centerline{\gnufigs\char1}
```

だけです。もちろん LaTeX では picture 環境を使って \makebox と \put マクロで任意の位置にグラフを配置することができます。

このやり方は、一度フォントを生成してしまえば、大幅に時間の節約になります: TeX はグラフを文字として使い、それを配置するにはごく少ない時間しか使用しませんし、グラフよりも文書の方が修正することが多いでしょうから。そしてこれは TeX のメモリの節約にもなります。METAFONT ドライバを使うもう最後の一つの利点は、生成される DVI ファイルが本来のデバイス非依存な形になるということです。それは eepic や tpic ドライバのような \special コマンドを全く使わないからです。

#### 50.44 Mgr

mgr ドライバは Mgr ウィンドウシステムをサポートします。オプションはありません。

#### 50.45 Mif

mif ドライバは Frame Maker MIF フォーマット (version 3.00) の出力を生成します。これは 15\*10 cm のサイズの MIF フレームを出力し、同じペンで書かれるグラフの基本要素は同じ MIF グループにグループ化されます。gnuplot の 1 ページにおけるグラフの基本要素は一つの MIF フレームに描画され、いくつかの MIF フレームは一つの大きな MIF フレーム内に集められます。文字列で使われる MIF フォントは "Times" です。

MIF 3.00 ドライバではいくつかのオプションが設定できます。

書式:

```
set terminal mif {color | colour | monochrome} {polyline | vectors}
                  {help | ?}
```

colour は線種 (line type)  $>= 0$  の線をカラー (MIF separation 2-7) で、monochrome は全ての線を黒 (MIF separation 0) で描画します。polyline は曲線を連続曲線として描画し、vectors は曲線をベクトルの集まりとして描画します。help と ? はオンラインヘルプを標準エラー出力に表示します。両者はその使用法の短い説明を出力し、help は更にオプションも表示します。

例:

```
set term mif colour polylines      # デフォルト
set term mif                      # デフォルト
set term mif vectors
set term mif help
```

#### 50.46 Mp

mp ドライバは Metapost プログラムへ入力することを意図した出力を生成し、そのファイルに対して Metapost を実行するとグラフを含む EPS ファイルが作られます。デフォルトでは Metapost は全ての文字列を TeX に通します。これはタイトルや見出しに任意の TeX の記号を本質的に使うことができる、という利点を持つことを意味します。

書式:

```

set term mp {color | colour | monochrome}
    {solid | dashed}
    {notex | tex | latex}
    {magnification <magsize>}
    {psnfss | psnfss-version7 | nopsnfss}
    {prologues <value>}
    {a4paper}
    {amstex}
    {"<fontname>"} {<fontsize>}

```

オプション `color` は線をカラーで書くことを意味し (それをサポートするプリントやディスプレイ上で)、`monochrome` (または何も指定しない場合) は黒の線が選択されます。オプション `solid` は線を実線で描き、`dashed` (または無指定) は線を異なるパターンの点線で描き分けます。`solid` が指定されてかつ `color` が指定されなかった場合、ほとんど全ての線が同じものになりますが、これも何かの場合には有用でしょうから認められています。

オプション `notex` は完全に TeX を迂回しますので、このオプションの元では見出しには TeX のコードは使うことができません。これは、古いグラフファイル、あるいは TeX では特殊記号として解釈されてしまう \$ や % のような一般的な文字をたくさん使うファイルのために用意されています。

オプション `tex` は、TeX で処理する文字列を出力するように設定します。

オプション `latex` は、LaTeX で処理する文字列を出力するように設定します。これによって TeX では使えないけれど LaTeX では使えるもの、例えば分数を `\frac` で書いたりすることができます。このオプションを使う場合は、環境変数 `TEX` に LaTeX の実行プログラム名 (通常は `latex`) を設定するか、あるいは `mpost -tex=<LaTeX の実行プログラム名> ...` とすることを忘れないでください。そうでないと `metapost` はテキストの処理に TeX を使おうとして失敗してしまうでしょう。

TeX におけるフォントサイズの変更は数式には効果がなく、そのような変更を行なうとても簡単な方法は、大域的に拡大率 (magnification factor) を設定する以外にはありません。それがオプション `magnification` の意味です。その場合は拡大率を後ろに指定する必要があります。全ての文字 (グラフではなく) はこの率で拡大されます。数式をデフォルトの 10pt 以外の他のサイズで書きたい場合はこれを使用してください。ただ残念なことに全ての数式が同じサイズになってしまいますが、しかし、以下の MP 出力の編集に関する説明を参照してください。`mag` は `notex` の元でも働きますが、それを行なう意味がないくらい (以下に述べる) フォントサイズオプションはうまく働きます。

オプション `psnfss` は postscript フォントを LaTeX と組み合わせて使用します。このオプションは LaTeX が使われる場合のみ意味を持ちますので、自動的に `latex` オプションが選択されます。このオプションは以下の LaTeX パッケージを使用します: `inputenc(latin1)`, `fontenc(T1)`, `mathptmx`, `helvet(scaled=09.2)`, `courier`, `latextsym`, `textcomp`

オプション `psnfss-version7` も postscript フォントを LaTeX と組み合わせて使用します (`latex` が自動的に選択されます) が、以下の LaTeX パッケージを使用します: `inputenc(latin1)`, `fontenc(T1)`, `times`, `mathptmx`, `helvet`, `courier`

オプション `nopsnfss` はデフォルトで、標準的なフォント (何も指定されていなければ `cmr10`) が使われます。

オプション `prologues` は追加の値を引数に持ち、`metapost` ファイルに `prologues:=<その値>` という行を追加します。値として 2 を指定すると `metapost` は `eps` ファイルを作るよう `postscript` フォントを使用し、それによりその結果は例えば `ghostscript` などで参照できるようになります。標準では `metapost` は TeX のフォントを使用しますので、それを参照するには (La)TeX のファイルに取り込む必要があります。

オプション `noprologues` はデフォルトで、`prologue` で指定したいかかる行も追加されません。

オプション `a4paper` は `[a4paper]` を `documentclass` に追加します。標準では `letter` 用紙 (デフォルト) が使われます。このオプションは LaTeX でのみ使われますので、自動的に `latex` オプションが選択されます。

オプション `amstex` は、自動的に `latex` オプションを選択し、以下の LaTeX パッケージを使用します: `amsfonts`, `amsmath(intlimits)`。デフォルトではこれらは使用されません。

引用符で囲まれた名前はフォント名を表し、`set label` や `set title` で明示的にフォントが与えられない場合はこのフォントが使われます。フォントは TeX が認識できる (TFM ファイルが存在する) ものを使う

必要があります。デフォルトでは `notex` が選択されていなければ "cmr10" が、そうでなければ "pcrr8r" (Courier) が使われます。`notex` の元でも、Metapost には TFM ファイルは必要です。`pcrr8r.tfm` は LaTeX psnfss パッケージの Courier フォント名として与えられています。`notex` のデフォルトからフォントを変更する場合は、少なくとも 32-126 のコード範囲は ASCII エンコーディングに一致するものを選んでください。`cmtt10` もほぼ使えますが、しかしこれはコード 32 (スペース) にスペースではない文字が入っています。

サイズは 5.0 から 99.99 の間の任意の数字を指定でき、省略された場合は 10.0 が使われます。なるべく `magstep` サイズ、つまり 1.2 の整数かまたは 0.5 乗の 10 倍を小数以下 2 桁未満を丸めた値を使用することをお勧めします。それはそれが TeX のシステムで最もよく使われるフォントのサイズだからです。

全てのオプションは省略可能です。フォントを指定する場合はそれは (必要ならサイズもつけて) 一番最後に指定する必要があります。フォント名にそのサイズ情報が含まれていたとしても、サイズを変えるにはフォントサイズを指定する必要があります。例えば `set term mp "cmtt12"` は `cmtt12` をデフォルトのサイズである 10 に縮めて使います。それは多分望まないことでしょうし `cmtt10` を使う方が良いでしょう。

以下の ascii 文字は、TeX では特別に扱われます:

```
$, &, #, %, _; |, <, >; ^, ~, \, {, }
```

`$, #, &, %, _` の 5 つは、例えば `\$` とすることで容易にそれをエスケープできます。`<, >, |` の 3 つは、例えば `$_<$` のように数式モードに入れてやればうまくいきます。残りのものに関しては少し TeX の回避策が必要になりますが、適当なよい TeX の本がそれを指導してくれるでしょう。

見出しを二重引用符で囲む場合、TeX コードのバックスラッシュはエスケープする (2 つ書く) 必要があります。単一引用符を使えばそれを避けることはできますが、今度は改行として `\n` を使えなくなります。これを書いている現在、gnuplot 3.7 は `plot` コマンドで与えられたタイトルは、別な場所で与えられた場合とは異なる処理をしますし、引用符のスタイルにかかわらず TeX コマンドのバックスラッシュは二重化した方が良さそうです。

Metapost の画像は TeX の文書内で一般に使われています。Metapost はフォントを TeX が行なうと全く同じ方法で扱い、それは他の大抵の文書整形プログラムとは異なっています。グラフが LaTeX の文書に `graphics` パッケージで取り込まれ、あるいは `epsf.tex` を使って plainTeX に取り込まれ、そして `dvips` (または他の `dvi` から `ps` への変換ソフト) で PostScript に変換される場合、そのグラフ内の文字は大抵は正しく扱われているでしょう。しかし、Metapost 出力をそのまま PostScript インタプリタに送っても、グラフ内の文字は出力されないでしょう。

### 50.46.1 Metapost Instructions

- まず terminal ドライバを Metapost に設定、例えば:

```
set terminal mp mono "cmtt12" 12
```

- 出力ファイルを選択、例えば:

```
set output "figure.mp"
```

- グラフを作成。各 `plot` (または `multiplot` の各グループ) はそれぞれ別な Metapost `beginfig...endfig` グループに分けられます。そのデフォルトのサイズは 5x3 インチですが、それは `set size 0.5,0.5` とか、そういうふうに割合をそのように指定することで変更できます。

- gnuplot を終了。

- gnuplot の出力ファイルに対して Metapost を実行して EPS ファイルを作成:

```
mpost figure.mp OR mp figure.mp
```

Metapost プログラム名はシステムに依存し、Unix では通常 `mpost` で、他の多くのシステムでは `mp` です。Metapost は各グラフに対して 1 つずつの EPS ファイルを生成します。

- そのグラフを文書に取り込むには LaTeX `graphics` パッケージや、plainTeX では `epsf.tex` を使用:

```
\usepackage{graphics} % LaTeX
\input epsf.tex      % plainTeX
```

TeX DVI 出力を PS に変換するのに、`dvips` 以外の DVI ドライバを使う場合は、LaTeX ファイルに以下の行を入れる必要があるかも知れません:

```
\DeclareGraphicsRule{*}{eps}{*}{}
```

作られた各グラフは分離したファイルになっていて、最初のグラフのファイルは、例えば figure.0, 2 つ目は例えば figure.1 のような名前になります。よって、3 つ目のグラフを文書に取り込むためにあなたがしなければいけないことは以下の通りです:

```
\includegraphics{figure.2} % LaTeX
\epsfbox{figure.2} % plainTeX
```

mp ドライバの postscript ドライバに代わる利点は、もしあるとすれば、それは編集可能な出力であるということでしょう。この出力を可能な限り綺麗にするための、かなりの努力が払われました。Metapost 言語に関するそういう知識のおかげで、デフォルトの線種や色は配列 `lt[]` や `col[]` を編集することで変更できるようになりました。実線/点線、カラー/白黒といった選択も、真偽値として定義されている `dashedlines` や `colorlines` を変更することで行なえます。デフォルトの `tex` オプションが有効な場合、ロック `verbatimtex...etex` を編集することで、ラベル文字フォントに対する大域的な変更が行なえます。特に、もし望むなら LaTeX のプリアンブルを追加することもでき、その場合 LaTeX の持つサイズ変更コマンドを使えるので最大の柔軟性を発揮できるでしょう。ただし、Metapost に plainTeX ではなく LaTeX を実行させるよう、適切な MP 設定変数を設定することを忘れないでください。

## 50.47 Mtos

`mtos` ドライバにはオプションはありません。これはデータを GPCLIENT という外部プログラムにパイプ経由で渡します。それは MULTIOS, Magic 3.x, MagicMAC, MiNT 上で動作します。もし GPCLIENT を見つけられなかったら、`dirk@lstm.uni-erlangen.de` へメールしてください。

## 50.48 Next

`next` ドライバには設定のためのいくつかのオプションがあります。

書式:

```
set terminal next {<mode>} {<type>} {<color>} {<dashed>}
    {"<fontname>"} {<fontsize>} title {"<newtitle>"}
```

`<mode>` は `default` のみ指定でき、その場合全てのオプションがデフォルトになります。`<type>` は `new` か `old` で、`old` は古い単一ウィンドウを要求します。`<color>` は `color` (カラー) か `monochrome` (白黒)、`<dashed>` は `solid` (実線のみ) か `dashed` (点線が有効)、"`<fontname>`" は有効な PostScript フォントの名前を、`<fontsize>` は PostScript ポイント単位でのフォントのサイズを、`<title>` は GnuTerm ウィンドウのタイトルをそれぞれ設定します。デフォルトは `new, monochrome, dashed, "Helvetica", 14pt` です。

例:

```
set term next default
set term next 22
set term next color "Times-Roman" 14
set term next color "Helvetica" 12 title "MyPlot"
set term next old
```

点の大きさは `set linestyle` で変更できます。

## 50.49 Openstep (next)

`openstep (next)` ドライバには設定のためのいくつかのオプションがあります。

書式:

```
set terminal openstep {<mode>} {<type>} {<color>} {<dashed>}
    {"<fontname>"} {<fontsize>} title {"<newtitle>"}
```

`<mode>` は `default` のみ指定でき、その場合全てのオプションがデフォルトになります。`<type>` は `new` か `old` で、`old` は古い単一ウィンドウを要求します。`<color>` は `color` (カラー) か `monochrome` (白

黒)、<dashed> は solid (実線のみ) か dashed (点線が有効)、"<fontname>" は有効な PostScript フォントの名前を、<fontsize> は PostScript ポイント単位でのフォントのサイズを、<title> は GnuTerm ウィンドウのタイトルをそれぞれ設定します。デフォルトは new, monochrome, dashed, "Helvetica", 14pt です。

例:

```
set term openstep default
set term openstep 22
set term openstep color "Times-Roman" 14
set term openstep color "Helvetica" 12 title "MyPlot"
set term openstep old
```

点の大きさは set linestyle で変更できます。

## 50.50 Pbm

pbm ドライバ — PBMplus 用ドライバ — にはいくつかのオプションが設定できます。

書式:

```
set terminal pbm {<fontsize>} {<mode>} {size <x>,<y>}
```

<fontsize> は small か medium か large で、<mode> は monochrome か gray か color です。デフォルトの描画サイズは 640 ピクセルの幅で 480 ピクセルの高さです。

pbm ドライバの出力は <mode> によります: monochrome は portable bitmap (PBM; 1 ピクセル 1 ビット) を、gray は portable graymap (PGM; 1 ピクセル 3 bit) を、color は portable pixmap (PPM; 1 ピクセル 4 ビット) を出力します。

このドライバの出力は、NETPBM によって提供される様々な画像変換、画像処理ツールで使うことができます。Jef Poskanzer の PBMLPLUS パッケージに基づく NETPBM は、上記の PBM 形式から GIF, TIFF, MacPaint, Macintosh PICT, PCX, X11 ピットマップ、その他多くの形式に変換するプログラムを提供します。完全な情報は <http://netpbm.sourceforge.net/> にあります。

例:

```
set terminal pbm small monochrome          # デフォルト
set terminal pbm color medium size 800,600
set output '| pnmscale -angle 45 | pnmrotate 45 | pnmtopng > tilted.png' # NETPBM を利用
```

## 50.51 Pdf

このドライバは Adobe PDF (Portable Document Format) 出力を生成し、それは Acrobat Reader のようなツールで表示、印刷ができます。

書式:

```
set terminal pdf {monochrome|color|colour}
    {{no}enhanced}
    {fname "<font>"} {fsizze <fontsize>}
    {font "<fontname>{,<fontsize>}"}
    {linewidth <lw>} {rounded|butt}
    {solid|dashed} {dl <dashlength>}}
    {size <XX>{unit},<YY>{unit}}
```

デフォルトでは、個々の線種に対して異なる色を用います。monochrome を選択すると全ての線種を黒で描きますので、線種を区別するために dashed を使うと良いでしょう。モノクロモードでも塗りつぶし領域やラインスタイルでは明示的に色を使用することができます。

<font> はデフォルトで使われるフォント名 (デフォルトでは Helvetica) で <fontsize> はポイント単位でのフォントサイズ (デフォルトでは 12) です。どのようなフォントが使って、新しいフォントをインストールするには、といったことに関しては、ローカルにインストールされている pdflib のドキュメントを参照してください。

オプション `enhanced` は、拡張テキスト処理機能（下付き文字、上付き文字、および複数のフォントの利用）を有効にします。`enhanced` 参照。

描画における全ての線の幅は `linewidth` で指定する因子 `<n>` で増加することができます。同様に、`dashlength` はデフォルトの点線の空白部分に対する積因子です。

`rounded` は、線の端や接合部を丸くし、デフォルトの `butt` は尖った端と角張った接合部を使用します。

PDF 出力のデフォルトのサイズは、`5inch x 3inch` です。オプション `size` は、これをユーザの指定するものへ変更します。デフォルトの X, Y サイズの単位はインチですが、他の単位も使用可能ですが（現在は cm のみ）。

## 50.52 Pm

`pm` ドライバは、グラフが描画される OS/2 プレゼンテーションマネージャウィンドウを提供します。そのウィンドウは最初のグラフが描画されたときに開かれます。このウィンドウは印刷、クリップボードへのコピー、いくつかの線種や色の調整のための機能、そしてそれ自身のオンラインヘルプを持っています。`multiplot` オプションもサポートされています。

書式:

```
set terminal pm {server {n}} {persist} {widelines} {enhanced} {"title"}
```

`persist` が指定されると、各グラフはそれぞれ自身のウィンドウを持ち、そのすべてのウィンドウは `gnuplot` が終了した後も開いたままになります。`server` が指定されると、全てのグラフは同じウィンドウ内に現われ、それは `gnuplot` 終了後も開いたままになります。このオプションは、さらに追加の数引数を取り、その数字はサーバプロセスのインスタンスになります。よって、同時に複数のサーバウィンドウを使うことができます。

`widelines` が指定されると、全てのグラフは幅の広い線で描かれます。`enhanced` が指定されると `enhanced postscript` オプション（詳細は `set terminal postscript enhanced` 参照）と同じ書式で上付き文字や下付き文字、複数のフォントを使うことができます。基本的な PostScript フォントのフォント名は 1 文字に省略できます。

`title` が指定されると、それは描画ウィンドウのタイトルとして使われます。それはサーバインスタンス名としても使われ、それは追加の数引数を上書きします。

線の幅は `set linestyle` で変更できます。

## 50.53 Png

書式:

```
set terminal png
{{no}transparent} {{no}interlace}
{{no}truecolor} {rounded|butt}
{tiny | small | medium | large | giant}
{font <face> {<pointsize>}}
{size <x>,<y>} {{no}crop}
{{no}enhanced}
{<color0> <color1> <color2> ...}
```

PNG 画像は libgd によって生成されますが、オプションで libfreetype を使って TrueType フォントや Adobe Type 1 フォントを使うこともできます。Version 1.8 以上の libgd が必要です。

`transparent` はドライバに透明化（transparent）PNG を生成するよう指示します。この場合最初の指定色が透明化される色になります。デフォルトでは `notransparent` です。

`interlace` はドライバにインターレース GIF を生成するよう指示します。デフォルトでは `nointerlace` です。

`butt` は線分の描画で、その端の点でのみだしを起こさない描画メソッドを使うようドライバに指示します。この設定は、線幅が 1 より大きい場合にのみ有効です。この設定は、水平線、垂直線の描画の場合に有用でしょう。デフォルトは `rounded`（丸め）で、このオプションには libgd の Version 2.0 以上が必要です。

PNG の描画を、ImageMagick パッケージのプログラム 'display' を使って、出力をパイプに渡すことで以下のようにうまく見ることができます:

```
set term png
set output '| display png:-'
```

連続する plot コマンドの出力は、display のウィンドウで対話的に <space> を打つことで見てください。その中から特定の一つをディスクに保存するには display ウィンドウ上でマウス左ボタンをクリックして save を選択してください。

gd library によって次の 5 つの基本フォントが直接サポートされています: tiny (5x8 ピクセル), small (6x12 ピクセル), medium, (7x13 Bold), large (8x16), giant (9x15 ピクセル)。これらのフォントは大きさを変更したり回転したり (純粋に水平か垂直な文字列しか使えません) することはできません。

gnuplot が TrueType (\*.ttf) フォントか Adobe Type 1 (\*.pfa) フォントをサポートするように構築されている場合、オプション font <face> {<pointsize>} でそれを選択できます。<face> は、そのフォントファイルのフルパス名かフォントフェース名 (font face name) ですが、後者は、環境変数 GDFONTPATH で指示されるディレクトリ中のファイル名の先頭部分であると見なされるものです。よって、'set term png font "Face"' は、<あるディレクトリ>/Face.ttf か <あるディレクトリ>/Face.pfa を探そうとします。TrueType も Adobe Type 1 フォントも自由に大きさを変更できますし、任意の角度で回転できます。フォントを指定しなかった場合、gnuplot はより良いデフォルトフォントを見つけるために環境変数 GNUPLOT\_DEFAULT\_GDFONT を参照します。

enhanced は拡張テキスト処理機能 (上付き文字、下付き文字、および複数のフォントの利用) を有効にします。詳細は enhanced を参照してください。PNG/JPEG 出力ドライバは、それ自身拡張処理モードの命令をすべてサポートしていますが、これらの機能のうちいくつかは、それが使用している libgd イブリのバージョンやどのようなフォントが利用可能になっているかに依存します。

サイズ <x,y> はピクセル単位で与えます。デフォルトは 640x480 です。ピクセル数は set size コマンドで縮尺することでも変更できます。crop は完成した描画グラフの端の空白領域を取り除いて、結果として画像サイズを小さくします。デフォルトは nocrop です。

個々の色は 'xrrggb' の形式で指定します。ここで、x はそのまま文字 'x' を意味し、'rrggb' は赤、緑、青の 16 進数での成分です。例えば 'x00ff00' は緑を表します。背景色を最初に指定し、その後ろに境界色、X 軸と Y 軸の色、グラフの描画色、と続きます。色数は最大 256 色まで設定できます。

例:

```
set terminal png medium size 640,480 \
    xffffff x000000 x404040 \
    xff0000 xffa500 x66cdaa xcdb5cd \
    xadd8e6 x0000ff xdda0dd x9500d3 # デフォルト
```

この例では透明化されない背景色が白で、境界は黒、軸は灰色、そして描画色として赤、橙色、青緑色 (medium aquamarine)、アザミ色 (thistle 3)、明青色 (light blue)、青、かきつばた色 (plum)、暗いすみれ色 (dark violet) の 8 色を使います。

```
set terminal png font arial 14 size 800,600
```

これは、環境変数 GDFONTPATH で指定されたディレクトリにある 'arial' という名前で 14pt のフォントサイズの TrueType フォントを検索します。

```
set terminal png transparent xffffff \
    x000000 x202020 x404040 x606060 \
    x808080 xA0A0A0 xC0C0C0 xE0E0E0
```

これは透明化される背景が白で、境界は黒、軸は暗い灰色、描画色はグレイスケールで 6 色を使うことになります。

## 50.54 Postscript

postscript ドライバではいくつかのオプションが設定できます。

書式:

```

set terminal postscript {default}
set terminal postscript {landscape | portrait | eps}
    {enhanced | noenhanced}
    {defaultplex | simplex | duplex}
    {fontfile [add | delete] "<filename>"}
        | nofontfiles}
    {level1 | leveldefault}
    {color | colour | monochrome}
    {solid | dashed}
    {dashlength | dl <DL>}
    {linewidth | lw <LW>}
    {rounded | butt}
    {palfuncparam <samples>{,<maxdeviation>}}
    {size <XX>{unit},<YY>{unit}}
    {blacktext | colortext | colourtext}
    {{font} "fontname{,fontsize}" <fontsize>}}

```

以下のようなエラーメッセージが出た場合:

```
"Can't find PostScript prologue file ... "
```

`postscript prologue` の項目を見て、その指示に従ってください。

`landscape` と `portrait` は出力が横置か、縦置かを選択します。`eps` モードは EPS (Encapsulated PostScript) 出力を生成しますが、これは通常の PostScript に、それを他の多くのアプリケーションで取り込むことができるよういくつかの行を追加したものです (追加される行は PostScript のコメント行なので、よってそれ自身もちゃんと印刷できます)。EPS 出力を得るには `eps` モードを使用し、1 つのファイルには 1 つのグラフのみ、としてください。`eps` モードではフォントも含めてグラフ全体がデフォルトの大きさの半分に縮められます。

`enhanced` は拡張テキストモード (上付き文字、下付き文字、および複数のフォントの利用) の機能を有効にします。詳細は `enhanced` を参照してください。`blacktext` は、たとえカラー モードでも全ての文字列を黒で書きます。

PostScript の両面印刷命令 (`duplex`) は、プリンタで 1 枚の紙に両面印刷することを可能にします。`defaultplex` はプリンタのデフォルトの設定を使用し、`simplex` は紙の片面のみ印刷、`duplex` は両面印刷を行ないます (あなたのプリンタがそれを行なえないなら無視されます)。

"`<fontname>`" は有効な PostScript フォントの名前で、`<fontsize>` は PostScript ポイント単位でのフォントの大きさです。標準的な `postscript` フォント以外に、数式を表現するのに便利な oblique Symbol フォント ("Symbol-Oblique") が定義されています。

`default` は全てのオプションを以下のデフォルトの値に設定します: `landscape`, `monochrome`, `dashed`, `dl 1.0`, `lw 1.0`, `defaultplex`, `noenhanced`, "Helvetica", 14pt。PostScript のグラフのデフォルトの大きさは、10 インチの幅で 7 インチの高さです。オプション `color` はカラーを有効にし、`monochrome` は各要素を黒と白描画します。さらに、`monochrome` は灰色の palette も使用しますが、これは、明示的に `colorspec` で指定された部品の色を変更しません。

`solid` は全てのグラフの点線のパターンを実線で上書きします。`dashlength` または `dl` は点線の線分の長さを `<DL>` (0 より大きい実数) に設定し、`linewidth` または `lw` は全ての線の幅を `<LW>` に設定します。

デフォルトでは、生成される PostScript コードは、特にフィルタリングや `filledcurves` のようなでこぼこな領域のパターン塗りつぶしにおいて、PostScript Level 2 として紹介されている言語機能を使います。PostScript Level 2 の機能は条件的に保護されていて、PostScript Level 1 のインタープリタがエラーを出さず、むしろメッセージか PostScript Level 1 による近似であることを表示するようになっています。`level1` オプションは、これらの機能を近似する PostScript Level 1 で代用し、PostScript Level 2 コードを一切使用しません。これは古いプリンタや、Adobe Illustrator の古いバージョンなどで必要になるかもしれません。このフラグ `level1` は出力された PostScript ファイルのある一行を手で編集することで、後から強制的に PostScript Level 1 機能を ON/OFF にすることもできます。`level2` のコードが含まれている場合、上の機能は現われないか、このフラグがセットされた場合、あるいは PostScript インタプリタプログラムが `level2` 以上の PostScript を解釈するとは言わなかった場合に警告文に置き換わります。

`rounded` は、線の端や接合部を丸くし、デフォルトの `butt` は尖った端と角張った接合部を使用します。

`palfuncparam` は `set palette functions` から出力の傾きをどのようにコード化するかを制御します。解析的な色の成分関数 (`set palatte functions` で設定される) は、postscript 出力では傾きの線形補完を用いてコード化されます：まず色の成分関数が `<samples>` 個の点で標本化され、そしてそれらの点は、結果として線形補完との偏差が `<maxdeviation>` 以内に収まるように削除されます。ほとんど全ての有効なパレットで、デフォルトの `<samples> =2000` と `<maxdeviation>=0.003` の値をそのまま使うのが良いでしょう。

PostScript 出力のデフォルトの大きさは 10 インチ x 7 インチです。EPS 出力のデフォルトの大きさは 5 x 3.5 インチです。オプション `size` はこれらをユーザが指定したものに変更します。デフォルトでは X と Y のサイズの単位はインチとみなされますが、他の単位 (現在は cm のみ) も使うことはできます。描画の BoundingBox (PostScript ファイルの外枠) は、サイズが変更された画像を丁度含むように正しく設定されます。スクリーン座標は、オプション `size` で指定された描画枠の全体が 0.0 から 1.0 になります。注意：これは、以前は、出力形式での設定よりも、コマンド `set size` で設定した方がいい、と言っていたことの変更を意味します。以前の方法では BoundingBox は変更されずに残ってしまい、スクリーン座標が実際の描画の限界に対応していませんでした。

`fontfile` や `fontfile add` で指定されたフォントは、そのフォントのフォント定義を直接 postscript Type 1, TrueType フォントから gnuplot の postscript 出力の中にカプセル化します。よって、その埋め込まれたフォントは見出し、タイトルなどに使うことができます。詳細は postscript fontfile を参照してください。`fontfile delete` によってフォントファイルを埋め込まれるファイルの一覧から取り除くことができます。`nofontfiles` は埋め込みフォントのリストをクリアします。

例：

```
set terminal postscript default      # 以前の postscript
set terminal postscript enhanced    # 以前の enhpost
set terminal postscript landscape 22 # 以前の psbig
set terminal postscript eps 14      # 以前の epsf1
set terminal postscript eps 22      # 以前の epsf2
set size 0.7,1.4; set term post portrait color "Times-Roman" 14
set term post "VAGRoundedBT-Regular" 14 fontfile "bvrr8a.pfa"
```

線の幅と点の大きさは `set style line` で変更できます。

postscript ドライバは約 70 種類の異なる点種をサポートしていて、これは `plot` や `set style line` の `pointtype` オプションで選択できます。

gnuplot と Postscript に関する多分有用と思われるファイルが gnuplot の配布物、またはその配布サイトの `/docs/psdos` サブディレクトリ内にいくつか含まれています。そこには "ps\_symbols.gpi" (実行すると postscript ドライバで使える全ての記号を紹介する "ps\_symbols.ps" というファイルを生成する gnuplot のコマンドファイル)、"ps\_guide.ps" (拡張された書式に関する要約と、文字列内で 8 進コードで生成されるもの、symbol フォント等を含む PostScript ファイル)、"ps\_file.doc" (gnuplot で作られる PostScript ファイルの構造の説明を含むテキストファイル)、"ps\_fontfile.doc.tex" (数式フォントの文字の一覧と LaTeX のフォントの埋め込みに関する短い説明を含む LaTeX ファイル) があります。

PostScript ファイルは編集可能で、一度 gnuplot でそれを作れば、それを望むように修正することは自由に行なえます。editing postscript の節を参照すればそのためのヒントがいくつか得られるでしょう。

#### 50.54.1 Enhanced postscript

いくつかの出力形式では拡張テキストモードをサポートしていて、これはテキスト文字列中に文書整形情報を取り加えて埋め込みます。

拡張テキスト制御記号		
制御記号	例	説明
^	a^x	上付き文字
-	a_x	下付き文字
@	@x or a@^b_c	空ボックス (幅がない)
&	&{space}	指定した長さのスペースを挿入
~	~a{.8-}	'a' の上に ' ' を、現在のフォントサイズの .8 倍持ち上げた位置に重ね書き

中カッコは一文字が期待される箇所に複数の文字を書く場合に使われます(例えば`2^{10}`)。フォント、およびそのサイズを変更するには、以下の形式を使ってください: `{/[fontname][=fontsize | *fontscale] text}` よって、例えば`{/Symbol=20 G}`は20ポイントのガンマに、`{/*0.75 K}`は現在の有効なフォントサイズの3/4の大きさのKになります(文字`'/'`は必ず`'{'`の後ろの最初の文字でなくなりません)。

`set encoding`によって文字エンコード系が変更されている場合、「/」の次に「\_」を書くとデフォルトの文字エンコード系が代わりに使われます。しかし、これは`Symbol`フォントを使う場合は必要ありません。それは`/Symbol`は自分自身のエンコード系を使い、`gnuplot`は他のエンコーディング系をそれには適用しないからです。

空ボックス(phantom box)は`a@^b_c`の上付き文字と下付き文字を揃えるときに有用ですが、文字にアクセント記号を重ねる場合にはうまく働きません(後者を行なうには`set encoding iso_8859_1`でISO Latin-1 エンコーディング系に変更し、そこに含まれるたくさんの種類のアクセントや他のダイアクリティカルマークのついた文字を利用する方がいいでしょう)。そのボックスはスペーシングが行なわれないので、ボックス内(つまり@の後ろ)の上付き文字や下付き文字を短く出力するのに適しています。

ある文字列と同じ長さのスペースを文字`'&'`を使うことで入れることができます。すなわち、

```
'abc&{def}ghi'
```

は以下を生成します(abcとghiの間は3文字分の空白):

```
'abc    ghi'
```

文字`'~'`は、次の文字、またはカッコで囲まれた文字列に、それに続く文字またはカッコで囲まれた文字列を重ね書きします。2番目の文字は最初の文字にあわせて水平方向にセンタリングされます。よって`'~a/'`は`'a'`を貫くようなスラッシュが得られます。2番目の文字は、その前に数字を置くことで垂直方向に移動させることができます。その数字は現在のフォントサイズに対する割合を意味し、それに従って文字が上がったり下がったりします。この場合数字と文字列は1文字より長くなるのでカッコで囲む必要があります。重ね書きされる文字列が数字から始まっている場合は、垂直にずらす値と文字列との間にスペースを入れてください(`'~{abc}{.5 000}'`)。それ以外はスペースは不要です(`'~{abc}{.5 — }'`)。一方、あるいは両方のフォントを変更することもできます(`'~a{.5 /*.2 o}' — 'a'`その1/5の大きさの'o' — この場合数字とスラッシュの間のスペースは必要です)が、その文字列が始まった後で変更することはできません。それぞれの文字列内で、他の特殊な書式を使うこともできません。もちろん、`'~a{\\"}`のように特殊な文字をエスケープすること(下記参照)は可能です。

\文字コード(8進数)を指定することで特殊な記号を指定することができます。例えば、`{/Symbol \245}`は無限大の記号になります。

制御文字は、`\\"`や`\{`などのように`\`を使ってエスケープできます。

しかし、二重引用符内の文字列は單一引用符内の文字列とは異なって解釈されることを知っておいてください。主な違いは、二重引用符内の文字列ではバックスラッシュは2つ重ねる必要があることです。

例(これらは言葉で説明するのは難しいのでとりあえずやってみてください):

```
set xlabel 'Time (10^6 {/Symbol m}s)'
set title '{/Symbol=18 \362@_{/=9.6 0}^{{/=12 x}} \{/Helvetica e^{-{/Symbol m}^2/2} d{/Symbol m},'
```

`gnuplot`ソース配布物内の`/docs/psdoc`サブディレクトリにあるファイル"ps-guide.ps"に、拡張された書式に関する例が更に載っています。

#### 50.54.2 Editing postscript

PostScript言語はとても複雑な言語で、ここで詳細を記述することはとてもできません。それでも、`gnuplot`で作られるPostScriptファイルには、致命的なエラーをそのファイルに導入してしまう危険性のない変更を行なうことが可能な部分があります。

例えば、PostScriptの文`"/Color true def"`(`set terminal postscript color`コマンドに答えてファイルに書き込まれます)を変更して、その描画を白黒のものにする方法はおわかりでしょう。同様に、線の色、文字の色、線の太さ(weight)、記号のサイズも、本当に簡単に書き換えられるでしょう。タイトルや見出しなどの文字列の誤植や、フォントの変更も編集可能でしょう。任意のものの配置も変更できますし、もちろん、任意のものを追加したり、削除したりもできますが、それらの修正はPostScript言語の深い知識が必要でしょう。

gnuplot によって作られる PostScript ファイルの構成に関しては、gnuplot のソース配布物内の docs/ps ディレクトリのテキストファイル "ps\_file.doc" に述べられています。

### 50.54.3 Postscript fontfile

オプション fontfile または fontfile add は 1 つのファイル名を引数として持ち、そのファイルを postscript 出力内にカプセル化して埋め込み、それによって様々な文字列要素（ラベル、目盛り見出し、タイトル等）をそのフォントで出力することを可能にします。オプション fontfile delete も 1 つのファイル名を引数に持ち、そのファイル名をカプセル化されるファイルのリストから削除します。

postscript 出力ドライバはいくつかのフォントファイル形式を認識します： ASCII 形式の Type 1 フォント（拡張子 ".pfa"）、バイナリ形式の Type 1 フォント（拡張子 ".pfb"）、TrueType フォント（拡張子 ".ttf"）。 pfa ファイルは直接認識されますが、pfb と ttf ファイルは gnuplot の実行中に並行して変換され、そのため適切な変換ツール（下記参照）がインストールされている必要があります。ファイル名は拡張子も含めて完全な形で指定する必要があります。各 fontfile オプションはちょうど一つのフォントファイル名に対応しますので、複数のフォントファイルを埋め込むためにはこのオプションを複数回使って下さい。

フォントファイルは、作業ディレクトリ、そして set fontpath で与えられるフォントパス一覧の全てのディレクトリが検索されます。さらに、環境変数 GNUPLOT\_FONTPATH でフォントパスを設定することもできます。それが設定されていない場合はデフォルトの検索リストが使われますが、これはシステムに依存します。詳細は set fontpath を参照して下さい。

埋め込まれたフォントファイルを使うには、フォント名（通常ファイル名と同じではありません）を指定する必要があります。対話モードで fontfile オプションを使ってフォントを埋め込んだ場合、フォント名はスクリーンに表示されます。例：

```
Font file 'p0520041.pfb' contains the font 'URWPalladioL-Bold'. Location:  
/usr/lib/X11/fonts/URW/p0520041.pfb
```

pfa や pfb フォントでは、フォント名はフォントファイル内に見つけることができます。フォントファイル中に "/FontName /URWPalladioL-Bold def" のような行がありますが、この真中の物から / を除いたものがフォント名です。この例の場合は "URWPalladioL-Bold" となります。TrueType フォントでは、フォント名はバイナリ形式で保存されているので見つけるのは容易ではありません。さらに、その名前は多くの場合、Type 1 フォント（実行中に TrueType が変換される形式である）ではサポートされていない、スペースを含んだ形式になっています。そのため、フォント名はそこからスペースを取り除いた形に変換されます。gnuplot で使うために生成されたフォント名が何であるかを知る最も簡単な方法は、gnuplot を対話モードで起動して、以下のように入力することです："set terminal postscript fontfile '<filename.ttf>'".

フォントファイル (ttf, pfb) を pfa 形式に変換するために、フォントファイルを読んで、そして変換結果を標準出力に吐き出す変換ツールが必要になります。その出力を標準出力に書き出すことができない場合、実行中の変換はできません。

pfb ファイルに対しては、例えば "pfbtrops" が使えます。それがシステムにインストールされていれば、実行中の変換はうまく行くはずです。pfb ファイルのカプセル化をちょっとやってみましょう。もしプログラムの変換時に正しくツールを呼び出している場合は、どのようにツールを呼び出したら良いかを環境変数 GNUPLOT\_PFBTOPFA に、例えば "pfbtrops %s" のように定義して下さい。%s はフォントファイル名に置き換えられますので、これはその文字列に必ず必要です。

実行中の変換をしたくなくて、けれども pfa 形式のファイルは必要である場合、"pfb2pfa" という C で書かれた簡単なツールを使えば良いでしょう。これは大抵の C コンパイラでコンパイルでき、たくさんの ftp サーバに置いてあります。例えば

```
ftp://ftp.dante.de/tex-archive/fonts/utilities/ps2mf/
```

実際に "pfbtrops" と "pfb2ps" は同じ作業を行います。"pfbtrops" は結果の pfa コードを標準出力に出力しますが、"pfbtopfa" はファイルに出力します。

TrueType フォントは、例えば "ttf2pt1" というツールを使って Type 1 pfa フォーマットに変換できます。これは以下にあります：

```
http://ttf2pt1.sourceforge.net/
```

もし gnuplot に組み込まれている変換手順がうまく行かない場合、変換コマンドは環境変数 GNUPLOT\_TTF2PT1 で変更できます。ttf2pt1 を使う場合は、それを "ttf2pt1 -a -e -W 0 %s -" のように設定して下さい。ここでも%s はファイル名を意味します。

特殊な用途のために、パイプも使えるようになっています（パイプをサポートしている OS 上で）。ファイル名を "<" で始め、その後にプログラム呼び出しを追加します。そのプログラム出力は標準出力への pfa データでなければいけません。結果として pfa ファイルを、例えば以下のようにしてアクセスできることになります: set fontfile "< cat garamond.pfa"。

Type 1 フォントを取り込むことは、例えば LaTeX 文書中に postscript ファイルを取り込む場合に使えます。pbf 形式の "european computer modern" フォント ("computer modern" フォントの一種) が各地の CTAN サーバに置かれています。

```
ftp://ftp.dante.de/tex-archive/fonts/ps-type1/cm-super/
```

例えば、ファイル "sfrm1000.pfb" は、中太、セリフ付き、立体の 10 ポイントのフォント（フォント名 "SFRM1000"）です。computer modern フォントは今でも数式を書くのに必要ですが、それは以下にあります:

```
ftp://ftp.dante.de/tex-archive/fonts/cm/ps-type1/bluesky
```

これらによって、TeX 用の任意の文字も使えます。しかし、computer modern フォントは少しエンコーディングがおかしくなっています（このため、文字列には cmr10.pfb の代わりに sfrm1000.pfb を使うべきです）。TeX フォントの使用法はいくつかのデモの一つで知ることができます。gnuplot のソース配布物の /docs/psdoc に含まれるファイル "ps\_fontfile\_doc.tex" に TeX 数学フォントの文字の一覧表が含まれています。

フォント "CMEX10"（ファイル "cmex10.pfb"）を埋め込むと、gnuplot は追加フォント "CMEX10-Baseline" も定義します。それは、他の文字にあうように垂直方向にずらされたものです（CMEX10 は、記号の天辺にベースラインがあります）。

#### 50.54.4 Postscript prologue

各 PostScript 出力ファイルは %%Prolog セクションを含みますし、例えば文字エンコーディングなどを含む追加ユーザ定義セクションを含むかもしれません。これらのセクションは、gnuplot の実行ファイル中にコンパイルされている、あるいはあなたのコンピュータの別のところに保存されている PostScript prologue ファイル群からコピーされます。この振舞いと、それらのファイルの置き場所のデフォルトのディレクトリは、gnuplot のコンパイル時にセットされます。しかしこれは、環境変数 GNUPLOT\_PS\_DIR を定義するかまたは gnuplot コマンドの set loadpath を使うことで制御できます。詳細は set loadpath を参照してください。

#### 50.55 Pslatex and pstex

pslatex ドライバは LaTeX で後処理される出力を生成し、pstex ドライバは TeX で後処理される出力を生成します。pslatex は dvips と xdvi で認識可能な \special 命令を使用します。pstex で生成される図は、任意の plain-TeX ベースの TeX (LaTeX もそうです) で取り込むことができます。

書式:

```
set terminal [pslatex | pstex] {default}
set terminal [pslatex | pstex]
    {rotate | norotate}
    {oldstyle | newstyle}
    {auxfile | noauxfile}
    {level1 | leveldefault}
    {color | colour | monochrome}
    {solid | dashed}
    {dashlength | dl <DL>}
    {linewidth | lw <LW>}
    {rounded | butt}
    {palfuncparam <samples>{,<maxdeviation>}}
    {size <XX>{unit},<YY>{unit}}
    {<font_size>}
```

以下のようなエラーメッセージが出た場合:

```
"Can't find PostScript prologue file ... "
```

`postscript prologue` の項目を見て、その指示に従ってください。

オプション `color` はカラーを有効にし、`monochrome` は各要素を黒と白描画します。さらに、`monochrome` は灰色の `palette` も使用しますが、これは、明示的に `colorspec` で指定された部品の色を変更しません。

`solid` は全てのグラフの点線のパターンを実線で上書きします。`dashlength` または `dl` は点線の線分の長さを `<DL>` (0 より大きい実数) に設定し、`linewidth` または `lw` は全ての線の幅を `<LW>` に設定します。

デフォルトでは、生成される PostScript コードは、特にフィルタリングや `filledcurves` のようなでこぼこな領域のパターン塗りつぶしにおいて、PostScript Level 2 として紹介されている言語機能を使います。PostScript Level 2 の機能は条件的に保護されていて、PostScript Level 1 のインタープリタがエラーを出さず、むしろメッセージか PostScript Level 1 による近似であることを表示するようになっています。`level1` オプションは、これらの機能を近似する PostScript Level 1 で代用し、PostScript Level 2 コードを一切使用しません。これは古いプリンタや、Adobe Illustrator の古いバージョンなどで必要になるかもしれません。このフラグ `level1` は出力された PostScript ファイルのある一行を手で編集することで、後から強制的に PostScript Level 1 機能を ON/OFF にすることもできます。`level2` のコードが含まれている場合、上の機能は現われないか、このフラグがセットされた場合、あるいは PostScript インタプリタプログラムが `level2` 以上の PostScript を解釈するとは言わなかった場合に警告文に置き換わります。

`rounded` は、線の端や接合部を丸くし、デフォルトの `butt` は尖った端と角張った接合部を使用します。

`palfuncparam` は `set palette functions` から出力の傾きをどのようにコード化するかを制御します。解析的な色の成分関数 (`set palette functions` で設定される) は、`postscript` 出力では傾きの線形補完を用いてコード化されます：まず色の成分関数が `<samples>` 個の点で標本化され、そしてそれらの点は、結果として線形補完との偏差が `<maxdeviation>` 以内に収まるように削除されます。ほとんど全ての有効なパレットで、デフォルトの `<samples> = 2000` と `<maxdeviation> = 0.003` の値をそのまま使うのが良いでしょう。

PostScript 出力のデフォルトの大きさは 10 インチ x 7 インチです。EPS 出力のデフォルトの大きさは 5 x 3.5 インチです。オプション `size` はこれらをユーザが指定したものに変更します。デフォルトでは X と Y のサイズの単位はインチとみなされますが、他の単位 (現在は cm のみ) も使うことはできます。描画の `BoundingBox` (PostScript ファイルの外枠) は、サイズが変更された画像を丁度含むように正しく設定されます。スクリーン座標は、オプション `size` で指定された描画枠の全体が 0.0 から 1.0 になります。注意：これは、以前は、出力形式での設定よりも、コマンド `set size` で設定した方がいい、と言っていたことの変更を意味します。以前の方法では `BoundingBox` は変更されずに残ってしまい、スクリーン座標が実際の描画の限界に対応していませんでした。

`rotate` が指定されると `y` 軸の見出しが回転されます。`<font_size>` は希望するフォントの (ポイント単位での) 大さです。

`auxfile` が指定されると、ドライバは PostScript コマンドを、LaTeX ファイルに直接出力する代わりに、補助ファイルに書き出すようになります。これは、`dvips` がそれを扱えないくらい大きいグラフである場合に有用です。補助 PostScript ファイルの名前は、`set output` コマンドで与えられる TeX ファイルの名前から導かれるもので、それはその最後の `.tex` の部分 (実際のファイル名の最後の拡張子の部分) を `.ps` で置き換えたもの、または、TeX ファイルに拡張子がないならば `.ps` を最後に付け足したものになります。`.ps` ファイルは `\special{psfile=...}` という命令で `.tex` ファイルに取り込まれます。`multiplot` モード以外では、次の描画を行なう前にその出力ファイルをクローズするのを忘れないでください。

version 4.2 より前の gnuplot は ps(la)tex 出力形式では 5 x 3 インチの出力でしたが、現在では 5 x 3.5 インチになっています。これは postscript eps 出力形式に合わせた変更です。加えて、文字幅は、以前の `epslatex` 出力形式はフォントサイズの 50% と見なしましたが、現在は 60% と評価しています。従来の形式に戻すには、オプション `oldstyle` を指定してください。

`pslatex` ドライバは文字列の配置の制御に特別な方法を提供します：(a) ‘{’ で始まる文字列は、‘}’ で閉じる必要がありますが、その文字列全体が LaTeX によって水平方向にも垂直方向にもセンタリングされます。(b) '[' で始まる文字列の場合は、位置の指定をする文字列 (t,b,l,r のうち 2 つまで) が続き、次に ']{'、文字列本体、で最後に '}' としますが、この文字列は LaTeX が LR-box として整形します。`\rule{}{}` を使えばさらに良い位置合わせが可能でしょう。

ここに記述されていないオプションは Postscript terminal のものと同一ですので、それらが何を行なうのかを知りたければそちらを参照してください。

例：

```
set term pslatex monochrome dashed rotate      # デフォルトに設定
```

PostScript コマンドを "foo.ps" に書き出す:

```
set term pslatex auxfile
set output "foo.tex"; plot ...; set output
```

見出しの位置合わせに関して: gnuplot のデフォルト (大抵それなりになるが、そうでないこともある):

```
set title '\LaTeX\ -- $ \gamma'
```

水平方向にも垂直方向にもセンタリング:

```
set label '\LaTeX\ -- $ \gamma' at 0,0
```

位置を明示的に指定 (上に合わせる):

```
set xlabel '[t]\LaTeX\ -- $ \gamma'
```

他の見出し - 目盛りの長い見出しに対する見積り:

```
set ylabel '[r]\LaTeX\ -- $ \gamma $\rule{7mm}{0pt}'
```

線幅と点の大きさは set style line で変更できます。

## 50.56 Pstricks

pstricks ドライバは LaTeX の "pstricks.sty" マクロパッケージと共に使われることを意図しています。これは eepic や latex ドライバに代わる選択肢の一つです。"pstricks.sty" は必要ですが、もちろん PostScript を解釈するプリンタ、または Ghostscript のような変換ソフトも必要です。

PSTricks は anonymous ftp で Princeton.EDU の /pub ディレクトリから取得できます。このドライバは、PSTricks パッケージの全ての能力を使おうなどとは全く考えてはいません。

書式:

```
set terminal pstricks {hacktext | nohacktext} {unit | nounit}
```

最初のオプションは、あまり綺麗ではない方法で数字のより良い出力を生成するもので、2つ目のオプションはグラフを縮尺する際には必要です。デフォルトでは hacktext と nounit です。

## 50.57 Qms

qms ドライバは QMS/QUIC レーザープリンタ、Talaris 1200、その他をサポートします。オプションはありません。

## 50.58 Regis

regis ドライバは REGIS グラフィック言語での出力を生成します。このドライバには色を 4 色使うか (デフォルト) 16 色使うかのオプションがあります。

書式:

```
set terminal regis {4 | 16}
```

## 50.59 Rgip

rgip と uniplex ドライバは RGIP メタファイルをサポートします。それらはいくつかのグラフを一つのページの中に入れることができます。一つの出力ファイル内には 1 つのページしか許されません。

書式:

```
set terminal rgip | uniplex {portrait | landscape}
{[<horiz>, <vert>]} {<fontsize>}
```

フォントサイズは 1–8 間での値が許されていて、デフォルトでは 1 です。デフォルトの方向 (layout) は横置き (landscape) です。グラフはそのページに horizxvert 個が格子状に配置されます。デフォルトは [1,1] です。

例:

```
set terminal uniplex portrait [2,3]
```

これは 1 ページ 6 つのグラフを横に 2 つ縦に 3 つ並べ、縦置き (portrait) に出力します。

## 50.60 Sun

sun ドライバは SunView ウィンドウシステムをサポートしています。オプションはありません。

## 50.61 Svg

このドライバは W3C SVG (Scalable Vector Graphics) フォーマットを生成します。

書式:

```
set terminal svg {size <x>,<y> {|fixed|dynamic}}
    {{no}enhanced}
    {fname "<font>"} {fsizze <fontsize>}
    {font "<fontname>{,<fontsize>}"}
    {fontfile <filename>}
    {rounded|butt} {linewidth <lw>}
```

ここで <x> と <y> は生成される SVG グラフのサイズですが、dynamic は svg ビューワに描画のリサイズを許し、fixed は絶対サイズを要求します (デフォルト)。

linewidth <w> は図の中で使用される全ての線の幅を因子 <w> だけ増加させます。

<font> はデフォルトとして使われるフォント名 (デフォルトでは Arial)、<fontsize> はポイント単位でのフォントサイズ (デフォルトは 12) です。ただし、gnuplot は、今のところ出力ファイルへのフォントの埋め込み機能をサポートしていませんので、svg ビューワソフトはそのファイルの表示の際には別の代用フォントを使うことになるでしょう。

svg 出力形式は拡張テキスト処理機能 (enhanced) をサポートしています。これはフォント指定や他の書式命令をラベルや他の文字列内に埋め込むことを可能にします。拡張テキスト処理モードの書式指定は他の出力形式の場合と同じです。詳細は enhanced を参照してください。

SVG では、SVG 文書中にフォントを直接埋め込むこともできますし、好きなフォントへのハイパーリンクを与えることもできます。fontfile オプションには、結果として出力される SVG ファイルの <defs> セクションの中にコピーされるローカルファイル名を指定します。このファイルは、それ自身がフォントを含んでいるか、または期待するフォントを参照するハイパーリンクを生成するための行を含むものどちらかです。gnuplot は、環境変数 GNUPLOT\_FONTPATH のディレクトリリストから要求されたファイルを探します。

## 50.62 Svga

svga ドライバは SVGA グラフィックの PC をサポートします。これは DJGPP でコンパイルされた場合のみ使うことができます。オプションはフォントに関するもののみです。

書式:

```
set terminal svga {"<fontname>"}
```

## 50.63 Tek40

このドライバ群は VT-ライクな端末のいくつかをサポートします。tek40xx は Tektronix 4010 とその他ほとんどの TEK エミュレータをサポートします。vttek は VT-ライクな tek40xx 端末エミュレータをサポートします。kc-tek40xx は、カラーの MS-DOS Kermit Tek4010 ターミナルエミュレータを、

km-tek40xx はその白黒版をサポートします。selanar は Selanar グラフィック端末をサポートします。bitgraph は BBN Bitgraph 端末をサポートします。いずれもオプションはありません。

### 50.64 Tek410x

tek410x ドライバは Tektronix 410x, 420x ファミリーをサポートしています。オプションはありません。

### 50.65 Texdraw

texdraw ドライバは LaTeX texdraw 環境をサポートします。それは texdraw パッケージの "texdraw.sty" と "texdraw.tex" と共に使用されることを仮定しています。

数ある中で、点 (point) は、LaTeX のコマンド "\Diamond", "\Box" などを使って描かれます。これらのコマンドは現在は LaTeX2e のコアには存在せず latexsym パッケージに含まれていますが、このパッケージ基本配布の一部であり、よって多くの LaTeX のシステムの一部になっています。このパッケージを使うことを忘れないでください。

オプションはありません。

### 50.66 Tgif

Tgif は X11 ベースのドローツールです — ただし、これは GIF に関して何かするわけではありません。

tgif ドライバは複数の点の大きさ (set pointsize で)、複数の見出し用のフォント、フォントサイズ (例えば set label "Hallo" at x,y font "Helvetica,34")、そして 1 ページ内の複数のグラフ描画をサポートします。軸の比率は変更されません。

書式:

```
set terminal tgif {portrait | landscape | default} {[<x,y>]}
               {monochrome | color}
               {{linewidth | lw} <LW>}
               {solid | dashed}
               {font "<fontname>"} {[<fontsize>]}
```

*[x,y]* にはそのページ内の x 方向、y 方向のグラフの数を指定し、color はカラー機能を有効にし、 linewidth は全ての線幅を *<LW>* 倍し、"*<fontname>*" には有効な PostScript フォント名、*<fontsize>* はその PostScript フォントの大きさを指定します。defaults は全てのオプションの値をデフォルトの値にセットします。デフォルトは portrait, [1,1], color, linewidth 1.0, dashed, "Helvetica", 18 です。

solid オプションは、編集作業中にそうであるように、線がカラーである場合に普通使われます。ハードコピーは白黒になることが多いので、その場合は dashed を選択すべきでしょう。

多重描画 (multiplot) は 2 種類の方法で実装されています。

その一つは、標準的な gnuplot の多重描画のやり方です:

```
set terminal tgif
set output "file.obj"
set multiplot
set origin x01,y01
set size xs,ys
plot ...
...
set origin x02,y02
plot ...
unset multiplot
```

より詳しい情報については set multiplot を参照してください。

もう一つの方法はドライバの [x,y] オプションです。この方法の長所は、原点 (origin) や大きさ (size) の設定をしなくても全てのグラフが自動的に縮尺され配置されることです。グラフの比 x/y は、自然な比 3/2 (または set size で設定されたもの) が保持されます。

両方の多重描画の実装が選択された場合、標準的なやり方の方が選択され、警告のメッセージが表示されます。

単一描画（または標準的な多重描画）の例:

```
set terminal tgif          # デフォルト
set terminal tgif "Times-Roman" 24
set terminal tgif landscape
set terminal tgif landscape solid
```

ドライバの持つ多重描画の仕組みを利用する例:

```
set terminal tgif portrait [2,4]  # 縦置、x-方向に 2 つ、y-方向
                                 # に 4 つのグラフ描画
set terminal tgif [1,2]          # 縦置、x-方向に 1 つ、y-方向
                                 # に 2 つのグラフ描画
set terminal tgif landscape [3,3] # 横置、両方の方向に 3 つのグ
                                 # ラフ描画
```

## 50.67 Tkcanvas

このドライバは Tcl/Tk ベース（デフォルト）、または Perl ベースの Tk canvas widget コマンドを生成します。これを使うには、"term.h" のこのドライバに対応する行のコメント記号を外すか適当な行を書き加えて gnuplot を make し直して、以下のように実行します。

```
gnuplot> set term tkcanvas {perlTk} {interactive}
gnuplot> set output 'plot.file'
```

そして "wish" を起動した後で、以下の Tcl/Tk コマンド列を実行してください:

```
% source plot.file
% canvas .c
% pack .c
% gnuplot .c
```

Perl/Tk の場合は以下のようにしてこのプログラムを使います:

```
use Tk;
my $top = MainWindow->new;
my $c = $top->Canvas->pack;
my $gnuplot = do "plot.pl";
$gnuplot->($c);
MainLoop;
```

gnuplot によって生成されたコードは "gnuplot" と呼ばれる手続きを作り、それはその引数を canvas の名前とします。その手続きが呼ばれると、それは canvas をクリアし、canvas のサイズを見つけ、その中に丁度収まるようにグラフを書きます。

2 次元の描画 (plot) では 2 つの手続きが追加されて定義されます: "gnuplot\_plotarea" は描画範囲の境界を含むリスト "xleft, xright, ytop, ybot" を canvas スクリーン座標で返し、2 つの対の軸の描画座標での範囲 "x1min, x1max, y1min, y1max, x2min, x2max, y2min, y2max" は手続き "gnuplot\_axisranges" を呼べば得られます。"interactive" オプションを指定すると、canvas の線分上でマウスをクリックするとその線分の中点の座標が標準出力に出力されるようになります。さらに、"user\_gnuplot\_coordinates" という手続きを定義することで、それに代わる新たな動作を起こすことも可能です。その手続きには以下の引数が渡されます: "win id x1s y1s x2s y2s x1e y1e x2e y2e x1m y1m x2m y2m"、これらは、canvas の名前、線分の id、2 つの座標系でのその線分の開始点の座標、終了点の座標、そして中点の座標です。中点の座標は対数軸に対してのみ与えられます。

tkcanvas の現在の版では multiplot も replot もサポートしていません。

## 50.68 Tpic

**tpic** ドライバは tpic \special の LaTeX picture 環境の描画をサポートします。これは latex や eepic ドライバに代わる別な選択肢です。点の大きさ (pointsize)、線の幅 (linewidth)、点線の点の間隔 (interval) に関するオプションがあります。

書式:

```
set terminal tpic <pointsize> <linewidth> <interval>
```

**pointsize** と **linewidth** は整数でミリインチ単位、**interval** は実数で単位はインチです。正でない値を指定するとデフォルトの値が使われます。デフォルトでは pointsize = 40, linewidth = 6, interval = 0.1 です。

LaTeX に関する全てのドライバは文字列の配置の制御に特別な方法を提供します: '{' で始まる文字列は、'}' で閉じる必要がありますが、その文字列全体が LaTeX によって水平方向にも垂直方向にもセンタリングされます。'{' で始まる文字列の場合は、位置の指定をする文字列 (t,b,l,r のうち 2 つまで) が続き、次に '}', 文字列本体、で最後に '}' としますが、この文字列は LaTeX が LR-box として整形します。 \rule{}{}{} を使えばさらに良い位置合わせが可能でしょう。

例: 見出しの位置合わせに関して: gnuplot のデフォルト (大抵それなりになるが、そうでないこともある):

```
set title '\LaTeX\ -- $ \gamma'
```

水平方向にも垂直方向にもセンタリング:

```
set label '{\LaTeX\ -- $ \gamma}' at 0,0
```

位置を明示的に指定 (上に合わせる):

```
set xlabel '[t]{\LaTeX\ -- $ \gamma}'
```

他の見出し - 目盛りの長い見出しに対する見積り:

```
set ylabel '[r]{\LaTeX\ -- $ \gamma$\rule{7mm}{0pt}}'
```

## 50.69 Unixpc

**unixpc** ドライバは AT&T 3b1 と AT&T 7300 Unix PC をサポートします。オプションはありません。

## 50.70 Unixplot

**unixplot** ドライバは Unix の "plot" グラフィック言語の出力を生成します。オプションはありません。このドライバは、GNU 版の plot を使っている環境ではコンパイルされません。その場合は代わりに gnugraph ドライバ (terminal) を利用してください。

## 50.71 Atari ST (via VDI)

**vdi** 出力ドライバは、AES-Windows ではなく VDI 経由でスクリーンに出力することを除けば atari 出力ドライバと全く同じです。

**vdi** 出力形式には、画面の色と文字のサイズを設定するオプションがあります。

書式:

```
set terminal vdi {<fontsize>} {<col0> <col1> ... <col15>}
```

色を指定する場合、必ず文字サイズを指定しなければいけません。個々の (16 色までの) 色は、3 枠の 16 進数で与え、各桁は赤、緑、青 (RGB) の強度を (この順で) あらわします。0-15 の範囲は、実際にスクリーンが持つカラーの範囲に調整されます。例えば通常の ST スクリーンでは、奇数と偶数の強度は同じになります。

例:

```

set terminal vdi 4    # 小さい (6x6) フォント
set terminal vdi 6 0  # 黒背景に白の白黒画面
set terminal vdi 13 0 fff f00 f0 ff f0f
    # 最初の 8 色を黒、白、赤、緑、青、シアン、紫とし、
    # 大きいフォント (8x16) を使用

```

また、環境変数 GNCOLORS が設定されていれば、それもオプション文字列として解釈されますが、明示的なターミナルオプションの方が優先されます。

## 50.72 Vgagl

ドライバ vgagl はマウスと pm3d を完全にサポートした、linux の高速なコンソールドライバです。デフォルトモードの設定には 環境変数 SVGALIB\_DEFAULT\_MODE を参照しますが、設定されていない場合は 256 色モードで有効な解像度のうち最も高いものを選択します。

書式:

```

set terminal vgagl \
    background [red] [[green] [blue]] \
    [uniform | interpolate] \
    [mode]

```

カラー モードは mode オプションで与えることもできます。G1024x768x256 のような記号的名称や整数で与えることができます。オプション background は [0, 255] の範囲の整数 1 つ、または 3 つの組を取ります。整数 1 つの場合はそれは背景の灰色の値と見なされ、3 つの組の場合はそれに対応した色が背景に取られます。相互に排他的なオプション interpolate と uniform は、三角形の塗りつぶしの際に色の補間が行うかどうか (デフォルトでは ON) を制御します。

高解像度モードを得るには、多分 libvga の設定ファイル (通常/etc/vga/libvga.conf) を修正する必要があるでしょう。VESA fb を使うのは良い選択ですが、それはカーネルのコンパイルが必要です。

vgagl ドライバは、以下のリストのうちの \*有効な\* vga モードの最初のものを使用します。

- vgagl の設定時に与えられるモード、例えば ‘set term vgagl G1024x768x256’ は最初に G1024x768x256 モードが有効かどうかチェックします。
- 環境変数 SVGALIB\_DEFAULT\_MODE
- G1024x768x256
- G800x600x256
- G640x480x256
- G320x200x256
- G1280x1024x256
- G1152x864x256
- G1360x768x256
- G1600x1200x256

## 50.73 VWS

VWS ドライバは VAX ウィンドウシステムをサポートします。オプションはありません。このドライバはディスプレイの状態 (白黒か、グレイスケールかカラーか) を自動検知します。全ての線種は実線で描画されます。

## 50.74 Vx384

vx384 ドライバは Vectrix 384 と Tandy カラープリンタをサポートします。オプションはありません。

## 50.75 Windows

windows ドライバでは 3 つのオプションを設定することができます。

書式:

```
set terminal windows {color | monochrome}
    {enhanced | noenhanced}
    {{font} "fontname{,fontsize}" {<fontsize>}}
```

`color, monochrome` は、カラー出力か白黒出力かの選択で、`enhanced` は拡張文字処理モード (enhanced text mode) の機能 (上付文字、下付文字やフォントの混在) を有効にします。詳細は `enhanced` を参照。`"<fontname>"` は有効な Windows のフォント名で、`<fontsize>` はポイント単位でのフォントの大きさです。

他のオプションもグラフメニュー、初期化ファイル、または `set linestyle` で設定できます。古い Windows GDI インターフェースによるある制限、すなわち、線幅の変更は実線でのみ可能で点線や破線では機能しないという制限があることに注意してください。

Windows 版は、非対話型モードでは通常、コマンドラインから与えたファイルの最後に達すると直ちに終了しますが、コマンドラインの最後に `-` を指定した場合は別です。また、このモードではテキストウィンドウは表示せず、グラフのみの表示となりますが、オプションとして `-persist` (x11 版の gnuplot と同じオプション; 従来の Windows のみのオプション `/noend` や `-noend` を使うこともできます) を指定すると gnuplot は終了しなくなります。この場合他の OS での gnuplot の挙動とは異なり、`-persist` オプション後も gnuplot の対話型コマンドラインを受け付けます。

### 50.75.1 Graph-menu

`gnuplot graph` ウィンドウでマウスの右ボタンを押すか、システムメニューから Options を選択すると以下のオプションを持つポップアップメニューが現われます:

Bring to Top チェックを入れるとグラフィックウィンドウを他の全ての描画ウィンドウの手前に表示

Color チェックを入れるとカラーの線種が有効、チェック無しでは白黒

Copy to Clipboard クリップボードにビットマップやメタファイルの画像をコピー

Background... ウィンドウ背景色の設定

Choose Font... グラフィックウィンドウで使うフォントの選択

Line Styles... 線の色や種類のカスタマイズ

Print... グラフィックウィンドウを Windows プリンタドライバでプリントアウト。プリンタと拡大率の選択が可能ですが、この Print オプションによる印刷結果は gnuplot の持つプリンタドライバによるもの程良くはありません。

Update wgnuplot.ini 現在のウィンドウの位置、ウィンドウの大きさ、テキストウィンドウのフォントとそのフォントサイズ、グラフィックウィンドウのフォントとそのフォントサイズ、背景色、線種を初期化ファイル WGNUPLOT.INI に保存

### 50.75.2 Printing

好みにより、グラフは以下のような方法で印刷できます。

1. gnuplot のコマンド `set terminal` でプリンタを選択し、`set output` で出力をファイルにリダイレクト
2. gnuplot graph ウィンドウから Print... コマンドを選択。テキストウィンドウからこれを行なう特別なコマンド screendump もある。
3. `set output "PRN"` とすると出力は一時ファイルに出力され、gnuplot を終了するかまたは `set output` コマンドで出力を他のものへ変更すると、ダイアログ(対話)ボックスが現われ、そこでプリンタポートを選択。そこで OK を選択すると、出力はプリントマネージャでは加工されずにそのまま選択されたポートでプリントアウトされる。これは偶然(または故意)に、あるプリンタ用の出力を、それに対応していないプリンタに送り得ることを意味する。

### 50.75.3 Text-menu

`gnuplot text` ウィンドウでマウスの右ボタンを押すか、システムメニューから Options を選択すると以下のオプションを持つポップアップメニューが現われます:

**Copy to Clipboard** マークしたテキストをクリップボードにコピー

**Paste** 打ち込んだのと同じようにクリップボードからテキストをコピー

**Choose Font...** テキストウィンドウで使うフォントの選択

**System Colors** 選択するとコントロールパネルで設定したシステムカラーをテキストウィンドウに与える。選択しなければ白背景で文字は黒か青。

**Update wgnuplot.ini** 現在のテキストウィンドウの位置、テキストウィンドウの大きさ、テキストウィンドウのフォントとそのフォントサイズを初期化ファイル WGNUPLOT.INI に保存

#### MENU BAR

メニューファイル WGNUPLOT.MNU が WGNUPLOT.EXE と同じディレクトリにある場合、WGNUPLOT.MNU に書かれているメニューが読み込まれます。メニュー命令は以下の通りです:

[Menu] 次の行の名前で新しいメニューを開始

[EndMenu] 現在のメニューを終了

[–] 水平なメニューの仕切りを入れる

[|] 垂直なメニューの仕切りを入れる

[Button] メニューにプッシュボタンを入れ、それに次のマクロを当てはめる

マクロは 2 行で書き、最初の行はマクロ名 (メニューの見出し)、2 行目がマクロ本体です。先頭の空白列は無視されます。マクロ命令は以下の通りです:

[INPUT] — [EOS] か {ENTER} までをプロンプトとして出力し文字列を入力

[EOS] — 文字列の終り (End Of String)。何も出力しない

[OPEN] — 開くファイル名をリストボックスから取得。[EOS] までがリストボックスのタイトル、その後から次の [EOS] か {ENTER} までがデフォルトのファイル名 (Windows 3.1 の COMMDLG.DLL を使用)。

[SAVE] — セーブファイル名を取得 ([OPEN] 同様)

マクロ文字の置き換えは以下の通りです:

{ENTER} — 復帰 '\r'

{TAB} — タブ '\011'

{ESC} — エスケープ '\033'

{^A} — '\001'

...

{^\_} — '\031'

マクロは展開後の文字数が最大 256 文字に制限されています。

#### 50.75.4 Wgnuplot.ini

Windows 版の gnuplot は Windows ディレクトリにある WGNUPLOT.INI の [WGNUPLOT] セクションからいくつかのオプションを読み込みます。WGNUPLOT.INI ファイルの例です:

```
[WGNUPLOT]
TextOrigin=0 0
TextSize=640 150
TextFont=Terminal,9
GraphOrigin=0 150
GraphSize=640 330
GraphFont=Arial,10
GraphColor=1
GraphToTop=1
GraphBackground=255 255 255
Border=0 0 0 0
Axis=192 192 192 2 2
```

```

Line1=0 0 255 0 0
Line2=0 255 0 0 1
Line3=255 0 0 0 2
Line4=255 0 255 0 3
Line5=0 0 128 0 4

```

GraphFont の項にはフォント名とポイント単位のフォントサイズを指定します。Border, Axis, Line には、赤の強度 (0–255)、緑の強度、青の強度、カラーの線種、白黒の線種の 5 つの数字を指定します。この線種 (linestyle) は、0=実線、1=鎖線、2=点線、3=一点鎖線、4=二点鎖線、となっています。上の例の WGNUPLOT.INI では、Line 2 はカラー モードでは緑の実線で、白黒 モードでは鎖線になります。デフォルトの線幅は 1 ピクセルです。線種が負の値の場合、その絶対値がピクセル単位での線幅である実線を意味します。Line1 と points スタイルで使われる線種は 1 ピクセル幅の実線でなければなりません。

### 50.75.5 Windows3.0

Windows 3.1 の方が良いのですが、以下のような制限の元で Windows 3.0 上でも WGNUPLOT は動作します: 1. COMMDLG.DLL と SHELL.DLL (Windows 3.1、または Borland C++ 3.1 にある) が windows ディレクトリにあること

2. Borland C++ 3.1 で作られた WGNUPLOT.HLP は Windows 3.1 フォーマットなので、それには Borland C++ 3.1 とともに提供される WINHELP.EXE を使う必要があります
3. リアルモードではメモリ不足のため動作しない
4. TrueType フォントはグラフィック ウィンドウ内では利用できない
5. ドラッグ-ドロップは働かない

### 50.76 Wxt

wxt 出力形式は、個々のウィンドウへの出力を生成します。ウィンドウは wxWidgets ライブラリで生成されます (これが wxt の名前の由来です)。実際の描画は、2D グラフィック ライブラリ cairo と、文字列配置/レンダリング ライブラリ pango が処理します。

書式:

```

set term wxt {<n>}
    {{no}enhanced}
    {font <font>}
    {title "title"}
    {{no}persist}
    {{no}raise}
    {{no}ctrl}
    {close}

```

複数の描画 ウィンドウがサポートされています: set terminal wxt <n> は出力を番号 n の描画 ウィンドウに向けます。

デフォルトの ウィンドウ タイトルは、この番号に基づいています。そのタイトルは "title" キーワードでも指定できます。

描画 ウィンドウは、gnuplot の出力形式を別なものに変更しても残ったままになります。それを閉じるには、その ウィンドウ に入力 フォーカスがある状態で 'q' を入力するか、 ウィンドウ マネージャ のメニューで close を選択するか、 set term wxt <n> close としてください。

ウィンドウの サイズを変更すると、描画 グラフも ウィンドウ の新しい サイズにぴったり合うようにすぐに伸縮されます。他の 対話型 出力形式と違い、wxt 出力形式は フォント、線幅も含めて 描画 全体を伸縮しますが、全体の アスペクト比 は一定に保って、空いたスペースは 灰色で塗り潰します。その後 replay と タイプするか ターミナルツールバー の replay アイコンをクリックするか 新たに plot コマンドを入力すると、その新しい 描画 では 完全にその ウィンドウ に 合わせられますが、フォント サイズや 線幅 は その デフォルト に リセットされます。

有効な描画ウィンドウ (set term wxt <n> で選択されたもの) は対話的でその挙動は、他の出力形式と共通です。詳細は `mouse` を参照。それには、追加のアイコンもいくつかついていますが、それらはそれ自体が説明的なものになっているはずです。

この出力形式は、拡張文字処理モード (enhanced text mode) をサポートしていて、フォントや書式コマンド (上付、下付など) をラベルや他の文字列に埋め込むことができます。拡張文字処理モードの書式は他の gnuplot の出力形式と共通です。詳細は `enhanced` 参照。

<font> は "FontFace,FontSize" の形式で、FontFace と FontSize とをコンマで分離して一つの文字列として書きます。FontFace は、'Arial' のような通常のフォント名です。FontFace を与えない場合は、wxt 出力形式は 'Sans' を使用します。FontSize は、ポイント単位のフォントサイズです。FontSize を与えない場合は、wxt 出力形式は 10 ポイントを使用します。

例:

```
set term wxt font "Arial,12"
set term wxt font "Arial" # フォント名のみ変更
set term wxt font ",12" # フォントサイズのみ変更
set term wxt font "" # フォント名、フォントサイズをリセット
```

フォントは通常のフォントサブシステムから取得します。MS-Windows 上ではコントロールパネルの "Fonts" エントリで検索されるので、そこに設定します。Unix 上では、フォントは "fontconfig" が処理します。

文字列のレイアウトに使用される pango ライブラリは utf-8 を基本としていますので、wxt 出力形式ではエンコーディングを utf-8 にする必要があります。デフォルトの入力エンコーディングは、システムの 'locale' によります。他のエンコーディングを使用したい場合は、それを gnuplot に知らせる必要があります。詳細は `encoding` を使用してください。

pango は、unicode マッピングでないフォントに対しては予期せぬ結果を与えるかもしれません。例えば Symbol フォントに対しては、wxt 出力形式は、文字コードを unicode に変換するために <http://www.unicode.org/> で提供されるマッピングを利用します。pango は、その文字を含むフォントを見つけるためにあなたの Symbol フォントを検索し、そして DejaVu フォントのように、幅広く unicode をカバーする他のフォントを探す、といった最善の作業を行おうとします。なお、"the Symbol font" は、Acrobat Reader と一緒に "SY\_\_\_\_\_PFB" として配布されている Adobe Symbol フォントであると解釈されることに注意してください。この代わりに、OpenOffice.org と一緒に "opens\_\_\_\_.ttf" として配布される OpenSymbol フォントが同じ文字を提供しています。Microsoft も Symbol フォント ("symbol.ttf") を配布していますが、これは異なる文字セットになっていて、いくつかは欠けていますし、いくつかは数式記号に変わっています。あなたのデフォルトの設定でなんらかの問題が起きた場合 (例えばデモスクリプト enhancedtext.dem がちゃんと表示されないといった場合) は、Adobe か OpenOffice の Symbol フォントをインストールして、Microsoft の Symbol フォントを削除しないといけないかもしれません。"windings" のような他の非標準のフォントでも動作することが報告されています。

描画のレンダリングは、ツールバーで対話的に変更できます。出力をより良くするためにこのレンダリングは、アンチエイリアス、オーバーサンプリング、ヒンティングの 3 つの機構を持っています。アンチエイリアスは、水平や垂直でない線を滑らかに表示します。オーバーサンプリングは、アンチエイリアスと組でピクセルよりも小さいサイズでの精度を提供し、gnuplot が非整数座標の直線を書けるようになります。これは、対角方向の直線 (例えば 'plot x') が左右に揺れるのを避けます。ヒンティングは、オーバーサンプリングによって引き起こされる水平、垂直方向の線分のぼかしを避けます。この出力形式は、これらの直線を整数座標に揃え、それにより、1 ピクセル幅の直線は本当に 1 つ (1 つよりも少なくもない) のピクセルで描画します。

デフォルトでは、描画が行われたときにウィンドウはデスクトップの一番上 (最前面) に表示されます。これは、キーワード "raise" で制御できます。キーワード "persist" は、すべての描画ウィンドウを明示的に閉じない間は、gnuplot が終了しないようにします。最後に、デフォルトでは <space> キーは gnuplot コンソールウィンドウを上に上げ、'q' は描画ウィンドウを閉じます。キーワード "ctrl" は、それらのキー割り当てを、それぞれ <ctrl>+<space> と <ctrl>+'q' に変更します。これらの 3 つのキーワード (raise, persist, ctrl) は、設定ダイアログ上のやりとりでも設定し、記憶させることができます。

## 50.77 X11

gnuplot は X サーバを利用する x11 出力ドライバを提供します。この出力形式は、環境変数 DISPLAY がセットされているか、環境変数 TERM が xterm にセットされているか、またはコマンドラインオプションとして -display が使われていれば起動時に自動的に選択されます。

書式:

```
set terminal x11 {<n>}
    {title "<string>"}
    {{no}enhanced}
    {font <fontspec>}
    {{no}persist} {{no}raise} {solid|dashed} {{no}ctrlq}
    {close}
set terminal x11 {reset}
```

複数のグラフ描画ウィンドウをサポートしています。`set terminal x11 <n>` は番号 *n* の描画ウィンドウに出力します。*n* が 0 でなければ、タイトルが明示的に指定されなければその番号がウィンドウタイトルとしてつけられ、アイコンには Gnuplot <n> とラベル付けされます。現在有効なウィンドウはカーソル記号の変化で区別できます（デフォルトカーソルから十字カーソルへ）。

*x11* 出力形式は、利用可能なフォントの元で拡張テキストモード（enhanced 参照）をサポートしています。文字列に埋め込まれ、様々な効果を与えるフォントサイズ命令のために、デフォルトの *x11* フォントがスケーラブルフォントである必要があります。よって、以下の最初の例はうまくいくでしょうが、2 番目のものはそうではないでしょう。

```
set term x11 enhanced font "arial,15"
set title '{/=20 Big} Medium {/=5 Small}'
set term x11 enhanced font "terminal-14"
set title '{/=20 Big} Medium {/=5 Small}'
```

gnuplot ドライバが別な出力ドライバに変更されても、描画ウィンドウは開いたままになります。描画ウィンドウは、そのウィンドウにカーソルを置いて文字 *q* を押すか、ウィンドウマネージャのメニューの *close* を選択すれば閉じることができます。`reset` を実行すれば全てのウィンドウを一度に閉じれます。それは実際にウィンドウを管理している子プロセスを終了します（もし *-persist* が指定されていなければ）。コマンド *close* は、個々の描画ウィンドウを番号を指定して閉じるのに使うことができます。しかし、*persist* のために残っているような描画ウィンドウは *close* コマンドでは閉じることはできません。番号を省略して *close* した場合には現在有効な描画ウィンドウを閉じます。

gnuplot の外にあるドライバ `gnuplot_x11` は、プログラムのコンパイル時に選択されたデフォルトの場所が検索されます。これは環境変数 `GNUPLOT_DRIVER_DIR` を異なる場所と定義することで変更できます。

描画ウィンドウは *-persist* オプションが与えられていないければ、対話の終了時に自動的に閉じられます。オプション *persist* と *raise* はデフォルトでは設定されていませんが、それは、デフォルトの値 (*persist == no* で *raise == yes*) か、コマンドラインオプション *-persist* / *-raise* の指定か、または X のリソース値が使われる、ということを意味します。`[no]persist` か `[no]raise` が指定されるとそれはコマンドラインオプションや X リソースの設定よりも優先されます。これらのオプションの設定は直ちに効力を持ちますので、既に起動しているドライバの挙動は変更されます。ウィンドウを前面に出せない場合は、*raise* の項目の記述を参照してください。

オプション `title "<title name>"` は現在の描画ウィンドウに、または番号を指定すればその番号の描画ウィンドウに対するウィンドウタイトル名をつけます。そのタイトルが表示される場所、または表示されるかどうかは、使っている X のウィンドウマネージャに依存します。

描画サイズとアスペクト比は、gnuplot のウィンドウをリサイズすることでも変更できます。

線の幅と点のサイズは gnuplot の `set linestyle` で変更可能です。

出力ドライバ *x11* に関しては、gnuplot は（起動時に）、コマンドライン、または設定ファイルから、`geometry` や `font`, `name` などの通常の X Toolkit オプションやリソースの指定を受け付けます。それらのオプションについては X(1) マニュアルページ（やそれと同等のもの）を参照してください。

他にも *x11* 出力形式用の多くの gnuplot のオプションがあります。これらは gnuplot を呼ぶときにコマンドラインオプションとして指定するか、または設定ファイル `".Xdefaults"` のリソースとして指定できます。これらは起動時に設定されるので、gnuplot 実行時には変更できません（*persist* と *raise* 以外は）。

### 50.77.1 X11\_fonts

初期起動時は、システムの設定か、ユーザの `.Xdefaults` ファイルの設定か、コマンドライン指定か、のいずれかによる X11 リソースによってデフォルトのフォントが選択されます。

例:

```
gnuplot*font: lucidasans-bold-12
```

新しいデフォルトのフォントを、gnuplot 内部から x11 ドライバに以下のようにして指示することもできます:

```
'set term x11 font "<fontspec>"'
```

まず x11 ドライバは、与えられたフォントの正式名を X サーバに尋ねます。この問い合わせが失敗した場合、<fontspec> を"<font>,<size>,<slant>,<weight>" と解釈し、以下の形の完全な X11 フォント名を生成しようとします:

```
-*-<font>-<weight>-<s>-*-*-<size>-*-*-*-*-<encoding>
```

<font> はフォントの基本名 (base name) (例: Times, Symbol)

<size> はポイントサイズ (指定がなければデフォルトは 12)

<s> は <slant>=="italic" なら 'i'、<slant>=="oblique" なら 'o'、その他は 'r'

<weight> は明示的に指定されれば 'medium' か 'bold'、その他は '\*'。

<encoding> は現在の文字集合に基づいて設定 ('set encoding' 参照)

よって set term x11 font "arial,15,italic" は (デフォルトの encoding だとすれば) -\*-arial-\*-\*-\*-15-\*-\*-\*-\*-\*iso8859-1 に変換されます。<size>, <slant>, <weight> 指定はいずれも必須ではありません。<slant> や <weight> を指定しなかった場合は、フォントサーバが最初に見つけた、変種のフォントを取得するかもしれません。デフォルトのエンコーディングは、対応する X11 リソースを使って設定することもできます。例:

```
gnuplot*encoding: iso8859-15
```

x11 ドライバは、一般的な PostScript フォント名も認識し、それと同等で有効な X11 フォントか TrueType フォントに置き換えます。これと同じ手順が、set label の要求によるフォントの生成でも使われています。

あなたの gnuplot が configure の -enable-x11-mbfonts オプションをつけてインストールされたものなら、フォント名の前に "mbfont:" をつけることでマルチバイトフォントを指定することができます。フォント名を複数指定する個ともできますが、その場合はセミコロンで区切れます。マルチバイトフォントのエンコーディングは locale の設定に従いますので、環境変数 LC\_CTYPE を適切な値 (例えば ja\_JP.eucJP, ko\_KR.EUC, zh\_CN.EUC など) に設定する必要があります。

例:

```
set term x11 font 'mbfont:kana14;k14'
# 'kana14' と 'k14' は日本語の X11 font エイリアス名、';
# はフォント名の区切れです。
set term x11 font 'mbfont:fixed,16,r,medium'
# <font>,<size>,<slant>,<weight> 形式も使用できます。
set title '(mb strings)' font 'mbfont:*-fixed-medium-r-normal--14-*'
```

同じ書式は X のリソースでのデフォルトフォントの設定でも有効です。例:

```
gnuplot*font: \
mbfont:-misc-fixed-medium-r-normal--14-*-*-c-*jisx0208.1983-0
```

gnuplot が -enable-x11-mbfonts でインストールされた場合、"mbfont:" をつけなくても 2 つの特別な PostScript フォント名 'Ryumin-Light-\*', 'GothicBBB-Medium-\*' (標準的な日本語 PS フォント) を使うことができます。

## 50.77.2 Command-line\_options

X Toolkit オプションに加え、以下のオプションが gnuplot の立ち上げ時のコマンドラインで、またはユーザのファイル ".Xdefaults" 内のリソースとして指定できます (raise と persist は set term x11 [no]raise [no]persist によって上書きされることに注意してください):

'-mono'	カラーディスプレイ上で強制的に白黒描画
'-gray'	グレイスケールまたはカラーディスプレイ上でのグレイスケール描画 (デフォルトではグレイスケールディスプレイは白黒描画を受け付ける)
'-clear'	新しい描画を表示する前に(瞬間的に)画面を消去
'-tvtwm'	geometry オプションによる位置の指定を、仮想ルートウィンドウ中の現在の表示部分に対する相対的な位置にする
'-raise'	各描画後に描画ウィンドウを最前面へ出す
'-noraise'	各描画後に描画ウィンドウを最前面へ出すことはしない
'-noevents'	マウス、キーの入力を処理しない
'-persist'	gnuplot プログラム終了後も描画ウィンドウを残す

上記のオプションはコマンドライン上での指定書式で、".Xdefaults" にリソースとして指定するときは異なる書式を使います。

例:

```
gnuplot*gray: on
gnuplot*ctrlq: on
```

gnuplot は描画スタイル points で描画する点のサイズの制御にも、コマンドラインオプション (-pointsize <v>) とリソース (gnuplot\*pointsize: <v>) を提供しています。値 v は点のサイズの拡大率として使われる実数値 ( $0 < v \leq 10$ ) で、例えば -pointsize 2 はデフォルトのサイズの 2 倍、-pointsize 0.5 は普通のサイズの半分の点が使われます。

-noevents はマウス、キーの全ての入力の処理を無効にします (q と <space> によるウィンドウの終了を除いて)。これは、gnuplot のメインプログラムとは独立した形で x11 ドライバを利用するプログラムに有用です。

-ctrlq スイッチは、描画ウィンドウを閉じるホットキーを q から <ctrl>q に変更します。これは、pause mouse keystroke によるキーストロークの保存機能を使っている場合には、他のアルファベット文字と同様に q を保存できるようになるので有用でしょう。同じ理由で、-ctrlq スイッチは <space> ホットキーも <ctrl><space> に置き換えます。

### 50.77.3 Monochrome\_options

白黒ディスプレイに対しては gnuplot は描画色 (foreground) も背景色 (background) も与えません。デフォルトでは背景は白、描画は黒です。-rv や gnuplot\*reverseVideo: on の場合には背景が黒で描画は白になります。

### 50.77.4 Color\_resources

カラーディスプレイに対しては、gnuplot は以下のリソース (ここではそのデフォルトの値を示します)、または白黒階調 (greyscale) のリソースを参照します。リソースの値はシステム上の X11 rgb.txt ファイルに書かれている色名、または 16 進の色指定 (X11 のマニュアルを参照) か、色名と強度 (0 から 1 の間の値) をコンマで区切った値を使用できます。例えば blue, 0.5 は半分の強度の青、を意味します。

gnuplot*background: white
gnuplot*textColor: black
gnuplot*borderColor: black
gnuplot*axisColor: black
gnuplot*line1Color: red
gnuplot*line2Color: green
gnuplot*line3Color: blue
gnuplot*line4Color: magenta
gnuplot*line5Color: cyan
gnuplot*line6Color: sienna
gnuplot*line7Color: orange
gnuplot*line8Color: coral

これらに関するコマンドラインの書式は、背景 (background) に関しては単純で通常の X11 toolkit オプションの "-bg" に直接対応します。他のものも、全て一般的なリソースの上書きオプション "-xrm" を使うことで設定できます。

例: 背景色を変更するには

```
gnuplot -background coral
```

線種 1 番目の色を書き換えるには

```
gnuplot -xrm 'gnuplot*line1Color:blue'
```

### 50.77.5 Grayscale\_resources

-gray を選択すると、gnuplot は、グレイスケールまたはカラーディスプレイに対して、以下のリソースを参照します (ここではそのデフォルトの値を示します)。デフォルトの背景色は黒であることに注意してください。

gnuplot*background: black
gnuplot{textGray: white}
gnuplot*borderGray: gray50
gnuplot*axisGray: gray50
gnuplot*line1Gray: gray100
gnuplot*line2Gray: gray60
gnuplot*line3Gray: gray80
gnuplot*line4Gray: gray40
gnuplot*line5Gray: gray90
gnuplot*line6Gray: gray50
gnuplot*line7Gray: gray70
gnuplot*line8Gray: gray30

### 50.77.6 Line\_resources

gnuplot は描画の線の幅 (ピクセル単位) の設定のために以下のリソースを参照します (ここではそのデフォルトの値を示します)。0 または 1 は最小の線幅の 1 ピクセル幅を意味します。2 または 3 の値によってグラフの外観を改善できる場合もあるでしょう。

gnuplot*borderWidth: 2
gnuplot*axisWidth: 0
gnuplot*line1Width: 0
gnuplot*line2Width: 0
gnuplot*line3Width: 0
gnuplot*line4Width: 0
gnuplot*line5Width: 0
gnuplot*line6Width: 0
gnuplot*line7Width: 0
gnuplot*line8Width: 0

gnuplot は線描画で使用する点線の形式の設定用に以下のリソースを参照します。0 は実線を意味します。2 行の 10 進数  $jk$  ( $j$  と  $k$  は 1 から 9 までの値) は、 $j$  個のピクセルの描画に  $k$  個の空白のピクセルが続くパターンの繰り返しからなる点線を意味します。例えば '16' は 1 個のピクセルの後に 6 つの空白が続くパターンの点線になります。さらに、4 行の 10 進数でより詳細なピクセルと空白の列のパターンを指定できます。例えば、'4441' は 4 つのピクセル、4 つの空白、4 つのピクセル、1 つの空白のパターンを意味します。以下のデフォルトのリソース値は、白黒ディスプレイ、あるいはカラーや白黒階調 (grayscale) ディスプレイ上の白黒描画における値です。カラーディスプレイでは dashed:off がデフォルトになっています。

```
gnuplot*dashed: off
gnuplot*borderDashes: 0
gnuplot*axisDashes: 16
gnuplot*line1Dashes: 0
gnuplot*line2Dashes: 42
gnuplot*line3Dashes: 13
gnuplot*line4Dashes: 44
gnuplot*line5Dashes: 15
gnuplot*line6Dashes: 4441
gnuplot*line7Dashes: 42
gnuplot*line8Dashes: 13
```

### 50.77.7 X11 pm3d\_resources

適切な visual クラスと色数を選択するのは、X11 アプリケーションにとって苦しく、ちょっと厄介なことです。それは X11 が異なる深度 (depth) の 6 つの visual 型をサポートしているからです。

デフォルトでは gnuplot はそのスクリーンのデフォルトの visual を使用します。割り当てることのできる色数は選択された visual クラスによって変わります。12bit を超える深度を持つ visual クラス上では、gnuplot は最大色数である 0x200 (=512) 色で起動します。8bit を超える (12bit 以下で) 深度の visual クラスでは最大色数は 0x100 (=256) 色、8bit 以下のディスプレイでは最大色は 240 (16 色は曲線の色用に取られる) になります。

gnuplot は最初に、上に述べたような最大色を割り当てようと起動します。これに失敗するとその色数は、gnuplot がその全部を割り付けることができるまで、1/2 ずつ減らされます。maxcolors を繰り返し 2 で割った結果、mincolors よりも小さい数字になった場合、gnuplot は private カラーマップを使おうとします。この場合、ウィンドウマネージャは、ポインタが X11 ドライバのウィンドウに入るか出るかでカラーマップを退避 (swapping) させる責任を持つことになります。

mincolors のデフォルトの値は maxcolors / (num\_colormaps > 1 ? 2 : 8) で、num\_colormaps は gnuplot が現在使用しているカラーマップの数で、これは、x11 のウィンドウが 1 つだけ開いているような通常の場合は 1 です。

複数の (異なる) visual クラスを、一つのスクリーン上でサポートするようなシステムもあります。このようなシステムでは、gnuplot に指定した visual クラスを強制的に使わせる必要があります。例えば、デフォルトの visual が 8bit PseudoColor だけれどもスクリーンは 24bit TrueColor をサポートしていてむしろこちらの方を選択すべきであるような場合です。

X サーバの能力に関する情報はプログラム xdpinfo で取得できます。visual 名は次のうちの一つが選択できます: StaticGray, GrayScale, StaticColor, PseudoColor, TrueColor, DirectColor。その X サーバが要求された visual 型の異なる複数の深度をサポートしている場合、gnuplot は最も大きい(最深の)深度の visual クラスを選択します。要求された visual クラスがデフォルトの visual とあっていて、その型の複数のクラスがサポートされている場合は、デフォルトの visual が選択されます。

例: 8bit PseudoColor の visual 上では、gnuplot\*maxcolors: 240、及び gnuplot\*mincolors: 240 と指定することで強制的に private カラーマップを使うようにできます。

```
gnuplot*maxcolors: 整数
gnuplot*mincolors: 整数
gnuplot*visual: visual 名
```

### 50.77.8 X11 other\_resources

デフォルトでは、現在の描画ウィンドウの内容は、ウィンドウの X イベントに従って X11 クリップボードに送られます。'gnuplot\*exportselection' のリソースの値を 'off' か 'false' と設定することによりこれを無効にできます。

デフォルトでは、文字の回転は速くそれを行なう方法が使われますが、背景色によってはその付近が汚れることがあります。これが起こる場合は、リソース 'gnuplot.fastrotate' を 'off' にしてみてください。

```
gnuplot*exportselection: off
gnuplot*fastrotate: on
gnuplot*ctrlq: off
```

## 50.78 Xlib

xlib ドライバは X11 Windows System をサポートしています。このドライバは gnuplot\_x11 への命令を生成しますが、`set output '<filename>'` を指定するとそれらをファイルに書き出します。`set term x11` は、`set terminal xlib; set output "|gnuplot_x11 -noevents"` と同値です。xlib には x11 と同じオプションの組を与えることができます。

## Part IV

# Graphical User Interfaces

gnuplot のために色々なグラフィカルユーザインターフェースが書かれてきました。その win32 用の物はこの配布版に含まれています。そして、Python によるインターフェースが

<http://py-gnuplot.darwinports.com/>  
に置いてあります。

X11 に対するインターフェースも幾つかあります。xgfe と呼ばれるものがその一つで、Qt ライブリを使用します。これは以下にあります。

<http://www.flash.net/~dmishee/xgfe/xgfe.html>  
さらに、通常の Tcl/Tk の配布場所に置かれている、3つの Tcl/Tk プログラムがあります。

Bruce Ravel ([ravel@phys.washington.edu](mailto:ravel@phys.washington.edu)) は、Gershon Elber による gnuplot.el を元に、GNU emacs と XEmacs に対する新しい版の gnuplot-mode を書きました。gnuplot の CVS リポジトリにはそのコピーが含まれているが、このパッケージの最新版は以下にもあります。

<http://feff.phys.washington.edu/~ravel/software/gnuplot-mode/>

## Part V

# Bugs

以前のものと現在のリリースに関して報告されたバグは SourceForge 上の公式配布サイトに置かれています。

バグリポートは e-mail で gnuplot-bugs メーリングリストへ送って下さい。または、SourceForge の gnuplot ウェブサイトにその報告を投稿してください。その際、あなたが使用している gnuplot のバージョンの完全な情報、そして可能ならばそのバグを実証するテストスクリプトを送ってください。Seeking-assistance 参照。

以下のセクションは、gnuplot 4.2 のリリース時現在に知られている問題の一覧を示しています。これらには外部のサポートライブラリのバグも含まれているので、gnuplot の修正とは関係なく改良されるかもしれません。

## 51 Gnuplot limitations

浮動小数計算例外 (浮動小数値が大きすぎる (または小さすぎる) 場合、0 で割算した場合など) は、ユーザ定義関数において時折発生します。特に、いくつかのデモで、浮動小数の範囲を超える数値を生成することが起こるようです。システムがそのような例外を無視する (gnuplot はそのような点を定義できないもの、と見なします) か、または gnuplot の実行を中止するかは、コンパイル時 (あるいは実行時) の環

境によります。

ガンマ関数とベッセル関数は複素数引数に対しては動作しません。

コマンドラインに一つの "load" コマンドが含まれている場合、そのコマンドラインの "load <filename>" 以降はすべて無視されます。

すべての x11 描画ウィンドウに対して、一度に 1 つのカラーパレットのみが有効です。これは、異なるパレットを使用する描画を含んだ multiplot では x11 上では正しく表示されない、ということを意味します。

"時刻" として指定された座標は 24 時で折り返し、精度の最小単位は 1 秒です。これは、特に地理上の座標に基づく時刻を使っている場合には制限になります。

誤差指示線 (errorbars) は、極座標/球面座標描画モードでは使えません。

nohidden3d は、全体的な設定である set hidden3d を、個々の描画に対して免除するためのオプションですが、これは媒介変数の曲線 (parametric) に対しては機能しません。

## 52 Specific terminals

HP プロッタの古いモデルの中にはページの排出コマンド 'PG' を持たないものがあります。現在の HPGL ドライバはこのコマンドを HPGL\_reset で使用していますが、そのようなプロッタには、このコマンドを取り除く必要があるでしょう。現在の PCL5 ドライバは、グラフィックと同様にテキストにも HPGL/2 を使用しています。これはスケーラブル PCL フォントを使うように修正されるべきでしょう。

## 53 External libraries

外部ライブラリ GD (PNG/JPEG/GIF ドライバで使用): バージョン 2.0.33までの libgd には、Adobe の Symbol フォントの文字のマッピングに関するいくつかのバグがあります。また、アンチエイリアスされた線分がキャンバスの上の角と交わる場合に、ライブラリがセグメンテーションフォルトを引き起すこともあります。

外部ライブラリ PDFlib (PDF ドライバで使用): gnuplot は libpdf のバージョン 4, 5, 6 のいずれかをリンクできます。しかし、これらはバージョンによってパイプされた入出力の処理が違っていますので、パイプを使って PDF を出力する gnuplot スクリプトは、PDFlib のあるバージョンでだけしかちゃんと動かないかもしれません。

外部ライブラリ svgalib (linux, vgacl ドライバで使用): これは gnuplot が root に setuid (嫌!) されることを要求しますし、ビデオカードや X11 で使用されるグラフィックドライバに特有の多くのバグがあります。

国際化 (ロケールの設定): gnuplot は、入出力の数、時刻、日付文字列のロケールに依存した書式の制御を、C ランタイムライブラリに含まれる setlocale() を用いて行うので、ロケールの有効性や、ロケール機能のサポートのレベル (例えば数字の 3 行毎の区切り文字など) などは、あなたのコンピュータが提供する国際化のサポートの度合いに依存します。

## Part VI

## Index

# Index

.gnuplot, 34  
abs, 26  
acos, 26  
acosh, 26  
acsplines, 57  
aed512, 147  
aed767, 147  
aifm, 147  
amiga, 148  
angles, 70  
apollo, 148  
Aqua, 148  
aqua, 148  
arg, 26  
arrow, 70  
arrowstyle, 115  
asin, 26  
asinh, 26  
atan, 26  
atan2, 26  
atanh, 26  
atari, 148  
autoscale, 72  
avs, 54  
  
backquotes, 36  
bars, 74  
batch/interactive, 22  
BE, 149  
be, 149  
besj0, 26  
besj1, 26  
bessel, 200  
besy0, 26  
besy1, 26  
bezier, 58  
binary, 52  
binary examples, 55  
bind, 32  
bitgraph, 186  
bitwise operators, 28  
bmargin, 74  
border, 74  
boxerrorbars, 119  
boxes, 119  
boxwidth, 75  
boxxyerrorbars, 120  
branch, 46  
bugs, 200  
  
call, 39  
candlesticks, 120  
canvas, 19  
cbdata, 139  
cbdtics, 140  
cblabel, 141  
cbmtics, 141  
cbrange, 141  
cbtics, 141  
cd, 39  
ceil, 26  
cgi, 151  
cgm, 151  
clabel, 76  
clear, 40  
clip, 76  
cntrparam, 77  
color resources, 197  
colorbox, 78  
colornames, 111  
colors, 31, 32, 108, 109  
colorspec, 32  
column, 27  
command line editing, 22  
command line options, 196  
commands, 39  
comments, 23  
commentschars, 81  
compatibility, 19  
contour, 79  
coordinates, 23  
copyright, 14  
corel, 155  
cos, 26  
cosh, 26  
csplines, 58  
  
data, 50  
data file, 50  
datafile, 50  
datastrings, 24  
date specifiers, 87  
debug, 155  
decimalsign, 82  
defined, 27  
degrees, 70  
dgrid3d, 83  
dospc, 155  
dots, 121  
dumb, 155  
dummy, 83  
dxf, 156  
dxy800a, 156  
  
edf, 54  
editing, 22  
editing postscript, 181  
eepic, 156

egalib, 166  
egamono, 166  
ehf, 54  
emf, 157  
emtex, 169  
emxvesa, 157  
emxvga, 157  
encoding, 84  
encodings, 84  
enhanced, 180  
enhanced postscript, 180  
environment, 24  
epslatex, 158  
epson 180dpi, 161  
epson 60dpi, 161  
epson lx800, 161  
erf, 26  
erfc, 26  
errorbars, 62  
errorlines, 62  
every, 56  
example, 56  
excl, 161  
exists, 27, 36  
exit, 40  
exp, 26  
exponentiation, 28  
expressions, 25  
  
factorial, 27  
fig, 161  
file, 50  
filledcurves, 121  
fillstyle, 116  
financebars, 122  
fit, 41  
fit parameters, 42  
fitting, 42  
floor, 26  
fontfile, 182  
fontpath, 85  
fonts, 163, 168, 178, 195  
format, 85  
format specifiers, 86  
fortran, 80  
frequency, 58  
fsteps, 122  
functions, 26  
  
gamma, 26, 200  
gamma correction, 110  
ggi, 162  
gif, 163  
glossary, 30  
gnugraph, 164  
gpbin, 143  
gpic, 165  
gpr, 165  
gprintf, 86  
graph menu, 191  
grass, 166  
grayscale resources, 198  
grid, 88  
grid data, 145  
gui's, 200  
guidelines, 44  
hcgi, 151  
help, 47  
help desk, 16  
hercules, 166  
hidden3d, 89  
histeps, 122  
histograms, 122  
history, 47  
historysize, 90  
hp2623a, 166  
hp2648, 166  
hp500c, 166  
hpdj, 167  
hpgl, 166  
hpljii, 167  
hppj, 167  
ibeta, 26  
if, 48  
igamma, 26  
imag, 26  
image, 124  
imagen, 167  
impulses, 124  
index, 57  
int, 26  
internationalization, 201  
introduction, 15  
inverf, 26  
invnorm, 26  
iris4d, 168  
isosamples, 90  
jpeg, 168  
kc tek40xx, 186  
key, 91  
km tek40xx, 186  
kyo, 169  
label, 94  
labels, 125  
lambertw, 26  
latex, 169  
least squares, 41  
legend, 91  
lgamma, 26  
libgd, 201

license, 14  
line editing, 22  
line resources, 198  
lines, 125  
linespoints, 125  
linestyle, 117  
linetype, 31  
linux, 170  
lmargin, 95  
load, 48, 200  
loadpath, 96  
locale, 82, 96, 201  
log, 26  
log10, 26  
logscale, 96  
lower, 49  
lp, 125  
  
macintosh, 170  
macros, 36  
mapping, 97  
margin, 97  
Marquardt, 41  
matrix, 145  
metafont, 171  
metapost, 172  
mf, 171  
mgr, 172  
mif, 172  
missing, 80  
mixing macros backquotes, 36  
modulo, 28  
monochrome options, 197  
mouse, 98  
mp, 172  
mtos, 175  
multi branch, 46  
multiplot, 99  
mx2tics, 100  
mxtics, 101  
my2tics, 101  
mytics, 101  
mztics, 101  
  
nec cp6, 161  
negation, 27  
new features, 16  
newhistogram, 124  
NEXT, 175  
next, 175  
noarrow, 70  
noautoscale, 72  
noborder, 74  
nocbdtics, 140  
nocbmtics, 141  
nocbtics, 141  
noclip, 76  
  
nocontour, 79  
nodgrid3d, 83  
nogrid, 88  
nohidden3d, 89, 200  
nohistorysize, 90  
nokey, 91  
nolabel, 94  
nologscale, 96  
nomouse, 98  
nomultiplot, 99  
nomx2tics, 100  
nomxtics, 101  
nomy2tics, 101  
nomytics, 101  
nomztics, 101  
nooffsets, 102  
noparametric, 103  
nopolar, 111  
norm, 26  
nosurface, 126  
notimestamp, 128  
nox2dtics, 132  
nox2mtics, 132  
nox2tics, 132  
nox2zeroaxis, 132  
noxdtics, 133  
noxmtics, 134  
noxtics, 135  
noxzeroaxis, 138  
noy2dtics, 138  
noy2mtics, 138  
noy2tics, 138  
noy2zeroaxis, 139  
noydtics, 139  
noymtics, 139  
noytics, 139  
noyzeroaxis, 139  
nozdtics, 139  
nozmtics, 140  
noztics, 141  
nozzeroaxis, 139  
  
object, 112  
offsets, 102  
okidata, 161  
one's complement, 27  
OpenStep, 175  
Openstep, 175  
openstep, 175  
operator precedence, 27  
operators, 27  
origin, 102  
output, 102  
  
palette, 106  
parametric, 103  
pause, 49

pbm, 176  
pcl5, 166  
pdf, 176, 201  
plot, 50  
plotting, 34  
pm, 177  
pm3d, 103  
pm3d resources, 199  
png, 177  
points, 125  
pointsize, 111  
polar, 111  
pop, 127  
postscript, 178  
practical guidelines, 44  
prescribe, 169  
print, 67  
printing, 191  
prologue, 183  
pslatex, 183  
pstex, 183  
pstricks, 185  
punctuation, 37  
push, 127  
pwd, 67  
  
qms, 185  
quit, 67  
quotes, 37  
  
raise, 67  
rand, 26  
random, 27  
ranges, 63  
real, 26  
rectangle, 112  
regis, 185  
replot, 68  
reread, 68  
reset, 69  
restore, 134  
rgbcolor, 32  
rgbformulae, 108  
rgbimage, 125  
rgip, 185  
rmargin, 113  
rrange, 113  
  
samples, 113  
save, 69  
sbezier, 58  
seeking assistance, 16  
selanar, 186  
separator, 81  
set, 70  
sgn, 26  
shell, 141  
show, 70  
  
sin, 26  
sinh, 26  
size, 114  
smooth, 57  
special filenames, 58  
specify, 37  
splot, 142  
sprintf, 27  
sqrt, 26  
ssvgalib, 166  
starc, 161  
start, 34  
starting values, 46  
startup, 34  
statistical overview, 43  
steps, 125  
string operators, 28  
stringcolumn, 27  
strings, 35  
strlen, 27  
strstr, 27  
style, 65  
substitution, 35  
substring, 27, 35  
sun, 186  
surface, 126  
svg, 186  
svga, 186  
svgalib, 166, 201  
syntax, 37  
system, 27, 146  
  
table, 127  
tan, 26  
tandy 60dpi, 161  
tanh, 26  
tek40, 186  
tek410x, 187  
term, 147  
terminal, 147  
termoption, 127  
ternary, 28  
test, 146  
texdraw, 187  
text menu, 191  
tgif, 187  
thru, 59  
tics, 128  
ticscale, 128  
ticslevel, 138  
time specifiers, 87  
time/date, 38  
timecolumn, 27  
timefmt, 129, 200  
timestamp, 128  
tips, 46  
title, 130

tkcanvas, 188  
tm hour, 27  
tm mday, 27  
tm min, 27  
tm mon, 27  
tm sec, 27  
tm wday, 27  
tm yday, 27  
tm year, 27  
tmargin, 130  
tpic, 189  
trange, 130  
  
unary, 27  
uniplex, 185  
unique, 58  
unixpc, 189  
unixplot, 189  
unset, 146  
update, 146  
urange, 130  
user defined, 29  
using, 60  
  
valid, 27  
variables, 29  
vdi, 189  
vectors, 125  
vgagl, 190  
vgal, 157  
vgalib, 166  
vgamono, 166  
view, 131  
vrangle, 131  
vttek, 186  
VWS, 190  
vx384, 190  
  
wgnuplot.ini, 192  
windows, 190  
windows3.0, 193  
with, 65  
word, 27  
words, 27  
writeback, 134  
wxt, 193  
  
X resources, 195, 197–199  
X11, 194  
x11, 194  
x11 fonts, 195  
x2data, 131  
x2dtics, 132  
x2label, 132  
x2mtics, 132  
x2range, 132  
x2tics, 132  
x2zeroaxis, 132  
  
xdata, 132  
xdtics, 133  
xerrorbars, 125  
xerrorlines, 126  
xfig, 161  
 xlabel, 133  
xlib, 200  
xmtics, 134  
xrange, 134  
xtics, 135  
xyerrorbars, 126  
xyerrorlines, 126  
xyplane, 138  
xzzeroaxis, 138  
  
y2data, 138  
y2dtics, 138  
y2label, 138  
y2mtics, 138  
y2range, 138  
y2tics, 138  
y2zeroaxis, 139  
ydata, 139  
ydtics, 139  
yerrorbars, 126  
yerrorlines, 126  
ylabel, 139  
ymtics, 139  
yrangle, 139  
ytics, 139  
yzzeroaxis, 139  
  
zdata, 139  
zdtics, 139  
zero, 140  
zeroaxis, 140  
zlabel, 140  
zmtics, 140  
zrange, 141  
ztics, 141  
zzeroaxis, 139